

Demostración Asistida por Ordenador con Isabelle/HOL

José A. Alonso Jiménez

Grupo de Lógica Computacional

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

Sevilla, 29 de enero de 2013 (versión del 16 de enero de 2024)

Esta obra está bajo una licencia Reconocimiento-NoComercial-CompartirIgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:

Reconocimiento. Debe reconocer los créditos de la obra de la manera especificada por el autor.



No comercial. No puede utilizar esta obra para fines comerciales.



Compartir bajo la misma licencia. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Índice general

1. Programación funcional con Isabelle/HOL	7
1.1. Tema: Programación funcional con Isabelle/HOL	7
1.2. Ejercicios: Programación funcional con Isabelle/HOL	14
2. Razonamiento sobre programas	19
2.1. Tema: Razonamiento sobre programas	19
2.2. Ejercicios: Razonamiento sobre programas	26
3. Deducción natural en lógica proposicional	31
3.1. Tema: Deducción natural en lógica proposicional	31
3.2. Ejercicios: Deducción natural en lógica proposicional	57
3.3. Ejercicios: Argumentación lógica proposicional	112
3.4. Ejercicios: Eliminación de conectivas	125
4. Deducción natural en lógica de primer orden	133
4.1. Tema: Deducción natural en lógica de primer orden	133
4.2. Ejercicios: Deducción natural en lógica de primer orden	154
4.3. Ejercicios: Argumentación lógica de primer orden	191
4.4. Ejercicios: Argumentación lógica de primer orden con igualdad	239
5. Resumen de Isabelle/Isar y de la lógica	259
6. Razonamiento por casos y por inducción	263
6.1. Ejercicios de inducción sobre listas	279
6.2. Ejercicios sobre árboles y otros tipos de datos inductivos	355
6.3. Ejercicios sobre aritmética	404
7. Caso de estudio: Compilación de expresiones	417
8. Conjuntos, funciones y relaciones	425
8.1. Gramáticas libres de contexto	440

Introducción

En este libro se presenta una introducción a la demostración asistida por ordenador usando [Isabelle/HOL](https://www.cl.cam.ac.uk/research/hvg/Isabelle/index.html)¹.

La presentación se realiza mediante teorías de ejemplos y de ejercicios.

El contenido del libro se ha utilizado en los cursos de [Razonamiento automático](http://www.cs.us.es/~jalonso/cursos/m-ra)² del [Máster Universitario en Lógica, computación e inteligencia artificial](http://www.cs.us.es/~jalonso/cursos/m-ra) de la [Universidad de Sevilla](http://www.cs.us.es/~jalonso/cursos/m-ra).

¹<https://www.cl.cam.ac.uk/research/hvg/Isabelle/index.html>

²<http://www.cs.us.es/~jalonso/cursos/m-ra>

Capítulo 1

Programación funcional con Isabelle/HOL

1.1. Tema: Programación funcional con Isabelle/HOL

```
chapter {* Tema 1: Programación funcional en Isabelle *}
```

```
theory T1
imports Main
begin
```

```
section {* Introducción *}
```

```
text {* Esta notas son una introducción a la demostración asistida
    utilizando el sistema Isabelle/HOL/Isar. La versión de Isabelle
    utilizada es la 2012.
```

*En este capítulos se presenta el lenguaje funcional que está incluido en Isabelle. El lenguaje funcional es muy parecido a Haskell. **

```
section {* Números naturales, enteros y booleanos *}
```

```
text {* En Isabelle están definidos los número naturales con la sintaxis
    de Peano usando dos constructores: 0 (cero) y Suc (el sucesor).
```

Los números como el 1 son abreviaturas de los correspondientes en la notación de Peano, en este caso "Suc 0".

El tipo de los números naturales es nat.

*Por ejemplo, el siguiente del 0 es el 1. **

```
value "Suc 0" -- "= 1"
```

text *{* En Isabelle está definida la suma de los números naturales:
(x + y) es la suma de x e y.*

*Por ejemplo, la suma de los números naturales 1 y 2 es el número natural 3. **

```
value "(1::nat) + 2" -- "= 3"
```

text *{* La notación del par de dos puntos se usa para asignar un tipo a un término (por ejemplo, (1::nat) significa que se considera que 1 es un número natural).*

*En Isabelle está definida el producto de los números naturales:
(x * y) es el producto de x e y.*

*Por ejemplo, el producto de los números naturales 2 y 3 es el número natural 6. **

```
value "(2::nat) * 3" -- "= 6"
```

text *{* En Isabelle está definida la división de números naturales:
(n div m) es el cociente entero de x entre y.*

*Por ejemplo, la división natural de 7 entre 3 es 2. **

```
value "(7::nat) div 3" -- "= 2"
```

text *{* En Isabelle está definida el resto de división de números naturales: (n mod m) es el resto de dividir n entre m.*

*Por ejemplo, el resto de dividir 7 entre 3 es 1. **

```
value "(7::nat) mod 3" -- "= 1"
```

text *{* En Isabelle también están definidos los números enteros. El tipo de los enteros se representa por int.*

*Por ejemplo, la suma de 1 y -2 es el número entero -1. **

```
value "(1::int) + -2" -- "= -1"
```

text *{* Los numerales están sobrecargados. Por ejemplo, el 1 puede ser un natural o un entero, dependiendo del contexto.*

Isabelle resuelve ambigüedades mediante inferencia de tipos.

A veces, es necesario usar declaraciones de tipo para resolver la ambigüedad.

En Isabelle están definidos los valores booleanos (True y False), las conectivas ($\neg$, $\wedge$, $\vee$, $\rightarrow$ y $\leftrightarrow$) y los cuantificadores ($\forall$ y $\exists$).

*El tipo de los booleanos es bool. **

text *{* La conjunción de dos fórmulas verdaderas es verdadera. **
value "True $\wedge$ True" -- "= True"

text *{* La conjunción de una fórmula verdadera y una falsa es falsa. **
value "True $\wedge$ False" -- "= False"

text *{* La disyunción de una fórmula verdadera y una falsa es verdadera. **
value "True $\vee$ False" -- "= True"

text *{* La disyunción de dos fórmulas falsas es falsa. **
value "False $\vee$ False" -- "= False"

text *{* La negación de una fórmula verdadera es falsa. **
value " $\neg$ True" -- "= False"

text *{* Una fórmula falsa implica una fórmula verdadera. **
value "False $\rightarrow$ True" -- "= True"

text *{* Un lema introduce una proposición seguida de una demostración.*

Isabelle dispone de varios procedimientos automáticos para generar demostraciones, uno de los cuales es el de simplificación (llamado simp).

*El procedimiento simp aplica un conjunto de reglas de reescritura, que inicialmente contiene un gran número de reglas relativas a los objetos definidos. **

text *{* Ej. de simp: Todo elemento es igual a sí mismo. **
lemma " $\forall x. x = x$ "

```

by simp

text {* Ej. de simp: Existe un elemento igual a 1. *}
lemma "∃x. x = 1"
by simp

section {* Definiciones no recursivas *}

text {* La disyunción exclusiva de A y B se verifica si una es verdadera
y la otra no lo es. *}

definition xor :: "bool ⇒ bool ⇒ bool" where
  "xor A B ≡ (A ∧ ¬B) ∨ (¬A ∧ B)"

text {* Prop.: La disyunción exclusiva de dos fórmulas verdaderas es
falsa.

  Dem.: Por simplificación, usando la definición de la disyunción
exclusiva.
*}

lemma "xor True True = False"
by (simp add: xor_def)

text {* Se añade la definición de la disyunción exclusiva al conjunto de
reglas de simplificación automáticas. *}

declare xor_def[simp]

lemma "xor True False = True"
by simp

section {* Definiciones locales *}

text {* Se puede asignar valores a variables locales mediante 'let' y
usarlo en las expresiones dentro de 'in'.

  Por ejemplo, si x es el número natural 3, entonces "x*x=9". *}

value "let x = 3::nat in x * x" -- "= 9"

section {* Pares *}

text {* Un par se representa escribiendo los elementos entre paréntesis
y separados por coma.
```

El tipo de los pares es el producto de los tipos.

La función fst devuelve el primer elemento de un par y la snd el segundo.

*Por ejemplo, si p es el par de números naturales (2,3), entonces la suma del primer elemento de p y 1 es igual al segundo elemento de p. **

```
value "let p = (2,3)::nat × nat in fst p + 1 = snd p"
```

```
section {* Listas *
```

```
text {* Una lista se representa escribiendo los elementos entre corchetes y separados por comas.
```

La lista vacía se representa por [].

Todos los elementos de una lista tienen que ser del mismo tipo.

El tipo de las listas de elementos del tipo a es (a list).

El término (x#xs) representa la lista obtenida añadiendo el elemento x al principio de la lista xs.

*Por ejemplo, la lista obtenida añadiendo sucesivamente a la lista vacía los elementos c, b y a es [a,b,c]. **

```
value "a#(b#(c#[]))" -- "= [a,b,c]"
```

```
text {* Funciones de descomposición de listas:
  · (hd xs) es el primer elemento de la lista xs.
  · (tl xs) es el resto de la lista xs.
```

*Por ejemplo, si xs es la lista [a,b,c], entonces el primero de xs es a y el resto de xs es [b,c]. **

```
value "let xs = [a,b,c] in hd xs = a ∧ tl xs = [b,c]" -- "= True"
```

```
text {* (length xs) es la longitud de la lista xs. Por ejemplo, la longitud de la lista [1,2,3] es 3. *
```

```
value "length [1,2,3]" -- "= 3"
```

```
text {* En la sesión 47 de "Isabelle/HOL – Higher-Order Logic"
      http://goo.gl/sFsFF se encuentran más definiciones y propiedades de
      las listas. *}

```

```
section {* Registros *}

```

```
text {* Un registro es una colección de campos y valores.

```

```
      Por ejemplo, los puntos del plano se pueden representar mediante
      registros con dos campos, las coordenadas, con valores enteros. *}

```

```
record punto =
  coordenada_x :: int
  coordenada_y :: int

```

```
text {* Ejemplo, el punto pt tiene de coordenadas 3 y 7. *}

```

```
definition pt :: punto where
  "pt ≡ (|coordenada_x = 3, coordenada_y = 7|)"

```

```
text {* Ejemplo, la coordenada x del punto pt es 3. *}

```

```
value "coordenada_x pt" -- "= 3"

```

```
text {* Ejemplo, sea pt2 el punto obtenido a partir del punto pt
      cambiando el valor de su coordenada x por 4. Entonces la coordenada x
      del punto pt2 es 4. *}

```

```
value "let pt2 = pt(|coordenada_x:=4|) in coordenada_x (pt2)" -- "= 4"

```

```
section {* Funciones anónimas *}

```

```
text {* En Isabelle pueden definirse funciones anónimas.

```

```
      Por ejemplo, el valor de la función que a un número le asigna su doble
      aplicada a 1 es 2. *}

```

```
value "(λx. x + x) 1::nat" -- "= 2"

```

```
section {* Condicionales *}

```

```
text {* El valor absoluto del entero x es x, si "x ≥ 0" y es -x en caso
      contrario. *}

```

```
definition absoluto :: "int ⇒ int" where

```

```

"absoluto  $x \equiv (\text{if } x \geq 0 \text{ then } x \text{ else } -x)"$ 

text {* Ejemplo, el valor absoluto de -3 es 3. *}

value "absoluto(-3)" -- "= 3"

text {* Def.: Un número natural  $n$  es un sucesor si es de la forma
  (Suc  $m$ ). *}

definition es_sucesor :: "nat  $\Rightarrow$  bool" where
  "es_sucesor  $n \equiv (\text{case } n \text{ of}$ 
    0       $\Rightarrow$  False
  | Suc  $m \Rightarrow$  True)"

text {* Ejemplo, el número 3 es sucesor. *}

value "es_sucesor 3" -- "= True"

section {* Tipos de datos y definiciones recursivas *}

text {* Una lista de elementos de tipo  $a$  es la lista Vacía o se obtiene
  añadiendo, con Cons, un elemento de tipo  $a$  a una lista de elementos de
  tipo  $a$ . *}

datatype 'a Lista = Vacía | Cons 'a "'a Lista"

text {* (conc  $xs\ ys$ ) es la concatenación de las lista  $xs$  e  $ys$ . Por
  ejemplo,
  conc (Cons  $a$  (Cons  $b$  Vacía)) (Cons  $c$  Vacía)
  = Cons  $a$  (Cons  $b$  (Cons  $c$  Vacía))
  *}

fun conc :: "'a Lista  $\Rightarrow$  'a Lista  $\Rightarrow$  'a Lista" where
  "conc Vacía  $ys = ys$ "
| "conc (Cons  $x\ xs$ )  $ys =$  Cons  $x$  (conc  $xs\ ys$ )"

value "conc (Cons  $a$  (Cons  $b$  Vacía)) (Cons  $c$  Vacía)"
-- "= Cons  $a$  (Cons  $b$  (Cons  $c$  Vacía))"

text {* (suma  $n$ ) es la suma de los primeros  $n$  números naturales. Por
  ejemplo,
  suma 3 = 6
  *}

fun suma :: "nat  $\Rightarrow$  nat" where

```

```

    "suma 0          = 0"
  | "suma (Suc m) = (Suc m) + suma m"

value "suma 3" -- "= 6"

text {* (sumaImpares n) es la suma de los n primeros números
      impares. Por ejemplo,
      sumaImpares 3 = 9
    *}

fun sumaImpares :: "nat ⇒ nat" where
  "sumaImpares 0          = 0"
| "sumaImpares (Suc n) = (2 * (Suc n) - 1) + sumaImpares n"

value "sumaImpares 3" -- "= 9"

end

```

1.2. Ejercicios: Programación funcional con Isabelle/HOL

```

chapter {* T1R1: Programación funcional en Isabelle *}

theory T1R1
imports Main
begin

text {* -----
      Ejercicio 1. Definir, por recursión, la función
      longitud :: 'a list ⇒ nat
      tal que (longitud xs) es la longitud de la listas xs. Por ejemplo,
      longitud [4,2,5] = 3
      ----- *}

fun longitud :: "'a list ⇒ nat" where
  "longitud []          = 0"
| "longitud (x#xs) = 1 + longitud xs"

value "longitud [4,2,5]" -- "= 3"

text {* -----
      Ejercicio 2. Definir la función

```

```

    fun intercambia :: 'a × 'b ⇒ 'b × 'a
    tal que (intercambia p) es el par obtenido intercambiando las
    componentes del par p. Por ejemplo,
    intercambia (u,v) = (v,u)
    ----- *}

```

```

fun intercambia :: "'a × 'b ⇒ 'b × 'a" where
  "intercambia (x,y) = (y,x)"

```

```

value "intercambia (u,v)" -- "= (v,u)"

```

```

text {* -----
  Ejercicio 3. Definir, por recursión, la función
  inversa :: 'a list ⇒ 'a list
  tal que (inversa xs) es la lista obtenida invirtiendo el orden de los
  elementos de xs. Por ejemplo,
  inversa [a,d,c] = [c,d,a]
  ----- *}

```

```

fun inversa :: "'a list ⇒ 'a list" where
  "inversa [] = []"
| "inversa (x#xs) = inversa xs @ [x]"

```

```

value "inversa [a,d,c]" -- "= [c,d,a]"

```

```

text {* -----
  Ejercicio 4. Definir la función
  repite :: nat ⇒ 'a ⇒ 'a list
  tal que (repite n x) es la lista formada por n copias del elemento
  x. Por ejemplo,
  repite 3 a = [a,a,a]
  ----- *}

```

```

fun repite :: "nat ⇒ 'a ⇒ 'a list" where
  "repite 0 x = []"
| "repite (Suc n) x = x # (repite n x)"

```

```

value "repite 3 a" -- "= [a,a,a]"

```

```

text {* -----
  Ejercicio 5. Definir la función
  conc :: 'a list ⇒ 'a list ⇒ 'a list
  tal que (conc xs ys) es la concatenación de las listas xs e ys. Por
  ejemplo,
  conc [a,d] [b,d,a,c] = [a,d,b,d,a,c]

```

```

----- *}

fun conc :: "'a list  $\Rightarrow$  'a list  $\Rightarrow$  'a list" where
  "conc []      ys = ys"
| "conc (x#xs) ys = x # (conc xs ys)"

value "conc [a,d] [b,d,a,c]" -- "= [a,d,b,d,a,c]"

text {* -----
  Ejercicio 6. Definir la función
    coge :: nat  $\Rightarrow$  'a list  $\Rightarrow$  'a list
  tal que (coge n xs) es la lista de los n primeros elementos de xs. Por
  ejemplo,
    coge 2 [a,c,d,b,e] = [a,c]
  ----- *}

fun coge :: "nat  $\Rightarrow$  'a list  $\Rightarrow$  'a list" where
  "coge n []      = []"
| "coge 0 xs      = []"
| "coge (Suc n) (x#xs) = x # (coge n xs)"

value "coge 2 [a,c,d,b,e]" -- "= [a,c]"

text {* -----
  Ejercicio 7. Definir la función
    elimina :: nat  $\Rightarrow$  'a list  $\Rightarrow$  'a list
  tal que (elimina n xs) es la lista obtenida eliminando los n primeros
  elementos de xs. Por ejemplo,
    elimina 2 [a,c,d,b,e] = [d,b,e]
  ----- *}

fun elimina :: "nat  $\Rightarrow$  'a list  $\Rightarrow$  'a list" where
  "elimina n []      = []"
| "elimina 0 xs      = xs"
| "elimina (Suc n) (x#xs) = elimina n xs"

value "elimina 2 [a,c,d,b,e]" -- "= [d,b,e]"

text {* -----
  Ejercicio 8. Definir la función
    esVacia :: 'a list  $\Rightarrow$  bool
  tal que (esVacia xs) se verifica si xs es la lista vacía. Por ejemplo,
    esVacia [] = True
    esVacia [1] = False
  ----- *}

```

```

fun esVacia :: "'a list  $\Rightarrow$  bool" where
  "esVacia []      = True"
| "esVacia (x#xs) = False"

value "esVacia []"  -- "= True"
value "esVacia [1]" -- "= False"

text {* -----
  Ejercicio 9. Definir la función
    inversaAc :: 'a list  $\Rightarrow$  'a list
  tal que (inversaAc xs) es a inversa de xs calculada usando
  acumuladores. Por ejemplo,
    inversaAc [a,c,b,e] = [e,b,c,a]
  ----- *}

fun inversaAcAux :: "'a list  $\Rightarrow$  'a list  $\Rightarrow$  'a list" where
  "inversaAcAux [] ys      = ys"
| "inversaAcAux (x#xs) ys = inversaAcAux xs (x#ys)"

fun inversaAc :: "'a list  $\Rightarrow$  'a list" where
  "inversaAc xs = inversaAcAux xs []"

value "inversaAc [a,c,b,e]" -- "= [e,b,c,a]"

text {* -----
  Ejercicio 10. Definir la función
    sum :: nat list  $\Rightarrow$  nat
  tal que (sum xs) es la suma de los elementos de xs. Por ejemplo,
    sum [3,2,5] = 10
  ----- *}

fun sum :: "nat list  $\Rightarrow$  nat" where
  "sum []      = 0"
| "sum (x#xs) = x + sum xs"

value "sum [3,2,5]" -- "= 10"

text {* -----
  Ejercicio 11. Definir la función
    map :: ('a  $\Rightarrow$  'b)  $\Rightarrow$  'a list  $\Rightarrow$  'b list
  tal que (map f xs) es la lista obtenida aplicando la función f a los
  elementos de xs. Por ejemplo,
    map ( $\lambda x. 2*x$ ) [3,2,5] = [6,4,10]
  ----- *}

```

```
fun map :: "('a ⇒ 'b) ⇒ 'a list ⇒ 'b list" where
  "map f []      = []"
| "map f (x#xs) = (f x) # map f xs"

value "map (λx. 2*x) [3::nat,2,5]" -- " = [6,4,10]"

end
```

Capítulo 2

Razonamiento sobre programas

2.1. Tema: Razonamiento sobre programas

```
chapter {* Tema 2: Razonamiento sobre programas *}

theory T2
imports Main
begin

text {*
  En este tema se demuestra con Isabelle las propiedades de los
  programas funcionales como se expone en el tema 8 del curso
  "Informática" que puede leerse en
  http://www.cs.us.es/~jalonso/cursos/ilm/temas/tema-8t.pdf

  Todas las demostraciones se hacen automáticamente por simplificación e
  inducción.
*}

text {* -----
  Ejemplo 1. Definir, por recursión, la función
    longitud :: 'a list  $\Rightarrow$  nat
  tal que (longitud xs) es la longitud de la listas xs. Por ejemplo,
    longitud [4,2,5] = 3
  ----- *}

fun longitud :: "'a list  $\Rightarrow$  nat" where
  "longitud []      = 0"
| "longitud (x#xs) = 1 + longitud xs"

value "longitud [4,2,5]" -- "= 3"
```

```

text {* -----
  Ejemplo 2. Demostrar que
    longitud [4,2,5] = 3
  ----- *}

```

```

lemma "longitud [4,2,5] = 3"
by simp

```

```

text {* -----
  Ejemplo 3. Definir la función
    fun intercambia :: 'a × 'b ⇒ 'b × 'a
  tal que (intercambia p) es el par obtenido intercambiando las
  componentes del par p. Por ejemplo,
    intercambia (u,v) = (v,u)
  ----- *}

```

```

fun intercambia :: "'a × 'b ⇒ 'b × 'a" where
  "intercambia (x,y) = (y,x)"

```

```

value "intercambia (u,v)" -- "= (v,u)"

```

```

text {* -----
  Ejemplo 4. Demostrar que
    intercambia (intercambia (x,y)) = (x,y)
  ----- *}

```

```

lemma "intercambia (intercambia (x,y)) = (x,y)"
by simp

```

```

text {* -----
  Ejemplo 5. Definir, por recursión, la función
    inversa :: 'a list ⇒ 'a list
  tal que (inversa xs) es la lista obtenida invirtiendo el orden de los
  elementos de xs. Por ejemplo,
    inversa [a,d,c] = [c,d,a]
  ----- *}

```

```

fun inversa :: "'a list ⇒ 'a list" where
  "inversa [] = []"
| "inversa (x#xs) = inversa xs @ [x]"

```

```

value "inversa [a,d,c]" -- "= [c,d,a]"

```

```

text {* -----
  Ejemplo 6. Demostrar que

```

```

    inversa [x] = [x]
    ----- *}

lemma "inversa [x] = [x]"
by simp

text {* -----
  Ejemplo 7. Definir la función
    repite :: nat ⇒ 'a ⇒ 'a list
  tal que (repite n x) es la lista formada por n copias del elemento
  x. Por ejemplo,
    repite 3 a = [a,a,a]
  ----- *}

fun repite :: "nat ⇒ 'a ⇒ 'a list" where
  "repite 0 x      = []"
| "repite (Suc n) x = x # (repite n x)"

value "repite 3 a" -- "= [a,a,a]"

text {* -----
  Ejemplo 8. Demostrar que
    longitud (repite n x) = n
  ----- *}

lemma "longitud (repite n x) = n"
by (induct n) auto

text {* -----
  Ejemplo 9. Definir la función
    conc :: 'a list ⇒ 'a list ⇒ 'a list
  tal que (conc xs ys) es la concatenación de las listas xs e ys. Por
  ejemplo,
    conc [a,d] [b,d,a,c] = [a,d,b,d,a,c]
  ----- *}

fun conc :: "'a list ⇒ 'a list ⇒ 'a list" where
  "conc []      ys = ys"
| "conc (x#xs) ys = x # (conc xs ys)"

value "conc [a,d] [b,d,a,c]" -- "= [a,d,b,d,a,c]"

text {* -----
  Ejemplo 10. Demostrar que
    conc xs (conc ys zs) = (conc xs ys) zs

```

```

----- *}

lemma "conc xs (conc ys zs) = conc (conc xs ys) zs"
by (induct xs) auto

text {* -----
  Ejemplo 11. Refutar que
    conc xs ys = conc ys xs
----- *}

lemma "conc xs ys = conc ys xs"
quickcheck
oops

text {* Encuentra el contraejemplo,
  xs = [a\<^isub>2]
  ys = [a\<^isub>1] *}

text {* -----
  Ejemplo 12. Demostrar que
    conc xs [] = xs
----- *}

lemma "conc xs [] = xs"
by (induct xs) auto

text {* -----
  Ejemplo 13. Demostrar que
    longitud (conc xs ys) = longitud xs + longitud ys
----- *}

lemma "longitud (conc xs ys) = longitud xs + longitud ys"
by (induct xs) auto

text {* -----
  Ejemplo 14. Definir la función
    coge :: nat ⇒ 'a list ⇒ 'a list
  tal que (coge n xs) es la lista de los n primeros elementos de xs. Por
  ejemplo,
    coge 2 [a,c,d,b,e] = [a,c]
----- *}

fun coge :: "nat ⇒ 'a list ⇒ 'a list" where
  "coge n []          = []"
| "coge 0 xs          = []"
```

```
| "coge (Suc n) (x#xs) = x # (coge n xs)"

value "coge 2 [a,c,d,b,e]" -- "= [a,c]"

text {* -----
  Ejemplo 15. Definir la función
    elimina :: nat ⇒ 'a list ⇒ 'a list
  tal que (elimina n xs) es la lista obtenida eliminando los n primeros
  elementos de xs. Por ejemplo,
    elimina 2 [a,c,d,b,e] = [d,b,e]
  ----- *}

fun elimina :: "nat ⇒ 'a list ⇒ 'a list" where
  "elimina n []          = []"
| "elimina 0 xs          = xs"
| "elimina (Suc n) (x#xs) = elimina n xs"

value "elimina 2 [a,c,d,b,e]" -- "= [d,b,e]"

text {* -----
  Ejemplo 16. Demostrar que
    conc (coge n xs) (elimina n xs) = xs
  ----- *}

lemma "conc (coge n xs) (elimina n xs) = xs"
by (induct rule: coge.induct) auto

text {* coge.induct es el esquema de inducción asociado a la definición
  de la función coge.
  
$$\begin{aligned} & \llbracket \bigwedge n. P\ n\ []; \\ & \quad \bigwedge x\ xs. P\ 0\ (x\#xs); \\ & \quad \bigwedge n\ x\ xs. P\ n\ xs \implies P\ (Suc\ n)\ (x\#xs) \rrbracket \\ & \implies P\ n\ x \end{aligned}$$

  Puede verse usando "thm coge.induct". *}

text {* -----
  Ejemplo 17. Definir la función
    esVacia :: 'a list ⇒ bool
  tal que (esVacia xs) se verifica si xs es la lista vacía. Por ejemplo,
    esVacia [] = True
    esVacia [1] = False
  ----- *}

fun esVacia :: "'a list ⇒ bool" where
  "esVacia [] = True"
```

```

| "esVacia (x#xs) = False"

value "esVacia []" -- "= True"
value "esVacia [1]" -- "= False"

text {* -----
  Ejemplo 18. Demostrar que
    esVacia xs = esVacia (conc xs xs)
  ----- *}

lemma "esVacia xs = esVacia (conc xs xs)"
by (induct xs) auto

text {* -----
  Ejemplo 19. Definir la función
    inversaAc :: 'a list ⇒ 'a list
  tal que (inversaAc xs) es a inversa de xs calculada usando
  acumuladores. Por ejemplo,
    inversaAc [a,c,b,e] = [e,b,c,a]
  ----- *}

fun inversaAcAux :: "'a list ⇒ 'a list ⇒ 'a list" where
  "inversaAcAux [] ys = ys"
| "inversaAcAux (x#xs) ys = inversaAcAux xs (x#ys)"

fun inversaAc :: "'a list ⇒ 'a list" where
  "inversaAc xs = inversaAcAux xs []"

value "inversaAc [a,c,b,e]" -- "= [e,b,c,a]"

text {* -----
  Ejemplo 20. Demostrar que
    inversaAcAux xs ys = (inversa xs)@ys
  ----- *}

lemma inversaAcAux_es_inversa:
  "inversaAcAux xs ys = (inversa xs)@ys"
by (induct xs arbitrary: ys) auto

text {* -----
  Ejemplo 21. Demostrar que
    inversaAc xs = inversa xs
  ----- *}

corollary "inversaAc xs = inversa xs"

```

```

by (simp add: inversaAcAux_es_inversa)

text {* -----
  Ejemplo 22. Definir la función
    sum :: nat list ⇒ nat
  tal que (sum xs) es la suma de los elementos de xs. Por ejemplo,
    sum [3,2,5] = 10
  ----- *}

fun sum :: "nat list ⇒ nat" where
  "sum []      = 0"
| "sum (x#xs) = x + sum xs"

value "sum [3,2,5]" -- "= 10"

text {* -----
  Ejemplo 23. Definir la función
    map :: ('a ⇒ 'b) ⇒ 'a list ⇒ 'b list
  tal que (map f xs) es la lista obtenida aplicando la función f a los
  elementos de xs. Por ejemplo,
    map (λx. 2*x) [3,2,5] = [6,4,10]
  ----- *}

fun map :: "('a ⇒ 'b) ⇒ 'a list ⇒ 'b list" where
  "map f []      = []"
| "map f (x#xs) = (f x) # map f xs"

value "map (λx. 2*x) [3::nat,2,5]" -- "= [6,4,10]"

text {* -----
  Ejemplo 24. Demostrar que
    sum (map (λx. 2*x) xs) = 2 * (sum xs)
  ----- *}

lemma "sum (map (λx. 2*x) xs) = 2 * (sum xs)"
by (induct xs) auto

text {* -----
  Ejemplo 25. Demostrar que
    longitud (map f xs) = longitud xs
  ----- *}

lemma "longitud (map f xs) = longitud xs"
by (induct xs) auto

```

```
section {* Referencias *}
```

```
text {*
```

- J.A. Alonso. "Razonamiento sobre programas" <http://goo.gl/R0603>
- G. Hutton. "Programming in Haskell". Cap. 13 "Reasoning about programs".
- S. Thompson. "Haskell: the Craft of Functional Programming, 3rd Edition. Cap. 8 "Reasoning about programs".
- L. Paulson. "ML for the Working Programmer, 2nd Edition". Cap. 6. "Reasoning about functional programs".

```
*}
```

```
end
```

2.2. Ejercicios: Razonamiento sobre programas

```
chapter {* T2R1: Razonamiento sobre programas *}
```

```
theory T2R1
```

```
imports Main
```

```
begin
```

```
text {* -----
  Ejercicio 1. Definir la función
    sumaImpares :: nat ⇒ nat
  tal que (sumaImpares n) es la suma de los n primeros números
  impares. Por ejemplo,
    sumaImpares 5 = 25
  ----- *}

```

```
fun sumaImpares :: "nat ⇒ nat" where
  "sumaImpares 0 = 0"
| "sumaImpares (Suc n) = sumaImpares n + (2*n+1)"

```

```
value "sumaImpares 5" -- "= 25"
```

```
text {* -----
  Ejercicio 2. Demostrar que
    sumaImpares n = n*n
  ----- *}

```

```
lemma "sumaImpares n = n*n"
by (induct n) auto
```

```

text {* -----
  Ejercicio 3. Definir la función
    sumaPotenciasDeDosMasUno :: nat ⇒ nat
  tal que
    (sumaPotenciasDeDosMasUno n) = 1 + 2^0 + 2^1 + 2^2 + ... + 2^n.
  Por ejemplo,
    sumaPotenciasDeDosMasUno 3 = 16
  ----- *}

fun sumaPotenciasDeDosMasUno :: "nat ⇒ nat" where
  "sumaPotenciasDeDosMasUno 0 = 2"
| "sumaPotenciasDeDosMasUno (Suc n) =
    sumaPotenciasDeDosMasUno n + 2^(n+1)"

value "sumaPotenciasDeDosMasUno 3" -- "= 16"

text {* -----
  Ejercicio 4. Demostrar que
    sumaPotenciasDeDosMasUno n = 2^(n+1)
  ----- *}

lemma "sumaPotenciasDeDosMasUno n = 2^(n+1)"
by (induct n) auto

text {* -----
  Ejercicio 5. Definir la función
    copia :: nat ⇒ 'a ⇒ 'a list
  tal que (copia n x) es la lista formado por n copias del elemento
  x. Por ejemplo,
    copia 3 x = [x,x,x]
  ----- *}

fun copia :: "nat ⇒ 'a ⇒ 'a list" where
  "copia 0 x = []"
| "copia (Suc n) x = x # copia n x"

value "copia 3 x" -- "= [x,x,x]"

text {* -----
  Ejercicio 6. Definir la función
    todos :: ('a ⇒ bool) ⇒ 'a list ⇒ bool
  tal que (todos p xs) se verifica si todos los elementos de xs cumplen
  la propiedad p. Por ejemplo,
    todos (λx. x>(1::nat)) [2,6,4] = True
    todos (λx. x>(2::nat)) [2,6,4] = False

```

Nota: La conjunción se representa por $\wedge$

```

----- *}

fun todos :: "('a  $\Rightarrow$  bool)  $\Rightarrow$  'a list  $\Rightarrow$  bool" where
  "todos p []      = True"
| "todos p (x#xs) = (p x  $\wedge$  todos p xs)"

value "todos ( $\lambda$ x. x>(1::nat)) [2,6,4]" -- "= True"
value "todos ( $\lambda$ x. x>(2::nat)) [2,6,4]" -- "= False"

text {* -----
  Ejercicio 7. Demostrar que todos los elementos de (copia n x) son
  iguales a x.
----- *}

lemma "todos ( $\lambda$ y. y=x) (copia n x)"
by (induct n) auto

text {* -----
  Ejercicio 8. Definir la función
    factR :: nat  $\Rightarrow$  nat
  tal que (factR n) es el factorial de n. Por ejemplo,
    factR 4 = 24
----- *}

fun factR :: "nat  $\Rightarrow$  nat" where
  "factR 0      = 1"
| "factR (Suc n) = Suc n * factR n"

value "factR 4" -- "= 24"

text {* -----
  Ejercicio 9. Se considera la siguiente definición iterativa de la
  función factorial
    factI :: "nat  $\Rightarrow$  nat" where
    factI n = factI' n 1

    factI' :: nat  $\Rightarrow$  nat  $\Rightarrow$  nat" where
    factI' 0      x = x
    factI' (Suc n) x = factI' n (Suc n)*x
  Demostrar que, para todo n y todo x, se tiene
    factI' n x = x * factR n
----- *}

fun factI' :: "nat  $\Rightarrow$  nat  $\Rightarrow$  nat" where

```

```

    "factI' 0          x = x"
| "factI' (Suc n) x = factI' n (Suc n)*x"

fun factI :: "nat  $\Rightarrow$  nat" where
    "factI n = factI' n 1"

value "factI 4" -- "= 24"

lemma fact: "factI' n x = x * factR n"
by (induct n arbitrary: x) auto

text {* -----
    Ejercicio 10. Demostrar que
        factI n = factR n
    ----- *}

corollary "factI n = factR n"
by (simp add: fact)

text {* -----
    Ejercicio 11. Definir, recursivamente y sin usar (@), la función
        amplia :: 'a list  $\Rightarrow$  'a  $\Rightarrow$  'a list
    tal que (amplia xs y) es la lista obtenida añadiendo el elemento y al
    final de la lista xs. Por ejemplo,
        amplia [d,a] t = [d,a,t]
    ----- *}

fun amplia :: "'a list  $\Rightarrow$  'a  $\Rightarrow$  'a list" where
    "amplia []      y = [y]"
| "amplia (x#xs) y = x # amplia xs y"

value "amplia [d,a] t" -- "= [d,a,t]"

text {* -----
    Ejercicio 12. Demostrar que
        amplia xs y = xs @ [y]
    ----- *}

lemma "amplia xs y = xs @ [y]"
by (induct xs) auto

end

```


Capítulo 3

Deducción natural en lógica proposicional

3.1. Tema: Deducción natural en lógica proposicional

```
chapter {* Tema 3: Deducción natural proposicional con Isabelle/HOL *}

theory T3
imports Main
begin

text {*
  En esta sección se presentan los ejemplos del tema de deducción natural
  proposicional siguiendo la presentación de Huth y Ryan en su libro
  "Logic in Computer Science" http://goo.gl/qsVpY y, más concretamente,
  a la forma como se explica en la asignatura de "Lógica informática" (LI)
  http://goo.gl/AwDiv

  La página al lado de cada ejemplo indica la página de las transparencias
  de LI donde se encuentra la demostración. *}

subsection {* Reglas de la conjunción *}

text {*
  Ejemplo 1 (p. 4). Demostrar que
   $p \wedge q, r \vdash q \wedge r$ .
  *}

-- "La demostración detallada es"
lemma ejemplo_1_1:
```

```

    assumes 1: "p ∧ q" and
             2: "r"
    shows "q ∧ r"
proof -
  have 3: "q" using 1 by (rule conjunct2)
  show 4: "q ∧ r" using 3 2 by (rule conjI)
qed

text {*
  Notas sobre el lenguaje: En la demostración anterior se ha usado
  · "assumes" para indicar las hipótesis,
  · "and" para separar las hipótesis,
  · "shows" para indicar la conclusión,
  · "proof" para iniciar la prueba,
  · "qed" para terminar la pruebas,
  · "-" (después de "proof") para no usar el método por defecto,
  · "have" para establecer un paso,
  · "using" para usar hechos en un paso,
  · "by (rule ..)" para indicar la regla con la que se prueba un hecho,
  · "show" para establecer la conclusión.

  Notas sobre la lógica: Las reglas de la conjunción son
  · conjI:       $\llbracket P; Q \rrbracket \Longrightarrow P \wedge Q$ 
  · conjunct1:   $P \wedge Q \Longrightarrow P$ 
  · conjunct2:   $P \wedge Q \Longrightarrow Q$ 
*}

text {* Se pueden dejar implícitas las reglas como sigue *}

lemma ejemplo_1_2:
  assumes 1: "p ∧ q" and
           2: "r"
  shows "q ∧ r"
proof -
  have 3: "q" using 1 ..
  show 4: "q ∧ r" using 3 2 ..
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "..." para indicar que se prueba por la regla correspondiente. *}

text {* Se pueden eliminar las etiquetas como sigue *}

lemma ejemplo_1_3:

```

```

    assumes "p ∧ q"
           "r"
    shows   "q ∧ r"
proof -
  have "q" using assms(1) ..
  thus "q ∧ r" using assms(2) ..
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "assms(n)" para indicar la hipótesis n y
  · "thus" para demostrar la conclusión usando el hecho anterior.
  Además, no es necesario usar and entre las hipótesis. *}

text {* Se puede automatizar la demostración como sigue *}

lemma ejemplo_1_4:
  assumes "p ∧ q"
         "r"
  shows   "q ∧ r"
using assms
by auto

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "assms" para indicar las hipótesis y
  · "by auto" para demostrar la conclusión automáticamente. *}

text {* Se puede automatizar totalmente la demostración como sigue *}

lemma ejemplo_1_5:
  "[p ∧ q; r] ⇒ q ∧ r"
by auto

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "[ ... ]" para representar las hipótesis,
  · ";" para separar las hipótesis y
  · "⇒" para separar las hipótesis de la conclusión. *}

text {* Se puede hacer la demostración por razonamiento hacia atrás,
  como sigue *}

lemma ejemplo_1_6:
  assumes "p ∧ q"

```

```

    and "r"
  shows "q  $\wedge$  r"
proof (rule conjI)
  show "q" using assms(1) by (rule conjunct2)
next
  show "r" using assms(2) by this
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "proof (rule r)" para indicar que se hará la demostración con la
    regla r,
  · "next" para indicar el comienzo de la prueba del siguiente
    subobjetivo,
  · "this" para indicar el hecho actual. *}

text {* Se pueden dejar implícitas las reglas como sigue *}

lemma ejemplo_1_7:
  assumes "p  $\wedge$  q"
    "r"
  shows "q  $\wedge$  r"
proof
  show "q" using assms(1) ..
next
  show "r" using assms(2) .
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "." para indicar por el hecho actual. *}

subsection {* Reglas de la doble negación *}

text {*
  La regla de eliminación de la doble negación es
  · notnotD:  $\neg\neg P \Rightarrow P$ 

  Para ajustarnos al tema de LI vamos a introducir la siguiente regla de
  introducción de la doble negación
  · notnotI:  $P \Rightarrow \neg\neg P$ 
  aunque, de momento, no detallamos su demostración.
  *}

lemma notnotI [intro!]: "P  $\Rightarrow \neg\neg P$ "

```

```

by auto

text {*
  Ejemplo 2. (p. 5)
   $p, \neg\neg(q \wedge r) \vdash \neg\neg p \wedge r$ 
*}

-- "La demostración detallada es"
lemma ejemplo_2_1:
  assumes 1: "p" and
          2: " $\neg\neg(q \wedge r)$ "
  shows   " $\neg\neg p \wedge r$ "
proof -
  have 3: " $\neg\neg p$ " using 1 by (rule notnotI)
  have 4: " $q \wedge r$ " using 2 by (rule notnotD)
  have 5: "r" using 4 by (rule conjunct2)
  show 6: " $\neg\neg p \wedge r$ " using 3 5 by (rule conjI)
qed

-- "La demostración estructurada es"
lemma ejemplo_2_2:
  assumes "p"
          " $\neg\neg(q \wedge r)$ "
  shows   " $\neg\neg p \wedge r$ "
proof -
  have " $\neg\neg p$ " using assms(1) ..
  have " $q \wedge r$ " using assms(2) by (rule notnotD)
  hence "r" ..
  with ' $\neg\neg p$ ' show " $\neg\neg p \wedge r$ " ..
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "hence" para indicar que se tiene por el hecho anterior,
  · '...' para referenciar un hecho y
  · "with P show Q" para indicar que con el hecho anterior junto con el
    hecho P se demuestra Q. *}

-- "La demostración automática es"
lemma ejemplo_2_3:
  assumes "p"
          " $\neg\neg(q \wedge r)$ "
  shows   " $\neg\neg p \wedge r$ "
using assms
by auto

```

```
text {* Se puede demostrar hacia atrás *}

```

```
lemma ejemplo_2_4:
  assumes "p"
           "¬¬(q ∧ r)"
  shows   "¬¬p ∧ r"
proof (rule conjI)
  show "¬¬p" using assms(1) by (rule notnotI)
next
  have "q ∧ r" using assms(2) by (rule notnotD)
  thus "r" by (rule conjunct2)
qed

```

```
text {* Se puede eliminar las reglas en la demostración anterior, como
sigue: *}

```

```
lemma ejemplo_2_5:
  assumes "p"
           "¬¬(q ∧ r)"
  shows   "¬¬p ∧ r"
proof
  show "¬¬p" using assms(1) ..
next
  have "q ∧ r" using assms(2) by (rule notnotD)
  thus "r" ..
qed

```

```
subsection {* Regla de eliminación del condicional *}

```

```
text {*
  La regla de eliminación del condicional es la regla del modus ponens
  · mp:  $\llbracket P \rightarrow Q; P \rrbracket \Longrightarrow Q$ 
*}

```

```
text {*
  Ejemplo 3. (p. 6) Demostrar que
   $\neg p \wedge q, \neg p \wedge q \rightarrow r \vee \neg p \vdash r \vee \neg p$ 
*}

```

```
-- "La demostración detallada es"

```

```
lemma ejemplo_3_1:
  assumes 1: "¬p ∧ q" and
          2: "¬p ∧ q → r ∨ ¬p"
  shows   "r ∨ ¬p"

```

```

proof -
  show "r v ¬p" using 2 1 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejemplo_3_2:
  assumes "¬p ∧ q"
    "¬p ∧ q → r v ¬p"
  shows "r v ¬p"
proof -
  show "r v ¬p" using assms(2,1) ..
qed

-- "La demostración automática es"
lemma ejemplo_3_3:
  assumes "¬p ∧ q"
    "¬p ∧ q → r v ¬p"
  shows "r v ¬p"
using assms
by auto

text {*
  Ejemplo 4 (p. 6) Demostrar que
     $p, p \rightarrow q, p \rightarrow (q \rightarrow r) \vdash r$ 
  *}

-- "La demostración detallada es"
lemma ejemplo_4_1:
  assumes 1: "p" and
    2: "p → q" and
    3: "p → (q → r)"
  shows "r"
proof -
  have 4: "q" using 2 1 by (rule mp)
  have 5: "q → r" using 3 1 by (rule mp)
  show 6: "r" using 5 4 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejemplo_4_2:
  assumes "p"
    "p → q"
    "p → (q → r)"
  shows "r"
proof -

```

```

    have "q" using assms(2,1) ..
    have "q → r" using assms(3,1) ..
    thus "r" using 'q' ..
qed

-- "La demostración automática es"
lemma ejemplo_4_3:
  "[[p; p → q; p → (q → r)]] ⇒ r"
by auto

subsection {* Regla derivada del modus tollens *}

text {*
  Para ajustarnos al tema de LI vamos a introducir la regla del modus
  tollens
  · mt:  $\llbracket F \rightarrow G; \neg G \rrbracket \Rightarrow \neg F$ 
  aunque, de momento, sin detallar su demostración.
*}

lemma mt: " $\llbracket F \rightarrow G; \neg G \rrbracket \Rightarrow \neg F$ "
by auto

text {*
  Ejemplo 5 (p. 7). Demostrar
   $p \rightarrow (q \rightarrow r), p, \neg r \vdash \neg q$ 
*}

-- "La demostración detallada es"
lemma ejemplo_5_1:
  assumes 1: "p → (q → r)" and
    2: "p" and
    3: "¬r"
  shows "¬q"
proof -
  have 4: "q → r" using 1 2 by (rule mp)
  show "¬q" using 4 3 by (rule mt)
qed

-- "La demostración estructurada es"
lemma ejemplo_5_2:
  assumes "p → (q → r)"
    "p"
    "¬r"
  shows "¬q"
proof -

```

```

    have "q → r" using assms(1,2) ..
    thus "¬q" using assms(3) by (rule mt)
qed

-- "La demostración automática es"
lemma ejemplo_5_3:
  assumes "p → (q → r)"
    "p"
    "¬r"
  shows "¬q"
using assms
by auto

text {*
  Ejemplo 6. (p. 7) Demostrar
     $\neg p \rightarrow q, \neg q \vdash p$ 
*}

-- "La demostración detallada es"
lemma ejemplo_6_1:
  assumes 1: "¬p → q" and
    2: "¬q"
  shows "p"
proof -
  have 3: "¬¬p" using 1 2 by (rule mt)
  show "p" using 3 by (rule notnotD)
qed

-- "La demostración estructurada es"
lemma ejemplo_6_2:
  assumes "¬p → q"
    "¬q"
  shows "p"
proof -
  have "¬¬p" using assms(1,2) by (rule mt)
  thus "p" by (rule notnotD)
qed

-- "La demostración automática es"
lemma ejemplo_6_3:
  "[¬p → q; ¬q] ⇒ p"
by auto

text {*
  Ejemplo 7. (p. 7) Demostrar

```

```

     $p \rightarrow \neg q, q \vdash \neg p$ 
  *}

-- "La demostración detallada es"
lemma ejemplo_7_1:
  assumes 1: " $p \rightarrow \neg q$ " and
    2: " $q$ "
  shows " $\neg p$ "
proof -
  have 3: " $\neg\neg q$ " using 2 by (rule notnotI)
  show " $\neg p$ " using 1 3 by (rule mt)
qed

-- "La demostración detallada es"
lemma ejemplo_7_2:
  assumes " $p \rightarrow \neg q$ "
    " $q$ "
  shows " $\neg p$ "
proof -
  have " $\neg\neg q$ " using assms(2) by (rule notnotI)
  with assms(1) show " $\neg p$ " by (rule mt)
qed

-- "La demostración estructurada es"
lemma ejemplo_7_3:
  "[ $p \rightarrow \neg q; q$ ]  $\implies \neg p$ "
by auto

subsection {* Regla de introducción del condicional *}

text {
  La regla de introducción del condicional es
  ·  $\text{impI}: (P \implies Q) \implies P \rightarrow Q$ 
}

text {
  Ejemplo 8. (p. 8) Demostrar
     $p \rightarrow q \vdash \neg q \rightarrow \neg p$ 
}

-- "La demostración detallada es"
lemma ejemplo_8_1:
  assumes 1: " $p \rightarrow q$ "
  shows " $\neg q \rightarrow \neg p$ "
proof -

```

```

{ assume 2: "¬q"
  have "¬p" using 1 2 by (rule mt) }
thus "¬q → ¬p" by (rule impI)
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "{ ... }" para representar una caja. *}

-- "La demostración estructurada es"
lemma ejemplo_8_2:
  assumes "p → q"
  shows "¬q → ¬p"
proof
  assume "¬q"
  with assms show "¬p" by (rule mt)
qed

-- "La demostración automática es"
lemma ejemplo_8_3:
  assumes "p → q"
  shows "¬q → ¬p"
using assms
by auto

text {*
  Ejemplo 9. (p. 9) Demostrar
     $\neg q \rightarrow \neg p \vdash p \rightarrow \neg \neg q$ 
  *}

-- "La demostración detallada es"
lemma ejemplo_9_1:
  assumes 1: "¬q → ¬p"
  shows "p → ¬¬q"
proof -
  { assume 2: "p"
    have 3: "¬¬p" using 2 by (rule notnotI)
    have "¬¬q" using 1 3 by (rule mt) }
  thus "p → ¬¬q" by (rule impI)
qed

-- "La demostración estructurada es"
lemma ejemplo_9_2:
  assumes "¬q → ¬p"
  shows "p → ¬¬q"

```

```

proof
  assume "p"
  hence "¬¬p" by (rule notnotI)
  with assms show "¬¬q" by (rule mt)
qed

-- "La demostración automática es"
lemma ejemplo_9_3:
  assumes "¬q → ¬p"
  shows "p → ¬¬q"
using assms
by auto

text {*
  Ejemplo 10 (p. 9). Demostrar
     $\vdash p \rightarrow p$ 
*}

-- "La demostración detallada es"
lemma ejemplo_10_1:
  "p → p"
proof -
  { assume 1: "p"
    have "p" using 1 by this }
  thus "p → p" by (rule impI)
qed

-- "La demostración estructurada es"
lemma ejemplo_10_2:
  "p → p"
proof (rule impI)
qed

-- "La demostración automática es"
lemma ejemplo_10_3:
  "p → p"
by auto

text {*
  Ejemplo 11 (p. 10) Demostrar
     $\vdash (q \rightarrow r) \rightarrow ((\neg q \rightarrow \neg p) \rightarrow (p \rightarrow r))$ 
*}

-- "La demostración detallada es"
lemma ejemplo_11_1:

```

```

"(q → r) → ((¬q → ¬p) → (p → r))"
proof -
  { assume 1: "q → r"
    { assume 2: "¬q → ¬p"
      { assume 3: "p"
        have 4: "¬¬p" using 3 by (rule notnotI)
        have 5: "¬¬q" using 2 4 by (rule mt)
        have 6: "q" using 5 by (rule notnotD)
        have "r" using 1 6 by (rule mp) }
      hence "p → r" by (rule impI) }
    hence "(¬q → ¬p) → p → r" by (rule impI) }
  thus "(q → r) → ((¬q → ¬p) → p → r)" by (rule impI)
qed

-- "La demostración hacia atrás es"
lemma ejemplo_11_2:
  "(q → r) → ((¬q → ¬p) → (p → r))"
proof (rule impI)
  assume 1: "q → r"
  show "(¬q → ¬p) → (p → r)"
  proof (rule impI)
    assume 2: "¬q → ¬p"
    show "p → r"
    proof (rule impI)
      assume 3: "p"
      have 4: "¬¬p" using 3 by (rule notnotI)
      have 5: "¬¬q" using 2 4 by (rule mt)
      have 6: "q" using 5 by (rule notnotD)
      show "r" using 1 6 by (rule mp)
    qed
  qed
qed

-- "La demostración hacia atrás con reglas implícitas es"
lemma ejemplo_11_3:
  "(q → r) → ((¬q → ¬p) → (p → r))"
proof
  assume 1: "q → r"
  show "(¬q → ¬p) → (p → r)"
  proof
    assume 2: "¬q → ¬p"
    show "p → r"
    proof
      assume 3: "p"
      have 4: "¬¬p" using 3 ..
    
```

```

    have 5: "¬¬q" using 2 4 by (rule mt)
    have 6: "q" using 5 by (rule notnotD)
    show "r" using 1 6 ..
  qed
qed
qed

-- "La demostración sin etiquetas es"
lemma ejemplo_11_4:
  "(q → r) → ((¬q → ¬p) → (p → r))"
proof
  assume "q → r"
  show "(¬q → ¬p) → (p → r)"
  proof
    assume "¬q → ¬p"
    show "p → r"
    proof
      assume "p"
      hence "¬¬p" ..
      with '¬q → ¬p' have "¬¬q" by (rule mt)
      hence "q" by (rule notnotD)
      with 'q → r' show "r" ..
    qed
  qed
qed
qed

-- "La demostración automática es"
lemma ejemplo_11_5:
  "(q → r) → ((¬q → ¬p) → (p → r))"
by auto

subsection {* Reglas de la disyunción *}

text {
  Las reglas de la introducción de la disyunción son
  · disjI1:  $P \implies P \vee Q$ 
  · disjI2:  $Q \implies P \vee Q$ 
  La regla de eliminación de la disyunción es
  · disjE:  $\llbracket P \vee Q; P \implies R; Q \implies R \rrbracket \implies R$ 
*}

text {
  Ejemplo 12 (p. 11). Demostrar
   $p \vee q \vdash q \vee p$ 
*}

```

```
-- "La demostración detallada es"
lemma ejemplo_12_1:
  assumes "p v q"
  shows "q v p"
proof -
  have "p v q" using assms by this
  moreover
  { assume 2: "p"
    have "q v p" using 2 by (rule disjI2) }
  moreover
  { assume 3: "q"
    have "q v p" using 3 by (rule disjI1) }
  ultimately show "q v p" by (rule disjE)
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "moreover" para separar los bloques y
  · "ultimately" para unir los resultados de los bloques. *}

-- "La demostración detallada con reglas implícitas es"
lemma ejemplo_12_2:
  assumes "p v q"
  shows "q v p"
proof -
  note 'p v q'
  moreover
  { assume "p"
    hence "q v p" .. }
  moreover
  { assume "q"
    hence "q v p" .. }
  ultimately show "q v p" ..
qed

text {*
  Nota sobre el lenguaje: En la demostración anterior se ha usado
  · "note" para copiar un hecho. *}

-- "La demostración hacia atrás es"
lemma ejemplo_12_3:
  assumes 1: "p v q"
  shows "q v p"
using 1
```

```

proof (rule disjE)
  { assume 2: "p"
    show "q v p" using 2 by (rule disjI2) }
next
  { assume 3: "q"
    show "q v p" using 3 by (rule disjI1) }
qed

-- "La demostración hacia atrás con reglas implícitas es"
lemma ejemplo_12_4:
  assumes "p v q"
  shows "q v p"
using assms
proof
  { assume "p"
    thus "q v p" .. }
next
  { assume "q"
    thus "q v p" .. }
qed

-- "La demostración automática es"
lemma ejemplo_12_5:
  assumes "p v q"
  shows "q v p"
using assms
by auto

text {*
  Ejemplo 13. (p. 12) Demostrar
     $q \rightarrow r \vdash p \vee q \rightarrow p \vee r$ 
*}

-- "La demostración detallada es"
lemma ejemplo_13_1:
  assumes 1: "q → r"
  shows "p v q → p v r"
proof (rule impI)
  assume 2: "p v q"
  thus "p v r"
  proof (rule disjE)
    { assume 3: "p"
      show "p v r" using 3 by (rule disjI1) }
  next
    { assume 4: "q"

```

```

      have 5: "r" using 1 4 by (rule mp)
      show "p v r" using 5 by (rule disjI2) }
qed
qed

```

```
-- "La demostración estructurada es"
```

```

lemma ejemplo_13_2:
  assumes "q → r"
  shows "p v q → p v r"
proof
  assume "p v q"
  thus "p v r"
  proof
    { assume "p"
      thus "p v r" .. }
    next
    { assume "q"
      have "r" using assms 'q' ..
      thus "p v r" .. }
  qed
qed

```

```
-- "La demostración automática es"
```

```

lemma ejemplo_13_3:
  assumes "q → r"
  shows "p v q → p v r"
using assms
by auto

```

```
subsection {* Regla de copia *}
```

```

text {
  Ejemplo 14 (p. 13). Demostrar
     $\vdash p \rightarrow (q \rightarrow p)$ 
*}

```

```
-- "La demostración detallada es"
```

```

lemma ejemplo_14_1:
  "p → (q → p)"
proof (rule impI)
  assume 1: "p"
  show "q → p"
  proof (rule impI)
    assume "q"
    show "p" using 1 by this
  qed
qed

```

```

    qed
qed

-- "La demostración estructurada es"
lemma ejemplo_14_2:
  "p → (q → p)"
proof
  assume "p"
  thus "q → p" ..
qed

-- "La demostración automática es"
lemma ejemplo_14_3:
  "p → (q → p)"
by auto

subsection {* Reglas de la negación *}

text {*
  La regla de eliminación de lo falso es
  · FalseE: False ⇒ P
  La regla de eliminación de la negación es
  · notE: [¬P; P] ⇒ R
  La regla de introducción de la negación es
  · notI: (P ⇒ False) ⇒ ¬P
*}

text {*
  Ejemplo 15 (p. 15). Demostrar
  ¬p ∨ q ⊢ p → q
*}

-- "La demostración detallada es"
lemma ejemplo_15_1:
  assumes 1: "¬p ∨ q"
  shows "p → q"
proof (rule impI)
  assume 2: "p"
  note 1
  thus "q"
  proof (rule disjE)
    { assume 3: "¬p"
      show "q" using 3 2 by (rule notE) }
  next
    { assume 4: "q"

```

```

        show "q" using 4 by this}
qed
qed

-- "La demostración estructurada es"
lemma ejemplo_15_2:
  assumes "¬p ∨ q"
  shows "p → q"
proof
  assume "p"
  note '¬p ∨ q'
  thus "q"
  proof
    { assume "¬p"
      thus "q" using 'p' .. }
  next
    { assume "q"
      thus "q" . }
  qed
qed

-- "La demostración automática es"
lemma ejemplo_15_3:
  assumes "¬p ∨ q"
  shows "p → q"
using assms
by auto

text {*
  Ejemplo 16 (p. 16). Demostrar
     $p \rightarrow q, p \rightarrow \neg q \vdash \neg p$ 
*}

-- "La demostración detallada es"
lemma ejemplo_16_1:
  assumes 1: "p → q" and
          2: "p → ¬q"
  shows "¬p"
proof (rule notI)
  assume 3: "p"
  have 4: "q" using 1 3 by (rule mp)
  have 5: "¬q" using 2 3 by (rule mp)
  show False using 5 4 by (rule notE)
qed

```

```
-- "La demostración estructurada es"
lemma ejemplo_16_2:
  assumes "p  $\rightarrow$  q"
           "p  $\rightarrow$   $\neg$ q"
  shows " $\neg$ p"
proof
  assume "p"
  have "q" using assms(1) 'p' ..
  have " $\neg$ q" using assms(2) 'p' ..
  thus False using 'q' ..
qed

-- "La demostración automática es"
lemma ejemplo_16_3:
  assumes "p  $\rightarrow$  q"
           "p  $\rightarrow$   $\neg$ q"
  shows " $\neg$ p"
using assms
by auto

subsection {* Reglas del bicondicional *}

text {
  La regla de introducción del bicondicional es
  · iffI:  $\llbracket P \implies Q; Q \implies P \rrbracket \implies P \leftrightarrow Q$ 
  Las reglas de eliminación del bicondicional son
  · iffD1:  $\llbracket Q \leftrightarrow P; Q \rrbracket \implies P$ 
  · iffD2:  $\llbracket P \leftrightarrow Q; Q \rrbracket \implies P$ 
*}

text {
  Ejemplo 17 (p. 17) Demostrar
*}

-- "La demostración detallada es"
lemma ejemplo_17_1:
  "(p  $\wedge$  q)  $\leftrightarrow$  (q  $\wedge$  p)"
proof (rule iffI)
  { assume 1: "p  $\wedge$  q"
    have 2: "p" using 1 by (rule conjunct1)
    have 3: "q" using 1 by (rule conjunct2)
    show "q  $\wedge$  p" using 3 2 by (rule conjI) }
next
  { assume 4: "q  $\wedge$  p"
    have 5: "q" using 4 by (rule conjunct1)
```

```

    have 6: "p" using 4 by (rule conjunct2)
    show "p ∧ q" using 6 5 by (rule conjI) }
qed

-- "La demostración estructurada es"
lemma ejemplo_17_2:
  "(p ∧ q) ↔ (q ∧ p)"
proof
  { assume 1: "p ∧ q"
    have "p" using 1 ..
    have "q" using 1 ..
    show "q ∧ p" using 'q' 'p' .. }
next
  { assume 2: "q ∧ p"
    have "q" using 2 ..
    have "p" using 2 ..
    show "p ∧ q" using 'p' 'q' .. }
qed

-- "La demostración automática es"
lemma ejemplo_17_3:
  "(p ∧ q) ↔ (q ∧ p)"
by auto

text {*
  Ejemplo 18 (p. 18). Demostrar
     $p \leftrightarrow q, p \vee q \vdash p \wedge q$ 
*}

-- "La demostración detallada es"
lemma ejemplo_18_1:
  assumes 1: "p ↔ q" and
    2: "p ∨ q"
  shows "p ∧ q"
using 2
proof (rule disjE)
  { assume 3: "p"
    have 4: "q" using 1 3 by (rule iffD1)
    show "p ∧ q" using 3 4 by (rule conjI) }
next
  { assume 5: "q"
    have 6: "p" using 1 5 by (rule iffD2)
    show "p ∧ q" using 6 5 by (rule conjI) }
qed

```

```
-- "La demostración estructurada es"
lemma ejemplo_18_2:
  assumes "p  $\leftrightarrow$  q"
           "p  $\vee$  q"
  shows "p  $\wedge$  q"
using assms(2)
proof
  { assume "p"
    with assms(1) have "q" ..
    with 'p' show "p  $\wedge$  q" .. }
next
  { assume "q"
    with assms(1) have "p" ..
    thus "p  $\wedge$  q" using 'q' .. }
qed

-- "La demostración automática es"
lemma ejemplo_18_3:
  assumes "p  $\leftrightarrow$  q"
           "p  $\vee$  q"
  shows "p  $\wedge$  q"
using assms
by auto

subsection {* Reglas derivadas *}

subsubsection {* Regla del modus tollens *}

text {*
  Ejemplo 19 (p. 20) Demostrar la regla del modus tollens a partir de
  las reglas básicas.
*}

-- "La demostración detallada es"
lemma ejemplo_20_1:
  assumes 1: "F  $\rightarrow$  G" and
           2: " $\neg$ G"
  shows " $\neg$ F"
proof (rule notI)
  assume 3: "F"
  have 4: "G" using 1 3 by (rule mp)
  show False using 2 4 by (rule notE)
qed

-- "La demostración estructurada es"
```

```

lemma ejemplo_20_2:
  assumes "F  $\rightarrow$  G"
           " $\neg$ G"
  shows   " $\neg$ F"
proof
  assume "F"
  with assms(1) have "G" ..
  with assms(2) show False ..
qed

-- "La demostración automática es"
lemma ejemplo_20_3:
  assumes "F  $\rightarrow$  G"
           " $\neg$ G"
  shows   " $\neg$ F"
using assms
by auto

subsubsection {* Regla de la introducción de la doble negación *}

text {
  Ejemplo 21 (p. 21) Demostrar la regla de introducción de la doble negación a partir de las reglas básicas.
*}

-- "La demostración detallada es"
lemma ejemplo_21_1:
  assumes 1: "F"
  shows   " $\neg\neg$ F"
proof (rule notI)
  assume 2: " $\neg$ F"
  show False using 2 1 by (rule notE)
qed

-- "La demostración estructurada es"
lemma ejemplo_21_2:
  assumes "F"
  shows   " $\neg\neg$ F"
proof
  assume " $\neg$ F"
  thus False using assms ..
qed

-- "La demostración automática es"
lemma ejemplo_21_3:

```

```

    assumes "F"
    shows "¬¬F"
using assms
by auto

subsubsection {* Regla de reducción al absurdo *}

text {*
  La regla de reducción al absurdo en Isabelle se corresponde con la
  regla clásica de contradicción
  · ccontr:  $(\neg P \implies \text{False}) \implies P$ 
*}

subsubsection {* Ley del tercio excluso *}

text {*
  La ley del tercio excluso es
  · excluded_middle:  $\neg P \vee P$ 
*}

text {*
  Ejemplo 22 (p. 23). Demostrar la ley del tercio excluso a partir de
  las reglas básicas.
*}

-- "La demostración detallada es"
lemma ejemplo_22_1:
  "F  $\vee$   $\neg F$ "
proof (rule ccontr)
  assume 1: "¬(F  $\vee$   $\neg F$ )"
  thus False
  proof (rule notE)
    show "F  $\vee$   $\neg F$ "
    proof (rule disjI2)
      show "¬F"
      proof (rule notI)
        assume 2: "F"
        hence 3: "F  $\vee$   $\neg F$ " by (rule disjI1)
        show False using 1 3 by (rule notE)
      qed
    qed
  qed
qed

-- "La demostración estructurada es"
```

```

lemma ejemplo_22_2:
  "F v ¬F"
proof (rule ccontr)
  assume "¬(F v ¬F)"
  thus False
  proof (rule notE)
    show "F v ¬F"
    proof (rule disjI2)
      show "¬F"
      proof (rule notI)
        assume "F"
        hence "F v ¬F" ..
        with '¬(F v ¬F)' show False ..
      qed
    qed
  qed
qed

-- "La demostración automática es"
lemma ejemplo_22_3:
  "F v ¬F"
using assms
by auto

text {*
  Ejemplo 23 (p. 24). Demostrar
     $p \rightarrow q \vdash \neg p \vee q$ 
*}

-- "La demostración detallada es"
lemma ejemplo_23_1:
  assumes 1: "p → q"
  shows "¬p v q"
proof -
  have "¬p v p" by (rule excluded_middle)
  thus "¬p v q"
  proof (rule disjE)
    { assume "¬p"
      thus "¬p v q" by (rule disjI1) }
    next
    { assume 2: "p"
      have "q" using 1 2 by (rule mp)
      thus "¬p v q" by (rule disjI2) }
  qed
qed

```

```

-- "La demostración estructurada es"
lemma ejemplo_23_2:
  assumes "p  $\rightarrow$  q"
  shows " $\neg$ p  $\vee$  q"
proof -
  have " $\neg$ p  $\vee$  p" ..
  thus " $\neg$ p  $\vee$  q"
  proof
    { assume " $\neg$ p"
      thus " $\neg$ p  $\vee$  q" .. }
  next
    { assume "p"
      with assms have "q" ..
      thus " $\neg$ p  $\vee$  q" .. }
  qed
qed

-- "La demostración automática es"
lemma ejemplo_23_3:
  assumes "p  $\rightarrow$  q"
  shows " $\neg$ p  $\vee$  q"
using assms
by auto

subsection {* Demostraciones por contradicción *}

text {*
  Ejemplo 24. Demostrar que
     $\neg$ p, p  $\vee$  q  $\vdash$  q
*}

-- "La demostración detallada es"
lemma ejemplo_24_1:
  assumes " $\neg$ p"
           "p  $\vee$  q"
  shows "q"
using 'p  $\vee$  q'
proof (rule disjE)
  assume "p"
  with assms(1) show "q" by contradiction
next
  assume "q"
  thus "q" by assumption
qed

```

```
-- "La demostración estructurada es"
lemma ejemplo_24_2:
  assumes "¬p"
           "p ∨ q"
  shows "q"
using 'p ∨ q'
proof
  assume "p"
  with assms(1) show "q" ..
next
  assume "q"
  thus "q" .
qed
```

```
-- "La demostración automática es"
lemma ejemplo_24_3:
  assumes "¬p"
           "p ∨ q"
  shows "q"
using assms
by auto

end
```

3.2. Ejercicios: Deducción natural en lógica proposicional

```
chapter {* T3R1: Deducción natural proposicional *}
```

```
theory T3R1
imports Main
begin
```

```
text {*
```

 El objetivo de esta relación es lemas usando sólo las reglas básicas de deducción natural de la lógica proposicional.

Los ejercicios son los de la asignatura de "Lógica informática" que se encuentran en <http://goo.gl/yrPLn>

Las reglas básicas de la deducción natural son las siguientes:

```

· conjI:       $\llbracket P; Q \rrbracket \Longrightarrow P \wedge Q$ 
· conjunct1:   $P \wedge Q \Longrightarrow P$ 
· conjunct2:   $P \wedge Q \Longrightarrow Q$ 
· notnotD:     $\neg\neg P \Longrightarrow P$ 
· notnotI:     $P \Longrightarrow \neg\neg P$ 
· mp:          $\llbracket P \rightarrow Q; P \rrbracket \Longrightarrow Q$ 
· mt:          $\llbracket F \rightarrow G; \neg G \rrbracket \Longrightarrow \neg F$ 
· impI:        $(P \Longrightarrow Q) \Longrightarrow P \rightarrow Q$ 
· disjI1:      $P \Longrightarrow P \vee Q$ 
· disjI2:      $Q \Longrightarrow P \vee Q$ 
· disjE:       $\llbracket P \vee Q; P \Longrightarrow R; Q \Longrightarrow R \rrbracket \Longrightarrow R$ 
· FalseE:      $\text{False} \Longrightarrow P$ 
· notE:        $\llbracket \neg P; P \rrbracket \Longrightarrow R$ 
· notI:        $(P \Longrightarrow \text{False}) \Longrightarrow \neg P$ 
· iffI:        $\llbracket P \Longrightarrow Q; Q \Longrightarrow P \rrbracket \Longrightarrow P = Q$ 
· iffD1:       $\llbracket Q = P; Q \rrbracket \Longrightarrow P$ 
· iffD2:       $\llbracket P = Q; Q \rrbracket \Longrightarrow P$ 
· ccontr:      $(\neg P \Longrightarrow \text{False}) \Longrightarrow P$ 
· excluded_middle:  $\neg P \vee P$ 
-----
*}

text {*
  Se usarán las reglas notnotI y mt que demostramos a continuación.
*}

lemma notnotI: "P  $\Longrightarrow$   $\neg\neg P$ "
by auto

lemma mt: " $\llbracket F \rightarrow G; \neg G \rrbracket \Longrightarrow \neg F$ "
by auto

section {* Implicaciones *}

text {* -----
  Ejercicio 1. Demostrar
     $p \rightarrow q, p \vdash q$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_1_1:
  assumes 1: "p  $\rightarrow$  q" and
          2: "p"
  shows "q"
proof -

```

```

    show "q" using 1 2 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_1_2:
  assumes "p → q"
    "p"
  shows "q"
proof -
  show "q" using assms ..
qed

-- "La demostración automática es"
lemma ejercicio_1_3:
  assumes "p → q"
    "p"
  shows "q"
using assms
by auto

text {* -----
  Ejercicio 2. Demostrar
     $p \rightarrow q, q \rightarrow r, p \vdash r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_2_1:
  assumes 1: "p → q" and
    2: "q → r" and
    3: "p"
  shows "r"
proof -
  have 4: "q" using 1 3 by (rule mp)
  show "r" using 2 4 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_2_2:
  assumes "p → q"
    "q → r"
    "p"
  shows "r"
proof -
  have "q" using assms(1,3) ..
  show "r" using assms(2) 'q' ..

```

qed

-- "La demostración automática es"

lemma ejercicio_2_3:

assumes "p $\rightarrow$ q"

"q $\rightarrow$ r"

"p"

shows "r"

using assms

by auto

text {* -----
Ejercicio 3. Demostrar
 $p \rightarrow (q \rightarrow r), p \rightarrow q, p \vdash r$
 ----- *} }

-- "La demostración detallada es"

lemma ejercicio_3_1:

assumes 1: "p $\rightarrow$ (q $\rightarrow$ r)" and

2: "p $\rightarrow$ q" and

3: "p"

shows "r"

proof -

have 4: "q" using 2 3 by (rule mp)

have 5: "q $\rightarrow$ r" using 1 3 by (rule mp)

show "r" using 5 4 by (rule mp)

qed

-- "La demostración estructurada es"

lemma ejercicio_3_2:

assumes "p $\rightarrow$ (q $\rightarrow$ r)"

"p $\rightarrow$ q"

"p"

shows "r"

proof -

have "q" using assms(2,3) ..

have "q $\rightarrow$ r" using assms(1,3) ..

thus "r" using 'q' ..

qed

-- "La demostración automática es"

lemma ejercicio_3_3:

assumes "p $\rightarrow$ (q $\rightarrow$ r)"

"p $\rightarrow$ q"

"p"

```

    shows    "r"
using assms
by auto

text {* -----
  Ejercicio 4. Demostrar
     $p \rightarrow q, q \rightarrow r \vdash p \rightarrow r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_4_1:
  assumes 1: "p  $\rightarrow$  q" and
          2: "q  $\rightarrow$  r"
  shows "p  $\rightarrow$  r"
proof (rule impI)
  assume 3: "p"
  have 4: "q" using 1 3 by (rule mp)
  show "r" using 2 4 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_4_2:
  assumes "p  $\rightarrow$  q"
          "q  $\rightarrow$  r"
  shows "p  $\rightarrow$  r"
proof
  assume "p"
  with assms(1) have "q" ..
  with assms(2) show "r" ..
qed

-- "La demostración automática es"
lemma ejercicio_4_3:
  assumes "p  $\rightarrow$  q"
          "q  $\rightarrow$  r"
  shows "p  $\rightarrow$  r"
using assms
by auto

text {* -----
  Ejercicio 5. Demostrar
     $p \rightarrow (q \rightarrow r) \vdash q \rightarrow (p \rightarrow r)$ 
  ----- *}

-- "La demostración detallada es"

```

```

lemma ejercicio_5_1:
  assumes 1: "p → (q → r)"
  shows   "q → (p → r)"
proof (rule impI)
  assume 2: "q"
  show "p → r"
  proof (rule impI)
    assume 3: "p"
    have "q → r" using 1 3 by (rule mp)
    thus "r" using 2 by (rule mp)
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_5_2:
  assumes "p → (q → r)"
  shows   "q → (p → r)"
proof
  assume "q"
  show "p → r"
  proof
    assume "p"
    with assms(1) have "q → r" ..
    thus "r" using 'q' ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_5_3:
  assumes "p → (q → r)"
  shows   "q → (p → r)"
using assms
by auto

text {
  -- "Ejercicio 6. Demostrar"
  -- "p → (q → r) ⊢ (p → q) → (p → r)"
  -- "-----"
  -- "}"
}

-- "La demostración detallada es"
lemma ejercicio_6_1:
  assumes 1: "p → (q → r)"
  shows   "(p → q) → (p → r)"
proof (rule impI)
  assume 2: "p → q"

```

```

show "p → r"
proof (rule impI)
  assume 3: "p"
  have 4: "q" using 2 3 by (rule mp)
  have 5: "q → r" using 1 3 by (rule mp)
  show "r" using 5 4 by (rule mp)
qed
qed

-- "La demostración estructurada es"
lemma ejercicio_6_2:
  assumes "p → (q → r)"
  shows "(p → q) → (p → r)"
proof
  assume "p → q"
  show "p → r"
  proof
    assume "p"
    with 'p → q' have "q" ..
    have "q → r" using assms(1) 'p' ..
    thus "r" using 'q' ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_6_3:
  assumes "p → (q → r)"
  shows "(p → q) → (p → r)"
using assms
by auto

text {* -----
  Ejercicio 7. Demostrar
   $p \vdash q \rightarrow p$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_7_1:
  assumes 1: "p"
  shows "q → p"
proof (rule impI)
  assume 2: "q"
  show "p" using 1 by this
qed

```

```
-- "La demostración estructurada es"
lemma ejercicio_7_2:
  assumes "p"
  shows   "q → p"
proof
  assume "q"
  show "p" using assms(1) .
qed

-- "La demostración automática es"
lemma ejercicio_7_3:
  assumes "p"
  shows   "q → p"
using assms
by auto

text {* -----
      Ejercicio 8. Demostrar
       $\vdash p \rightarrow (q \rightarrow p)$ 
      ----- *}

-- "La demostración detallada es"
lemma ejercicio_8_1:
  "p → (q → p)"
proof (rule impI)
  assume 1: "p"
  show "q → p"
  proof (rule impI)
    assume 2: "q"
    show "p" using 1 by this
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_8_2:
  "p → (q → p)"
proof
  assume "p"
  show "q → p"
  proof
    assume "q"
    show "p" using 'p' .
  qed
qed
```

```
-- "La demostración automática es"
lemma ejercicio_8_3:
  "p → (q → p)"
by auto

text {* -----
  Ejercicio 9. Demostrar
     $p \rightarrow q \vdash (q \rightarrow r) \rightarrow (p \rightarrow r)$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_9_1:
  assumes 1: "p → q"
  shows    "(q → r) → (p → r)"
proof (rule impI)
  assume 2: "q → r"
  show "p → r"
  proof (rule impI)
    assume 3: "p"
    have 4: "q" using 1 3 by (rule mp)
    show "r" using 2 4 by (rule mp)
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_9_2:
  assumes "p → q"
  shows    "(q → r) → (p → r)"
proof
  assume "q → r"
  show "p → r"
  proof
    assume "p"
    with assms(1) have "q" ..
    with 'q → r' show "r" ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_9_3:
  assumes "p → q"
  shows    "(q → r) → (p → r)"
using assms
by auto
```

```

text {* -----
  Ejercicio 10. Demostrar
     $p \rightarrow (q \rightarrow (r \rightarrow s)) \vdash r \rightarrow (q \rightarrow (p \rightarrow s))$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_10_1:
  assumes 1: "p → (q → (r → s))"
  shows    "r → (q → (p → s))"
proof -
  { assume 2: "r"
    { assume 3: "q"
      { assume 4: "p"
        have 5: "q → (r → s)" using 1 4 by (rule mp)
        have 6: "r → s" using 5 3 by (rule mp)
        have 7: "s" using 6 2 by (rule mp) }
      hence 8: "p → s" by (rule impI) }
      hence 9: "q → (p → s)" by (rule impI) }
    thus 10: "r → (q → (p → s))" by (rule impI)
  }
qed

-- "Una variante de la demostración anterior es"
lemma ejercicio_10_1b:
  assumes 1: "p → (q → (r → s))"
  shows    "r → (q → (p → s))"
proof (rule impI)
  assume 2: "r"
  show "q → (p → s)"
  proof (rule impI)
    assume 3: "q"
    show "p → s"
    proof (rule impI)
      assume 4: "p"
      have 5: "q → (r → s)" using 1 4 by (rule mp)
      have 6: "r → s" using 5 3 by (rule mp)
      show "s" using 6 2 by (rule mp)
    qed
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_10_2:
  assumes "p → (q → (r → s))"
  shows    "r → (q → (p → s))"
proof

```

```

assume "r"
show "q → (p → s)"
proof
  assume "q"
  show "p → s"
  proof
    assume "p"
    with assms(1) have "q → (r → s)" ..
    hence "r → s" using 'q' ..
    thus "s" using 'r' ..
  qed
qed
qed

-- "La demostración automática es"
lemma ejercicio_10_3:
  assumes "p → (q → (r → s))"
  shows   "r → (q → (p → s))"
using assms
by auto

text {* -----
  Ejercicio 11. Demostrar
     $\vdash (p \rightarrow (q \rightarrow r)) \rightarrow ((p \rightarrow q) \rightarrow (p \rightarrow r))$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_11_1:
  "(p → (q → r)) → ((p → q) → (p → r))"
proof -
  { assume 1: "p → (q → r)"
    { assume 2: "p → q"
      { assume 3: "p"
        have 4: "q" using 2 3 by (rule mp)
        have 5: "q → r" using 1 3 by (rule mp)
        have 6: "r" using 5 4 by (rule mp) }
      hence 7: "p → r" by (rule impI) }
    hence 8: "(p → q) → (p → r)" by (rule impI) }
  thus "(p → (q → r)) → ((p → q) → (p → r))" by (rule impI)
qed

-- "La demostración estructurada es"
lemma ejercicio_11_2:
  "(p → (q → r)) → ((p → q) → (p → r))"
proof

```

```

assume "p → (q → r)"
{ assume "p → q"
  { assume "p"
    with 'p → q' have "q" ..
    have "q → r" using 'p → (q → r)' 'p' ..
    hence "r" using 'q' .. }
  hence "p → r" .. }
thus "(p → q) → (p → r)" ..
qed

-- "La demostración automática es"
lemma ejercicio_11_3:
  "(p → (q → r)) → ((p → q) → (p → r))"
by auto

text {
  -- "Ejercicio 12. Demostrar"
  -- "(p → q) → r ⊢ p → (q → r)"
  -- "}"
}

-- "La demostración detallada es"
lemma ejercicio_12_1:
  assumes 1: "(p → q) → r"
  shows "p → (q → r)"
proof -
  { assume 2: "p"
    { assume 3: "q"
      { assume 4: "p"
        have 5: "q" using 3 by this }
        hence 6: "p → q" by (rule impI)
        have 7: "r" using 1 6 by (rule mp) }
        hence 8: "q → r" by (rule impI) }
    thus 9: "p → (q → r)" by (rule impI)
  }
qed

-- "La demostración estructurada es"
lemma ejercicio_12_2:
  assumes "(p → q) → r"
  shows "p → (q → r)"
proof
  assume "p"
  { assume "q"
    { assume "p"
      have "q" using 'q' . }
    hence "p → q" ..
  }

```

```

    with assms(1) have "r" .. }
  thus "q → r" ..
qed

-- "La demostración automática es"
lemma ejercicio_12_3:
  assumes "(p → q) → r"
  shows "p → (q → r)"
using assms
by auto

section {* Conjunciones *}

text {
  -----
  Ejercicio 13. Demostrar
     $p, q \vdash p \wedge q$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_13_1:
  assumes 1: "p" and
          2: "q"
  shows "p ∧ q"
proof -
  show "p ∧ q" using assms by (rule conjI)
qed

-- "La demostración estructurada es"
lemma ejercicio_13_2:
  assumes "p"
          "q"
  shows "p ∧ q"
proof
  show "p" using assms(1) .
next
  show "q" using assms(2) .
qed

-- "La demostración automática es"
lemma ejercicio_13_3:
  assumes "p"
          "q"
  shows "p ∧ q"
using assms
by auto

```

```

text {* -----
  Ejercicio 14. Demostrar
     $p \wedge q \vdash p$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_14_1:
  assumes 1: " $p \wedge q$ "
  shows   " $p$ "
proof -
  show " $p$ " using 1 by (rule conjunct1)
qed

-- "La demostración estructurada es"
lemma ejercicio_14_2:
  assumes " $p \wedge q$ "
  shows   " $p$ "
proof -
  show " $p$ " using assms ..
qed

-- "La demostración automática es"
lemma ejercicio_14_3:
  assumes " $p \wedge q$ "
  shows   " $p$ "
using assms
by auto

text {* -----
  Ejercicio 15. Demostrar
     $p \wedge q \vdash q$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_15_1:
  assumes 1: " $p \wedge q$ "
  shows   " $q$ "
proof -
  show " $q$ " using 1 by (rule conjunct2)
qed

-- "La demostración estructurada es"
lemma ejercicio_15_2:
  assumes " $p \wedge q$ "

```

```

    shows    "q"
proof -
  show "q" using assms ..
qed

-- "La demostración automática es"
lemma ejercicio_15_3:
  assumes "p ∧ q"
  shows    "q"
using assms
by auto

text {* -----
  Ejercicio 16. Demostrar
     $p \wedge (q \wedge r) \vdash (p \wedge q) \wedge r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_16_1:
  assumes 1: "p ∧ (q ∧ r)"
  shows    "(p ∧ q) ∧ r"
proof -
  have 2: "p" using 1 by (rule conjunct1)
  have 3: "q ∧ r" using 1 by (rule conjunct2)
  have 4: "q" using 3 by (rule conjunct1)
  have 5: "p ∧ q" using 2 4 by (rule conjI)
  have 6: "r" using 3 by (rule conjunct2)
  show 7: "(p ∧ q) ∧ r" using 5 6 by (rule conjI)
qed

-- "La demostración estructurada es"
lemma ejercicio_16_2:
  assumes "p ∧ (q ∧ r)"
  shows    "(p ∧ q) ∧ r"
proof
  show "p ∧ q"
  proof
    show "p" using assms ..
  next
    have "q ∧ r" using assms ..
    thus "q" ..
  qed
next
  have "q ∧ r" using assms ..
  thus "r" ..

```

qed

-- "La demostración automática es"

```
lemma ejercicio_16_3:
  assumes "p ∧ (q ∧ r)"
  shows   "(p ∧ q) ∧ r"
using assms
by auto
```

```
text {* -----
  Ejercicio 17. Demostrar
   $(p \wedge q) \wedge r \vdash p \wedge (q \wedge r)$ 
  ----- *}

```

-- "La demostración detallada es"

```
lemma ejercicio_17_1:
  assumes 1: "(p ∧ q) ∧ r"
  shows   "p ∧ (q ∧ r)"
proof -
  have 2: "p ∧ q" using 1 by (rule conjunct1)
  have 3: "p" using 2 by (rule conjunct1)
  have 4: "q" using 2 by (rule conjunct2)
  have 5: "r" using 1 by (rule conjunct2)
  have 6: "q ∧ r" using 4 5 by (rule conjI)
  show 7: "p ∧ (q ∧ r)" using 3 6 by (rule conjI)
```

qed

-- "La demostración estructurada es"

```
lemma ejercicio_17_2:
  assumes "(p ∧ q) ∧ r"
  shows   "p ∧ (q ∧ r)"
proof
  have "p ∧ q" using assms ..
  thus "p" ..
next
  show "q ∧ r"
  proof
    have "p ∧ q" using assms ..
    thus "q" ..
  next
    show "r" using assms ..
  qed
qed
```

-- "La demostración automática es"

```

lemma ejercicio_17_3:
  assumes "(p ∧ q) ∧ r"
  shows   "p ∧ (q ∧ r)"
using assms
by auto

text {* -----
  Ejercicio 18. Demostrar
     $p \wedge q \vdash p \rightarrow q$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_18_1:
  assumes 1: "p ∧ q"
  shows   "p → q"
proof (rule impI)
  assume 2: "p"
  show "q" using 1 by (rule conjunct2)
qed

-- "La demostración estructurada es"
lemma ejercicio_18_2:
  assumes "p ∧ q"
  shows   "p → q"
proof
  assume "p"
  show "q" using assms ..
qed

-- "La demostración automática es"
lemma ejercicio_18_3:
  assumes "p ∧ q"
  shows   "p → q"
using assms
by auto

text {* -----
  Ejercicio 19. Demostrar
     $(p \rightarrow q) \wedge (p \rightarrow r) \vdash p \rightarrow q \wedge r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_19_1:
  assumes 1: "(p → q) ∧ (p → r)"
  shows   "p → q ∧ r"

```

```

proof (rule impI)
  assume 2: "p"
  have 3: "p → q" using 1 by (rule conjunct1)
  have 4: "q" using 3 2 by (rule mp)
  have 5: "p → r" using 1 by (rule conjunct2)
  have 6: "r" using 5 2 by (rule mp)
  show 7: "q ∧ r" using 4 6 by (rule conjI)
qed

-- "La demostración estructurada es"
lemma ejercicio_19_2:
  assumes "(p → q) ∧ (p → r)"
  shows "p → q ∧ r"
proof
  assume "p"
  show "q ∧ r"
  proof
    have "p → q" using assms ..
    thus "q" using 'p' ..
  next
    have "p → r" using assms ..
    thus "r" using 'p' ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_19_3:
  assumes "(p → q) ∧ (p → r)"
  shows "p → q ∧ r"
using assms
by auto

text {* -----
  Ejercicio 20. Demostrar
     $p \rightarrow q \wedge r \vdash (p \rightarrow q) \wedge (p \rightarrow r)$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_20_1:
  assumes 1: "p → q ∧ r"
  shows "(p → q) ∧ (p → r)"
proof (rule conjI)
  show "p → q"
  proof (rule impI)
    assume 2: "p"

```

```

    have "q ∧ r" using 1 2 by (rule mp)
    thus "q" by (rule conjunct1)
  qed
next
  show "p → r"
  proof (rule impI)
    assume 3: "p"
    have "q ∧ r" using 1 3 by (rule mp)
    thus "r" by (rule conjunct2)
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_20_2:
  assumes "p → q ∧ r"
  shows "(p → q) ∧ (p → r)"
proof
  show "p → q"
  proof
    assume "p"
    have "q ∧ r" using assms 'p' ..
    thus "q" ..
  qed
next
  show "p → r"
  proof
    assume "p"
    have "q ∧ r" using assms 'p' ..
    thus "r" ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_20_3:
  assumes "p → q ∧ r"
  shows "(p → q) ∧ (p → r)"
using assms
by auto

text {
  -- "La demostración detallada es"
  -- "Ejercicio 21. Demostrar"
  -- " $p \rightarrow (q \rightarrow r) \vdash p \wedge q \rightarrow r$ "
  -- "-----"
  -- "}"
}

```

```

lemma ejercicio_21_1:
  assumes 1: "p → (q → r)"
  shows   "p ∧ q → r"
proof (rule impI)
  assume 2: "p ∧ q"
  have 3: "p" using 2 by (rule conjunct1)
  have 4: "q" using 2 by (rule conjunct2)
  have 5: "q → r" using 1 3 by (rule mp)
  show 6: "r" using 5 4 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_21_2:
  assumes "p → (q → r)"
  shows   "p ∧ q → r"
proof
  assume "p ∧ q"
  hence "p" ..
  have "q" using 'p ∧ q' ..
  have "q → r" using assms(1) 'p' ..
  thus "r" using 'q' ..
qed

-- "La demostración automática es"
lemma ejercicio_21_3:
  assumes "p → (q → r)"
  shows   "p ∧ q → r"
using assms
by auto

text {
  -- "Ejercicio 22. Demostrar"
  -- "p ∧ q → r ⊢ p → (q → r)"
  -- "-----"
  -- "}"
}

-- "La demostración detallada es"
lemma ejercicio_22_1:
  assumes 1: "p ∧ q → r"
  shows   "p → (q → r)"
proof (rule impI)
  assume 2: "p"
  { assume 3: "q"
    have 4: "p ∧ q" using 2 3 by (rule conjI)
    have "r" using 1 4 by (rule mp) }
  thus "q → r" by (rule impI)

```

```

qed

-- "La demostración estructurada es"
lemma ejercicio_22_2:
  assumes "p ∧ q → r"
  shows "p → (q → r)"
proof
  assume "p"
  show "q → r"
  proof
    assume "q"
    with 'p' have "p ∧ q" ..
    with assms(1) show "r" ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_22_3:
  assumes "p ∧ q → r"
  shows "p → (q → r)"
using assms
by auto

text {* -----
  Ejercicio 23. Demostrar
     $(p \rightarrow q) \rightarrow r \vdash p \wedge q \rightarrow r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_23_1:
  assumes 1: "(p → q) → r"
  shows "p ∧ q → r"
proof (rule impI)
  assume 2: "p ∧ q"
  have 3: "p → q"
  proof (rule impI)
    assume 4: "p"
    show "q" using 2 by (rule conjunct2)
  qed
  show "r" using 1 3 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_23_2:
  assumes "(p → q) → r"

```

```

    shows   "p ∧ q → r"
proof
  assume "p ∧ q"
  have "p → q"
  proof
    assume "p"
    show "q" using 'p ∧ q' ..
  qed
  with assms show "r" ..
qed

-- "La demostración automática es"
lemma ejercicio_23_3:
  assumes "(p → q) → r"
  shows   "p ∧ q → r"
using assms
by auto

text {* -----
  Ejercicio 24. Demostrar
    p ∧ (q → r) ⊢ (p → q) → r
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_24_1:
  assumes 1: "p ∧ (q → r)"
  shows   "(p → q) → r"
proof (rule impI)
  assume 2: "p → q"
  have 3: "p" using 1 by (rule conjunct1)
  have 4: "q" using 2 3 by (rule mp)
  have 5: "q → r" using 1 by (rule conjunct2)
  thus "r" using 4 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejercicio_24_2:
  assumes "p ∧ (q → r)"
  shows   "(p → q) → r"
proof
  assume "p → q"
  have "p" using assms ..
  with 'p → q' have "q" ..
  have "q → r" using assms ..
  thus "r" using 'q' ..

```

qed

-- "La demostración automática es"

```
lemma ejercicio_24_3:
  assumes "p ∧ (q → r)"
  shows   "(p → q) → r"
using assms
by auto
```

section *{* Disyunciones **}

```
text {* -----
Ejercicio 25. Demostrar
  p ⊢ p ∨ q
----- *}
```

-- "La demostración detallada es"

```
lemma ejercicio_25_1:
  assumes 1: "p"
  shows   "p ∨ q"
proof -
  show "p ∨ q" using 1 by (rule disjI1)
qed
```

-- "La demostración estructurada es"

```
lemma ejercicio_25_2:
  assumes "p"
  shows   "p ∨ q"
proof -
  show "p ∨ q" using assms ..
qed
```

-- "La demostración estructurada es"

```
lemma ejercicio_25_3:
  assumes "p"
  shows   "p ∨ q"
using assms
by auto
```

```
text {* -----
Ejercicio 26. Demostrar
  q ⊢ p ∨ q
----- *}
```

-- "La demostración detallada es"

```

lemma ejercicio_26_1:
  assumes 1: "q"
  shows    "p ∨ q"
proof -
  show "p ∨ q" using 1 by (rule disjI2)
qed

-- "La demostración estructurada es"
lemma ejercicio_26_2:
  assumes "q"
  shows    "p ∨ q"
proof -
  show "p ∨ q" using assms ..
qed

-- "La demostración estructurada es"
lemma ejercicio_26_3:
  assumes "q"
  shows    "p ∨ q"
using assms
by auto

text {* -----
  Ejercicio 27. Demostrar
     $p \vee q \vdash q \vee p$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_27_1:
  assumes 1: "p ∨ q"
  shows    "q ∨ p"
using 1
proof (rule disjE)
  assume "p"
  thus "q ∨ p" by (rule disjI2)
next
  assume "q"
  thus "q ∨ p" by (rule disjI1)
qed

-- "La demostración estructurada es"
lemma ejercicio_27_2:
  assumes "p ∨ q"
  shows    "q ∨ p"
using assms

```

```

proof
  assume "p"
  thus "q v p" ..
next
  assume "q"
  thus "q v p" ..
qed

-- "La demostración estructurada es"
lemma ejercicio_27_3:
  assumes "p v q"
  shows   "q v p"
using assms
by auto

text {* -----
  Ejercicio 28. Demostrar
     $q \rightarrow r \vdash p \vee q \rightarrow p \vee r$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_28_1:
  assumes 1: "q  $\rightarrow$  r"
  shows   "p v q  $\rightarrow$  p v r"
proof (rule impI)
  assume "p v q"
  thus "p v r"
  proof (rule disjE)
    assume "p"
    thus "p v r" by (rule disjI1)
  next
    assume "q"
    have "r" using assms 'q' by (rule mp)
    thus "p v r" by (rule disjI2)
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_28_2:
  assumes "q  $\rightarrow$  r"
  shows   "p v q  $\rightarrow$  p v r"
proof
  assume "p v q"
  thus "p v r"
  proof

```

```

    assume "p"
    thus "p v r" ..
next
    assume "q"
    have "r" using assms 'q' ..
    thus "p v r" ..
qed
qed

-- "La demostración automática es"
lemma ejercicio_28_3:
  assumes "q → r"
  shows "p v q → p v r"
using assms
by auto

text {* -----
  Ejercicio 29. Demostrar
     $p \vee p \vdash p$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_29_1:
  assumes 1: "p v p"
  shows "p"
using 1
proof (rule disjE)
  assume "p"
  thus "p" by this
next
  assume "p"
  thus "p" by this
qed

-- "La demostración estructurada es"
lemma ejercicio_29_2:
  assumes "p v p"
  shows "p"
using assms
proof
  assume "p"
  thus "p" .
next
  assume "p"
  thus "p" .

```

qed

-- "La demostración estructurada es"

lemma ejercicio_29_3:

assumes "p v p"

shows "p"

using assms

by auto

text {* -----
Ejercicio 30. Demostrar
 $p \vdash p \vee p$
 ----- *} }

-- "La demostración detallada es"

lemma ejercicio_30_1:

assumes 1: "p"

shows "p v p"

proof -

show "p v p" using 1 by (rule disjI1)

qed

-- "La demostración estructurada es"

lemma ejercicio_30_2:

assumes "p"

shows "p v p"

proof -

show "p v p" using assms ..

qed

-- "La demostración estructurada es"

lemma ejercicio_30_3:

assumes "p"

shows "p v p"

using assms

by auto

text {* -----
Ejercicio 31. Demostrar
 $p \vee (q \vee r) \vdash (p \vee q) \vee r$
 ----- *} }

-- "La demostración detallada es"

lemma ejercicio_31_1:

assumes 1: "p v (q v r)"

```

    shows      "(p v q) v r"
using 1
proof (rule disjE)
  assume "p"
  hence "p v q" by (rule disjI1)
  thus "(p v q) v r" by (rule disjI1)
next
  assume "q v r"
  thus "(p v q) v r"
proof (rule disjE)
  assume "q"
  hence "p v q" by (rule disjI2)
  thus "(p v q) v r" by (rule disjI1)
next
  assume "r"
  thus "(p v q) v r" by (rule disjI2)
qed
qed

```

-- "La demostración estructurada es"

```

lemma ejercicio_31_2:
  assumes "p v (q v r)"
  shows   "(p v q) v r"
using assms
proof
  assume "p"
  hence "p v q" ..
  thus "(p v q) v r" ..
next
  assume "q v r"
  thus "(p v q) v r"
proof
  assume "q"
  hence "p v q" ..
  thus "(p v q) v r" ..
next
  assume "r"
  thus "(p v q) v r" ..
qed
qed

```

-- "La demostración estructurada es"

```

lemma ejercicio_31_3:
  assumes "p v (q v r)"
  shows   "(p v q) v r"

```

```

using assms
by auto

text {
  -----
  Ejercicio 32. Demostrar
   $(p \vee q) \vee r \vdash p \vee (q \vee r)$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_32_1:
  assumes 1: "(p v q) v r"
  shows    "p v (q v r)"
using 1
proof (rule disjE)
  assume "p v q"
  thus "p v (q v r)"
  proof (rule disjE)
    assume "p"
    thus "p v (q v r)" by (rule disjI1)
  next
    assume "q"
    hence "q v r" by (rule disjI1)
    thus "p v (q v r)" by (rule disjI2)
  qed
next
  assume "r"
  hence "q v r" by (rule disjI2)
  thus "p v (q v r)" by (rule disjI2)
qed

-- "La demostración estructurada es"
lemma ejercicio_32_2:
  assumes "(p v q) v r"
  shows    "p v (q v r)"
using assms
proof
  assume "p v q"
  thus "p v (q v r)"
  proof
    assume "p"
    thus "p v (q v r)" ..
  next
    assume "q"
    hence "q v r" ..
    thus "p v (q v r)" ..
  qed

```

```

    qed
next
  assume "r"
  hence "q v r" ..
  thus "p v (q v r)" ..
qed

-- "La demostración estructurada es"
lemma ejercicio_32_3:
  assumes "(p v q) v r"
  shows "p v (q v r)"
using assms
by auto

text {* -----
  Ejercicio 33. Demostrar
     $p \wedge (q \vee r) \vdash (p \wedge q) \vee (p \wedge r)$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_33_1:
  assumes 1: "p  $\wedge$  (q v r)"
  shows " (p  $\wedge$  q) v (p  $\wedge$  r)"
proof -
  have 2: "p" using 1 by (rule conjunct1)
  have 3: "q v r" using 1 by (rule conjunct2)
  thus "(p  $\wedge$  q) v (p  $\wedge$  r)"
  proof (rule disjE)
    assume 4: "q"
    have "p  $\wedge$  q" using 2 4 by (rule conjI)
    thus "(p  $\wedge$  q) v (p  $\wedge$  r)" by (rule disjI1)
  next
    assume 5: "r"
    have "q v r" using 3 by (rule conjunct2)
    thus "(p  $\wedge$  q) v (p  $\wedge$  r)"
    proof (rule disjE)
      assume "q"
      have "p  $\wedge$  q" using 'p' 'q' by (rule conjI)
      thus "(p  $\wedge$  q) v (p  $\wedge$  r)" by (rule disjI1)
    next
      assume "r"
      have "p  $\wedge$  r" using 'p' 'r' by (rule conjI)
      thus "(p  $\wedge$  q) v (p  $\wedge$  r)" by (rule disjI2)
    qed
  qed
qed
qed

```

qed

-- "La demostración estructurada es"

```
lemma ejercicio_33_2:
  assumes "p ∧ (q ∨ r)"
  shows "(p ∧ q) ∨ (p ∧ r)"
proof -
  have "p" using assms ..
  have "q ∨ r" using assms ..
  thus "(p ∧ q) ∨ (p ∧ r)"
  proof
    assume "q"
    with 'p' have "p ∧ q" ..
    thus "(p ∧ q) ∨ (p ∧ r)" ..
  next
    assume "r"
    hence "q ∨ r" ..
    thus "(p ∧ q) ∨ (p ∧ r)"
    proof
      assume "q"
      with 'p' have "p ∧ q" ..
      thus "(p ∧ q) ∨ (p ∧ r)" ..
    next
      assume "r"
      with 'p' have "p ∧ r" ..
      thus "(p ∧ q) ∨ (p ∧ r)" ..
    qed
  qed
qed
```

-- "La demostración automática es"

```
lemma ejercicio_33_3:
  assumes "p ∧ (q ∨ r)"
  shows "(p ∧ q) ∨ (p ∧ r)"
using assms
by auto
```

```
text {* -----
  Ejercicio 34. Demostrar
     $(p \wedge q) \vee (p \wedge r) \vdash p \wedge (q \vee r)$ 
  ----- *}

```

-- "La demostración detallada es"

```
lemma ejercicio_34_1:
  assumes "(p ∧ q) ∨ (p ∧ r)"
```

```

    shows "p ∧ (q ∨ r)"
using assms
proof (rule disjE)
  assume 1: "p ∧ q"
  show "p ∧ (q ∨ r)"
  proof (rule conjI)
    show "p" using 1 by (rule conjunct1)
  next
    have "q" using 1 by (rule conjunct2)
    thus "q ∨ r" by (rule disjI1)
  qed
next
  assume 2: "p ∧ r"
  hence 3: "p" by (rule conjunct1)
  have "r" using 2 by (rule conjunct2)
  hence 4: "q ∨ r" by (rule disjI2)
  show "p ∧ (q ∨ r)" using 3 4 by (rule conjI)
qed

```

-- "La demostración estructurada es"

```

lemma ejercicio_34_2:
  assumes "(p ∧ q) ∨ (p ∧ r)"
  shows "p ∧ (q ∨ r)"
using assms
proof
  assume "p ∧ q"
  show "p ∧ (q ∨ r)"
  proof
    show "p" using 'p ∧ q' ..
  next
    have "q" using 'p ∧ q' ..
    thus "q ∨ r" ..
  qed
next
  assume "p ∧ r"
  hence "p" ..
  have "r" using 'p ∧ r' ..
  hence "q ∨ r" ..
  with 'p' show "p ∧ (q ∨ r)" ..
qed

```

-- "La demostración estructurada es"

```

lemma ejercicio_34_3:
  assumes "(p ∧ q) ∨ (p ∧ r)"
  shows "p ∧ (q ∨ r)"

```

```

using assms
by auto

text {* -----
  Ejercicio 35. Demostrar
     $p \vee (q \wedge r) \vdash (p \vee q) \wedge (p \vee r)$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_35_1:
  assumes "p  $\vee$  (q  $\wedge$  r)"
  shows "(p  $\vee$  q)  $\wedge$  (p  $\vee$  r)"
using assms
proof
  assume "p"
  show "(p  $\vee$  q)  $\wedge$  (p  $\vee$  r)"
  proof
    show "p  $\vee$  q" using 'p' ..
  next
    show "p  $\vee$  r" using 'p' ..
  qed
next
  assume "q  $\wedge$  r"
  show "(p  $\vee$  q)  $\wedge$  (p  $\vee$  r)"
  proof
    have "q" using 'q  $\wedge$  r' ..
    thus "p  $\vee$  q" ..
  next
    have "r" using 'q  $\wedge$  r' ..
    thus "p  $\vee$  r" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_35_2:
  assumes "p  $\vee$  (q  $\wedge$  r)"
  shows "(p  $\vee$  q)  $\wedge$  (p  $\vee$  r)"
using assms
proof (rule disjE)
  assume "p"
  show "(p  $\vee$  q)  $\wedge$  (p  $\vee$  r)"
  proof (rule conjI)
    show "p  $\vee$  q" using 'p' by (rule disjI1)
  next
    show "p  $\vee$  r" using 'p' by (rule disjI1)
  qed
qed

```

```

    qed
next
  assume "q ∧ r"
  show "(p ∨ q) ∧ (p ∨ r)"
  proof (rule conjI)
    have "q" using 'q ∧ r' ..
    thus "p ∨ q" by (rule disjI2)
  next
    have "r" using 'q ∧ r' by (rule conjunct2)
    thus "p ∨ r" by (rule disjI2)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_35_3:
  assumes "p ∨ (q ∧ r)"
  shows "(p ∨ q) ∧ (p ∨ r)"
using assms
by auto

text {* -----
  Ejercicio 36. Demostrar
    (p ∨ q) ∧ (p ∨ r) ⊢ p ∨ (q ∧ r)
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_36_1:
  assumes "(p ∨ q) ∧ (p ∨ r)"
  shows "p ∨ (q ∧ r)"
proof -
  have "p ∨ q" using assms ..
  thus "p ∨ (q ∧ r)"
  proof
    assume "p"
    thus "p ∨ (q ∧ r)" ..
  next
    assume "q"
    have "p ∨ r" using assms ..
    thus "p ∨ (q ∧ r)"
    proof
      assume "p"
      thus "p ∨ (q ∧ r)" ..
    next
      assume "r"
      with 'q' have "q ∧ r" ..
    qed
  qed

```

```

        thus "p v (q ∧ r)" ..
      qed
    qed
  qed

-- "La demostración detallada es"
lemma ejercicio_36_2:
  assumes "(p v q) ∧ (p v r)"
  shows   "p v (q ∧ r)"
proof -
  have "p v q" using assms by (rule conjunct1)
  thus "p v (q ∧ r)"
  proof (rule disjE)
    assume "p"
    thus "p v (q ∧ r)" by (rule disjI1)
  next
    assume "q"
    have "p v r" using assms by (rule conjunct2)
    thus "p v (q ∧ r)"
    proof (rule disjE)
      assume "p"
      thus "p v (q ∧ r)" by (rule disjI1)
    next
      assume "r"
      have "q ∧ r" using 'q' 'r' by (rule conjI)
      thus "p v (q ∧ r)" by (rule disjI2)
    qed
  qed
qed
qed

-- "La demostración automática es"
lemma ejercicio_36_3:
  assumes "(p v q) ∧ (p v r)"
  shows   "p v (q ∧ r)"
using assms
by auto

text {* -----
  Ejercicio 37. Demostrar
     $(p \rightarrow r) \wedge (q \rightarrow r) \vdash p \vee q \rightarrow r$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_37_1:
  assumes "(p → r) ∧ (q → r)"

```

```

    shows    "p ∨ q → r"
proof
  assume "p ∨ q"
  thus "r"
  proof
    assume "p"
    have "p → r" using assms ..
    thus "r" using 'p' ..
  next
    assume "q"
    have "q → r" using assms ..
    thus "r" using 'q' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_37_2:
  assumes "(p → r) ∧ (q → r)"
  shows    "p ∨ q → r"
proof (rule impI)
  assume "p ∨ q"
  thus "r"
  proof (rule disjE)
    assume "p"
    have "p → r" using assms by (rule conjunct1)
    thus "r" using 'p' by (rule mp)
  next
    assume "q"
    have "q → r" using assms by (rule conjunct2)
    thus "r" using 'q' by (rule mp)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_37_3:
  assumes "(p → r) ∧ (q → r)"
  shows    "p ∨ q → r"
using assms
by auto

text {* -----
  Ejercicio 38. Demostrar
     $p \vee q \rightarrow r \vdash (p \rightarrow r) \wedge (q \rightarrow r)$ 
  ----- *}

```

```
-- "La demostración estructurada es"
lemma ejercicio_38_1:
  assumes "p ∨ q → r"
  shows "(p → r) ∧ (q → r)"
proof
  show "p → r"
  proof
    assume "p"
    hence "p ∨ q" ..
    with assms show "r" ..
  qed
next
  show "q → r"
  proof
    assume "q"
    hence "p ∨ q" ..
    with assms show "r" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_38_2:
  assumes "p ∨ q → r"
  shows "(p → r) ∧ (q → r)"
proof (rule conjI)
  show "p → r"
  proof (rule impI)
    assume "p"
    hence "p ∨ q" by (rule disjI1)
    show "r" using assms 'p ∨ q' by (rule mp)
  qed
next
  show "q → r"
  proof (rule impI)
    assume "q"
    hence "p ∨ q" by (rule disjI2)
    show "r" using assms 'p ∨ q' by (rule mp)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_38_3:
  assumes "p ∨ q → r"
  shows "(p → r) ∧ (q → r)"
using assms
```

```
by auto
```

```
section {* Negación *
```

```
text {* -----
  Ejercicio 39. Demostrar
     $p \vdash \neg\neg p$ 
  ----- *}

```

```
-- "La demostración detallada es"
```

```
lemma ejercicio_39_1:
  assumes "p"
  shows   " $\neg\neg p$ "
proof -
  show " $\neg\neg p$ " using assms by (rule notnotI)
qed

```

```
-- "La demostración automática es"
```

```
lemma ejercicio_39_2:
  assumes "p"
  shows   " $\neg\neg p$ "
using assms
by auto

```

```
text {* -----
  Ejercicio 40. Demostrar
     $\neg p \vdash p \rightarrow q$ 
  ----- *}

```

```
-- "La demostración estructurada es"
```

```
lemma ejercicio_40_1:
  assumes " $\neg p$ "
  shows   " $p \rightarrow q$ "
proof
  assume "p"
  with assms(1) show "q" ..
qed

```

```
-- "La demostración detallada es"
```

```
lemma ejercicio_40_2:
  assumes " $\neg p$ "
  shows   " $p \rightarrow q$ "
proof (rule impI)
  assume "p"
  show "q" using assms(1) 'p' by (rule notE)

```

qed

-- "La demostración automática es"

lemma ejercicio_40_3:

assumes "¬p"

shows "p → q"

using assms

by auto

```
text {* -----
  Ejercicio 41. Demostrar
     $p \rightarrow q \vdash \neg q \rightarrow \neg p$ 
  ----- *}

```

-- "La demostración estructurada es"

lemma ejercicio_41_1:

assumes "p → q"

shows "¬q → ¬p"

proof

assume "¬q"

with assms(1) show "¬p" by (rule mt)

qed

-- "La demostración detallada es"

lemma ejercicio_41_2:

assumes "p → q"

shows "¬q → ¬p"

proof (rule impI)

assume "¬q"

show "¬p" using assms(1) '¬q' by (rule mt)

qed

-- "La demostración automática es"

lemma ejercicio_41_3:

assumes "p → q"

shows "¬q → ¬p"

using assms

by auto

```
text {* -----
  Ejercicio 42. Demostrar
     $p \vee q, \neg q \vdash p$ 
  ----- *}

```

-- "La demostración estructurada es"

```

lemma ejercicio_42_1:
  assumes "p ∨ q"
    "¬q"
  shows "p"
using assms(1)
proof
  assume "p"
  thus "p" .
next
  assume "q"
  with assms(2) show "p" ..
qed

-- "La demostración detallada es"
lemma ejercicio_42_2:
  assumes "p ∨ q"
    "¬q"
  shows "p"
using assms(1)
proof (rule disjE)
  assume "p"
  thus "p" by this
next
  assume "q"
  show "p" using assms(2) 'q' by (rule notE)
qed

-- "La demostración automática es"
lemma ejercicio_42_3:
  assumes "p ∨ q"
    "¬q"
  shows "p"
using assms
by auto

text {* -----
  Ejercicio 42. Demostrar
     $p \vee q, \neg p \vdash q$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_43_1:
  assumes "p ∨ q"
    "¬p"
  shows "q"

```

```

using assms(1)
proof
  assume "p"
  with assms(2) show "q" ..
next
  assume "q"
  thus "q" .
qed

-- "La demostración detallada es"
lemma ejercicio_43_2:
  assumes "p ∨ q"
           "¬p"
  shows   "q"
using assms(1)
proof (rule disjE)
  assume "p"
  show "q" using assms(2) 'p' by (rule notE)
next
  assume "q"
  thus "q" by this
qed

-- "La demostración automática es"
lemma ejercicio_43_3:
  assumes "p ∨ q"
           "¬p"
  shows   "q"
using assms
by auto

text {* -----
  Ejercicio 44. Demostrar
     $p \vee q \vdash \neg(\neg p \wedge \neg q)$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_44_1:
  assumes "p ∨ q"
  shows   "¬(¬p ∧ ¬q)"
proof
  assume "¬p ∧ ¬q"
  note 'p ∨ q'
  thus "False"
proof

```

```

    assume "p"
    have "¬p" using '¬p ∧ ¬q' ..
    thus "False" using 'p' ..
  next
    assume "q"
    have "¬q" using '¬p ∧ ¬q' ..
    thus "False" using 'q' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_44_2:
  assumes "p ∨ q"
  shows "¬(¬p ∧ ¬q)"
proof (rule notI)
  assume "¬p ∧ ¬q"
  note 'p ∨ q'
  thus "False"
  proof
    assume "p"
    have "¬p" using '¬p ∧ ¬q' by (rule conjunct1)
    thus "False" using 'p' by (rule notE)
  next
    assume "q"
    have "¬q" using '¬p ∧ ¬q' by (rule conjunct2)
    thus "False" using 'q' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_44_3:
  assumes "p ∨ q"
  shows "¬(¬p ∧ ¬q)"
using assms
by auto

text {* -----
  Ejercicio 45. Demostrar
     $p \wedge q \vdash \neg(\neg p \vee \neg q)$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_45_1:
  assumes "p ∧ q"
  shows "¬(¬p ∨ ¬q)"

```

```

proof
  assume "¬p ∨ ¬q"
  thus "False"
  proof
    assume "¬p"
    have "p" using assms ..
    with '¬p' show "False" ..
  next
    assume "¬q"
    have "q" using assms ..
    with '¬q' show "False" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_45_2:
  assumes "p ∧ q"
  shows "¬(¬p ∨ ¬q)"
proof (rule notI)
  assume "¬p ∨ ¬q"
  thus "False"
  proof
    assume "¬p"
    have "p" using assms by (rule conjunct1)
    show "False" using '¬p' 'p' by (rule notE)
  next
    assume "¬q"
    have "q" using assms by (rule conjunct2)
    show "False" using '¬q' 'q' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_45_3:
  assumes "p ∧ q"
  shows "¬(¬p ∨ ¬q)"
using assms
by auto

text {
  -----
  Ejercicio 46. Demostrar
  
$$\neg(p \vee q) \vdash \neg p \wedge \neg q$$

  ----- *}

-- "La demostración estructurada es"

```

```

lemma ejercicio_46_1:
  assumes "¬(p ∨ q)"
  shows "¬p ∧ ¬q"
proof
  show "¬p"
  proof
    assume "p"
    hence "p ∨ q" ..
    with assms show "False" ..
  qed
next
  show "¬q"
  proof
    assume "q"
    hence "p ∨ q" ..
    with assms show "False" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_46_2:
  assumes "¬(p ∨ q)"
  shows "¬p ∧ ¬q"
proof (rule conjI)
  show "¬p"
  proof (rule notI)
    assume "p"
    hence "p ∨ q" by (rule disjI1)
    show "False" using assms 'p ∨ q' by (rule notE)
  qed
next
  show "¬q"
  proof (rule notI)
    assume "q"
    hence "p ∨ q" by (rule disjI2)
    show "False" using assms 'p ∨ q' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_46_3:
  assumes "¬(p ∨ q)"
  shows "¬p ∧ ¬q"
using assms
by auto

```

```

text {* -----
  Ejercicio 47. Demostrar
     $\neg p \wedge \neg q \vdash \neg(p \vee q)$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_47_1:
  assumes " $\neg p \wedge \neg q$ "
  shows   " $\neg(p \vee q)$ "
proof
  assume " $p \vee q$ "
  thus False
  proof
    assume "p"
    have " $\neg p$ " using assms ..
    thus False using 'p' ..
  next
    assume "q"
    have " $\neg q$ " using assms ..
    thus False using 'q' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_47_2:
  assumes " $\neg p \wedge \neg q$ "
  shows   " $\neg(p \vee q)$ "
proof (rule notI)
  assume " $p \vee q$ "
  thus False
  proof (rule disjE)
    assume "p"
    have " $\neg p$ " using assms by (rule conjunct1)
    thus False using 'p' by (rule notE)
  next
    assume "q"
    have " $\neg q$ " using assms by (rule conjunct2)
    thus False using 'q' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_47_3:
  assumes " $\neg p \wedge \neg q$ "

```

```

    shows    "¬(p ∨ q)"
using assms
by auto

text {* -----
Ejercicio 48. Demostrar
  ¬p ∨ ¬q ⊢ ¬(p ∧ q)
----- *}

-- "La demostración estructurada es"
lemma ejercicio_48_1:
  assumes "¬p ∨ ¬q"
  shows   "¬(p ∧ q)"
proof
  assume "p ∧ q"
  note '¬p ∨ ¬q'
  thus False
  proof
    assume "¬p"
    have "p" using 'p ∧ q' ..
    with '¬p' show False ..
  next
    assume "¬q"
    have "q" using 'p ∧ q' ..
    with '¬q' show False ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_48_2:
  assumes "¬p ∨ ¬q"
  shows   "¬(p ∧ q)"
proof (rule notI)
  assume "p ∧ q"
  note '¬p ∨ ¬q'
  thus False
  proof (rule disjE)
    assume "¬p"
    have "p" using 'p ∧ q' by (rule conjunct1)
    show False using '¬p' 'p' by (rule notE)
  next
    assume "¬q"
    have "q" using 'p ∧ q' by (rule conjunct2)
    show False using '¬q' 'q' by (rule notE)
  qed

```

qed

-- "La demostración automática es"

lemma ejercicio_48_3:

assumes " $\neg p \vee \neg q$ "

shows " $\neg(p \wedge q)$ "

using assms

by auto

text {* -----
Ejercicio 49. Demostrar
 $\vdash \neg(p \wedge \neg p)$
 ----- *} }

-- "La demostración estructurada es"

lemma ejercicio_49_1:

" $\neg(p \wedge \neg p)$ "

proof

assume " $p \wedge \neg p$ "

hence "p" ..

have " $\neg p$ " using 'p $\wedge$ $\neg p$ ' ..

thus False using 'p' ..

qed

-- "La demostración detallada es"

lemma ejercicio_49_2:

" $\neg(p \wedge \neg p)$ "

proof (rule notI)

assume " $p \wedge \neg p$ "

hence "p" by (rule conjunct1)

have " $\neg p$ " using 'p $\wedge$ $\neg p$ ' by (rule conjunct2)

thus False using 'p' by (rule notE)

qed

-- "La demostración automática es"

lemma ejercicio_49_3:

" $\neg(p \wedge \neg p)$ "

by auto

text {* -----
Ejercicio 50. Demostrar
 $p \wedge \neg p \vdash q$
 ----- *} }

-- "La demostración estructurada es"

```

lemma ejercicio_50_1:
  assumes "p ∧ ¬p"
  shows   "q"
proof -
  have "p" using assms ..
  have "¬p" using assms ..
  thus "q" using 'p' ..
qed

-- "La demostración detallada es"
lemma ejercicio_50_2:
  assumes "p ∧ ¬p"
  shows   "q"
proof -
  have "p" using assms by (rule conjunct1)
  have "¬p" using assms by (rule conjunct2)
  thus "q" using 'p' by (rule notE)
qed

-- "La demostración automática es"
lemma ejercicio_50_3:
  assumes "p ∧ ¬p"
  shows   "q"
using assms
by auto

text {* -----
  Ejercicio 51. Demostrar
     $\neg p \vdash p$ 
  ----- *}

-- "La demostración detallada es"
lemma ejercicio_51_1:
  assumes "¬¬p"
  shows   "p"
using assms
by (rule notnotD)

-- "La demostración automática es"
lemma ejercicio_51_2:
  assumes "¬¬p"
  shows   "p"
using assms
by auto

```

```

text {* -----
  Ejercicio 52. Demostrar
     $\vdash p \vee \neg p$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_52_1:
  "p v ¬p"
proof -
  have "¬¬p v ¬p" ..
  thus "p v ¬p" by simp
qed

-- "La demostración detallada es"
lemma ejercicio_52_2:
  "p v ¬p"
proof -
  have "¬¬p v ¬p" by (rule excluded_middle)
  thus "p v ¬p" by simp
qed

-- "La demostración automática es"
lemma ejercicio_52_3:
  "p v ¬p"
by auto

text {* -----
  Ejercicio 53. Demostrar
     $\vdash ((p \rightarrow q) \rightarrow p) \rightarrow p$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_53_1:
  " $((p \rightarrow q) \rightarrow p) \rightarrow p$ "
proof
  assume "(p → q) → p"
  show "p"
  proof (rule ccontr)
    assume "¬p"
    have "¬(p → q)" using '(p → q) → p' '¬p' by (rule mt)
    have "p → q"
    proof
      assume "p"
      with '¬p' show "q" ..
    qed
  qed

```

```

    show False using '¬(p → q)' 'p → q' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_53_2:
  "((p → q) → p) → p"
proof (rule impI)
  assume "(p → q) → p"
  show "p"
  proof (rule ccontr)
    assume "¬p"
    have "¬(p → q)" using '(p → q) → p' '¬p' by (rule mt)
    have "p → q"
    proof (rule impI)
      assume "p"
      show "q" using '¬p' 'p' by (rule notE)
    qed
    show False using '¬(p → q)' 'p → q' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_53_3:
  "((p → q) → p) → p"
by auto

text {* -----
  Ejercicio 54. Demostrar
     $\neg q \rightarrow \neg p \vdash p \rightarrow q$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_54_1:
  assumes "¬q → ¬p"
  shows "p → q"
proof
  assume "p"
  show "q"
  proof (rule ccontr)
    assume "¬q"
    with assms have "¬p" ..
    thus False using 'p' ..
  qed
qed

```

```
-- "La demostración detallada es"
lemma ejercicio_54_2:
  assumes "¬q → ¬p"
  shows "p → q"
proof (rule impI)
  assume "p"
  show "q"
  proof (rule ccontr)
    assume "¬q"
    have "¬p" using assms '¬q' by (rule mp)
    thus False using 'p' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_54_3:
  assumes "¬q → ¬p"
  shows "p → q"
using assms
by auto

text {* -----
  Ejercicio 55. Demostrar
     $\neg(\neg p \wedge \neg q) \vdash p \vee q$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_55_1:
  assumes "¬(¬p ∧ ¬q)"
  shows "p ∨ q"
proof -
  have "¬p ∨ p" ..
  thus "p ∨ q"
  proof
    assume "¬p"
    have "¬q ∨ q" ..
    thus "p ∨ q"
    proof
      assume "¬q"
      with '¬p' have "¬p ∧ ¬q" ..
      with assms show "p ∨ q" ..
    next
      assume "q"
      thus "p ∨ q" ..
    qed
  qed

```

```

    qed
  next
    assume "p"
    thus "p ∨ q" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_55_2:
  assumes "¬(¬p ∧ ¬q)"
  shows "p ∨ q"
proof -
  have "¬p ∨ p" by (rule excluded_middle)
  thus "p ∨ q"
  proof
    assume "¬p"
    have "¬q ∨ q" by (rule excluded_middle)
    thus "p ∨ q"
    proof
      assume "¬q"
      have "¬p ∧ ¬q" using '¬p' '¬q' by (rule conjI)
      show "p ∨ q" using assms '¬p ∧ ¬q' by (rule notE)
    next
      assume "q"
      thus "p ∨ q" by (rule disjI2)
    qed
  qed
next
  assume "p"
  thus "p ∨ q" by (rule disjI1)
qed
qed

-- "La demostración automática es"
lemma ejercicio_55_3:
  assumes "¬(¬p ∧ ¬q)"
  shows "p ∨ q"
using assms
by auto

text {* -----
  Ejercicio 56. Demostrar
     $\neg(\neg p \vee \neg q) \vdash p \wedge q$ 
  ----- *}

-- "La demostración estructurada es"
```

```

lemma ejercicio_56_1:
  assumes " $\neg(\neg p \vee \neg q)$ "
  shows " $p \wedge q$ "
proof
  show "p"
  proof (rule ccontr)
    assume " $\neg p$ "
    hence " $\neg p \vee \neg q$ " ..
    with assms show False ..
  qed
next
  show "q"
  proof (rule ccontr)
    assume " $\neg q$ "
    hence " $\neg p \vee \neg q$ " ..
    with assms show False ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_56_2:
  assumes " $\neg(\neg p \vee \neg q)$ "
  shows " $p \wedge q$ "
proof (rule conjI)
  show "p"
  proof (rule ccontr)
    assume " $\neg p$ "
    hence " $\neg p \vee \neg q$ " by (rule disjI1)
    show False using assms ' $\neg p \vee \neg q$ ' by (rule notE)
  qed
next
  show "q"
  proof (rule ccontr)
    assume " $\neg q$ "
    hence " $\neg p \vee \neg q$ " by (rule disjI2)
    show False using assms ' $\neg p \vee \neg q$ ' by (rule notE)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_56_3:
  assumes " $\neg(\neg p \vee \neg q)$ "
  shows " $p \wedge q$ "
using assms
by auto

```

```

text {* -----
  Ejercicio 57. Demostrar
     $\neg(p \wedge q) \vdash \neg p \vee \neg q$ 
  ----- *}

```

```
-- "La demostración estructurada es"
```

```
lemma ejercicio_57_1:
```

```
  assumes " $\neg(p \wedge q)$ "
```

```
  shows   " $\neg p \vee \neg q$ "
```

```
proof -
```

```
  have " $\neg p \vee p$ " ..
```

```
  thus " $\neg p \vee \neg q$ "
```

```
  proof
```

```
    assume " $\neg p$ "
```

```
    thus " $\neg p \vee \neg q$ " ..
```

```
  next
```

```
    assume "p"
```

```
    have " $\neg q \vee q$ " ..
```

```
    thus " $\neg p \vee \neg q$ "
```

```
    proof
```

```
      assume " $\neg q$ "
```

```
      thus " $\neg p \vee \neg q$ " ..
```

```
    next
```

```
      assume "q"
```

```
      with 'p' have " $p \wedge q$ " ..
```

```
      with assms show " $\neg p \vee \neg q$ " ..
```

```
    qed
```

```
  qed
```

```
qed
```

```
-- "La demostración detallada es"
```

```
lemma ejercicio_57_2:
```

```
  assumes " $\neg(p \wedge q)$ "
```

```
  shows   " $\neg p \vee \neg q$ "
```

```
proof -
```

```
  have " $\neg p \vee p$ " by (rule excluded_middle)
```

```
  thus " $\neg p \vee \neg q$ "
```

```
  proof (rule disjE)
```

```
    assume " $\neg p$ "
```

```
    thus " $\neg p \vee \neg q$ " by (rule disjI1)
```

```
  next
```

```
    assume "p"
```

```
    have " $\neg q \vee q$ " by (rule excluded_middle)
```

```
    thus " $\neg p \vee \neg q$ "
```

```

proof
  assume "¬q"
  thus "¬p ∨ ¬q" by (rule disjI2)
next
  assume "q"
  have "p ∧ q" using 'p' 'q' by (rule conjI)
  show "¬p ∨ ¬q" using assms 'p ∧ q' by (rule notE)
qed
qed
qed

-- "La demostración automática es"
lemma ejercicio_57_3:
  assumes "¬(p ∧ q)"
  shows "¬p ∨ ¬q"
using assms
by auto

text {* -----
  Ejercicio 58. Demostrar
     $\vdash (p \rightarrow q) \vee (q \rightarrow p)$ 
  ----- *}

-- "La demostración estructurada es"
lemma ejercicio_58_1:
  "(p → q) ∨ (q → p)"
proof -
  have "¬p ∨ p" ..
  thus "(p → q) ∨ (q → p)"
  proof
    assume "¬p"
    have "p → q"
    proof
      assume "p"
      with '¬p' show "q" ..
    qed
    thus "(p → q) ∨ (q → p)" ..
  next
    assume "p"
    have "q → p"
    proof
      assume "q"
      show "p" using 'p' .
    qed
    thus "(p → q) ∨ (q → p)" ..
  qed

```

```

    qed
qed

-- "La demostración detallada es"
lemma ejercicio_58_2:
  "(p → q) ∨ (q → p)"
proof -
  have "¬p ∨ p" by (rule excluded_middle)
  thus "(p → q) ∨ (q → p)"
  proof
    assume "¬p"
    have "p → q"
    proof (rule impI)
      assume "p"
      show "q" using '¬p' 'p' by (rule notE)
    qed
    thus "(p → q) ∨ (q → p)" by (rule disjI1)
  next
    assume "p"
    have "q → p"
    proof
      assume "q"
      show "p" using 'p' by this
    qed
    thus "(p → q) ∨ (q → p)" by (rule disjI2)
  qed
qed

-- "La demostración automática es"
lemma ejercicio_58_3:
  "(p → q) ∨ (q → p)"
by auto

end

```

3.3. Ejercicios: Argumentación lógica proposicional

```

chapter {* T3R2: Argumentación proposicional *}

theory T3R2
imports Main
begin

```

text {*

 El objetivo de esta es relación formalizar y demostrar la corrección de los argumentos usando sólo las reglas básicas de deducción natural de la lógica proposicional (sin usar el método auto).

Las reglas básicas de la deducción natural son las siguientes:

- conjI: $\llbracket P; Q \rrbracket \Rightarrow P \wedge Q$
- conjunct1: $P \wedge Q \Rightarrow P$
- conjunct2: $P \wedge Q \Rightarrow Q$
- notnotD: $\neg\neg P \Rightarrow P$
- notnotI: $P \Rightarrow \neg\neg P$
- mp: $\llbracket P \rightarrow Q; P \rrbracket \Rightarrow Q$
- mt: $\llbracket F \rightarrow G; \neg G \rrbracket \Rightarrow \neg F$
- impI: $(P \Rightarrow Q) \Rightarrow P \rightarrow Q$
- disjI1: $P \Rightarrow P \vee Q$
- disjI2: $Q \Rightarrow P \vee Q$
- disjE: $\llbracket P \vee Q; P \Rightarrow R; Q \Rightarrow R \rrbracket \Rightarrow R$
- FalseE: $\text{False} \Rightarrow P$
- notE: $\llbracket \neg P; P \rrbracket \Rightarrow R$
- notI: $(P \Rightarrow \text{False}) \Rightarrow \neg P$
- iffI: $\llbracket P \Rightarrow Q; Q \Rightarrow P \rrbracket \Rightarrow P = Q$
- iffD1: $\llbracket Q = P; Q \rrbracket \Rightarrow P$
- iffD2: $\llbracket P = Q; Q \rrbracket \Rightarrow P$
- ccontr: $(\neg P \Rightarrow \text{False}) \Rightarrow P$

*}

text {*

Se usarán las reglas notnotI y mt que demostramos a continuación.

*}

lemma notnotI: " $P \Rightarrow \neg\neg P$ "

by auto

lemma mt: " $\llbracket F \rightarrow G; \neg G \rrbracket \Rightarrow \neg F$ "

by auto

text {*

 Ejercicio 1. Formalizar, y demostrar la corrección, del siguiente argumento

Si no está el mañana ni el ayer escrito, entonces no está el mañana escrito.

Usar M: El mañana está escrito.

```

    A: El ayer está escrito.
    ----- *}

-- "La demostración automática es:"
lemma ejercicio_1_1:
  "¬M ∧ ¬A ⇒ ¬M"
by auto

-- "La demostración estructurada es:"
lemma ejercicio_1_2:
  assumes "¬M ∧ ¬A"
  shows   "¬M"
using assms ..

-- "La demostración detallada es:"
lemma ejercicio_1_3:
  assumes "¬M ∧ ¬A"
  shows   "¬M"
using assms by (rule conjunct1)

text {* -----
  Ejercicio 2. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Siempre que un número x es divisible por 10, acaba en 0. El número
    x no acaba en 0. Por lo tanto, x no es divisible por 10.
  Usar D para "el número es divisible por 10" y
    C para "el número acaba en cero".
  ----- *}

-- "La demostración automática es"
lemma ejercicio_2_1:
  "[C → D; ¬D] ⇒ ¬C"
by auto

-- "La demostración estructurada es"
lemma ejercicio_2_2:
  assumes "C → D"
  assume "¬D"
  shows   "¬C"
proof
  assume "C"
  with assms(1) have "D" ..
  with assms(2) show False ..
qed

```

```
-- "La demostración detallada es"
lemma ejercicio_2_3:
  assumes "C  $\rightarrow$  D"
          " $\neg$ D"
  shows   " $\neg$ C"
proof
  assume "C"
  with assms(1) have "D" by (rule mp)
  with assms(2) show False by (rule notE)
qed

text {* -----
  Ejercicio 3. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Si no hay control de nacimientos, entonces la población crece
    ilimitadamente; pero si la población crece ilimitadamente,
    aumentará el índice de pobreza. Por consiguiente, si no hay control
    de nacimientos, aumentará el índice de pobreza.
  Usar N: Hay control de nacimientos.
      P: La población crece ilimitadamente,
      I: Aumentará el índice de pobreza.
  ----- *}

-- "La demostración automática es"
lemma ejercicio_3_1:
  "[ $\neg$ N  $\rightarrow$  P; P  $\rightarrow$  I]  $\implies$   $\neg$ N  $\rightarrow$  I"
by auto

-- "La demostración estructurada es"
lemma ejercicio_3_2:
  assumes " $\neg$ N  $\rightarrow$  P"
          "P  $\rightarrow$  I"
  shows   " $\neg$ N  $\rightarrow$  I"
proof
  assume " $\neg$ N"
  with assms(1) have "P" ..
  with assms(2) show "I" ..
qed

-- "La demostración detallada es"
lemma ejercicio_3_3:
  assumes " $\neg$ N  $\rightarrow$  P"
          "P  $\rightarrow$  I"
  shows   " $\neg$ N  $\rightarrow$  I"
proof (rule impI)
```

```

    assume "¬N"
    with assms(1) have "P" by (rule mp)
    with assms(2) show "I" by (rule mp)
qed

text {* -----
  Ejercicio 4. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Si te llamé por teléfono, entonces recibiste mi llamada y no es
    cierto que no te avisé del peligro que corrías. Por consiguiente,
    como te llamé, es cierto que te avisé del peligro que corrías.
  Usar T: Te llamé por teléfono.
      R: Recibiste mi llamada.
      P: Te avisé del peligro que corrías.
  ----- *}

-- "La demostración automática es:"
lemma ejercicio_4_1:
  "T → R ∧ ¬¬A ⇒ T → A"
by auto

-- "La demostración estructurada es:"
lemma ejercicio_4_2:
  assumes "T → R ∧ ¬¬A"
  shows   "T → A"
proof
  assume "T"
  with assms have "R ∧ ¬¬A" ..
  hence "¬¬A" ..
  thus "A" by (rule notnotD)
qed

-- "La demostración detallada es:"
lemma ejercicio_4_3:
  assumes "T → R ∧ ¬¬A"
  shows   "T → A"
proof (rule impI)
  assume "T"
  with assms(1) have "R ∧ ¬¬A" by (rule mp)
  hence "¬¬A" by (rule conjunct2)
  thus "A" by (rule notnotD)
qed

text {* -----
  Ejercicio 5. Formalizar, y demostrar la corrección, del siguiente

```

argumento

Si la válvula está abierta o la monitorización está preparada, entonces se envía una señal de reconocimiento y un mensaje de funcionamiento al controlador del ordenador. Si se envía un mensaje de funcionamiento al controlador del ordenador o el sistema está en estado normal, entonces se aceptan las órdenes del operador. Por lo tanto, si la válvula está abierta, entonces se aceptan las órdenes del operador.

Usar A : La válvula está abierta.

P : La monitorización está preparada.

R : Envía una señal de reconocimiento.

F : Envía un mensaje de funcionamiento.

N : El sistema está en estado normal.

O : Se aceptan órdenes del operador.

----- *}

-- "La demostración automática es"

lemma ejercicio_5_1:

" $[A \vee P \rightarrow R \wedge F; F \vee N \rightarrow O] \Rightarrow A \rightarrow O$ "

by auto

-- "La demostración estructurada es"

lemma ejercicio_5_2:

assumes "A $\vee$ P $\rightarrow$ R $\wedge$ F"

"F $\vee$ N $\rightarrow$ O"

shows "A $\rightarrow$ O"

proof

assume "A"

hence "A $\vee$ P" ..

with assms(1) **have** "R $\wedge$ F" ..

hence "F" ..

hence "F $\vee$ N" ..

with assms(2) **show** "O" ..

qed

-- "La demostración detallada es"

lemma ejercicio_5_3:

assumes "A $\vee$ P $\rightarrow$ R $\wedge$ F"

"F $\vee$ N $\rightarrow$ O"

shows "A $\rightarrow$ O"

proof (rule impI)

assume "A"

hence "A $\vee$ P" **by** (rule disjI1)

with assms(1) **have** "R $\wedge$ F" **by** (rule mp)

hence "F" **by** (rule conjunct2)

```

    hence "F v N" by (rule disjI1)
    with assms(2) show "Or" by (rule mp)
qed

text {* -----
  Ejercicio 6. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Cuando tanto la temperatura como la presión atmosférica permanecen
    constantes, no llueve. La temperatura permanece constante. Por lo
    tanto, en caso de que llueva, la presión atmosférica no permanece
    constante.
  Usar T para "La temperatura permanece constante",
    P para "La presión atmosférica permanece constante" y
    L para "Llueve".
  ----- *}

-- "La demostración automática es"
lemma ejercicio_6_1:
  "[[T ∧ P → ¬L; T]] ⇒ L → ¬P"
by auto

-- "La demostración estructurada es"
lemma ejercicio_6_2:
  assumes "T ∧ P → ¬L"
    "T"
  shows "L → ¬P"
proof
  assume "L"
  show "¬P"
  proof
    assume "P"
    with 'T' have "T ∧ P" ..
    with assms(1) have "¬L" ..
    thus "False" using 'L' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_6_3:
  assumes "T ∧ P → ¬L"
    "T"
  shows "L → ¬P"
proof (rule impI)
  assume "L"
  show "¬P"

```

```

proof
  assume "P"
  with 'T' have "T ∧ P" by (rule conjI)
  with assms(1) have "¬L" by (rule mp)
  thus "False" using 'L' by (rule notE)
qed
qed

text {* -----
  Ejercicio 7. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Si el general era leal, hubiera obedecido las órdenes, y si era
    inteligente las hubiera comprendido. O el general desobedeció las
    órdenes o no las comprendió. Luego, el general era desleal o no era
    inteligente.
  Usar L: El general es leal.
      Ob: El general obedece las órdenes.
      I: El general es inteligente.
      C: El general comprende las órdenes.
  ----- *}

-- "La demostración automática es"
lemma ejercicio_7_1:
  "[ (L → Ob) ∧ (I → C); ¬Ob ∨ ¬C ] ⇒ ¬L ∨ ¬I"
by auto

-- "La demostración estructurada es"
lemma ejercicio_7_2:
  assumes "(L → Ob) ∧ (I → C)"
  "¬Ob ∨ ¬C"
  shows "¬L ∨ ¬I"
using assms(2)
proof
  assume "¬Ob"
  have "L → Ob" using assms(1) ..
  hence "¬L" using '¬Ob' by (rule mt)
  thus "¬L ∨ ¬I" ..
next
  assume "¬C"
  have "I → C" using assms(1) ..
  hence "¬I" using '¬C' by (rule mt)
  thus "¬L ∨ ¬I" ..
qed

-- "La demostración detallada es"

```

```

lemma ejercicio_7_3:
  assumes "(L → 0b) ∧ (I → C)"
    "¬0b ∨ ¬C"
  shows "¬L ∨ ¬I"
using assms(2)
proof (rule disjE)
  assume "¬0b"
  have "L → 0b" using assms(1) by (rule conjunct1)
  hence "¬L" using '¬0b' by (rule mt)
  thus "¬L ∨ ¬I" by (rule disjI1)
next
  assume "¬C"
  have "I → C" using assms(1) by (rule conjunct2)
  hence "¬I" using '¬C' by (rule mt)
  thus "¬L ∨ ¬I" by (rule disjI2)
qed

text {* -----
  Ejercicio 8. Formalizar, y demostrar la corrección, del siguiente
  argumento
    Me matan si no trabajo y si trabajo me matan. Me matan siempre me
    matan.
  Usar M: Me matan.
    T: Trabajo.
  ----- *}

-- "La demostración automática es"
lemma ejercicio_8_1:
  "(¬T → M) ∧ (T → M) ⇒ M"
by auto

-- "La demostración estructurada es"
lemma ejercicio_8_2:
  assumes "(¬T → M) ∧ (T → M)"
  shows "M"
proof -
  have "¬T ∨ T" ..
  thus "M"
  proof
    assume "¬T"
    have "¬T → M" using assms ..
    thus "M" using '¬T' ..
  next
    assume "T"
    have "T → M" using assms ..
  qed

```

```

    thus "M" using 'T' ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_8_3:
  assumes "(¬T → M) ∧ (T → M)"
  shows "M"
proof -
  have "¬T ∨ T" by (rule excluded_middle)
  thus "M"
  proof (rule disjE)
    assume "¬T"
    have "¬T → M" using assms by (rule conjunct1)
    thus "M" using '¬T' by (rule mp)
  next
    assume "T"
    have "T → M" using assms by (rule conjunct2)
    thus "M" using 'T' by (rule mp)
  qed
qed

text {* -----
  Ejercicio 9. Formalizar, y demostrar la corrección, del siguiente
  argumento
    En cierto experimento, cuando hemos empleado un fármaco A, el
    paciente ha mejorado considerablemente en el caso, y sólo en el
    caso, en que no se haya empleado también un fármaco B. Además, o se
    ha empleado el fármaco A o se ha empleado el fármaco B. En
    consecuencia, podemos afirmar que si no hemos empleado el fármaco
    B, el paciente ha mejorado considerablemente.
  Usar A: Hemos empleado el fármaco A.
    B: Hemos empleado el fármaco B.
    M: El paciente ha mejorado notablemente.
  ----- *}

-- "La demostración automática es"
lemma ejercicio_9_1:
  assumes "A → (M ↔ ¬B)"
    "A ∨ B"
  shows "¬B → M"
using assms
by auto

-- "La demostración estructurada es"

```

```

lemma ejercicio_9_2:
  assumes "A  $\rightarrow$  (M  $\leftrightarrow$   $\neg$ B)"
          "A  $\vee$  B"
  shows   " $\neg$ B  $\rightarrow$  M"
proof
  assume " $\neg$ B"
  note 'A  $\vee$  B'
  hence "A"
  proof
    assume "A"
    thus "A" .
  next
    assume "B"
    with ' $\neg$ B' show "A" ..
  qed
  have "M  $\leftrightarrow$   $\neg$ B" using assms(1) 'A' ..
  thus "M" using ' $\neg$ B' ..
qed

-- "La demostración detallada es"
lemma ejercicio_9_3:
  assumes "A  $\rightarrow$  (M  $\leftrightarrow$   $\neg$ B)"
          "A  $\vee$  B"
  shows   " $\neg$ B  $\rightarrow$  M"
proof (rule impI)
  assume " $\neg$ B"
  note 'A  $\vee$  B'
  hence "A"
  proof (rule disjE)
    assume "A"
    thus "A" by this
  next
    assume "B"
    with ' $\neg$ B' show "A" by (rule notE)
  qed
  have "M  $\leftrightarrow$   $\neg$ B" using assms(1) 'A' by (rule mp)
  thus "M" using ' $\neg$ B' by (rule iffD2)
qed

text {* -----
  Ejercicio 10. Formalizar, y demostrar la corrección, del siguiente
  argumento
  Si trabajo gano dinero, pero si no trabajo gozo de la vida. Sin
  embargo, si trabajo no gozo de la vida, mientras que si no trabajo
  no gano dinero. Por lo tanto, gozo de la vida si y sólo si no gano

```

```

dinero.
Usar p: Trabajo
      q: Gano dinero.
      r: Gozo de la vida.
----- *}

-- "La demostración automática es"
lemma ejercicio_61_1:
  "[[p → q) ∧ (¬p → r); (p → ¬r) ∧ (¬p → ¬q)] ⇒ r ↔ ¬q"
by auto

-- "La demostración estructurada es"
lemma ejercicio_61_2:
  assumes "(p → q) ∧ (¬p → r)"
          "(p → ¬r) ∧ (¬p → ¬q)"
  shows   "r ↔ ¬q"
proof
  assume "r"
  hence "¬¬r" by (rule notnotI)
  have "p → ¬r" using assms(2) ..
  hence "¬p" using '¬¬r' by (rule mt)
  have "¬p → ¬q" using assms(2) ..
  thus "¬q" using '¬p' ..
next
  assume "¬q"
  have "p → q" using assms(1) ..
  hence "¬p" using '¬q' by (rule mt)
  have "¬p → r" using assms(1) ..
  thus "r" using '¬p' ..
qed

-- "La demostración detallada es"
lemma ejercicio_61_3:
  assumes "(p → q) ∧ (¬p → r)"
          "(p → ¬r) ∧ (¬p → ¬q)"
  shows   "r ↔ ¬q"
proof (rule iffI)
  assume "r"
  hence "¬¬r" by (rule notnotI)
  have "p → ¬r" using assms(2) by (rule conjunct1)
  hence "¬p" using '¬¬r' by (rule mt)
  have "¬p → ¬q" using assms(2) by (rule conjunct2)
  thus "¬q" using '¬p' by (rule mp)
next
  assume "¬q"

```

```

have "p → q" using assms(1) by (rule conjunct1)
hence "¬p" using '¬q' by (rule mt)
have "¬p → r" using assms(1) by (rule conjunct2)
thus "r" using '¬p' by (rule mp)
qed

text {* -----
Ejercicio 11. Formalizar, y demostrar la corrección, del siguiente
argumento
    Si Dios fuera capaz de evitar el mal y quisiera hacerlo, lo
    haría. Si Dios fuera incapaz de evitar el mal, no sería
    omnipotente; si no quisiera evitar el mal sería malévolo. Dios no
    evita el mal. Si Dios existe, es omnipotente y no es
    malévolo. Luego, Dios no existe.
Usar C: Dios es capaz de evitar el mal.
    Q: Dios quiere evitar el mal.
    Om: Dios es omnipotente.
    M: Dios es malévolo.
    P: Dios evita el mal.
    E: Dios existe.
----- *}

-- "La demostración automática es"
lemma ejercicio_11_1:
  "[C ∧ Q → P; (¬C → ¬Om) ∧ (¬Q → M); ¬P; E → Om ∧ ¬M] ⇒ ¬E"
by auto

-- "La demostración estructurada es"
lemma ejercicio_11_2:
  assumes "C ∧ Q → P"
    " (¬C → ¬Om) ∧ (¬Q → M) "
    "¬P"
    "E → Om ∧ ¬M"
  shows "¬E"
proof
  assume "E"
  have "Om ∧ ¬M" using assms(4) 'E' ..
  hence "Om" ..
  hence "¬¬Om" by (rule notnotI)
  have "¬C → ¬Om" using assms(2) ..
  hence "¬¬C" using '¬¬Om' by (rule mt)
  hence "C" by (rule notnotD)
  have "¬M" using 'Om ∧ ¬M' ..
  have "¬Q → M" using assms(2) ..
  hence "¬¬Q" using '¬M' by (rule mt)

```

```

    hence "Q" by (rule notnotD)
  with 'C' have "C ∧ Q" ..
  with assms(1) have "P" ..
  with assms(3) show False ..
qed

-- "La demostración detallada es"
lemma ejercicio_11_3:
  assumes "C ∧ Q → P"
          "(¬C → ¬O) ∧ (¬Q → M)"
          "¬P"
          "E → O ∧ ¬M"
  shows   "¬E"
proof (rule notI)
  assume "E"
  with assms(4) have "O ∧ ¬M" by (rule mp)
  hence "O" by (rule conjunct1)
  hence "¬¬O" by (rule notnotI)
  have "¬C → ¬O" using assms(2) by (rule conjunct1)
  hence "¬¬C" using '¬¬O' by (rule mt)
  hence "C" by (rule notnotD)
  have "¬M" using 'O ∧ ¬M' by (rule conjunct2)
  have "¬Q → M" using assms(2) by (rule conjunct2)
  hence "¬¬Q" using '¬M' by (rule mt)
  hence "Q" by (rule notnotD)
  with 'C' have "C ∧ Q" by (rule conjI)
  with assms(1) have "P" by (rule mp)
  with assms(3) show False by (rule notE)
qed

end

```

3.4. Ejercicios: Eliminación de conectivas

```
chapter {* T3R3: Eliminación de conectivas *
```

```
theory T3R3
imports Main
begin

```

```
text {*

```

 El objetivo de esta es relación es demostrar cómo a partir de las
 conectivas False, $\wedge$ y $\rightarrow$ pueden definirse las restantes.

```

-----
*}

text {* -----
  Ejercicio 1. Definir  $\leftrightarrow$  usando  $\wedge$  y  $\rightarrow$ ; es decir, sustituir en
     $(A \leftrightarrow B) = \text{indefinida}$ 
  la indefinida por una fórmula que sólo usa las conectivas  $\wedge$  y  $\leftrightarrow$  y
  demostrar la equivalencia.
----- *}

-- "La demostración automática es"
lemma bicondicional: "
   $(A \leftrightarrow B) = ((A \rightarrow B) \wedge (B \rightarrow A))$ "
by auto

-- "La demostración estructurada es"
lemma bicondicional_2: "
   $(A \leftrightarrow B) = ((A \rightarrow B) \wedge (B \rightarrow A))$ "
proof
  assume "A  $\leftrightarrow$  B"
  show " $(A \rightarrow B) \wedge (B \rightarrow A)$ "
  proof
    show "A  $\rightarrow$  B"
    proof
      assume "A"
      with 'A  $\leftrightarrow$  B' show "B" ..
    qed
  next
    show "B  $\rightarrow$  A"
    proof
      assume "B"
      with 'A  $\leftrightarrow$  B' show "A" ..
    qed
  qed
next
assume 1: " $(A \rightarrow B) \wedge (B \rightarrow A)$ "
show "A  $\leftrightarrow$  B"
proof
  assume "A"
  have "A  $\rightarrow$  B" using 1 ..
  thus "B" using 'A' ..
next
  assume "B"
  have "B  $\rightarrow$  A" using 1 ..

```

```

    thus "A" using 'B' ..
  qed
qed

-- "La demostración detallada es"
lemma bicondicional_3: "
  (A  $\leftrightarrow$  B) = ((A  $\rightarrow$  B)  $\wedge$  (B  $\rightarrow$  A))"
proof (rule iffI)
  assume "A  $\leftrightarrow$  B"
  show "(A  $\rightarrow$  B)  $\wedge$  (B  $\rightarrow$  A)"
  proof (rule conjI)
    show "A  $\rightarrow$  B"
    proof (rule impI)
      assume "A"
      with 'A  $\leftrightarrow$  B' show "B" by (rule iffD1)
    qed
  next
    show "B  $\rightarrow$  A"
    proof (rule impI)
      assume "B"
      with 'A  $\leftrightarrow$  B' show "A" by (rule iffD2)
    qed
  qed
next
  assume 1: "(A  $\rightarrow$  B)  $\wedge$  (B  $\rightarrow$  A)"
  show "A  $\leftrightarrow$  B"
  proof (rule iffI)
    assume "A"
    have "A  $\rightarrow$  B" using 1 by (rule conjunct1)
    thus "B" using 'A' by (rule mp)
  next
    assume "B"
    have "B  $\rightarrow$  A" using 1 by (rule conjunct2)
    thus "A" using 'B' by (rule mp)
  qed
qed

text {* -----
  Ejercicio 2. Definir  $\neg$  usando  $\rightarrow$  y False; es decir, sustituir en
    ( $\neg$ A) = indefinida
  la indefinida por una fórmula que sólo usa las conectivas  $\rightarrow$  y False
  y demostrar la equivalencia.
  ----- *}

-- "La demostración automática es"
```

```

lemma negacion:
  "(\neg A) = (A \rightarrow False)"
by auto

-- "La demostración estructurada es"
lemma negacion_2:
  "(\neg A) = (A \rightarrow False)"
proof
  assume "\neg A"
  show "A \rightarrow False"
  proof
    assume "A"
    with '\neg A' show False ..
  qed
next
  assume "A \rightarrow False"
  show "\neg A"
  proof
    assume "A"
    with 'A \rightarrow False' show False ..
  qed
qed

-- "La demostración detallada es"
lemma negacion_3:
  "(\neg A) = (A \rightarrow False)"
proof (rule iffI)
  assume "\neg A"
  show "A \rightarrow False"
  proof (rule impI)
    assume "A"
    with '\neg A' show False by (rule notE)
  qed
next
  assume "A \rightarrow False"
  show "\neg A"
  proof (rule notI)
    assume "A"
    with 'A \rightarrow False' show False by (rule mp)
  qed
qed

text {* -----
  Ejercicio 3. Definir  $\vee$  usando  $\rightarrow$  y  $\text{False}$ ; es decir, sustituir en
   $(A \vee B) = \text{indefinida}$ 

```

la indefinida por una fórmula que sólo usa las conectivas $\rightarrow$ y False y demostrar la equivalencia.

```
----- *}

-- "La demostración automática es"
lemma disyuncion:
  "(A v B) = ((A → False) → B)"
by auto

-- "La demostración estructurada es"
lemma disyuncion_2:
  "(A v B) = ((A → False) → B)"
proof
  assume "A v B"
  show "(A → False) → B"
  proof
    assume "A → False"
    note 'A v B'
    thus "B"
    proof
      assume "A"
      with 'A → False' have False ..
      thus "B" ..
    next
      assume "B"
      thus "B" .
    qed
  qed
next
  assume "(A → False) → B"
  show "A v B"
  proof -
    have "¬A v A" ..
    thus "A v B"
    proof
      assume "¬A"
      have "A → False"
      proof
        assume "A"
        with '¬A' show False ..
      qed
      with '(A → False) → B' have "B" ..
      thus "A v B" ..
    next
      assume "A"

```

```

        thus "A ∨ B" ..
      qed
    qed
  qed

-- "La demostración detallada es"
lemma disyuncion_3:
  "(A ∨ B) = ((A → False) → B)"
proof (rule iffI)
  assume "A ∨ B"
  show "(A → False) → B"
  proof (rule impI)
    assume "A → False"
    note 'A ∨ B'
    thus "B"
    proof (rule disjE)
      assume "A"
      with 'A → False' have False by (rule mp)
      thus "B" by (rule FalseE)
    next
      assume "B"
      thus "B" by this
    qed
  qed
qed
next
assume "(A → False) → B"
show "A ∨ B"
proof -
  have "¬A ∨ A" by (rule excluded_middle)
  thus "A ∨ B"
  proof (rule disjE)
    assume "¬A"
    have "A → False"
    proof (rule impI)
      assume "A"
      with '¬A' show False by (rule notE)
    qed
    with '(A → False) → B' have "B" by (rule mp)
    thus "A ∨ B" by (rule disjI2)
  next
    assume "A"
    thus "A ∨ B" by (rule disjI1)
  qed
qed
qed
qed

```

```

text {* -----
  Ejercicio 4. Encontrar una fórmula equivalente a
     $(A \vee (B \wedge C)) \leftrightarrow A$ 
  que sólo use las conectivas False,  $\wedge$  y  $\rightarrow$  y demostrar la
  equivalencia.
  ----- *}

-- "Se puede buscar la fórmula con"
lemma ejercicio_4a:
  " $(A \vee (B \wedge C)) \leftrightarrow A$ "
apply (simp only: bicondicional disyuncion)
oops

-- "La fórmula obtenida es"
-- " $((A \rightarrow \text{False}) \rightarrow B \wedge C) \rightarrow A \wedge (A \rightarrow (A \rightarrow \text{False}) \rightarrow B \wedge C)$ "

-- "La demostración de la equivalencia es"
lemma ejercicio_4b:
  " $((A \vee (B \wedge C)) \leftrightarrow A) \leftrightarrow$ 
     $((((A \rightarrow \text{False}) \rightarrow B \wedge C) \rightarrow A) \wedge (A \rightarrow (A \rightarrow \text{False}) \rightarrow B \wedge C))$ "
by (auto simp add: bicondicional disyuncion)

end

```


Capítulo 4

Deducción natural en lógica de primer orden

4.1. Tema: Deducción natural en lógica de primer orden

```
chapter {* Tema 4: Deducción natural en lógica de primer orden *
```

```
theory T4
imports Main
begin
```

```
text {*
  El objetivo de este tema es presentar la deducción natural en
  lógica de primer orden con Isabelle/HOL. La presentación se
  basa en los ejemplos de tema 2 del curso LMF que se encuentra
  en http://goo.gl/uJj8d (que a su vez se basa en el libro de
  Huth y Ryan "Logic in Computer Science" http://goo.gl/qsVpY ).
```

```
  La página al lado de cada ejemplo indica la página de las
  transparencias de LMF donde se encuentra la demostración. *}

```

```
section {* Reglas del cuantificador universal *
```

```
text {*
  Las reglas del cuantificador universal son
  · allE:  $\llbracket \forall x. P\ x; P\ a \rrbracket \Longrightarrow R$ 
  · allI:  $(\bigwedge x. P\ x) \Longrightarrow \forall x. P\ x$ 
  *}

```

```
text {*
```

```

Ejemplo 1 (p. 10). Demostrar que
   $P(c), \forall x. (P(x) \rightarrow \neg Q(x)) \vdash \neg Q(c)$ 
*)}

-- "La demostración detallada es"
lemma ejemplo_1a:
  assumes 1: "P(c)" and
    2: " $\forall x. (P(x) \rightarrow \neg Q(x))$ "
  shows " $\neg Q(c)$ "
proof -
  have 3: " $P(c) \rightarrow \neg Q(c)$ " using 2 by (rule allE)
  show 4: " $\neg Q(c)$ " using 3 1 by (rule mp)
qed

-- "La demostración estructurada es"
lemma ejemplo_1b:
  assumes "P(c)"
    " $\forall x. (P(x) \rightarrow \neg Q(x))$ "
  shows " $\neg Q(c)$ "
proof -
  have " $P(c) \rightarrow \neg Q(c)$ " using assms(2) ..
  thus " $\neg Q(c)$ " using assms(1) ..
qed

-- "La demostración automática es"
lemma ejemplo_1c:
  assumes "P(c)"
    " $\forall x. (P(x) \rightarrow \neg Q(x))$ "
  shows " $\neg Q(c)$ "
using assms
by auto

text {*
  Ejemplo 2 (p. 11). Demostrar que
     $\forall x. (P x \rightarrow \neg(Q x)), \forall x. P x \vdash \forall x. \neg(Q x)$ 
  *)}

-- "La demostración detallada es"
lemma ejemplo_2a:
  assumes 1: " $\forall x. (P x \rightarrow \neg(Q x))$ " and
    2: " $\forall x. P x$ "
  shows " $\forall x. \neg(Q x)$ "
proof -
  { fix a
    have 3: " $P a \rightarrow \neg(Q a)$ " using 1 by (rule allE)

```

```

    have 4: "P a" using 2 by (rule allE)
    have 5: "¬(Q a)" using 3 4 by (rule mp) }
  thus "∀x. ¬(Q x)" by (rule allI)
qed

```

-- "La demostración detallada hacia atrás es"

```

lemma ejemplo_2b:
  assumes 1: "∀x. (P x → ¬(Q x))" and
          2: "∀x. P x"
  shows "∀x. ¬(Q x)"
proof (rule allI)
  fix a
  have 3: "P a → ¬(Q a)" using 1 by (rule allE)
  have 4: "P a" using 2 by (rule allE)
  show 5: "¬(Q a)" using 3 4 by (rule mp)
qed

```

-- "La demostración estructurada es"

```

lemma ejemplo_2c:
  assumes "∀x. (P x → ¬(Q x))"
          "∀x. P x"
  shows "∀x. ¬(Q x)"
proof
  fix a
  have "P a" using assms(2) ..
  have "P a → ¬(Q a)" using assms(1) ..
  thus "¬(Q a)" using 'P a' ..
qed

```

-- "La demostración automática es"

```

lemma ejemplo_2d:
  assumes "∀x. (P x → ¬(Q x))"
          "∀x. P x"
  shows "∀x. ¬(Q x)"
using assms
by auto

```

section *{* Reglas del cuantificador existencial **}

```

text {*
  Las reglas del cuantificador existencial son
  · exI:       $P\ a \implies \exists x. P\ x$ 
  · exE:       $\llbracket \exists x. P\ x; \bigwedge x. P\ x \implies Q \rrbracket \implies Q$ 

```

En la regla exE la nueva variable se introduce mediante la declaración

```

    "obtain ... where ... by (rule exE)"
  *}

text {*
  Ejemplo (p. 12). Demostrar que
     $\forall x. P\ x \vdash \exists x. P\ x$ 
  *}

-- "La demostración detallada es"
lemma ejemplo_3a:
  assumes "∀x. P x"
  shows "∃x. P x"
proof -
  fix a
  have "P a" using assms by (rule allE)
  thus "∃x. P x" by (rule exI)
qed

-- "La demostración estructurada es"
lemma ejemplo_3b:
  assumes "∀x. P x"
  shows "∃x. P x"
proof -
  fix a
  have "P a" using assms ..
  thus "∃x. P x" ..
qed

-- "La demostración estructurada se puede simplificar"
lemma ejemplo_3c:
  assumes "∀x. P x"
  shows "∃x. P x"
proof (rule exI)
  fix a
  show "P a" using assms ..
qed

-- "La demostración estructurada se puede simplificar aún más"
lemma ejemplo_3d:
  assumes "∀x. P x"
  shows "∃x. P x"
proof
  fix a
  show "P a" using assms ..
qed

```

```

-- "La demostración automática es"
lemma ejemplo_3e:
  assumes "∀x. P x"
  shows "∃x. P x"
using assms
by auto

text {*
  Ejemplo 4 (p. 13). Demostrar
   $\forall x. (P x \rightarrow Q x), \exists x. P x \vdash \exists x. Q x$ 
*}

-- "La demostración detallada es"
lemma ejemplo_4a:
  assumes 1: "∀x. (P x → Q x)" and
  2: "∃x. P x"
  shows "∃x. Q x"
proof -
  obtain a where 3: "P a" using 2 by (rule exE)
  have 4: "P a → Q a" using 1 by (rule allE)
  have 5: "Q a" using 4 3 by (rule mp)
  thus 6: "∃x. Q x" by (rule exI)
qed

-- "La demostración estructurada es"
lemma ejemplo_4b:
  assumes "∀x. (P x → Q x)"
  "∃x. P x"
  shows "∃x. Q x"
proof -
  obtain a where "P a" using assms(2) ..
  have "P a → Q a" using assms(1) ..
  hence "Q a" using 'P a' ..
  thus "∃x. Q x" ..
qed

-- "La demostración automática es"
lemma ejemplo_4c:
  assumes "∀x. (P x → Q x)"
  "∃x. P x"
  shows "∃x. Q x"
using assms
by auto

```

```

section {* Demostración de equivalencias *}

text {*
  Ejemplo 5.1 (p. 15). Demostrar
     $\neg \forall x. P\ x \vdash \exists x. \neg(P\ x)$  *}

-- "La demostración detallada es"
lemma ejemplo_5_1a:
  assumes " $\neg(\forall x. P(x))$ "
  shows   " $\exists x. \neg P(x)$ "
proof (rule ccontr)
  assume " $\neg(\exists x. \neg P(x))$ "
  have " $\forall x. P(x)$ "
  proof (rule allI)
    fix a
    show " $P(a)$ "
    proof (rule ccontr)
      assume " $\neg P(a)$ "
      hence " $\exists x. \neg P(x)$ " by (rule exI)
      with ' $\neg(\exists x. \neg P(x))$ ' show False by (rule notE)
    qed
  qed
  with assms show False by (rule notE)
qed

-- "La demostración estructurada es"
lemma ejemplo_5_1b:
  assumes " $\neg(\forall x. P(x))$ "
  shows   " $\exists x. \neg P(x)$ "
proof (rule ccontr)
  assume " $\neg(\exists x. \neg P(x))$ "
  have " $\forall x. P(x)$ "
  proof
    fix a
    show " $P(a)$ "
    proof (rule ccontr)
      assume " $\neg P(a)$ "
      hence " $\exists x. \neg P(x)$ " ..
      with ' $\neg(\exists x. \neg P(x))$ ' show False ..
    qed
  qed
  with assms show False ..
qed

-- "La demostración automática es"

```

```

lemma ejemplo_5_1c:
  assumes "¬(∀x. P(x))"
  shows   "∃x. ¬P(x)"
using assms
by auto

text {*
  Ejemplo 5.2 (p. 16). Demostrar
     $\exists x. \neg(P\ x) \vdash \neg\forall x. P\ x$  *}

-- "La demostración detallada es"
lemma ejemplo_5_2a:
  assumes "∃x. ¬P(x)"
  shows   "¬(∀x. P(x))"
proof (rule notI)
  assume "∀x. P(x)"
  obtain a where "¬P(a)" using assms by (rule exE)
  have "P(a)" using '∀x. P(x)' by (rule allE)
  with '¬P(a)' show False by (rule notE)
qed

-- "La demostración estructurada es"
lemma ejemplo_5_2b:
  assumes "∃x. ¬P(x)"
  shows   "¬(∀x. P(x))"
proof
  assume "∀x. P(x)"
  obtain a where "¬P(a)" using assms ..
  have "P(a)" using '∀x. P(x)' ..
  with '¬P(a)' show False ..
qed

-- "La demostración automática es"
lemma ejemplo_5_2c:
  assumes "∃x. ¬P(x)"
  shows   "¬(∀x. P(x))"
using assms
by auto

text {*
  Ejemplo 5.3 (p. 17). Demostrar
     $\vdash \neg\forall x. P\ x \leftrightarrow \exists x. \neg(P\ x)$  *}

-- "La demostración detallada es"
lemma ejemplo_5_3a:

```

```

    "(¬(∀x. P(x))) ↔ (∃x. ¬P(x))"
proof (rule iffI)
  assume "¬(∀x. P(x))"
  thus "∃x. ¬P(x)" by (rule ejemplo_5_1a)
next
  assume "∃x. ¬P(x)"
  thus "¬(∀x. P(x))" by (rule ejemplo_5_2a)
qed

-- "La demostración automática es"
lemma ejemplo_5_3b:
  "(¬(∀x. P(x))) ↔ (∃x. ¬P(x))"
by auto

text {*
  Ejemplo 6.1 (p. 18). Demostrar
     $\forall x. P(x) \wedge Q(x) \vdash (\forall x. P(x)) \wedge (\forall x. Q(x))$  *}

-- "La demostración detallada es"
lemma ejemplo_6_1a:
  assumes "∀x. P(x) ∧ Q(x)"
  shows   "(∀x. P(x)) ∧ (∀x. Q(x))"
proof (rule conjI)
  show "∀x. P(x)"
  proof (rule allI)
    fix a
    have "P(a) ∧ Q(a)" using assms by (rule allE)
    thus "P(a)" by (rule conjunct1)
  qed
next
  show "∀x. Q(x)"
  proof (rule allI)
    fix a
    have "P(a) ∧ Q(a)" using assms by (rule allE)
    thus "Q(a)" by (rule conjunct2)
  qed
qed

-- "La demostración estructurada es"
lemma ejemplo_6_1b:
  assumes "∀x. P(x) ∧ Q(x)"
  shows   "(∀x. P(x)) ∧ (∀x. Q(x))"
proof
  show "∀x. P(x)"
proof

```

```

    fix a
    have "P(a) ∧ Q(a)" using assms ..
    thus "P(a)" ..
qed
next
show "∀x. Q(x)"
proof
  fix a
  have "P(a) ∧ Q(a)" using assms ..
  thus "Q(a)" ..
qed
qed

-- "La demostración automática es"
lemma ejemplo_6_1c:
  assumes "∀x. P(x) ∧ Q(x)"
  shows   "(∀x. P(x)) ∧ (∀x. Q(x))"
using assms
by auto

text {*
  Ejemplo 6.2 (p. 19). Demostrar
  (∀x. P(x)) ∧ (∀x. Q(x)) ⊢ ∀x. P(x) ∧ Q(x)  *}

-- "La demostración detallada es"
lemma ejemplo_6_2a:
  assumes "(∀x. P(x)) ∧ (∀x. Q(x))"
  shows   "∀x. P(x) ∧ Q(x)"
proof (rule allI)
  fix a
  have "∀x. P(x)" using assms by (rule conjunct1)
  hence "P(a)" by (rule allE)
  have "∀x. Q(x)" using assms by (rule conjunct2)
  hence "Q(a)" by (rule allE)
  with 'P(a)' show "P(a) ∧ Q(a)" by (rule conjI)
qed

-- "La demostración estructurada es"
lemma ejemplo_6_2b:
  assumes "(∀x. P(x)) ∧ (∀x. Q(x))"
  shows   "∀x. P(x) ∧ Q(x)"
proof
  fix a
  have "∀x. P(x)" using assms ..
  hence "P(a)" by (rule allE)

```

```

    have "∀x. Q(x)" using assms ..
    hence "Q(a)" ..
    with 'P(a)' show "P(a) ∧ Q(a)" ..
qed

-- "La demostración automática es"
lemma ejemplo_6_2c:
  assumes "(∀x. P(x)) ∧ (∀x. Q(x))"
  shows    "∀x. P(x) ∧ Q(x)"
using assms
by auto

text {*
  Ejemplo 6.3 (p. 20). Demostrar
     $\vdash \forall x. P(x) \wedge Q(x) \leftrightarrow (\forall x. P(x)) \wedge (\forall x. Q(x))$  *}

-- "La demostración detallada es"
lemma ejemplo_6_3a:
  "(∀x. P(x) ∧ Q(x)) ↔ ((∀x. P(x)) ∧ (∀x. Q(x)))"
proof (rule iffI)
  assume "∀x. P(x) ∧ Q(x)"
  thus "(∀x. P(x)) ∧ (∀x. Q(x))" by (rule ejemplo_6_1a)
next
  assume "(∀x. P(x)) ∧ (∀x. Q(x))"
  thus "∀x. P(x) ∧ Q(x)" by (rule ejemplo_6_2a)
qed

text {*
  Ejemplo 7.1 (p. 21). Demostrar
     $(\exists x. P(x)) \vee (\exists x. Q(x)) \vdash \exists x. P(x) \vee Q(x)$  *}

-- "La demostración detallada es"
lemma ejemplo_7_1a:
  assumes "(∃x. P(x)) ∨ (∃x. Q(x))"
  shows    "∃x. P(x) ∨ Q(x)"
using assms
proof (rule disjE)
  assume "∃x. P(x)"
  then obtain a where "P(a)" by (rule exE)
  hence "P(a) ∨ Q(a)" by (rule disjI1)
  thus "∃x. P(x) ∨ Q(x)" by (rule exI)
next
  assume "∃x. Q(x)"
  then obtain a where "Q(a)" by (rule exE)
  hence "P(a) ∨ Q(a)" by (rule disjI2)

```

```

    thus "∃x. P(x) ∨ Q(x)" by (rule exI)
qed

-- "La demostración estructurada es"
lemma ejemplo_7_1b:
  assumes "(∃x. P(x)) ∨ (∃x. Q(x))"
  shows "∃x. P(x) ∨ Q(x)"
using assms
proof
  assume "∃x. P(x)"
  then obtain a where "P(a)" ..
  hence "P(a) ∨ Q(a)" ..
  thus "∃x. P(x) ∨ Q(x)" ..
next
  assume "∃x. Q(x)"
  then obtain a where "Q(a)" ..
  hence "P(a) ∨ Q(a)" ..
  thus "∃x. P(x) ∨ Q(x)" ..
qed

-- "La demostración automática es"
lemma ejemplo_7_1c:
  assumes "(∃x. P(x)) ∨ (∃x. Q(x))"
  shows "∃x. P(x) ∨ Q(x)"
using assms
by auto

text {
  Ejemplo 7.2 (p. 22). Demostrar
   $\exists x. P(x) \vee Q(x) \vdash (\exists x. P(x)) \vee (\exists x. Q(x))$  *}

-- "La demostración detallada es"
lemma ejemplo_7_2a:
  assumes "∃x. P(x) ∨ Q(x)"
  shows "(∃x. P(x)) ∨ (∃x. Q(x))"
proof -
  obtain a where "P(a) ∨ Q(a)" using assms by (rule exE)
  thus "(∃x. P(x)) ∨ (∃x. Q(x))"
  proof (rule disjE)
    assume "P(a)"
    hence "∃x. P(x)" by (rule exI)
    thus "(∃x. P(x)) ∨ (∃x. Q(x))" by (rule disjI1)
  next
    assume "Q(a)"
    hence "∃x. Q(x)" by (rule exI)
  qed

```

```

    thus "( $\exists x. P(x)$ )  $\vee$  ( $\exists x. Q(x)$ )" by (rule disjI2)
  qed
qed

-- "La demostración estructurada es"
lemma ejercicio_7_2b:
  assumes " $\exists x. P(x) \vee Q(x)$ "
  shows   " $(\exists x. P(x)) \vee (\exists x. Q(x))$ "
proof -
  obtain a where "P(a)  $\vee$  Q(a)" using assms ..
  thus "( $\exists x. P(x)$ )  $\vee$  ( $\exists x. Q(x)$ )"
  proof
    assume "P(a)"
    hence " $\exists x. P(x)$ " ..
    thus "( $\exists x. P(x)$ )  $\vee$  ( $\exists x. Q(x)$ )" ..
  next
    assume "Q(a)"
    hence " $\exists x. Q(x)$ " ..
    thus "( $\exists x. P(x)$ )  $\vee$  ( $\exists x. Q(x)$ )" ..
  qed
qed

-- "La demostración automática es"
lemma ejercicio_7_2c:
  assumes " $\exists x. P(x) \vee Q(x)$ "
  shows   " $(\exists x. P(x)) \vee (\exists x. Q(x))$ "
using assms
by auto

text {*
  Ejemplo 7.3 (p. 23). Demostrar
     $\vdash ((\exists x. P(x)) \vee (\exists x. Q(x))) \leftrightarrow (\exists x. P(x) \vee Q(x))$  *}

-- "La demostración detallada es"
lemma ejemplo_7_3a:
  " $((\exists x. P(x)) \vee (\exists x. Q(x))) \leftrightarrow (\exists x. P(x) \vee Q(x))$ "
proof (rule iffI)
  assume " $(\exists x. P(x)) \vee (\exists x. Q(x))$ "
  thus " $\exists x. P(x) \vee Q(x)$ " by (rule ejemplo_7_1a)
next
  assume " $\exists x. P(x) \vee Q(x)$ "
  thus " $(\exists x. P(x)) \vee (\exists x. Q(x))$ " by (rule ejemplo_7_2a)
qed

-- "La demostración automática es"

```

```

lemma ejemplo_7_3b:
  "(( $\exists x. P(x)$ )  $\vee$  ( $\exists x. Q(x)$ ))  $\leftrightarrow$  ( $\exists x. P(x) \vee Q(x)$ )"
using assms
by auto

text {*
  Ejemplo 8.1 (p. 24). Demostrar
     $\exists x y. P(x,y) \vdash \exists y x. P(x,y)$  *}

-- "La demostración detallada es"
lemma ejemplo_8_1a:
  assumes " $\exists x y. P(x,y)$ "
  shows   " $\exists y x. P(x,y)$ "
proof -
  obtain a where " $\exists y. P(a,y)$ " using assms by (rule exE)
  then obtain b where " $P(a,b)$ " by (rule exE)
  hence " $\exists x. P(x,b)$ " by (rule exI)
  thus " $\exists y x. P(x,y)$ " by (rule exI)
qed

-- "La demostración estructurada es"
lemma ejemplo_8_1b:
  assumes " $\exists x y. P(x,y)$ "
  shows   " $\exists y x. P(x,y)$ "
proof -
  obtain a where " $\exists y. P(a,y)$ " using assms ..
  then obtain b where " $P(a,b)$ " ..
  hence " $\exists x. P(x,b)$ " ..
  thus " $\exists y x. P(x,y)$ " ..
qed

-- "La demostración automática es"
lemma ejemplo_8_1c:
  assumes " $\exists x y. P(x,y)$ "
  shows   " $\exists y x. P(x,y)$ "
using assms
by auto

text {*
  Ejemplo 8.2. Demostrar
     $\exists y x. P(x,y) \vdash \exists x y. P(x,y)$  *}

-- "La demostración detallada es"
lemma ejemplo_8_2a:
  assumes " $\exists y x. P(x,y)$ "

```

```

    shows   "∃x y. P(x,y)"
proof -
  obtain b where "∃x. P(x,b)" using assms by (rule exE)
  then obtain a where "P(a,b)" by (rule exE)
  hence "∃y. P(a,y)" by (rule exI)
  thus "∃x y. P(x,y)" by (rule exI)
qed

```

-- "La demostración estructurada es"

```

lemma ejemplo_8_2b:
  assumes "∃y x. P(x,y)"
  shows   "∃x y. P(x,y)"
proof -
  obtain b where "∃x. P(x,b)" using assms ..
  then obtain a where "P(a,b)" ..
  hence "∃y. P(a,y)" ..
  thus "∃x y. P(x,y)" ..
qed

```

-- "La demostración estructurada es"

```

lemma ejemplo_8_2c:
  assumes "∃y x. P(x,y)"
  shows   "∃x y. P(x,y)"
using assms
by auto

```

```

text {*
  Ejemplo 8.3 (p. 25). Demostrar
    ⊢ (∃x y. P(x,y)) ↔ (∃y x. P(x,y))  *}

```

-- "La demostración detallada es"

```

lemma ejemplo_8_3a:
  "(∃x y. P(x,y)) ↔ (∃y x. P(x,y))"
proof (rule iffI)
  assume "∃x y. P(x,y)"
  thus "∃y x. P(x,y)" by (rule ejemplo_8_1a)
next
  assume "∃y x. P(x,y)"
  thus "∃x y. P(x,y)" by (rule ejemplo_8_2a)
qed

```

-- "La demostración automática es"

```

lemma ejemplo_8_3b:
  "(∃x y. P(x,y)) ↔ (∃y x. P(x,y))"
by auto

```

```

section {* Reglas de la igualdad *}

text {*
  Las reglas básicas de la igualdad son:
  · refl:  $t = t$ 
  · subst:  $\llbracket s = t; P\ s \rrbracket \implies P\ t$ 
*}

text {*
  Ejemplo 9 (p. 27). Demostrar
   $x+1 = 1+x, x+1 > 1 \rightarrow x+1 > 0 \vdash 1+x > 1 \rightarrow 1+x > 0$ 
*}

-- "La demostración detallada es"
lemma ejemplo_9a:
  assumes "x+1 = 1+x"
           "x+1 > 1  $\rightarrow$  x+1 > 0"
  shows   "1+x > 1  $\rightarrow$  1+x > 0"
proof -
  show "1+x > 1  $\rightarrow$  1+x > 0" using assms by (rule subst)
qed

-- "La demostración estructurada es"
lemma ejemplo_9b:
  assumes "x+1 = 1+x"
           "x+1 > 1  $\rightarrow$  x+1 > 0"
  shows   "1+x > 1  $\rightarrow$  1+x > 0"
using assms
by (rule subst)

-- "La demostración automática es"
lemma ejemplo_9c:
  assumes "x+1 = 1+x"
           "x+1 > 1  $\rightarrow$  x+1 > 0"
  shows   "1+x > 1  $\rightarrow$  1+x > 0"
using assms
by auto

text {*
  Ejemplo 10 (p. 27). Demostrar
   $x = y, y = z \vdash x = z$ 
*}

-- "La demostración detallada es"

```

```

lemma ejemplo_10a:
  assumes "x = y"
          "y = z"
  shows   "x = z"
proof -
  show "x = z" using assms(2,1) by (rule subst)
qed

```

-- "La demostración estructurada es"

```

lemma ejemplo_10b:
  assumes "x = y"
          "y = z"
  shows   "x = z"
using assms(2,1)
by (rule subst)

```

-- "La demostración automática es"

```

lemma ejemplo_10c:
  assumes "x = y"
          "y = z"
  shows   "x = z"
using assms
by auto

```

```

text {*
  Ejemplo 11 (p. 28). Demostrar
     $s = t \vdash t = s$ 
*}

```

-- "La demostración detallada es"

```

lemma ejemplo_11a:
  assumes "s = t"
  shows   "t = s"
proof -
  have "s = s" by (rule refl)
  with assms show "t = s" by (rule subst)
qed

```

-- "La demostración automática es"

```

lemma ejemplo_11b:
  assumes "s = t"
  shows   "t = s"
using assms
by auto

```

```

text {*
  Algunas reglas derivadas de la igualdad son:
  · trans:       $\llbracket r = s; s = t \rrbracket \implies r = t$ 
  · sym:         $s = t \implies t = s$ 
  · not_sym:     $t \neq s \implies s \neq t$ 
  · ssubst:      $\llbracket t = s; P\ s \rrbracket \implies P\ t$ 
  · box_equals:  $\llbracket a = b; a = c; b = d \rrbracket \implies c = d$ 
  · arg_cong:    $x = y \implies f\ x = f\ y$ 
  · fun_cong:    $f = g \implies f\ x = g\ x$ 
  · cong:        $\llbracket f = g; x = y \rrbracket \implies f\ x = g\ y$ 
*}

-- "La demostración estructurada de not_sym es"
lemma not_sym_1:
  assumes "t ≠ s"
  shows   "s ≠ t"
proof
  assume "s = t"
  hence "t = s" ..
  show False using assms(1) 't = s' ..
qed

-- "La demostración detallada de not_sym es"
lemma not_sym_2:
  assumes "t ≠ s"
  shows   "s ≠ t"
proof (rule notI)
  assume "s = t"
  hence "t = s" by (rule sym)
  show False using assms(1) 't = s' by (rule notE)
qed

-- "La demostración automática de not_sym es"
lemma not_sym_3:
  assumes "t ≠ s"
  shows   "s ≠ t"
using assms
by auto

-- "La demostración estructurada de ssubst es"
lemma sssubs_1:
  assumes "t = s"
         "P s"
  shows   "P t"
proof -

```

```

    have "s = t" using assms(1) ..
    thus "P t" using assms(2) by (rule subst)
qed

-- "La demostración detallada de ssubst es"
lemma sssubs_2:
  assumes "t = s"
    "P s"
  shows "P t"
proof -
  have "s = t" using assms(1) by (rule sym)
  thus "P t" using assms(2) by (rule subst)
qed

-- "La demostración automática de ssubst es"
lemma ssubst_3:
  assumes "t = s"
    "P s"
  shows "P t"
using assms
by auto

-- "La demostración detallada de box_equals es"
lemma box_equals_1:
  assumes "a = b"
    "a = c"
    "b = d"
  shows "c = d"
proof -
  have "c = b" using assms(2,1) by (rule subst)
  with assms(3) show "c = d" by (rule subst)
qed

-- "La demostración semiautomática de box_equals es"
lemma box_equals_2:
  assumes "a = b"
    "a = c"
    "b = d"
  shows "c = d"
using assms
by (rule box_equals)

-- "La demostración automática de box_equals es"
lemma box_equals_3:
  assumes "a = b"

```

```

      "a = c"
      "b = d"
    shows   "c = d"
  using assms
  by auto

-- "La demostración detallada de arg_cong es"
lemma arg_cong_1:
  assumes "x = y"
  shows   "f x = f y"
proof -
  have "f x = f x" by (rule refl)
  with assms(1) show "f x = f y" by (rule subst)
qed

-- "La demostración semiautomática de arg_cong es"
lemma arg_cong_2:
  assumes "x = y"
  shows   "f x = f y"
using assms
by (rule arg_cong)

-- "La demostración automática de arg_cong es"
lemma arg_cong_3:
  assumes "x = y"
  shows   "f x = f y"
using assms
by auto

-- "La demostración detallada de fun_cong es"
lemma fun_cong_1:
  assumes "f = g"
  shows   "f x = g x"
proof -
  have "f x = f x" by (rule refl)
  with assms(1) show "f x = g x" by (erule_tac P = "λh. f x = h x" in subst)
qed

-- "La demostración semiautomática de fun_cong es"
lemma fun_cong_2:
  assumes "f = g"
  shows   "f x = g x"
using assms
by (rule fun_cong)

```

```

-- "La demostración automática de fun_cong es"
lemma fun_cong_3:
  assumes "f = g"
  shows   "f x = g x"
using assms
by auto

-- "La demostración detallada de cong es"
lemma cong_1:
  assumes "f = g"
         "x = y"
  shows   "f x = g y"
proof -
  have "f x = f x" by (rule refl)
  with assms(2) have "f x = f y" by (rule subst)
  with assms(1) show "f x = g y" by (erule_tac P = "λh. f x = h y" in subst)
qed

-- "La demostración semiautomática de cong es"
lemma cong_2:
  assumes "f = g"
         "x = y"
  shows   "f x = g y"
using assms
by (rule cong)

-- "La demostración automática de cong es"
lemma cong_3:
  assumes "f = g"
         "x = y"
  shows   "f x = g y"
using assms
by auto

section {* Ejemplo de razonamiento sobre programa *}

text {
  Definición. El número natural  $x$  divide al número natural  $y$  si existe
  un natural  $k$  tal que  $kx = y$ .
}

definition divide :: "nat  $\Rightarrow$  nat  $\Rightarrow$  bool" where
  "divide x y  $\equiv \exists k. k * x = y$ "

text {

```

```

    La definición de divide se añade a las reglas de simplificación.
*}

declare divide_def[simp]

text {*
    Ejemplo 8 [Transitividad de la divisibilidad]. Sean  $a$ ,  $b$  y  $c$  números
    naturales. Si  $b$  es divisible por  $a$  y  $c$  es divisible por  $b$ , entonces  $c$ 
    es divisible por  $a$ .
*}

-- "La demostración estructurada es"
lemma ejemplo_8_1:
  assumes "divide a b"
          "divide b c"
  shows   "divide a c"
proof -
  obtain m where "m*a = b" using assms(1) by auto
  obtain n where "n*b = c" using assms(2) by auto
  hence "m*n*a = c" using 'm*a = b' by auto
  hence "∃k. k*a = c" by (rule exI)
  thus "divide a c" by simp
qed

-- "La demostración puede simplificarse"
lemma ejemplo_8_2:
  assumes "divide a b"
          "divide b c"
  shows   "divide a c"
proof simp
  obtain m where "m*a = b" using assms(1) by auto
  obtain n where "n*b = c" using assms(2) by auto
  hence "m*n*a = c" using 'm*a = b' by auto
  thus "∃k. k*a = c" ..
qed

-- "La demostración automática es"
lemma ejemplo_8_3:
  assumes "divide a b"
          "divide b c"
  shows   "divide a c"
using assms
by auto

section {* Razonamiento ecuacional *}

```

```

text {*
  El razonamiento ecuacional se realiza usando la combinación de "also"
  (además) y "finally" (finalmente).
*}

text {*
  Ejemplo 9 [Razonamiento ecuacional]. Si  $a=b$ ,  $b=c$  y  $c=d$ , entonces  $a=d$ .
*}

-- "La demostración detallada es"
lemma ejemplo_9_1:
  assumes "a = b"
          "b = c"
          "c = d"
  shows   "a = d"
proof -
  have "a = b" by (rule assms(1))
  also have "... = c" by (rule assms(2))
  also have "... = d" by (rule assms(3))
  finally show "a = d" .
qed

-- "La demostración automática es"
lemma ejemplo_9_2:
  assumes "a = b"
          "b = c"
          "c = d"
  shows   "a = d"
using assms
by metis

end

```

4.2. Ejercicios: Deducción natural en lógica de primer orden

```

chapter {* T4R1: Deducción natural de primer orden *}

theory T4R1
imports Main
begin

```

text {*

Demostrar o refutar los siguientes lemas usando sólo las reglas básicas de deducción natural de la lógica proposicional, de los cuantificadores y de la igualdad:

- conjI: $\llbracket P; Q \rrbracket \Rightarrow P \wedge Q$
- conjunct1: $P \wedge Q \Rightarrow P$
- conjunct2: $P \wedge Q \Rightarrow Q$
- notnotD: $\neg\neg P \Rightarrow P$
- mp: $\llbracket P \rightarrow Q; P \rrbracket \Rightarrow Q$
- impI: $(P \Rightarrow Q) \Rightarrow P \rightarrow Q$
- disjI1: $P \Rightarrow P \vee Q$
- disjI2: $Q \Rightarrow P \vee Q$
- disjE: $\llbracket P \vee Q; P \Rightarrow R; Q \Rightarrow R \rrbracket \Rightarrow R$
- FalseE: $\text{False} \Rightarrow P$
- notE: $\llbracket \neg P; P \rrbracket \Rightarrow R$
- notI: $(P \Rightarrow \text{False}) \Rightarrow \neg P$
- iffI: $\llbracket P \Rightarrow Q; Q \Rightarrow P \rrbracket \Rightarrow P = Q$
- iffD1: $\llbracket Q = P; Q \rrbracket \Rightarrow P$
- iffD2: $\llbracket P = Q; Q \rrbracket \Rightarrow P$
- ccontr: $(\neg P \Rightarrow \text{False}) \Rightarrow P$

- allI: $\llbracket \forall x. P\ x; P\ x \Rightarrow R \rrbracket \Rightarrow R$
- allE: $(\bigwedge x. P\ x) \Rightarrow \forall x. P\ x$
- exI: $P\ x \Rightarrow \exists x. P\ x$
- exE: $\llbracket \exists x. P\ x; \bigwedge x. P\ x \Rightarrow Q \rrbracket \Rightarrow Q$

- refl: $t = t$
- subst: $\llbracket s = t; P\ s \rrbracket \Rightarrow P\ t$
- trans: $\llbracket r = s; s = t \rrbracket \Rightarrow r = t$
- sym: $s = t \Rightarrow t = s$
- not_sym: $t \neq s \Rightarrow s \neq t$
- ssubst: $\llbracket t = s; P\ s \rrbracket \Rightarrow P\ t$
- box_equals: $\llbracket a = b; a = c; b = d \rrbracket \Rightarrow a = d$
- arg_cong: $x = y \Rightarrow f\ x = f\ y$
- fun_cong: $f = g \Rightarrow f\ x = g\ x$
- cong: $\llbracket f = g; x = y \rrbracket \Rightarrow f\ x = g\ y$

*}

text {*

Se usarán las reglas notnotI y mt que demostramos a continuación.

*}

lemma notnotI: " $P \Rightarrow \neg\neg P$ "

by auto

```

lemma mt: "[[F → G; ¬G]] ⇒ ¬F"
by auto

text {* -----
  Ejercicio 1. Demostrar
     $\forall x. P\ x \rightarrow Q\ x \vdash (\forall x. P\ x) \rightarrow (\forall x. Q\ x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_1a:
  " $\forall x. P\ x \rightarrow Q\ x \Rightarrow (\forall x. P\ x) \rightarrow (\forall x. Q\ x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_1b:
  assumes " $\forall x. P\ x \rightarrow Q\ x$ "
  shows   " $(\forall x. P\ x) \rightarrow (\forall x. Q\ x)$ "
proof
  assume " $\forall x. P\ x$ "
  show " $\forall x. Q\ x$ "
  proof
    fix a
    have "P a" using ' $\forall x. P\ x$ ' ..
    have "P a → Q a" using assms(1) ..
    thus "Q a" using 'P a' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_1c:
  assumes " $\forall x. P\ x \rightarrow Q\ x$ "
  shows   " $(\forall x. P\ x) \rightarrow (\forall x. Q\ x)$ "
proof (rule impI)
  assume " $\forall x. P\ x$ "
  show " $\forall x. Q\ x$ "
  proof (rule allI)
    fix a
    have "P a" using ' $\forall x. P\ x$ ' by (rule allE)
    have "P a → Q a" using assms(1) by (rule allE)
    thus "Q a" using 'P a' by (rule mp)
  qed
qed

text {* -----
  Ejercicio 2. Demostrar

```

```

       $\exists x. \neg(P\ x) \vdash \neg(\forall x. P\ x)$ 
----- *}

-- "La demostración automática es"
lemma ejercicio_2a: " $\exists x. \neg(P\ x) \implies \neg(\forall x. P\ x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_2b:
  assumes " $\exists x. \neg(P\ x)$ "
  shows   " $\neg(\forall x. P\ x)$ "
proof
  assume " $\forall x. P\ x$ "
  obtain a where " $\neg(P\ a)$ " using assms(1) ..
  have "P a" using ' $\forall x. P\ x$ ' ..
  with ' $\neg(P\ a)$ ' show False ..
qed

-- "La demostración detallada es"
lemma ejercicio_2c:
  assumes " $\exists x. \neg(P\ x)$ "
  shows   " $\neg(\forall x. P\ x)$ "
proof (rule notI)
  assume " $\forall x. P\ x$ "
  obtain a where " $\neg(P\ a)$ " using assms(1) by (rule exE)
  have "P a" using ' $\forall x. P\ x$ ' by (rule allE)
  with ' $\neg(P\ a)$ ' show False by (rule notE)
qed

text {* -----
  Ejercicio 3. Demostrar
     $\forall x. P\ x \vdash \forall y. P\ y$ 
----- *}

-- "La demostración automática es"
lemma ejercicio_3a: " $\forall x. P\ x \implies \forall y. P\ y$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_3b:
  assumes " $\forall x. P\ x$ "
  shows   " $\forall y. P\ y$ "
proof
  fix a
  show "P a" using assms ..

```

qed

-- "La demostración estructurada es"

```
lemma ejercicio_3c:
  assumes "∀x. P x"
  shows   "∀y. P y"
proof (rule allI)
  fix a
  show "P a" using assms by (rule allE)
```

qed

```
text {* -----
  Ejercicio 4. Demostrar
     $\forall x. P x \rightarrow Q x \vdash (\forall x. \neg(Q x)) \rightarrow (\forall x. \neg(P x))$ 
  ----- *}

```

-- "La demostración automática es"

```
lemma ejercicio_4a:
  "∀x. P x → Q x ⇒ (∀x. ¬(Q x)) → (∀x. ¬(P x))"
by auto
```

-- "La demostración estructurada es"

```
lemma ejercicio_4b:
  assumes "∀x. P x → Q x"
  shows   "(∀x. ¬(Q x)) → (∀x. ¬(P x))"
proof
  assume "∀x. ¬(Q x)"
  show "∀x. ¬(P x)"
  proof
    fix a
    show "¬(P a)"
    proof
      assume "P a"
      have "P a → Q a" using assms ..
      hence "Q a" using 'P a' ..
      have "¬(Q a)" using '∀x. ¬(Q x)' ..
      thus False using 'Q a' ..
```

qed

qed

qed

-- "La demostración detallada es"

```
lemma ejercicio_4c:
  assumes "∀x. P x → Q x"
  shows   "(∀x. ¬(Q x)) → (∀x. ¬(P x))"
```

```

proof (rule impI)
  assume "∀x. ¬(Q x)"
  show "∀x. ¬(P x)"
  proof (rule allI)
    fix a
    show "¬(P a)"
    proof
      assume "P a"
      have "P a → Q a" using assms by (rule allE)
      hence "Q a" using 'P a' by (rule mp)
      have "¬(Q a)" using '∀x. ¬(Q x)' by (rule allE)
      thus False using 'Q a' by (rule notE)
    qed
  qed
qed

text {* -----
  Ejercicio 5. Demostrar
     $\forall x. P x \rightarrow \neg(Q x) \vdash \neg(\exists x. P x \wedge Q x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_5a:
  "∀x. P x → ¬(Q x) ⇒ ¬(∃x. P x ∧ Q x)"
by auto

-- "La demostración estructurada es"
lemma ejercicio_5b:
  assumes "∀x. P x → ¬(Q x)"
  shows "¬(∃x. P x ∧ Q x)"
proof
  assume "∃x. P x ∧ Q x"
  then obtain a where "P a ∧ Q a" ..
  hence "P a" ..
  have "P a → ¬(Q a)" using assms ..
  hence "¬(Q a)" using 'P a' ..
  have "Q a" using 'P a ∧ Q a' ..
  with '¬(Q a)' show False ..
qed

-- "La demostración estructurada es"
lemma ejercicio_5c:
  assumes "∀x. P x → ¬(Q x)"
  shows "¬(∃x. P x ∧ Q x)"
proof (rule notI)

```

```

assume "∃x. P x ∧ Q x"
then obtain a where "P a ∧ Q a" by (rule exE)
hence "P a" by (rule conjunct1)
have "P a → ¬(Q a)" using assms by (rule allE)
hence "¬(Q a)" using 'P a' by (rule mp)
have "Q a" using 'P a ∧ Q a' by (rule conjunct2)
with '¬(Q a)' show False by (rule notE)
qed

text {* -----
  Ejercicio 6. Demostrar
     $\forall x y. P x y \vdash \forall u v. P u v$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_6a:
  "∀x y. P x y ⇒ ∀u v. P u v"
by auto

-- "La demostración estructurada es"
lemma ejercicio_6b:
  assumes "∀x y. P x y"
  shows "∀u v. P u v"
proof
  fix a
  show "∀v. P a v"
  proof
    fix b
    have "∀y. P a y" using assms ..
    thus "P a b" ..
  qed
qed

-- "La demostración estructurada simplificada es"
lemma ejercicio_6b2:
  assumes "∀x y. P x y"
  shows "∀u v. P u v"
proof (rule allI)+
  fix a b
  have "∀y. P a y" using assms ..
  thus "P a b" ..
qed

-- "La demostración detallada es"
lemma ejercicio_6c:

```

```

    assumes "∀x y. P x y"
    shows   "∀u v. P u v"
proof (rule allI)+
  fix a b
  have "∀y. P a y" using assms by (rule allE)
  thus "P a b" by (rule allE)
qed

text {* -----
  Ejercicio 7. Demostrar
     $\exists x y. P x y \implies \exists u v. P u v$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_7a:
  " $\exists x y. P x y \implies \exists u v. P u v$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_7b:
  assumes " $\exists x y. P x y$ "
  shows   " $\exists u v. P u v$ "
proof -
  obtain a where " $\exists y. P a y$ " using assms ..
  then obtain b where "P a b" ..
  hence " $\exists v. P a v$ " ..
  thus " $\exists u v. P u v$ " ..
qed

text {* -----
  Ejercicio 8. Demostrar
     $\exists x. \forall y. P x y \vdash \forall y. \exists x. P x y$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_8a:
  " $\exists x. \forall y. P x y \implies \forall y. \exists x. P x y$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_8b:
  assumes " $\exists x. \forall y. P x y$ "
  shows   " $\forall y. \exists x. P x y$ "
proof
  fix b

```

```

    obtain a where "∀y. P a y" using assms ..
    hence "P a b" ..
    thus "∃x. P x b" ..
qed

-- "La demostración detallada es"
lemma ejercicio_8c:
  assumes "∃x. ∀y. P x y"
  shows   "∀y. ∃x. P x y"
proof (rule allI)
  fix b
  obtain a where "∀y. P a y" using assms by (rule exE)
  hence "P a b" by (rule allE)
  thus "∃x. P x b" by (rule exI)
qed

text {* -----
  Ejercicio 9. Demostrar
     $\exists x. P a \rightarrow Q x \vdash P a \rightarrow (\exists x. Q x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_9a:
  "∃x. P a → Q x  $\implies$  P a → (∃x. Q x)"
by auto

-- "La demostración estructurada es"
lemma ejercicio_9b:
  assumes "∃x. P a → Q x"
  shows   "P a → (∃x. Q x)"
proof
  assume "P a"
  obtain b where "P a → Q b" using assms ..
  hence "Q b" using 'P a' ..
  thus "∃x. Q x" ..
qed

-- "La demostración detallada es"
lemma ejercicio_9c:
  assumes "∃x. P a → Q x"
  shows   "P a → (∃x. Q x)"
proof (rule impI)
  assume "P a"
  obtain b where "P a → Q b" using assms by (rule exE)
  hence "Q b" using 'P a' by (rule mp)

```

```

    thus "∃x. Q x" by (rule exI)
qed

text {* -----
  Ejercicio 10. Demostrar
     $P a \rightarrow (\exists x. Q x) \vdash \exists x. P a \rightarrow Q x$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_10a:
  "P a → (∃x. Q x) ⇒ ∃x. P a → Q x"
by auto

-- "La demostración estructurada es"
lemma ejercicio_10b:
  fixes P Q :: "'b ⇒ bool"
  assumes "P a → (∃x. Q x)"
  shows "∃x. P a → Q x"
proof -
  have "¬(P a) ∨ P a" ..
  thus "∃x. P a → Q x"
  proof
    assume "¬(P a)"
    have "P a → Q a"
    proof
      assume "P a"
      with '¬(P a)' show "Q a" ..
    qed
    thus "∃x. P a → Q x" ..
  next
    assume "P a"
    with assms have "∃x. Q x" by (rule mp)
    then obtain b where "Q b" ..
    have "P a → Q b"
    proof
      assume "P a"
      note 'Q b'
      thus "Q b" .
    qed
    thus "∃x. P a → Q x" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_10c:

```

```

fixes P Q :: "'b  $\Rightarrow$  bool"
assumes "P a  $\rightarrow$  ( $\exists$ x. Q x)"
shows    " $\exists$ x. P a  $\rightarrow$  Q x"
proof -
  have " $\neg$ (P a)  $\vee$  P a" by (rule excluded_middle)
  thus " $\exists$ x. P a  $\rightarrow$  Q x"
  proof (rule disjE)
    assume " $\neg$ (P a)"
    have "P a  $\rightarrow$  Q a"
    proof (rule impI)
      assume "P a"
      with ' $\neg$ (P a)' show "Q a" by (rule notE)
    qed
    thus " $\exists$ x. P a  $\rightarrow$  Q x" by (rule exI)
  next
    assume "P a"
    with assms have " $\exists$ x. Q x" by (rule mp)
    then obtain b where "Q b" by (rule exE)
    have "P a  $\rightarrow$  Q b"
    proof (rule impI)
      assume "P a"
      note 'Q b'
      thus "Q b" by this
    qed
    thus " $\exists$ x. P a  $\rightarrow$  Q x" by (rule exI)
  qed
qed

text {* -----
  Ejercicio 11. Demostrar
     $(\exists x. P x) \rightarrow Q a \vdash \forall x. P x \rightarrow Q a$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_11a:
  " $(\exists x. P x) \rightarrow Q a \Longrightarrow \forall x. P x \rightarrow Q a$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_11b:
  assumes " $(\exists x. P x) \rightarrow Q a$ "
  shows    " $\forall x. P x \rightarrow Q a$ "
proof
  fix b
  show "P b  $\rightarrow$  Q a"

```

```

proof
  assume "P b"
  hence "∃x. P x" ..
  with assms show "Q a" ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_11c:
  assumes "(∃x. P x) → Q a"
  shows "∀x. P x → Q a"
proof (rule allI)
  fix b
  show "P b → Q a"
  proof (rule impI)
    assume "P b"
    hence "∃x. P x" by (rule exI)
    with assms show "Q a" by (rule mp)
  qed
qed

text {* -----
  Ejercicio 12. Demostrar
     $\forall x. P x \rightarrow Q a \vdash \exists x. P x \rightarrow Q a$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_12a:
  "∀x. P x → Q a ⇒ ∃x. P x → Q a"
by auto

-- "La demostración estructurada es"
lemma ejercicio_12b:
  assumes "∀x. P x → Q a"
  shows "∃x. P x → Q a"
proof -
  have "P b → Q a" using assms ..
  thus "∃x. P x → Q a" ..
qed

-- "La demostración detallada es"
lemma ejercicio_12c:
  assumes "∀x. P x → Q a"
  shows "∃x. P x → Q a"
proof -

```

```

have "P b → Q a" using assms by (rule allE)
thus "∃x. P x → Q a" by (rule exI)
qed

text {* -----
  Ejercicio 13. Demostrar
     $(\forall x. P x) \vee (\forall x. Q x) \vdash \forall x. P x \vee Q x$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_13a:
  " $(\forall x. P x) \vee (\forall x. Q x) \implies \forall x. P x \vee Q x$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_13b:
  assumes " $(\forall x. P x) \vee (\forall x. Q x)$ "
  shows   " $\forall x. P x \vee Q x$ "
proof
  fix a
  note assms
  thus "P a ∨ Q a"
  proof
    assume "∀x. P x"
    hence "P a" ..
    thus "P a ∨ Q a" ..
  next
    assume "∀x. Q x"
    hence "Q a" ..
    thus "P a ∨ Q a" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_13c:
  assumes " $(\forall x. P x) \vee (\forall x. Q x)$ "
  shows   " $\forall x. P x \vee Q x$ "
proof (rule allI)
  fix a
  note assms
  thus "P a ∨ Q a"
  proof (rule disjE)
    assume "∀x. P x"
    hence "P a" by (rule allE)
    thus "P a ∨ Q a" by (rule disjI1)

```

```

next
  assume "∀x. Q x"
  hence "Q a" by (rule allE)
  thus "P a ∨ Q a" by (rule disjI2)
qed
qed

text {* -----
  Ejercicio 14. Demostrar
     $\exists x. P x \wedge Q x \vdash (\exists x. P x) \wedge (\exists x. Q x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_14a:
  " $\exists x. P x \wedge Q x \implies (\exists x. P x) \wedge (\exists x. Q x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_14b:
  assumes "∃x. P x ∧ Q x"
  shows   "(∃x. P x) ∧ (∃x. Q x)"
proof
  obtain a where "P a ∧ Q a" using assms ..
  hence "P a" ..
  thus "∃x. P x" ..
next
  obtain a where "P a ∧ Q a" using assms ..
  hence "Q a" ..
  thus "∃x. Q x" ..
qed

-- "La demostración detallada es"
lemma ejercicio_14c:
  assumes "∃x. P x ∧ Q x"
  shows   "(∃x. P x) ∧ (∃x. Q x)"
proof (rule conjI)
  obtain a where "P a ∧ Q a" using assms by (rule exE)
  hence "P a" by (rule conjunct1)
  thus "∃x. P x" by (rule exI)
next
  obtain a where "P a ∧ Q a" using assms by (rule exE)
  hence "Q a" by (rule conjunct2)
  thus "∃x. Q x" by (rule exI)
qed

```

```

text {* -----
  Ejercicio 15. Demostrar
     $\forall x y. P y \rightarrow Q x \vdash (\exists y. P y) \rightarrow (\forall x. Q x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_15a:
  " $\forall x y. P y \rightarrow Q x \implies (\exists y. P y) \rightarrow (\forall x. Q x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_15b:
  assumes " $\forall x y. P y \rightarrow Q x$ "
  shows   " $(\exists y. P y) \rightarrow (\forall x. Q x)$ "
proof
  assume " $\exists y. P y$ "
  then obtain b where "P b" ..
  show " $\forall x. Q x$ "
  proof
    fix a
    have " $\forall y. P y \rightarrow Q a$ " using assms ..
    hence " $P b \rightarrow Q a$ " ..
    thus "Q a" using 'P b' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_15c:
  assumes " $\forall x y. P y \rightarrow Q x$ "
  shows   " $(\exists y. P y) \rightarrow (\forall x. Q x)$ "
proof (rule impI)
  assume " $\exists y. P y$ "
  then obtain b where "P b" by (rule exE)
  show " $\forall x. Q x$ "
  proof (rule allI)
    fix a
    have " $\forall y. P y \rightarrow Q a$ " using assms by (rule allE)
    hence " $P b \rightarrow Q a$ " by (rule allE)
    thus "Q a" using 'P b' by (rule mp)
  qed
qed

text {* -----
  Ejercicio 16. Demostrar
     $\neg(\forall x. \neg(P x)) \vdash \exists x. P x$ 
  ----- *}

```

```

----- *}

-- "La demostración automática es"
lemma ejercicio_16a:
  " $\neg(\forall x. \neg(P\ x)) \implies \exists x. P\ x$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_16b:
  assumes " $\neg(\forall x. \neg(P\ x))$ "
  shows " $\exists x. P\ x$ "
proof (rule ccontr)
  assume " $\neg(\exists x. P\ x)$ "
  have " $\forall x. \neg(P\ x)$ "
  proof
    fix a
    show " $\neg(P\ a)$ "
    proof
      assume "P a"
      hence " $\exists x. P\ x$ " ..
      with ' $\neg(\exists x. P\ x)$ ' show False ..
    qed
  qed
  with assms show False ..
qed

-- "La demostración detallada es"
lemma ejercicio_16c:
  assumes " $\neg(\forall x. \neg(P\ x))$ "
  shows " $\exists x. P\ x$ "
proof (rule ccontr)
  assume " $\neg(\exists x. P\ x)$ "
  have " $\forall x. \neg(P\ x)$ "
  proof (rule allI)
    fix a
    show " $\neg(P\ a)$ "
    proof
      assume "P a"
      hence " $\exists x. P\ x$ " by (rule exI)
      with ' $\neg(\exists x. P\ x)$ ' show False by (rule notE)
    qed
  qed
  with assms show False by (rule notE)
qed

```

```

text {* -----
  Ejercicio 17. Demostrar
     $\forall x. \neg(P\ x) \vdash \neg(\exists x. P\ x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_17a:
  " $\forall x. \neg(P\ x) \implies \neg(\exists x. P\ x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_17b:
  assumes " $\forall x. \neg(P\ x)$ "
  shows   " $\neg(\exists x. P\ x)$ "
proof
  assume " $\exists x. P\ x$ "
  then obtain a where "P a" ..
  have " $\neg(P\ a)$ " using assms ..
  thus False using 'P a' ..
qed

-- "La demostración detallada es"
lemma ejercicio_17c:
  assumes " $\forall x. \neg(P\ x)$ "
  shows   " $\neg(\exists x. P\ x)$ "
proof (rule notI)
  assume " $\exists x. P\ x$ "
  then obtain a where "P a" by (rule exE)
  have " $\neg(P\ a)$ " using assms by (rule allE)
  thus False using 'P a' by (rule notE)
qed

text {* -----
  Ejercicio 18. Demostrar
     $\exists x. P\ x \vdash \neg(\forall x. \neg(P\ x))$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_18a:
  " $\exists x. P\ x \implies \neg(\forall x. \neg(P\ x))$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_18b:
  assumes " $\exists x. P\ x$ "

```

```

    shows "¬(∀x. ¬(P x))"
proof
  assume "∀x. ¬(P x)"
  obtain a where "P a" using assms ..
  have "¬(P a)" using '∀x. ¬(P x)' ..
  thus False using 'P a' ..
qed

-- "La demostración detallada es"
lemma ejercicio_18c:
  assumes "∃x. P x"
  shows "¬(∀x. ¬(P x))"
proof (rule notI)
  assume "∀x. ¬(P x)"
  obtain a where "P a" using assms by (rule exE)
  have "¬(P a)" using '∀x. ¬(P x)' by (rule allE)
  thus False using 'P a' by (rule notE)
qed

text {* -----
  Ejercicio 19. Demostrar
     $P a \rightarrow (\forall x. Q x) \vdash \forall x. P a \rightarrow Q x$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_19a:
  "P a → (∀x. Q x) ⇒ ∀x. P a → Q x"
by auto

-- "La demostración estructurada es"
lemma ejercicio_19b:
  assumes "P a → (∀x. Q x)"
  shows "∀x. P a → Q x"
proof
  fix b
  show "P a → Q b"
  proof
    assume "P a"
    with assms have "∀x. Q x" ..
    thus "Q b" ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_19c:

```

```

    assumes "P a  $\rightarrow$  ( $\forall x. Q x$ )"
    shows   " $\forall x. P a \rightarrow Q x$ "
proof (rule allI)
  fix b
  show "P a  $\rightarrow$  Q b"
proof (rule impI)
  assume "P a"
  with assms have " $\forall x. Q x$ " by (rule mp)
  thus "Q b" by (rule allE)
qed
qed

text {* -----
  Ejercicio 20. Demostrar
    { $\forall x y z. R x y \wedge R y z \rightarrow R x z$ ,
       $\forall x. \neg(R x x)$ }
     $\vdash \forall x y. R x y \rightarrow \neg(R y x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_20a:
  " $\llbracket \forall x y z. R x y \wedge R y z \rightarrow R x z; \forall x. \neg(R x x) \rrbracket \implies \forall x y. R x y \rightarrow \neg(R y x)$ "
by metis

-- "La demostración estructurada es"
lemma ejercicio_20b:
  assumes " $\forall x y z. R x y \wedge R y z \rightarrow R x z$ "
          " $\forall x. \neg(R x x)$ "
  shows   " $\forall x y. R x y \rightarrow \neg(R y x)$ "
proof (rule allI)+
  fix a b
  show "R a b  $\rightarrow$   $\neg(R b a)$ "
proof
  assume "R a b"
  show " $\neg(R b a)$ "
proof
    assume "R b a"
    show False
  proof -
    have "R a b  $\wedge$  R b a" using 'R a b' 'R b a' ..
    have " $\forall y z. R a y \wedge R y z \rightarrow R a z$ " using assms(1) ..
    hence " $\forall z. R a b \wedge R b z \rightarrow R a z$ " ..
    hence "R a b  $\wedge$  R b a  $\rightarrow$  R a a" ..
    hence "R a a" using 'R a b  $\wedge$  R b a' ..
    have " $\neg(R a a)$ " using assms(2) ..
  qed
qed

```

```

        thus False using 'R a a' ..
      qed
    qed
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_20c:
  assumes "∀x y z. R x y ∧ R y z → R x z"
    "∀x. ¬(R x x)"
  shows "∀x y. R x y → ¬(R y x)"
proof (rule allI)+
  fix a b
  show "R a b → ¬(R b a)"
proof (rule impI)
  assume "R a b"
  show "¬(R b a)"
proof (rule notI)
  assume "R b a"
  show False
proof -
  have "R a b ∧ R b a" using 'R a b' 'R b a' by (rule conjI)
  have "∀y z. R a y ∧ R y z → R a z" using assms(1) by (rule allE)
  hence "∀z. R a b ∧ R b z → R a z" by (rule allE)
  hence "R a b ∧ R b a → R a a" by (rule allE)
  hence "R a a" using 'R a b ∧ R b a' by (rule mp)
  have "¬(R a a)" using assms(2) by (rule allE)
  thus False using 'R a a' by (rule notE)
qed
qed
qed
qed

text {* -----
  Ejercicio 21. Demostrar
    {∀x. P x ∨ Q x, ∃x. ¬(Q x), ∀x. R x → ¬(P x)} ⊢ ∃x. ¬(R x)
  ----- *}

-- "La demostración automática es"
lemma ejercicio_21a:
  "[[∀x. P x ∨ Q x; ∃x. ¬(Q x); ∀x. R x → ¬(P x)] ⇒ ∃x. ¬(R x)]"
by auto

-- "La demostración estructurada es"
lemma ejercicio_21b:

```

```

assumes "∀x. P x ∨ Q x"
        "∃x. ¬(Q x)"
        "∀x. R x → ¬(P x)"
shows   "∃x. ¬(R x)"
proof -
  obtain a where "¬(Q a)" using assms(2) ..
  have "P a ∨ Q a" using assms(1) ..
  hence "P a"
  proof
    assume "P a"
    thus "P a" .
  next
    assume "Q a"
    with '¬(Q a)' show "P a" ..
  qed
  hence "¬¬(P a)" by (rule notnotI)
  have "R a → ¬(P a)" using assms(3) ..
  hence "¬(R a)" using '¬¬(P a)' by (rule mt)
  thus "∃x. ¬(R x)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_21c:
  assumes "∀x. P x ∨ Q x"
          "∃x. ¬(Q x)"
          "∀x. R x → ¬(P x)"
  shows   "∃x. ¬(R x)"
proof -
  obtain a where "¬(Q a)" using assms(2) by (rule exE)
  have "P a ∨ Q a" using assms(1) by (rule allE)
  hence "P a"
  proof (rule disjE)
    assume "P a"
    thus "P a" by this
  next
    assume "Q a"
    with '¬(Q a)' show "P a" by (rule notE)
  qed
  hence "¬¬(P a)" by (rule notnotI)
  have "R a → ¬(P a)" using assms(3) by (rule allE)
  hence "¬(R a)" using '¬¬(P a)' by (rule mt)
  thus "∃x. ¬(R x)" by (rule exI)
qed

text {* -----

```

Ejercicio 22. Demostrar

$\{\forall x. P\ x \rightarrow Q\ x \vee R\ x, \neg(\exists x. P\ x \wedge R\ x)\} \vdash \forall x. P\ x \rightarrow Q\ x$

----- *

-- "La demostración automática es"

lemma ejercicio_22a:

" $\forall x. P\ x \rightarrow Q\ x \vee R\ x; \neg(\exists x. P\ x \wedge R\ x) \implies \forall x. P\ x \rightarrow Q\ x$ "

by auto

-- "La demostración estructurada es"

lemma ejercicio_22b:

assumes " $\forall x. P\ x \rightarrow Q\ x \vee R\ x$ "

" $\neg(\exists x. P\ x \wedge R\ x)$ "

shows " $\forall x. P\ x \rightarrow Q\ x$ "

proof

fix a

show " $P\ a \rightarrow Q\ a$ "

proof

assume " $P\ a$ "

have " $P\ a \rightarrow Q\ a \vee R\ a$ " **using** assms(1) ..

hence " $Q\ a \vee R\ a$ " **using** 'P a' ..

thus " $Q\ a$ "

proof

assume " $Q\ a$ "

thus " $Q\ a$ " .

next

assume " $R\ a$ "

with 'P a' **have** " $P\ a \wedge R\ a$ " ..

hence " $\exists x. P\ x \wedge R\ x$ " ..

with assms(2) **show** " $Q\ a$ " ..

qed

qed

qed

-- "La demostración detallada es"

lemma ejercicio_22c:

assumes " $\forall x. P\ x \rightarrow Q\ x \vee R\ x$ "

" $\neg(\exists x. P\ x \wedge R\ x)$ "

shows " $\forall x. P\ x \rightarrow Q\ x$ "

proof (rule allI)

fix a

show " $P\ a \rightarrow Q\ a$ "

proof (rule impI)

assume " $P\ a$ "

have " $P\ a \rightarrow Q\ a \vee R\ a$ " **using** assms(1) **by** (rule allE)

```

hence "Q a  $\vee$  R a" using 'P a' by (rule mp)
thus "Q a"
proof (rule disjE)
  assume "Q a"
  thus "Q a" by this
next
  assume "R a"
  with 'P a' have "P a  $\wedge$  R a" by (rule conjI)
  hence " $\exists x. P x \wedge R x$ " by (rule exI)
  with assms(2) show "Q a" by (rule notE)
qed
qed
qed

text {* -----
  Ejercicio 23. Demostrar
     $\exists x y. R x y \vee R y x \vdash \exists x y. R x y$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_23a:
  " $\exists x y. R x y \vee R y x \implies \exists x y. R x y$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_23b:
  assumes " $\exists x y. R x y \vee R y x$ "
  shows   " $\exists x y. R x y$ "
proof -
  obtain a where " $\exists y. R a y \vee R y a$ " using assms ..
  then obtain b where " $R a b \vee R b a$ " ..
  thus " $\exists x y. R x y$ "
  proof
    assume "R a b"
    hence " $\exists y. R a y$ " ..
    thus " $\exists x y. R x y$ " ..
  next
    assume "R b a"
    hence " $\exists y. R b y$ " ..
    thus " $\exists x y. R x y$ " ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_23c:

```

```

    assumes "∃x y. R x y ∨ R y x"
    shows   "∃x y. R x y"
proof -
  obtain a where "∃y. R a y ∨ R y a" using assms by (rule exE)
  then obtain b where "R a b ∨ R b a" by (rule exE)
  thus "∃x y. R x y"
proof (rule disjE)
  assume "R a b"
  hence "∃y. R a y" by (rule exI)
  thus "∃ x y. R x y" by (rule exI)
next
  assume "R b a"
  hence "∃y. R b y" by (rule exI)
  thus "∃ x y. R x y" by (rule exI)
qed
qed

text {* -----
  Ejercicio 24. Demostrar
     $(\exists x. \forall y. P x y) \rightarrow (\forall y. \exists x. P x y)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_24a: "(∃x. ∀y. P x y) → (∀y. ∃x. P x y)"
by auto

-- "La demostración estructurada es"
lemma ejercicio_24b: "(∃x. ∀y. P x y) → (∀y. ∃x. P x y)"
proof
  assume "∃x. ∀y. P x y"
  then obtain a where "∀y. P a y" ..
  show "∀y. ∃x. P x y"
proof
  fix b
  have "P a b" using '∀y. P a y' ..
  thus "∃x. P x b" ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_24c: "(∃x. ∀y. P x y) → (∀y. ∃x. P x y)"
proof (rule impI)
  assume "∃x. ∀y. P x y"
  then obtain a where "∀y. P a y" by (rule exE)
  show "∀y. ∃x. P x y"

```

```

proof (rule allI)
  fix b
  have "P a b" using '∀y. P a y' by (rule allE)
  thus "∃x. P x b" by (rule exI)
qed
qed

text {* -----
  Ejercicio 25. Demostrar
     $(\forall x. P x \rightarrow Q) \leftrightarrow ((\exists x. P x) \rightarrow Q)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_25a: "(∀x. P x → Q) ↔ ((∃x. P x) → Q)"
by auto

-- "La demostración estructurada es"
lemma ejercicio_25b: "(∀x. P x → Q) ↔ ((∃x. P x) → Q)"
proof
  assume "∀x. P x → Q"
  show "(∃x. P x) → Q"
  proof
    assume "∃x. P x"
    then obtain a where "P a" ..
    have "P a → Q" using '∀x. P x → Q' ..
    thus "Q" using 'P a' ..
  qed
qed
next
  assume "(∃x. P x) → Q"
  show "∀x. P x → Q"
  proof
    fix b
    show "P b → Q"
    proof
      assume "P b"
      hence "∃x. P x" ..
      with '(∃x. P x) → Q' show "Q" ..
    qed
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_25c: "(∀x. P x → Q) ↔ ((∃x. P x) → Q)"
proof (rule iffI)
  assume "∀x. P x → Q"

```

```

show "( $\exists x. P x$ )  $\rightarrow$  Q"
proof (rule impI)
  assume " $\exists x. P x$ "
  then obtain a where "P a" by (rule exE)
  have "P a  $\rightarrow$  Q" using ' $\forall x. P x \rightarrow Q$ ' by (rule allE)
  thus "Q" using 'P a' by (rule mp)
qed
next
assume "( $\exists x. P x$ )  $\rightarrow$  Q"
show " $\forall x. P x \rightarrow Q$ "
proof (rule allI)
  fix b
  show "P b  $\rightarrow$  Q"
  proof (rule impI)
    assume "P b"
    hence " $\exists x. P x$ " by (rule exI)
    with ' $(\exists x. P x) \rightarrow Q$ ' show "Q" by (rule mp)
  qed
qed
qed
qed

text {* -----
  Ejercicio 26. Demostrar
     $((\forall x. P x) \wedge (\forall x. Q x)) \leftrightarrow (\forall x. P x \wedge Q x)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_26a: " $((\forall x. P x) \wedge (\forall x. Q x)) \leftrightarrow (\forall x. P x \wedge Q x)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_26b: " $((\forall x. P x) \wedge (\forall x. Q x)) \leftrightarrow (\forall x. P x \wedge Q x)$ "
proof
  assume " $(\forall x. P x) \wedge (\forall x. Q x)$ "
  show " $\forall x. P x \wedge Q x$ "
  proof
    fix a
    have " $\forall x. P x$ " using ' $(\forall x. P x) \wedge (\forall x. Q x)$ ' ..
    have " $\forall x. Q x$ " using ' $(\forall x. P x) \wedge (\forall x. Q x)$ ' ..
    hence "Q a" ..
    have "P a" using ' $\forall x. P x$ ' ..
    thus "P a  $\wedge$  Q a" using 'Q a' ..
  qed
next
  assume " $\forall x. P x \wedge Q x$ "

```

```

have "∀x. P x"
proof
  fix a
  have "P a ∧ Q a" using '∀x. P x ∧ Q x' ..
  thus "P a" ..
qed
moreover have "∀x. Q x"
proof
  fix a
  have "P a ∧ Q a" using '∀x. P x ∧ Q x' ..
  thus "Q a" ..
qed
ultimately show "(∀x. P x) ∧ (∀x. Q x)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_26c: "((∀x. P x) ∧ (∀x. Q x)) = (∀x. P x ∧ Q x)"
proof (rule iffI)
  assume "(∀x. P x) ∧ (∀x. Q x)"
  show "∀x. P x ∧ Q x"
  proof (rule allI)
    fix a
    have "∀x. P x" using '(∀x. P x) ∧ (∀x. Q x)' by (rule conjunct1)
    have "∀x. Q x" using '(∀x. P x) ∧ (∀x. Q x)' by (rule conjunct2)
    hence "Q a" by (rule allE)
    have "P a" using '∀x. P x' by (rule allE)
    thus "P a ∧ Q a" using 'Q a' by (rule conjI)
  qed
next
  assume "∀x. P x ∧ Q x"
  have "∀x. P x"
  proof (rule allI)
    fix a
    have "P a ∧ Q a" using '∀x. P x ∧ Q x' by (rule allE)
    thus "P a" by (rule conjunct1)
  qed
  moreover have "∀x. Q x"
  proof (rule allI)
    fix a
    have "P a ∧ Q a" using '∀x. P x ∧ Q x' by (rule allE)
    thus "Q a" by (rule conjunct2)
  qed
  ultimately show "(∀x. P x) ∧ (∀x. Q x)" by (rule conjI)
qed

```

```
text {* -----
  Ejercicio 27. Demostrar o refutar
     $((\forall x. P\ x) \vee (\forall x. Q\ x)) \leftrightarrow (\forall x. P\ x \vee Q\ x)$ 
  ----- *}

```

```
lemma ejercicio_27: " $((\forall x. P\ x) \vee (\forall x. Q\ x)) \leftrightarrow (\forall x. P\ x \vee Q\ x)$ "
oops

```

```
(*
Auto Quickcheck found a counterexample:
P = {a|<^isub>1}
Q = {a|<^isub>2}
*)

```

```
text {* -----
  Ejercicio 28. Demostrar o refutar
     $((\exists x. P\ x) \vee (\exists x. Q\ x)) \leftrightarrow (\exists x. P\ x \vee Q\ x)$ 
  ----- *}

```

```
-- "La demostración automática es"

```

```
lemma ejercicio_28a:
  " $((\exists x. P\ x) \vee (\exists x. Q\ x)) \leftrightarrow (\exists x. P\ x \vee Q\ x)$ "
by auto

```

```
-- "La demostración estructurada es"

```

```
lemma ejercicio_28b:
  " $((\exists x. P\ x) \vee (\exists x. Q\ x)) \leftrightarrow (\exists x. P\ x \vee Q\ x)$ "

```

```
proof
  assume " $(\exists x. P\ x) \vee (\exists x. Q\ x)$ "
  thus " $\exists x. P\ x \vee Q\ x$ "
  proof
    assume " $\exists x. P\ x$ "
    then obtain a where "P a" ..
    hence "P a  $\vee$  Q a" ..
    thus " $\exists x. P\ x \vee Q\ x$ " ..
  next
    assume " $\exists x. Q\ x$ "
    then obtain a where "Q a" ..
    hence "P a  $\vee$  Q a" ..
    thus " $\exists x. P\ x \vee Q\ x$ " ..
  qed
next
  assume " $\exists x. P\ x \vee Q\ x$ "
  then obtain a where "P a  $\vee$  Q a" ..
  thus " $(\exists x. P\ x) \vee (\exists x. Q\ x)$ "

```

```

proof
  assume "P a"
  hence "∃x. P x" ..
  thus "(∃x. P x) ∨ (∃x. Q x)" ..
next
  assume "Q a"
  hence "∃x. Q x" ..
  thus "(∃x. P x) ∨ (∃x. Q x)" ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_28c:
  "((∃x. P x) ∨ (∃x. Q x)) ↔ (∃x. P x ∨ Q x)"
proof (rule iffI)
  assume "(∃x. P x) ∨ (∃x. Q x)"
  thus "∃x. P x ∨ Q x"
  proof (rule disjE)
    assume "∃x. P x"
    then obtain a where "P a" by (rule exE)
    hence "P a ∨ Q a" by (rule disjI1)
    thus "∃x. P x ∨ Q x" by (rule exI)
  next
    assume "∃x. Q x"
    then obtain a where "Q a" by (rule exE)
    hence "P a ∨ Q a" by (rule disjI2)
    thus "∃x. P x ∨ Q x" by (rule exI)
  qed
next
  assume "∃x. P x ∨ Q x"
  then obtain a where "P a ∨ Q a" by (rule exE)
  thus "(∃x. P x) ∨ (∃x. Q x)"
  proof (rule disjE)
    assume "P a"
    hence "∃x. P x" by (rule exI)
    thus "(∃x. P x) ∨ (∃x. Q x)" by (rule disjI1)
  next
    assume "Q a"
    hence "∃x. Q x" by (rule exI)
    thus "(∃x. P x) ∨ (∃x. Q x)" by (rule disjI2)
  qed
qed
qed

text {* -----
  Ejercicio 29. Demostrar o refutar

```

```

      (∀x. ∃y. P x y) → (∃y. ∀x. P x y)
----- *}

lemma ejercicio_29:
  "(∀x. ∃y. P x y) → (∃y. ∀x. P x y)"
quickcheck
  (*
Quickcheck found a counterexample:

P = (λx. undefined)(a\<^isub> := {b}, b := {a})
*)
oops

text {* -----
  Ejercicio 30. Demostrar o refutar
    (¬(∀x. P x)) ↔ (∃x. ¬P x)
----- *}

-- "La demostración automática es"
lemma ejercicio_30a:
  "(¬(∀x. P x)) ↔ (∃x. ¬P x)"
by auto

-- "La demostración estructurada es"
lemma ejercicio_30b:
  "(¬(∀x. P x)) ↔ (∃x. ¬P x)"
proof
  assume "¬(∀x. P x)"
  show "∃x. ¬P x"
  proof (rule ccontr)
    assume "¬(∃x. ¬P x)"
    have "∀x. P x"
    proof
      fix a
      show "P a"
      proof (rule ccontr)
        assume "¬P a"
        hence "∃x. ¬P x" ..
        with '¬(∃x. ¬P x)' show False ..
      qed
    qed
    with '¬(∀x. P x)' show False ..
  qed
next
  assume "∃x. ¬P x"

```

```

then obtain a where "¬P a" ..
show "¬(∀x. P x)"
proof
  assume "∀x. P x"
  hence "P a" ..
  with '¬P a' show False ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_30c:
  "(¬(∀x. P x)) ↔ (∃x. ¬P x)"
proof (rule iffI)
  assume "¬(∀x. P x)"
  show "∃x. ¬P x"
  proof (rule ccontr)
    assume "¬(∃x. ¬P x)"
    have "∀x. P x"
    proof (rule allI)
      fix a
      show "P a"
      proof (rule ccontr)
        assume "¬P a"
        hence "∃x. ¬P x" by (rule exI)
        with '¬(∃x. ¬P x)' show False by (rule notE)
      qed
    qed
    with '¬(∀x. P x)' show False by (rule notE)
  qed
next
  assume "∃x. ¬P x"
  then obtain a where "¬P a" by (rule exE)
  show "¬(∀x. P x)"
  proof (rule notI)
    assume "∀x. P x"
    hence "P a" by (rule allE)
    show False using '¬P a' 'P a' by (rule notE)
  qed
qed

```

section *{* Ejercicios sobre igualdad **}

```

text {* -----
Ejercicio 31. Demostrar o refutar
 $P a \implies \forall x. x = a \rightarrow P x$ 

```

```

----- *}

-- "La demostración automática es"
lemma ejercicio_31a:
  "P a  $\implies \forall x. x = a \rightarrow P x$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_31b:
  assumes "P a"
  shows "  $\forall x. x = a \rightarrow P x$ "
proof
  fix b
  show "b = a  $\rightarrow P b$ "
  proof
    assume "b = a"
    thus "P b" using assms by (rule ssubst)
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_31c:
  assumes "P a"
  shows "  $\forall x. x = a \rightarrow P x$ "
proof (rule allI)
  fix b
  show "b = a  $\rightarrow P b$ "
  proof (rule impI)
    assume "b = a"
    thus "P b" using assms by (rule ssubst)
  qed
qed

text {
  -----
  Ejercicio 32. Demostrar o refutar
   $\exists x y. R x y \vee R y x; \neg(\exists x. R x x) \implies \exists x y. x \neq y$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_32a:
  fixes R :: "'c  $\Rightarrow$  'c  $\Rightarrow$  bool"
  assumes "  $\exists x y. R x y \vee R y x$ "
    "  $\neg(\exists x. R x x)$ "
  shows "  $\exists (x::'c) y. x \neq y$ "
using assms

```

by metis

-- "La demostración estructurada es"

lemma ejercicio_32b:

fixes R :: "'c $\Rightarrow$ 'c $\Rightarrow$ bool"

assumes " $\exists x y. R x y \vee R y x$ "

" $\neg(\exists x. R x x)$ "

shows " $\exists(x::'c) y. x \neq y$ "

proof -

obtain a where " $\exists y. R a y \vee R y a$ " using assms(1) ..

then obtain b where " $R a b \vee R b a$ " ..

hence " $a \neq b$ "

proof

assume " $R a b$ "

show " $a \neq b$ "

proof

assume " $a = b$ "

hence " $R b b$ " using 'R a b' by (rule subst)

hence " $\exists x. R x x$ " ..

with assms(2) show False ..

qed

next

assume " $R b a$ "

show " $a \neq b$ "

proof

assume " $a = b$ "

hence " $R a a$ " using 'R b a' by (rule ssubst)

hence " $\exists x. R x x$ " ..

with assms(2) show False ..

qed

qed

hence " $\exists y. a \neq y$ " ..

thus " $\exists(x::'c) y. x \neq y$ " ..

qed

-- "La demostración detallada es"

lemma ejercicio_32c:

fixes R :: "'c $\Rightarrow$ 'c $\Rightarrow$ bool"

assumes " $\exists x y. R x y \vee R y x$ "

" $\neg(\exists x. R x x)$ "

shows " $\exists(x::'c) y. x \neq y$ "

proof -

obtain a where " $\exists y. R a y \vee R y a$ " using assms(1) by (rule exE)

then obtain b where " $R a b \vee R b a$ " by (rule exE)

hence " $a \neq b$ "

```

proof (rule disjE)
  assume "R a b"
  show "a ≠ b"
  proof (rule notI)
    assume "a = b"
    hence "R b b" using 'R a b' by (rule subst)
    hence "∃x. R x x" by (rule exI)
    with assms(2) show False by (rule notE)
  qed
next
  assume "R b a"
  show "a ≠ b"
  proof (rule notI)
    assume "a = b"
    hence "R a a" using 'R b a' by (rule ssubst)
    hence "∃x. R x x" by (rule exI)
    with assms(2) show False by (rule notE)
  qed
qed
hence "∃y. a ≠ y" by (rule exI)
thus "∃(x::'c) y. x ≠ y" by (rule exI)
qed

text {* -----
  Ejercicio 33. Demostrar o refutar
  { $\forall x. P a x x,$ 
     $\forall x y z. P x y z \rightarrow P (f x) y (f z)$ }
   $\vdash P (f a) a (f a)$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_33a:
  "[ $\forall x. P a x x; \forall x y z. P x y z \rightarrow P (f x) y (f z)$ ]  $\Rightarrow P (f a) a (f a)$ "
by auto

-- "La demostración estructura es"
lemma ejercicio_33b:
  assumes "∑x. P a x x"
    "∑x y z. P x y z  $\rightarrow$  P (f x) y (f z)"
  shows "P (f a) a (f a)"
proof -
  have "P a a a" using assms(1) ..
  have "∑y z. P a y z  $\rightarrow$  P (f a) y (f z)" using assms(2) ..
  hence "∑z. P a a z  $\rightarrow$  P (f a) a (f z)" ..
  hence "P a a a  $\rightarrow$  P (f a) a (f a)" ..

```

```

    thus "P (f a) a (f a)" using 'P a a a' ..
qed

-- "La demostración detallada es"
lemma ejercicio_33c:
  assumes "∀x. P a x x"
    "∀x y z. P x y z → P (f x) y (f z)"
  shows "P (f a) a (f a)"
proof -
  have "P a a a" using assms(1) by (rule allE)
  have "∀y z. P a y z → P (f a) y (f z)" using assms(2) by (rule allE)
  hence "∀z. P a a z → P (f a) a (f z)" by (rule allE)
  hence "P a a a → P (f a) a (f a)" by (rule allE)
  thus "P (f a) a (f a)" using 'P a a a' by (rule mp)
qed

text {*-----
  Ejercicio 34. Demostrar o refutar
  {∀x. P a x x,
   ∀x y z. P x y z → P (f x) y (f z)}
  ⊢ ∃z. P (f a) z (f (f a))
  ----- *}

-- "La demostración automática es"
lemma ejercicio_34a:
  "[[∀x. P a x x; ∀x y z. P x y z → P (f x) y (f z)]]
   ⇒ ∃z. P (f a) z (f (f a))"
by metis

-- "La demostración estructura es"
lemma ejercicio_34b:
  assumes "∀x. P a x x"
    "∀x y z. P x y z → P (f x) y (f z)"
  shows "∃z. P (f a) z (f (f a))"
proof -
  have "P a (f a) (f a)" using assms(1) ..
  have "∀y z. P a y z → P (f a) y (f z)" using assms(2) ..
  hence "∀z. P a (f a) z → P (f a) (f a) (f z)" ..
  hence "P a (f a) (f a) → P (f a) (f a) (f (f a))" ..
  hence "P (f a) (f a) (f (f a))" using 'P a (f a) (f a)' ..
  thus "∃z. P (f a) z (f (f a))" ..
qed

-- "La demostración detallada es"
lemma ejercicio_34c:

```

```

    assumes "∀x. P a x x"
      "∀x y z. P x y z → P (f x) y (f z)"
    shows   "∃z. P (f a) z (f (f a))"
  proof -
    have "P a (f a) (f a)" using assms(1) by (rule allE)
    have "∀y z. P a y z → P (f a) y (f z)" using assms(2) by (rule allE)
    hence "∀z. P a (f a) z → P (f a) (f a) (f z)" by (rule allE)
    hence "P a (f a) (f a) → P (f a) (f a) (f (f a))" by (rule allE)
    hence "P (f a) (f a) (f (f a))" using 'P a (f a) (f a)' by (rule mp)
    thus "∃z. P (f a) z (f (f a))" by (rule exI)
  qed

text {
  -----
  Ejercicio 35. Demostrar o refutar
  {
    ∀y. Q a y,
    ∀x y. Q x y → Q (s x) (s y)
  }
  ⊢ ∃z. Q a z ∧ Q z (s (s a))
  ----- *}

-- "La demostración automática es"
lemma ejercicio_35a:
  "[(∀y. Q a y; ∀x y. Q x y → Q (s x) (s y))] ⇒ ∃z. Q a z ∧ Q z (s (s a))"
by auto

-- "La demostración estructural es"
lemma ejercicio_35b:
  assumes "∀y. Q a y"
    "∀x y. Q x y → Q (s x) (s y)"
  shows   "∃z. Q a z ∧ Q z (s (s a))"
  proof -
    have "Q a (s a)" using assms(1) ..
    have "∀y. Q a y → Q (s a) (s y)" using assms(2) ..
    hence "Q a (s a) → Q (s a) (s (s a))" ..
    hence "Q (s a) (s (s a))" using 'Q a (s a)' ..
    with 'Q a (s a)' have "Q a (s a) ∧ Q (s a) (s (s a))" ..
    thus "∃z. Q a z ∧ Q z (s (s a))" ..
  qed

-- "La demostración detallada es"
lemma ejercicio_35c:
  assumes "∀y. Q a y"
    "∀x y. Q x y → Q (s x) (s y)"
  shows   "∃z. Q a z ∧ Q z (s (s a))"
  proof -
    have "Q a (s a)" using assms(1) by (rule allE)

```

```

have "∀y. Q a y → Q (s a) (s y)" using assms(2) by (rule allE)
hence "Q a (s a) → Q (s a) (s (s a))" by (rule allE)
hence "Q (s a) (s (s a))" using 'Q a (s a)' by (rule mp)
with 'Q a (s a)' have "Q a (s a) ∧ Q (s a) (s (s a))" by (rule conjI)
thus "∃z. Q a z ∧ Q z (s (s a))" by (rule exI)
qed

text {* -----
  Ejercicio 36. Demostrar o refutar
    {x = f x, odd (f x)} ⊢ odd x
  ----- *}

-- "La demostración automática es"
lemma ejercicio_36a:
  "[x = f x; odd (f x)] ⇒ odd x"
by auto

-- "La demostración semiautomática es"
lemma ejercicio_36b:
  "[x = f x; odd (f x)] ⇒ odd x"
by (rule ssubst)

-- "La demostración estructurada es"
lemma ejercicio_36c:
  assumes "x = f x"
  and     "odd (f x)"
  shows   "odd x"
proof -
  show "odd x" using assms by (rule ssubst)
qed

text {* -----
  Ejercicio 37. Demostrar o refutar
    {x = f x, triple (f x) (f x) x} ⊢ triple x x x
  ----- *}

-- "La demostración automática es"
lemma ejercicio_37a:
  "[x = f x; triple (f x) (f x) x] ⇒ triple x x x"
by auto

-- "La demostración semiautomática es"
lemma ejercicio_37b:
  "[x = f x; triple (f x) (f x) x] ⇒ triple x x x"
by (rule ssubst)

```

```
-- "La demostración estructurada es"
lemma ejercicio_37c:
  assumes "x = f x"
    "triple (f x) (f x) x"
  shows "triple x x x"
proof -
  show "triple x x x" using assms by (rule ssubst)
qed

end
```

4.3. Ejercicios: Argumentación lógica de primer orden

```
chapter {* T4R2: Argumentación en lógica de primer orden *}
```

```
theory T4R2
imports Main
begin
```

```
text {*
```

El objetivo de esta relación es formalizar y decidir la corrección de los argumentos. En el caso de que sea correcto, demostrarlo usando sólo las reglas básicas de deducción natural de la lógica de primer orden (sin usar el método auto). En el caso de que sea incorrecto, calcular un contraejemplo con QuickCheck.

Las reglas básicas de la deducción natural son las siguientes:

- conjI: $\llbracket P; Q \rrbracket \Longrightarrow P \wedge Q$
- conjunct1: $P \wedge Q \Longrightarrow P$
- conjunct2: $P \wedge Q \Longrightarrow Q$
- notnotD: $\neg\neg P \Longrightarrow P$
- notnotI: $P \Longrightarrow \neg\neg P$
- mp: $\llbracket P \rightarrow Q; P \rrbracket \Longrightarrow Q$
- mt: $\llbracket F \rightarrow G; \neg G \rrbracket \Longrightarrow \neg F$
- impI: $(P \Longrightarrow Q) \Longrightarrow P \rightarrow Q$
- disjI1: $P \Longrightarrow P \vee Q$
- disjI2: $Q \Longrightarrow P \vee Q$
- disjE: $\llbracket P \vee Q; P \Longrightarrow R; Q \Longrightarrow R \rrbracket \Longrightarrow R$
- FalseE: $\text{False} \Longrightarrow P$
- notE: $\llbracket \neg P; P \rrbracket \Longrightarrow R$

```

· notI:      (P  $\implies$  False)  $\implies$   $\neg$ P
· iffI:       $\llbracket P \implies Q; Q \implies P \rrbracket \implies P = Q$ 
· iffD1:      $\llbracket Q = P; Q \rrbracket \implies P$ 
· iffD2:      $\llbracket P = Q; Q \rrbracket \implies P$ 
· ccontr:    ( $\neg$ P  $\implies$  False)  $\implies$  P
· excluded_middle:  $\neg$ P  $\vee$  P

· allI:       $\llbracket \forall x. P\ x; P\ x \implies R \rrbracket \implies R$ 
· allE:       $(\bigwedge x. P\ x) \implies \forall x. P\ x$ 
· exI:       P x  $\implies \exists x. P\ x$ 
· exE:        $\llbracket \exists x. P\ x; \bigwedge x. P\ x \implies Q \rrbracket \implies Q$ 

-----
*}

text {*
  Se usarán las reglas notnotI y mt que demostramos a continuación.
*}

lemma notnotI: "P  $\implies$   $\neg\neg$  P"
by auto

lemma mt: " $\llbracket F \rightarrow G; \neg G \rrbracket \implies \neg F$ "
by auto

lemma no_ex: " $\neg(\exists x. P(x)) \implies \forall x. \neg P(x)$ "
by auto

lemma no_para_todo: " $\neg(\forall x. P(x)) \implies \exists x. \neg P(x)$ "
by auto

text {* -----
  Ejercicio 1. Formalizar, y decidir la corrección, del siguiente
  argumento
    Sócrates es un hombre.
    Los hombres son mortales.
    Luego, Sócrates es mortal.
  Usar s      para Sócrates
    H(x) para x es un hombre
    M(x) para x es mortal
  -----
*}

-- "La demostración automática es"
lemma ejercicio_la:
  " $\llbracket H(s); \forall x. H(x) \rightarrow M(x) \rrbracket \implies M(s)$ "

```

```

by auto

-- "La demostración estructurada es"
lemma ejercicio_1b:
  assumes "H(s)"
    "∀x. H(x) → M(x)"
  shows "M(s)"
proof -
  have "H(s) → M(s)" using assms(2) ..
  thus "M(s)" using assms(1) ..
qed

-- "La demostración detallada es"
lemma ejercicio_1c:
  assumes "H(s)"
    "∀x. H(x) → M(x)"
  shows "M(s)"
proof -
  have "H(s) → M(s)" using assms(2) by (rule allE)
  thus "M(s)" using assms(1) by (rule mp)
qed

text {* -----
  Ejercicio 2. Formalizar, y decidir la corrección, del siguiente
  argumento
    Hay estudiantes inteligentes y hay estudiantes trabajadores. Por
    tanto, hay estudiantes inteligentes y trabajadores.
  Usar  $I(x)$  para  $x$  es inteligente
     $T(x)$  para  $x$  es trabajador
  ----- *}

-- "La refutación automática es"
lemma ejercicio_2a:
  "( $\exists x. I(x)$ )  $\wedge$  ( $\exists x. T(x)$ )  $\implies \exists x. I(x) \wedge T(x)$ "
quickcheck
oops

text {*
  El argumento es incorrecto como muestra el siguiente contraejemplo:
     $I = \{a\}^{isub>1}$ 
     $T = \{a\}^{isub>2}$ 
  *}

text {* -----
  Ejercicio 3. Formalizar, y decidir la corrección, del siguiente

```

```

argumento
  Todos los participantes son vencedores. Hay como máximo un
  vencedor. Hay como máximo un participante. Por lo tanto, hay
  exactamente un participante.
  Usar  $P(x)$  para  $x$  es un participante
        $V(x)$  para  $x$  es un vencedor
  ----- *}

-- "La refutación automática es"
lemma ejercicio_3a:
  "[ $\forall x. P(x) \rightarrow V(x);$ 
     $\forall x y. V(x) \wedge V(y) \rightarrow x=y;$ 
     $\forall x y. P(x) \wedge P(y) \rightarrow x=y$ ]"
   $\Rightarrow \exists x. P(x) \wedge (\forall y. P(y) \rightarrow x=y)$ "
quickcheck
oops

text {*
  El argumento es incorrecto como muestra el siguiente contraejemplo:
     $V = \{\}$ 
     $P = \{\}$ 
  *}

text {* -----
  Ejercicio 4. Formalizar, y decidir la corrección, del siguiente
  argumento
    Todo aquel que entre en el país y no sea un VIP será cacheado por
    un aduanero. Hay un contrabandista que entra en el país y que solo
    podrá ser cacheado por contrabandistas. Ningún contrabandista es un
    VIP. Por tanto, algún aduanero es contrabandista.
  Usar  $A(x)$  para  $x$  es aduanero
        $Ca(x,y)$  para  $x$  cachea a  $y$ 
        $Co(x)$  para  $x$  es contrabandista
        $E(x)$  para  $x$  entra en el país
        $V(x)$  para  $x$  es un VIP
  ----- *}

-- "La demostración automática es"
lemma ejercicio_4a:
  "[ $\forall x. E(x) \wedge \neg V(x) \rightarrow (\exists y. A(y) \wedge Ca(y,x));$ 
     $\exists x. Co(x) \wedge E(x) \wedge (\forall y. Ca(y,x) \rightarrow Co(y));$ 
     $\neg(\exists x. Co(x) \wedge V(x))$ ]"
   $\Rightarrow \exists x. A(x) \wedge Co(x)$ "
by metis

```

```
-- "La demostración estructurada es"
lemma ejercicio_4b:
  assumes "∀x. E(x) ∧ ¬V(x) → (∃y. A(y) ∧ Ca(y,x))"
    "∃x. Co(x) ∧ E(x) ∧ (∀y. Ca(y,x) → Co(y))"
    "¬(∃x. Co(x) ∧ V(x))"
  shows "∃x. A(x) ∧ Co(x)"
proof -
  obtain a where a: "Co(a) ∧ E(a) ∧ (∀y. Ca(y,a) → Co(y))"
    using assms(2) ..
  have "∃y. A(y) ∧ Ca(y,a)"
  proof -
    have "E(a) ∧ ¬V(a) → (∃y. A(y) ∧ Ca(y,a))" using assms(1) ..
    moreover
    have "E(a) ∧ ¬V(a)"
    proof
      have "E(a) ∧ (∀y. Ca(y,a) → Co(y))" using a ..
      thus "E(a)" ..
    next
      have "∀x. ¬(Co(x) ∧ V(x))" using assms(3) by (rule no_ex)
      hence "¬(Co(a) ∧ V(a))" ..
      have "Co(a)" using a ..
      show "¬V(a)"
    proof
      assume "V(a)"
      with 'Co(a)' have "Co(a) ∧ V(a)" ..
      with '¬(Co(a) ∧ V(a))' show False ..
    qed
  qed
  ultimately show "∃y. A(y) ∧ Ca(y,a)" ..
qed
then obtain b where "A(b) ∧ Ca(b,a)" ..
hence "A(b)" ..
moreover
have "Co(b)"
proof -
  have "E(a) ∧ (∀y. Ca(y,a) → Co(y))" using a ..
  hence "∀y. Ca(y,a) → Co(y)" ..
  hence "Ca(b,a) → Co(b)" ..
  have "Ca(b,a)" using 'A(b) ∧ Ca(b,a)' ..
  with 'Ca(b,a) → Co(b)' show "Co(b)" ..
qed
ultimately have "A(b) ∧ Co(b)" ..
thus "∃x. A(x) ∧ Co(x)" ..
qed
```

```

-- "La demostración detallada es"
lemma ejercicio_4c:
  assumes "∀x. E(x) ∧ ¬V(x) → (∃y. A(y) ∧ Ca(y,x))"
    "∃x. Co(x) ∧ E(x) ∧ (∀y. Ca(y,x) → Co(y))"
    "¬(∃x. Co(x) ∧ V(x))"
  shows "∃x. A(x) ∧ Co(x)"
proof -
  obtain a where a: "Co(a) ∧ E(a) ∧ (∀y. Ca(y,a) → Co(y))"
    using assms(2) by (rule exE)
  have "∃y. A(y) ∧ Ca(y,a)"
  proof -
    have "E(a) ∧ ¬V(a) → (∃y. A(y) ∧ Ca(y,a))" using assms(1) by (rule allE)
    moreover
    have "E(a) ∧ ¬V(a)"
    proof
      have "E(a) ∧ (∀y. Ca(y,a) → Co(y))" using a by (rule conjunct2)
      thus "E(a)" by (rule conjunct1)
    next
      have "∀x. ¬(Co(x) ∧ V(x))" using assms(3) by (rule no_ex)
      hence "¬(Co(a) ∧ V(a))" by (rule allE)
      have "Co(a)" using a by (rule conjunct1)
      show "¬V(a)"
    proof (rule notI)
      assume "V(a)"
      with 'Co(a)' have "Co(a) ∧ V(a)" by (rule conjI)
      with '¬(Co(a) ∧ V(a))' show False by (rule notE)
    qed
  qed
  ultimately show "∃y. A(y) ∧ Ca(y,a)" by (rule mp)
qed
then obtain b where "A(b) ∧ Ca(b,a)" by (rule exE)
hence "A(b)" by (rule conjunct1)
moreover
have "Co(b)"
proof -
  have "E(a) ∧ (∀y. Ca(y,a) → Co(y))" using a by (rule conjunct2)
  hence "∀y. Ca(y,a) → Co(y)" by (rule conjunct2)
  hence "Ca(b,a) → Co(b)" by (rule allE)
  have "Ca(b,a)" using 'A(b) ∧ Ca(b,a)' by (rule conjunct2)
  with 'Ca(b,a) → Co(b)' show "Co(b)" by (rule mp)
qed
ultimately have "A(b) ∧ Co(b)" by (rule conjI)
thus "∃x. A(x) ∧ Co(x)" by (rule exI)
qed

```

```

text {* -----
  Ejercicio 5. Formalizar, y decidir la corrección, del siguiente
  argumento
    Juan teme a María. Pedro es temido por Juan. Luego, alguien teme a
    María y a Pedro.
  Usar j      para Juan
      m      para María
      p      para Pedro
      T(x,y) para x teme a y
  ----- *}

-- "La demostración automática es"
lemma ejercicio_5a:
  "[T(j,m); T(j,p)]  $\implies \exists x. T(x,m) \wedge T(x,p)$ "
by auto

-- "La demostración estructurada es"
lemma ejercicio_5b:
  assumes "T(j,m)"
          "T(j,p)"
  shows   " $\exists x. T(x,m) \wedge T(x,p)$ "
proof
  show "T(j,m)  $\wedge$  T(j,p)" using assms ..
qed

-- "La demostración detallada es"
lemma ejercicio_5c:
  assumes "T(j,m)"
          "T(j,p)"
  shows   " $\exists x. T(x,m) \wedge T(x,p)$ "
proof (rule exI)
  show "T(j,m)  $\wedge$  T(j,p)" using assms by (rule conjI)
qed

text {* -----
  Ejercicio 6. Formalizar, y decidir la corrección, del siguiente
  argumento
    Los hermanos tienen el mismo padre. Juan es hermano de Luis. Carlos
    es padre de Luis. Por tanto, Carlos es padre de Juan.
  Usar H(x,y) para x es hermano de y
      P(x,y) para x es padre de y
      j      para Juan
      l      para Luis
      c      para Carlos
  ----- *}

```

```

-- "La demostración automática es"
lemma ejercicio_6a:
  assumes "∀x y. H(x,y) → H(y,x)"
          "∀x y z. H(x,y) ∧ P(z,x) → P(z,y)"
          "H(j,l)"
          "P(c,l)"
  shows   "P(c,j)"
using assms
by metis

-- "La demostración estructurada es"
lemma ejercicio_6b:
  fixes H P :: "'a × 'a ⇒ bool"
  assumes "∀x y. H(x,y) → H(y,x)"
          "∀x y z. H(x,y) ∧ P(z,x) → P(z,y)"
          "H(j,l)"
          "P(c,l)"
  shows   "P(c,j)"
proof -
  have "H(l,j) ∧ P(c,l) → P(c,j)"
  proof -
    have "∀y z. H(l,y) ∧ P(z,l) → P(z,y)" using assms(2) ..
    hence "∀z. H(l,j) ∧ P(z,l) → P(z,j)" ..
    thus "H(l,j) ∧ P(c,l) → P(c,j)" ..
  qed
  moreover
  have "H(l,j) ∧ P(c,l)"
  proof
    have "∀y. H(j,y) → H(y,j)" using assms(1) ..
    hence "H(j,l) → H(l,j)" ..
    thus "H(l,j)" using assms(3) ..
  next
    show "P(c,l)" using assms(4) .
  qed
  ultimately show "P(c,j)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_6c:
  fixes H P :: "'a × 'a ⇒ bool"
  assumes "∀x y. H(x,y) → H(y,x)"
          "∀x y z. H(x,y) ∧ P(z,x) → P(z,y)"
          "H(j,l)"
          "P(c,l)"

```

```

shows    "P(c,j)"
proof -
  have "H(l,j)  $\wedge$  P(c,l)  $\rightarrow$  P(c,j)"
  proof -
    have " $\forall y z. H(l,y) \wedge P(z,l) \rightarrow P(z,y)$ " using assms(2) by (rule allE)
    hence " $\forall z. H(l,j) \wedge P(z,l) \rightarrow P(z,j)$ " by (rule allE)
    thus "H(l,j)  $\wedge$  P(c,l)  $\rightarrow$  P(c,j)" by (rule allE)
  qed
  moreover
  have "H(l,j)  $\wedge$  P(c,l)"
  proof (rule conjI)
    have " $\forall y. H(j,y) \rightarrow H(y,j)$ " using assms(1) by (rule allE)
    hence "H(j,l)  $\rightarrow$  H(l,j)" by (rule allE)
    thus "H(l,j)" using assms(3) by (rule mp)
  next
    show "P(c,l)" using assms(4) by this
  qed
  ultimately show "P(c,j)" by (rule mp)
qed

```

```

text {* -----
Ejercicio 7. Formalizar, y decidir la corrección, del siguiente
argumento
  La existencia de algún canal de TV pública, supone un acicate para
  cualquier canal de TV privada; el que un canal de TV tenga un
  acicate, supone una gran satisfacción para cualquiera de sus
  directivos; en Madrid hay varios canales públicos de TV; TV5 es un
  canal de TV privada; por tanto, todos los directivos de TV5 están
  satisfechos.
Usar Pu(x) para x es un canal de TV pública
    Pr(x) para x es un canal de TV privada
    A(x) para x posee un acicate
    D(x,y) para x es un directivo del canal y
    S(x) para x está satisfecho
    t para TV5
----- *}

```

```

-- "La demostración automática es"
lemma ejercicio_7a:
  assumes "( $\exists x. Pu(x)$ )  $\rightarrow$  ( $\forall x. Pr(x) \rightarrow A(x)$ )"
    " $\forall x. A(x) \rightarrow (\forall y. D(y,x) \rightarrow S(y))$ "
    " $\exists x. Pu(x)$ "
    "Pr(t)"
  shows " $\forall x. D(x,t) \rightarrow S(x)$ "
using assms

```

by auto

-- "La demostración estructurada es"

lemma ejercicio_7b:

assumes "($\exists x. Pu(x) \rightarrow (\forall x. Pr(x) \rightarrow A(x))$)"
 " $\forall x. A(x) \rightarrow (\forall y. D(y,x) \rightarrow S(y))$ "
 " $\exists x. Pu(x)$ "
 " $Pr(t)$ "

shows " $\forall x. D(x,t) \rightarrow S(x)$ "

proof

fix a

show " $D(a,t) \rightarrow S(a)$ "

proof

assume " $D(a,t)$ "

have " $\forall x. Pr(x) \rightarrow A(x)$ " using assms(1) assms(3) ..

hence " $Pr(t) \rightarrow A(t)$ " ..

hence " $A(t)$ " using assms(4) ..

have " $A(t) \rightarrow (\forall y. D(y,t) \rightarrow S(y))$ " using assms(2) ..

hence " $\forall y. D(y,t) \rightarrow S(y)$ " using 'A(t)' ..

hence " $D(a,t) \rightarrow S(a)$ " ..

thus " $S(a)$ " using 'D(a,t)' ..

qed

qed

-- "La demostración detallada es"

lemma ejercicio_7c:

assumes "($\exists x. Pu(x) \rightarrow (\forall x. Pr(x) \rightarrow A(x))$)"
 " $\forall x. A(x) \rightarrow (\forall y. D(y,x) \rightarrow S(y))$ "
 " $\exists x. Pu(x)$ "
 " $Pr(t)$ "

shows " $\forall x. D(x,t) \rightarrow S(x)$ "

proof (rule allI)

fix a

show " $D(a,t) \rightarrow S(a)$ "

proof (rule impI)

assume " $D(a,t)$ "

have " $\forall x. Pr(x) \rightarrow A(x)$ " using assms(1) assms(3) by (rule mp)

hence " $Pr(t) \rightarrow A(t)$ " by (rule allE)

hence " $A(t)$ " using assms(4) by (rule mp)

have " $A(t) \rightarrow (\forall y. D(y,t) \rightarrow S(y))$ " using assms(2) by (rule allE)

hence " $\forall y. D(y,t) \rightarrow S(y)$ " using 'A(t)' by (rule mp)

hence " $D(a,t) \rightarrow S(a)$ " by (rule allE)

thus " $S(a)$ " using 'D(a,t)' by (rule mp)

qed

qed

```

text {* -----
  Ejercicio 8. Formalizar, y decidir la corrección, del siguiente
  argumento
    Quien intente entrar en un país y no tenga pasaporte, encontrará
    algún aduanero que le impida el paso. A algunas personas
    motorizadas que intentan entrar en un país le impiden el paso
    únicamente personas motorizadas. Ninguna persona motorizada tiene
    pasaporte. Por tanto, ciertos aduaneros están motorizados.
  Usar  $E(x)$  para  $x$  entra en un país
        $P(x)$  para  $x$  tiene pasaporte
        $A(x)$  para  $x$  es aduanero
        $I(x,y)$  para  $x$  impide el paso a  $y$ 
        $M(x)$  para  $x$  está motorizada
  ----- *}

-- "La demostración automática es"
lemma ejercicio_8a:
  assumes "∀x.  $E(x) \wedge \neg P(x) \rightarrow (\exists y. A(y) \wedge I(y,x))$ "
        "∀x.  $M(x) \wedge E(x) \wedge (\forall y. I(y,x) \rightarrow M(y))$ "
        "∀x.  $M(x) \rightarrow \neg P(x)$ "
  shows "∃x.  $A(x) \wedge M(x)$ "
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_8b:
  assumes "∀x.  $E(x) \wedge \neg P(x) \rightarrow (\exists y. A(y) \wedge I(y,x))$ "
        "∀x.  $M(x) \wedge E(x) \wedge (\forall y. I(y,x) \rightarrow M(y))$ "
        "∀x.  $M(x) \rightarrow \neg P(x)$ "
  shows "∃x.  $A(x) \wedge M(x)$ "
proof -
  obtain a where a: " $M(a) \wedge E(a) \wedge (\forall y. I(y,a) \rightarrow M(y))$ " using assms(2) ..
  hence "M(a)" ..
  have " $M(a) \rightarrow \neg P(a)$ " using assms(3) ..
  hence " $\neg P(a)$ " using 'M(a)' ..
  have a1: " $E(a) \wedge (\forall y. I(y,a) \rightarrow M(y))$ " using a ..
  hence "E(a)" ..
  hence " $E(a) \wedge \neg P(a)$ " using ' $\neg P(a)$ ' ..
  have " $E(a) \wedge \neg P(a) \rightarrow (\exists y. A(y) \wedge I(y,a))$ " using assms(1) ..
  hence "∃y.  $A(y) \wedge I(y,a)$ " using 'E(a) ∧ ¬P(a)' ..
  then obtain b where b: " $A(b) \wedge I(b,a)$ " ..
  have " $A(b) \wedge M(b)$ "
proof
  show "A(b)" using b ..

```

```

next
  have "I(b,a)" using b ..
  have "∀y. I(y,a) → M(y)" using a1 ..
  hence "I(b,a) → M(b)" ..
  thus "M(b)" using 'I(b,a)' ..
qed
thus "∃x. A(x) ∧ M(x)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_8c:
  assumes "∀x. E(x) ∧ ¬P(x) → (∃y. A(y) ∧ I(y,x))"
    "∃x. M(x) ∧ E(x) ∧ (∀y. I(y,x) → M(y))"
    "∀x. M(x) → ¬P(x)"
  shows "∃x. A(x) ∧ M(x)"
proof -
  obtain a where a: "M(a) ∧ E(a) ∧ (∀y. I(y,a) → M(y))"
    using assms(2) by (rule exE)
  hence "M(a)" by (rule conjunct1)
  have "M(a) → ¬P(a)" using assms(3) by (rule allE)
  hence "¬P(a)" using 'M(a)' by (rule mp)
  have a1: "E(a) ∧ (∀y. I(y,a) → M(y))" using a by (rule conjunct2)
  hence "E(a)" by (rule conjunct1)
  hence "E(a) ∧ ¬P(a)" using '¬P(a)' by (rule conjI)
  have "E(a) ∧ ¬P(a) → (∃y. A(y) ∧ I(y,a))"
    using assms(1) by (rule allE)
  hence "∃y. A(y) ∧ I(y,a)" using 'E(a) ∧ ¬P(a)' by (rule mp)
  then obtain b where b: "A(b) ∧ I(b,a)" by (rule exE)
  have "A(b) ∧ M(b)"
  proof (rule conjI)
    show "A(b)" using b by (rule conjunct1)
  next
    have "I(b,a)" using b by (rule conjunct2)
    have "∀y. I(y,a) → M(y)" using a1 by (rule conjunct2)
    hence "I(b,a) → M(b)" by (rule allE)
    thus "M(b)" using 'I(b,a)' by (rule mp)
  qed
  thus "∃x. A(x) ∧ M(x)" by (rule exI)
qed

text {* -----
  Ejercicio 9. Formalizar, y decidir la corrección, del siguiente
  argumento
  Los aficionados al fútbol aplauden a cualquier futbolista
  extranjero. Juanito no aplaude a futbolistas extranjeros. Por

```

```

    tanto, si hay algún futbolista extranjero nacionalizado español,
    Juanito no es aficionado al fútbol.
Usar Af(x)   para x es aficionado al fútbol
    Ap(x,y)  para x aplaude a y
    E(x)     para x es un futbolista extranjero
    N(x)     para x es un futbolista nacionalizado español
    j        para Juanito
----- *}

-- "La demostración automática es"
lemma ejercicio_9a:
  assumes "∀x. Af(x) → (∀y. E(y) → Ap(x,y))"
    "∀x. E(x) → ¬Ap(j,x)"
  shows "(∃x. E(x) ∧ N(x)) → ¬Af(j)"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_9b:
  assumes "∀x. Af(x) → (∀y. E(y) → Ap(x,y))"
    "∀x. E(x) → ¬Ap(j,x)"
  shows "(∃x. E(x) ∧ N(x)) → ¬Af(j)"
proof
  assume "∃x. E(x) ∧ N(x)"
  then obtain a where a: "E(a) ∧ N(a)" ..
  show "¬Af(j)"
  proof
    assume "Af(j)"
    have "Af(j) → (∀y. E(y) → Ap(j,y))" using assms(1) ..
    hence "∀y. E(y) → Ap(j,y)" using 'Af(j)' ..
    hence "E(a) → Ap(j,a)" ..
    have "E(a)" using a ..
    with 'E(a) → Ap(j,a)' have "Ap(j,a)" ..
    have "E(a) → ¬Ap(j,a)" using assms(2) ..
    hence "¬Ap(j,a)" using 'E(a)' ..
    thus False using 'Ap(j,a)' ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_9c:
  assumes "∀x. Af(x) → (∀y. E(y) → Ap(x,y))"
    "∀x. E(x) → ¬Ap(j,x)"
  shows "(∃x. E(x) ∧ N(x)) → ¬Af(j)"
proof (rule impI)

```

```

assume "∃x. E(x) ∧ N(x)"
then obtain a where a: "E(a) ∧ N(a)" by (rule exE)
show "¬Af(j)"
proof (rule notI)
  assume "Af(j)"
  have "Af(j) → (∀y. E(y) → Ap(j,y))" using assms(1) by (rule allE)
  hence "∀y. E(y) → Ap(j,y)" using 'Af(j)' by (rule mp)
  hence "E(a) → Ap(j,a)" by (rule allE)
  have "E(a)" using a by (rule conjunct1)
  with 'E(a) → Ap(j,a)' have "Ap(j,a)" by (rule mp)
  have "E(a) → ¬Ap(j,a)" using assms(2) by (rule allE)
  hence "¬Ap(j,a)" using 'E(a)' by (rule mp)
  thus False using 'Ap(j,a)' by (rule notE)
qed
qed

text {*-----
  Ejercicio 10. Formalizar, y decidir la corrección, del siguiente
  argumento
    Ningún aristócrata debe ser condenado a galeras a menos que sus
    crímenes sean vergonzosos y lleve una vida licenciosa. En la ciudad
    hay aristócratas que han cometido crímenes vergonzosos aunque su
    forma de vida no sea licenciosa. Por tanto, hay algún aristócrata
    que no está condenado a galeras.
  Usar A(x) para x es aristócrata
      G(x) para x está condenado a galeras
      L(x) para x lleva una vida licenciosa
      V(x) para x ha cometido crímenes vergonzoso
  ----- *}

-- "La demostración automática es"
lemma ejercicio_10a:
  assumes "∀x. A(x) ∧ G(x) → L(x) ∧ V(x)"
          "∃x. A(x) ∧ V(x) ∧ ¬L(x)"
  shows   "∃x. A(x) ∧ ¬G(x)"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_10b:
  assumes "∀x. A(x) ∧ G(x) → L(x) ∧ V(x)"
          "∃x. A(x) ∧ V(x) ∧ ¬L(x)"
  shows   "∃x. A(x) ∧ ¬G(x)"
proof -
  obtain b where b: "A(b) ∧ V(b) ∧ ¬L(b)" using assms(2) ..

```

```

have "A(b) ∧ ¬G(b)"
proof
  show "A(b)" using b ..
next
  show "¬G(b)"
  proof
    assume "G(b)"
    have "¬L(b)"
    proof -
      have "V(b) ∧ ¬L(b)" using b ..
      thus "¬L(b)" ..
    qed
    moreover have "L(b)"
    proof -
      have "A(b) ∧ G(b) → L(b) ∧ V(b)" using assms(1) ..
      have "A(b)" using b ..
      hence "A(b) ∧ G(b)" using 'G(b)' ..
      with 'A(b) ∧ G(b) → L(b) ∧ V(b)' have "L(b) ∧ V(b)" ..
      thus "L(b)" ..
    qed
  ultimately show False ..
qed
qed
thus "∃x. A(x) ∧ ¬G(x)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_10c:
  assumes "∀x. A(x) ∧ G(x) → L(x) ∧ V(x)"
    "∃x. A(x) ∧ V(x) ∧ ¬L(x)"
  shows "∃x. A(x) ∧ ¬G(x)"
proof -
  obtain b where b: "A(b) ∧ V(b) ∧ ¬L(b)" using assms(2) by (rule exE)
  have "A(b) ∧ ¬G(b)"
  proof (rule conjI)
    show "A(b)" using b by (rule conjunct1)
  next
    show "¬G(b)"
    proof (rule notI)
      assume "G(b)"
      have "¬L(b)"
      proof -
        have "V(b) ∧ ¬L(b)" using b by (rule conjunct2)
        thus "¬L(b)" by (rule conjunct2)
      qed
    qed
  qed

```

```

    moreover have "L(b)"
    proof -
      have "A(b) ∧ G(b) → L(b) ∧ V(b)" using assms(1) by (rule allE)
      have "A(b)" using b by (rule conjunct1)
      hence "A(b) ∧ G(b)" using 'G(b)' by (rule conjI)
      with 'A(b) ∧ G(b) → L(b) ∧ V(b)' have "L(b) ∧ V(b)" by (rule mp)
      thus "L(b)" by (rule conjunct1)
    qed
    ultimately show False by (rule notE)
  qed
qed
thus "∃x. A(x) ∧ ¬G(x)" by (rule exI)
qed

text {
  -----
  Ejercicio 11. Formalizar, y decidir la corrección, del siguiente argumento
  Todo individuo que esté conforme con el contenido de cualquier
  acuerdo internacional lo apoya o se inhibe en absoluto de asuntos
  políticos. Cualquiera que se inhiba de los asuntos políticos, no
  participará en el próximo referéndum. Todo español, está conforme
  con el acuerdo internacional de Maastricht, al que sin embargo no
  apoya. Por tanto, cualquier individuo o no es español, o en otro
  caso, está conforme con el contenido del acuerdo internacional de
  Maastricht y no participará en el próximo referéndum.
  Usar C(x,y) para la persona x conforme con el contenido del acuerdo y
  A(x,y) para la persona x apoya el acuerdo y
  I(x) para la persona x se inhibe de asuntos políticos
  R(x) para la persona x participará en el próximo referéndum
  E(x) para la persona x es española
  m para el acuerdo de Maastricht
  ----- *}

-- "La demostración automática es"
lemma ejercicio_11a:
  assumes "∀x y. C(x,y) → A(x,y) ∨ I(x)"
    "∀x. I(x) → ¬R(x)"
    "∀x. E(x) → C(x,m) ∧ ¬A(x,m)"
  shows "∀x. ¬E(x) ∨ (C(x,m) ∧ ¬R(x))"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_11b:
  assumes "∀x y. C(x,y) → A(x,y) ∨ I(x)"

```

```

      "∀x. I(x) → ¬R(x)"
      "∀x. E(x) → C(x,m) ∧ ¬A(x,m)"
shows   "∀x. ¬E(x) ∨ (C(x,m) ∧ ¬R(x))"
proof
  fix a
  have "¬E(a) ∨ E(a)" ..
  thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))"
  proof
    assume "¬E(a)"
    thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))" ..
  next
    assume "E(a)"
    have "E(a) → C(a,m) ∧ ¬A(a,m)" using assms(3) ..
    hence "C(a,m) ∧ ¬A(a,m)" using 'E(a)' ..
    hence "C(a,m)" ..
    have "∀y. C(a,y) → A(a,y) ∨ I(a)" using assms(1) ..
    hence "C(a,m) → A(a,m) ∨ I(a)" ..
    hence "A(a,m) ∨ I(a)" using 'C(a,m)' ..
    hence "¬R(a)"
    proof
      assume "A(a,m)"
      have "¬A(a,m)" using 'C(a,m) ∧ ¬A(a,m)' ..
      thus "¬R(a)" using 'A(a,m)' ..
    next
      assume "I(a)"
      have "I(a) → ¬R(a)" using assms(2) ..
      thus "¬R(a)" using 'I(a)' ..
    qed
  with 'C(a,m)' have "C(a,m) ∧ ¬R(a)" ..
  thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))" ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_11c:
  assumes "∀x y. C(x,y) → A(x,y) ∨ I(x)"
          "∀x. I(x) → ¬R(x)"
          "∀x. E(x) → C(x,m) ∧ ¬A(x,m)"
  shows   "∀x. ¬E(x) ∨ (C(x,m) ∧ ¬R(x))"
proof (rule allI)
  fix a
  have "¬E(a) ∨ E(a)" by (rule excluded_middle)
  thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))"
  proof (rule disjE)
    assume "¬E(a)"

```

```

    thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))" by (rule disjI1)
next
  assume "E(a)"
  have "E(a) → C(a,m) ∧ ¬A(a,m)" using assms(3) by (rule allE)
  hence "C(a,m) ∧ ¬A(a,m)" using 'E(a)' by (rule mp)
  hence "C(a,m)" by (rule conjunct1)
  have "∀y. C(a,y) → A(a,y) ∨ I(a)" using assms(1) by (rule allE)
  hence "C(a,m) → A(a,m) ∨ I(a)" by (rule allE)
  hence "A(a,m) ∨ I(a)" using 'C(a,m)' by (rule mp)
  hence "¬R(a)"
proof (rule disjE)
  assume "A(a,m)"
  have "¬A(a,m)" using 'C(a,m) ∧ ¬A(a,m)' by (rule conjunct2)
  thus "¬R(a)" using 'A(a,m)' by (rule notE)
next
  assume "I(a)"
  have "I(a) → ¬R(a)" using assms(2) by (rule allE)
  thus "¬R(a)" using 'I(a)' by (rule mp)
qed
with 'C(a,m)' have "C(a,m) ∧ ¬R(a)" by (rule conjI)
thus "¬E(a) ∨ (C(a,m) ∧ ¬R(a))" by (rule disjI2)
qed
qed

text {* -----
  Ejercicio 12. Formalizar, y decidir la corrección, del siguiente
  argumento
    Toda persona pobre tiene un padre rico. Por tanto, existe una
    persona rica que tiene un abuelo rico.
  Usar  $R(x)$  para  $x$  es rico
     $p(x)$  para el padre de  $x$ 
  ----- *}

-- "La demostración automática es"
lemma ejercicio_12a:
  assumes "∀x. ¬R(x) → R(p(x))"
  shows "∃x. R(x) ∧ R(p(p(x)))"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_12b:
  assumes "∀x. ¬R(x) → R(p(x))"
  shows "∃x. R(x) ∧ R(p(p(x)))"
proof -

```

```

have "∃x. R(x)"
proof (rule ccontr)
  assume "¬(∃x. R(x))"
  hence "∀x. ¬R(x)" by (rule no_ex)
  hence "¬R(a)" ..
  have "¬R(a) → R(p(a))" using assms ..
  hence "R(p(a))" using '¬R(a)' ..
  hence "∃x. R(x)" ..
  with '¬(∃x. R(x))' show False ..
qed
then obtain a where "R(a)" ..
have "¬R(p(p(a))) ∨ R(p(p(a)))" by (rule excluded_middle)
thus "∃x. R(x) ∧ R(p(p(x)))"
proof
  assume "¬R(p(p(a)))"
  have "R(p(a)) ∧ R(p(p(p(a))))"
  proof
    show "R(p(a))"
    proof (rule ccontr)
      assume "¬R(p(a))"
      have "¬R(p(a)) → R(p(p(a)))" using assms ..
      hence "R(p(p(a)))" using '¬R(p(a))' ..
      with '¬R(p(p(a)))' show False ..
    qed
  next
    show "R(p(p(p(a))))"
    proof -
      have "¬R(p(p(a))) → R(p(p(p(a))))" using assms ..
      thus "R(p(p(p(a))))" using '¬R(p(p(a)))' ..
    qed
  qed
  thus "∃x. R(x) ∧ R(p(p(x)))" ..
next
  assume "R(p(p(a)))"
  with 'R(a)' have "R(a) ∧ R(p(p(a)))" ..
  thus "∃x. R(x) ∧ R(p(p(x)))" ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_12c:
  assumes "∀x. ¬R(x) → R(p(x))"
  shows "∃x. R(x) ∧ R(p(p(x)))"
proof -
  have "∃x. R(x)"

```

```

proof (rule ccontr)
  assume "¬(∃x. R(x))"
  hence "∀x. ¬R(x)" by (rule no_ex)
  hence "¬R(a)" by (rule allE)
  have "¬R(a) → R(p(a))" using assms by (rule allE)
  hence "R(p(a))" using '¬R(a)' by (rule mp)
  hence "∃x. R(x)" by (rule exI)
  with '¬(∃x. R(x))' show False by (rule notE)
qed
then obtain a where "R(a)" by (rule exE)
have "¬R(p(p(a))) ∨ R(p(p(a)))" by (rule excluded_middle)
thus "∃x. R(x) ∧ R(p(p(x)))"
proof (rule disjE)
  assume "¬R(p(p(a)))"
  have "R(p(a)) ∧ R(p(p(p(a))))"
  proof (rule conjI)
    show "R(p(a))"
    proof (rule ccontr)
      assume "¬R(p(a))"
      have "¬R(p(a)) → R(p(p(a)))" using assms by (rule allE)
      hence "R(p(p(a)))" using '¬R(p(a))' by (rule mp)
      with '¬R(p(p(a)))' show False by (rule notE)
    qed
  qed
  next
    show "R(p(p(p(a))))"
    proof -
      have "¬R(p(p(a))) → R(p(p(p(a))))" using assms by (rule allE)
      thus "R(p(p(p(a))))" using '¬R(p(p(a)))' by (rule mp)
    qed
  qed
  thus "∃x. R(x) ∧ R(p(p(x)))" by (rule exI)
next
  assume "R(p(p(a)))"
  with 'R(a)' have "R(a) ∧ R(p(p(a)))" by (rule conjI)
  thus "∃x. R(x) ∧ R(p(p(x)))" by (rule exI)
qed
qed

```

text {* -----
Ejercicio 13. Formalizar, y decidir la corrección, del siguiente argumento
Todo deprimido que estima a un submarinista es listo. Cualquiera que se estime a sí mismo es listo. Ningún deprimido se estima a sí mismo. Por tanto, ningún deprimido estima a un submarinista.
Usar $D(x)$ para x está deprimido

```

    E(x,y) para x estima a y
    L(x)    para x es listo
    S(x)    para x es submarinista
    ----- *}

-- "La demostración automática es"
lemma ejercicio_13a:
  assumes "∀x. D(x) ∧ (∃y. S(y) ∧ E(x,y)) → L(x)"
          "∀x. E(x,x) → L(x)"
          "¬(∃x. D(x) ∧ E(x,x))"
  shows   "¬(∃x. D(x) ∧ (∃y. S(y) ∧ E(x,y)))"
quickcheck
oops

text {*
  El argumento es incorrecto como muestra el siguiente contraejemplo:
  Quickcheck found a counterexample:
  D = {a|<^isub>2}
  S = {a|<^isub>1}
  E = {(a|<^isub>2, a|<^isub>1)}
  L = {a|<^isub>2}
*}

text {* -----
  Ejercicio 14. Formalizar, y decidir la corrección, del siguiente
  argumento
  Todos los robots obedecen a los amigos del programador jefe.
  Alvaro es amigo del programador jefe, pero Benito no le
  obedece. Por tanto, Benito no es un robot.
  Usar R(x)    para x es un robot
       Ob(x,y) para x obedece a y
       A(x)    para x es amigo del programador jefe
       b       para Benito
       a       para Alvaro
  ----- *}

-- "La demostración automática es"
lemma ejercicio_14a:
  assumes "∀x y. R(x) ∧ A(y) → Ob(x,y)"
          "A(a)"
          "¬Ob(b,a)"
  shows   "¬R(b)"
using assms
by blast

```

```
-- "La demostración estructurada es"
lemma ejercicio_14b:
  fixes R A :: "'a  $\Rightarrow$  bool" and
    Ob :: "'a  $\times$  'a  $\Rightarrow$  bool"
  assumes "∀x y. R(x)  $\wedge$  A(y)  $\rightarrow$  Ob(x,y)"
    "A(a)"
    "¬Ob(b,a)"
  shows "¬R(b)"
proof
  assume "R(b)"
  have "∀y. R(b)  $\wedge$  A(y)  $\rightarrow$  Ob(b,y)" using assms(1) ..
  hence "R(b)  $\wedge$  A(a)  $\rightarrow$  Ob(b,a)" ..
  have "R(b)  $\wedge$  A(a)" using 'R(b)' assms(2) ..
  with 'R(b)  $\wedge$  A(a)  $\rightarrow$  Ob(b,a)' have "Ob(b,a)" ..
  with assms(3) show False ..
qed

-- "La demostración detallada es"
lemma ejercicio_14c:
  fixes R A :: "'a  $\Rightarrow$  bool" and
    Ob :: "'a  $\times$  'a  $\Rightarrow$  bool"
  assumes "∀x y. R(x)  $\wedge$  A(y)  $\rightarrow$  Ob(x,y)"
    "A(a)"
    "¬Ob(b,a)"
  shows "¬R(b)"
proof (rule notI)
  assume "R(b)"
  have "∀y. R(b)  $\wedge$  A(y)  $\rightarrow$  Ob(b,y)" using assms(1) by (rule allE)
  hence "R(b)  $\wedge$  A(a)  $\rightarrow$  Ob(b,a)" by (rule allE)
  have "R(b)  $\wedge$  A(a)" using 'R(b)' assms(2) by (rule conjI)
  with 'R(b)  $\wedge$  A(a)  $\rightarrow$  Ob(b,a)' have "Ob(b,a)" by (rule mp)
  with assms(3) show False by (rule notE)
qed

text {* -----
  Ejercicio 15. Formalizar, y decidir la corrección, del siguiente
  argumento
    En una pecera nadan una serie de peces. Se observa que:
    * Hay algún pez x que para cualquier pez y, si el pez x no se come
      al pez y entonces existe un pez z tal que z es un tiburón o bien
      z protege al pez y.
    * No hay ningún pez que se coma a todos los demás.
    * Ningún pez protege a ningún otro.
    Por tanto, existe algún tiburón en la pecera.
  Usar C(x,y) para x se come a y
```

```

    P(x,y) para x protege a y
    T(x)   para x es un tiburón
    ----- *}

-- "La demostración automática es"
lemma ejercicio_15a:
  assumes "∃x. ∀y. ¬C(x,y) → (∃z. T(z) ∨ P(z,y))"
          "∀x. ∃y. ¬C(x,y)"
          "∀x y. ¬P(x,y)"
  shows   "∃x. T(x)"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_15b:
  assumes "∃x. ∀y. ¬C(x,y) → (∃z. T(z) ∨ P(z,y))"
          "∀x. ∃y. ¬C(x,y)"
          "∀x y. ¬P(x,y)"
  shows   "∃x. T(x)"
proof -
  obtain a where a: "∀y. ¬C(a,y) → (∃z. T(z) ∨ P(z,y))"
    using assms(1) ..
  have "∃y. ¬C(a,y)" using assms(2) ..
  then obtain b where "¬C(a,b)" ..
  have "¬C(a,b) → (∃z. T(z) ∨ P(z,b))" using a ..
  hence "∃z. T(z) ∨ P(z,b)" using '¬C(a,b)' ..
  then obtain c where "T(c) ∨ P(c,b)" ..
  thus "∃x. T(x)"
proof
  assume "T(c)"
  thus "∃x. T(x)" ..
next
  assume "P(c,b)"
  have "∀y. ¬P(c,y)" using assms(3) ..
  hence "¬P(c,b)" ..
  thus "∃x. T(x)" using 'P(c,b)' ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_15c:
  assumes "∃x. ∀y. ¬C(x,y) → (∃z. T(z) ∨ P(z,y))"
          "∀x. ∃y. ¬C(x,y)"
          "∀x y. ¬P(x,y)"
  shows   "∃x. T(x)"

```

```

proof -
  obtain a where a: "∀y. ¬C(a,y) → (∃z. T(z) ∨ P(z,y))"
    using assms(1) by (rule exE)
  have "∃y. ¬C(a,y)" using assms(2) by (rule allE)
  then obtain b where "¬C(a,b)" by (rule exE)
  have "¬C(a,b) → (∃z. T(z) ∨ P(z,b))" using a by (rule allE)
  hence "∃z. T(z) ∨ P(z,b)" using '¬C(a,b)' by (rule mp)
  then obtain c where "T(c) ∨ P(c,b)" by (rule exE)
  thus "∃x. T(x)"
proof (rule disjE)
  assume "T(c)"
  thus "∃x. T(x)" by (rule exI)
next
  assume "P(c,b)"
  have "∀y. ¬P(c,y)" using assms(3) by (rule allE)
  hence "¬P(c,b)" by (rule allE)
  thus "∃x. T(x)" using 'P(c,b)' by (rule notE)
qed
qed

text {* -----
  Ejercicio 16. Formalizar, y decidir la corrección, del siguiente
  argumento
    Supongamos conocidos los siguientes hechos acerca del número de
    aprobados de dos asignaturas A y B:
    * Si todos los alumnos aprueban la asignatura A, entonces todos
      aprueban la asignatura B.
    * Si algún delegado de la clase aprueba A y B, entonces todos los
      alumnos aprueban A.
    * Si nadie aprueba B, entonces ningún delegado aprueba A.
    * Si Manuel no aprueba B, entonces nadie aprueba B.
    Por tanto, si Manuel es un delegado y aprueba la asignatura A,
    entonces todos los alumnos aprueban las asignaturas A y B.
  Usar A(x,y) para x aprueba la asignatura y
    D(x)   para x es delegado
    m      para Manuel
    a      para la asignatura A
    b      para la asignatura B
  ----- *}

-- "La demostración automática es"
lemma ejercicio_16a:
  assumes "(∀x. A(x,a)) → (∀x. A(x,b))"
    "(∃x. D(x) ∧ A(x,a) ∧ A(x,b)) → (∀x. A(x,a))"
    "(∀x. ¬A(x,b)) → (∀x. D(x) → ¬A(x,a))"

```

```

    "¬A(m,b) → (∀x. ¬A(x,b))"
  shows "D(m) ∧ A(m,a) → (∀x. A(x,a) ∧ A(x,b))"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_16b:
  assumes "(∀x. A(x,a)) → (∀x. A(x,b))"
    "(∃x. D(x) ∧ A(x,a) ∧ A(x,b)) → (∀x. A(x,a))"
    "(∀x. ¬A(x,b)) → (∀x. D(x) → ¬A(x,a))"
    "¬A(m,b) → (∀x. ¬A(x,b))"
  shows "D(m) ∧ A(m,a) → (∀x. A(x,a) ∧ A(x,b))"
proof
  assume "D(m) ∧ A(m,a)"
  hence "D(m)" ..
  have "A(m,a)" using 'D(m) ∧ A(m,a)' ..
  have "A(m,b)"
  proof (rule ccontr)
    assume "¬A(m,b)"
    with assms(4) have "∀x. ¬A(x,b)" ..
    with assms(3) have "∀x. D(x) → ¬A(x,a)" ..
    hence "D(m) → ¬A(m,a)" ..
    hence "¬A(m,a)" using 'D(m)' ..
    thus False using 'A(m,a)' ..
  qed
  with 'A(m,a)' have "A(m,a) ∧ A(m,b)" ..
  with 'D(m)' have "D(m) ∧ A(m,a) ∧ A(m,b)" ..
  hence "∃x. D(x) ∧ A(x,a) ∧ A(x,b)" ..
  with assms(2) have "∀x. A(x,a)" ..
  with assms(1) have "∀x. A(x,b)" ..
  show "∀x. A(x,a) ∧ A(x,b)"
proof
  fix c
  show "A(c,a) ∧ A(c,b)"
proof
  show "A(c,a)" using '∀x. A(x,a)' ..
  next
  show "A(c,b)" using '∀x. A(x,b)' ..
qed
qed
qed

-- "La demostración detallada es"
lemma ejercicio_16c:
  assumes "(∀x. A(x,a)) → (∀x. A(x,b))"

```

```

    "( $\exists x. D(x) \wedge A(x,a) \wedge A(x,b)$ )  $\rightarrow$  ( $\forall x. A(x,a)$ )"
    "( $\forall x. \neg A(x,b)$ )  $\rightarrow$  ( $\forall x. D(x) \rightarrow \neg A(x,a)$ )"
    " $\neg A(m,b) \rightarrow (\forall x. \neg A(x,b))$ "
  shows "D(m)  $\wedge$  A(m,a)  $\rightarrow$  ( $\forall x. A(x,a) \wedge A(x,b)$ )"
proof (rule impI)
  assume "D(m)  $\wedge$  A(m,a)"
  hence "D(m)" by (rule conjunct1)
  have "A(m,a)" using 'D(m)  $\wedge$  A(m,a)' by (rule conjunct2)
  have "A(m,b)"
  proof (rule ccontr)
    assume " $\neg A(m,b)$ "
    with assms(4) have " $\forall x. \neg A(x,b)$ " by (rule mp)
    with assms(3) have " $\forall x. D(x) \rightarrow \neg A(x,a)$ " by (rule mp)
    hence "D(m)  $\rightarrow$   $\neg A(m,a)$ " by (rule allE)
    hence " $\neg A(m,a)$ " using 'D(m)' by (rule mp)
    thus False using 'A(m,a)' by (rule notE)
  qed
  with 'A(m,a)' have "A(m,a)  $\wedge$  A(m,b)" by (rule conjI)
  with 'D(m)' have "D(m)  $\wedge$  A(m,a)  $\wedge$  A(m,b)" by (rule conjI)
  hence " $\exists x. D(x) \wedge A(x,a) \wedge A(x,b)$ " by (rule exI)
  with assms(2) have " $\forall x. A(x,a)$ " by (rule mp)
  with assms(1) have " $\forall x. A(x,b)$ " by (rule mp)
  show " $\forall x. A(x,a) \wedge A(x,b)$ "
proof (rule allI)
  fix c
  show "A(c,a)  $\wedge$  A(c,b)"
proof (rule conjI)
  show "A(c,a)" using ' $\forall x. A(x,a)$ ' by (rule allE)
next
  show "A(c,b)" using ' $\forall x. A(x,b)$ ' by (rule allE)
qed
qed
qed

text {* -----
  Ejercicio 17. Formalizar, y decidir la corrección, del siguiente
  argumento
  En cierto país oriental se ha celebrado la fase final del
  campeonato mundial de fútbol. Cierta diario deportivo ha publicado
  las siguientes estadísticas de tan magno acontecimiento:
  * A todos los porteros que no vistieron camiseta negra les marcó un
    gol algún delantero europeo.
  * Algún portero jugó con botas blancas y sólo le marcaron goles
    jugadores con botas blancas.
  * Ningún portero se marcó un gol a sí mismo.

```

```

* Ningún jugador con botas blancas vistió camiseta negra.
Por tanto, algún delantero europeo jugó con botas blancas.
Usar  $P(x)$  para  $x$  es portero
 $D(x)$  para  $x$  es delantero europeo
 $N(x)$  para  $x$  viste camiseta negra
 $B(x)$  para  $x$  juega con botas blancas
 $M(x,y)$  para  $x$  marcó un gol a  $y$ 
----- *}

-- "La demostración automática es"
lemma ejercicio_17a:
  assumes "∀x. P(x) ∧ ¬N(x) → (∃y. D(y) ∧ M(y,x))"
    "∃x. P(x) ∧ B(x) ∧ (∀y. M(y,x) → B(y))"
    "¬(∃x. P(x) ∧ M(x,x))"
    "¬(∃x. B(x) ∧ N(x))"
  shows "∃x. D(x) ∧ B(x)"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_17b:
  assumes "∀x. P(x) ∧ ¬N(x) → (∃y. D(y) ∧ M(y,x))"
    "∃x. P(x) ∧ B(x) ∧ (∀y. M(y,x) → B(y))"
    "¬(∃x. P(x) ∧ M(x,x))"
    "¬(∃x. B(x) ∧ N(x))"
  shows "∃x. D(x) ∧ B(x)"
proof -
  obtain a where a: "P(a) ∧ B(a) ∧ (∀y. M(y,a) → B(y))"
    using assms(2) ..
  have "P(a) ∧ ¬N(a)"
  proof
    show "P(a)" using a ..
  next
    have "B(a) ∧ (∀y. M(y,a) → B(y))" using a ..
    hence "B(a)" ..
    have "∀x. ¬(B(x) ∧ N(x))" using assms(4) by (rule no_ex)
    hence "¬(B(a) ∧ N(a))" ..
    show "¬N(a)"
  proof
    assume "N(a)"
    with 'B(a)' have "B(a) ∧ N(a)" ..
    with '¬(B(a) ∧ N(a))' show False ..
  qed
qed
hence "∃y. D(y) ∧ M(y,a)"

```

```

proof -
  have "P(a) ∧ ¬N(a) → (∃y. D(y) ∧ M(y,a))" using assms(1) ..
  thus "∃y. D(y) ∧ M(y,a)" using 'P(a) ∧ ¬N(a)' ..
qed
then obtain b where "D(b) ∧ M(b,a)" ..
have "D(b) ∧ B(b)"
proof
  show "D(b)" using 'D(b) ∧ M(b,a)' ..
next
  have "M(b,a)" using 'D(b) ∧ M(b,a)' ..
  have "B(a) ∧ (∀y. M(y,a) → B(y))" using a ..
  hence "∀y. M(y,a) → B(y)" ..
  hence "M(b,a) → B(b)" ..
  thus "B(b)" using 'M(b,a)' ..
qed
thus "∃x. D(x) ∧ B(x)" ..
qed

-- "La demostración estructurada es"
lemma ejercicio_17c:
  assumes "∀x. P(x) ∧ ¬N(x) → (∃y. D(y) ∧ M(y,x))"
    "∃x. P(x) ∧ B(x) ∧ (∀y. M(y,x) → B(y))"
    "¬(∃x. P(x) ∧ M(x,x))"
    "¬(∃x. B(x) ∧ N(x))"
  shows "∃x. D(x) ∧ B(x)"
proof -
  obtain a where a: "P(a) ∧ B(a) ∧ (∀y. M(y,a) → B(y))"
    using assms(2) by (rule exE)
  have "P(a) ∧ ¬N(a)"
  proof (rule conjI)
    show "P(a)" using a by (rule conjunct1)
  next
    have "B(a) ∧ (∀y. M(y,a) → B(y))" using a by (rule conjunct2)
    hence "B(a)" by (rule conjunct1)
    have "∀x. ¬(B(x) ∧ N(x))" using assms(4) by (rule no_ex)
    hence "¬(B(a) ∧ N(a))" by (rule allE)
    show "¬N(a)"
  proof (rule notI)
    assume "N(a)"
    with 'B(a)' have "B(a) ∧ N(a)" by (rule conjI)
    with '¬(B(a) ∧ N(a))' show False by (rule notE)
  qed
qed
hence "∃y. D(y) ∧ M(y,a)"
proof -

```

```

    have "P(a) ∧ ¬N(a) → (∃y. D(y) ∧ M(y,a))"
      using assms(1) by (rule allE)
    thus "∃y. D(y) ∧ M(y,a)" using 'P(a) ∧ ¬N(a)' by (rule mp)
  qed
  then obtain b where "D(b) ∧ M(b,a)" by (rule exE)
  have "D(b) ∧ B(b)"
  proof (rule conjI)
    show "D(b)" using 'D(b) ∧ M(b,a)' by (rule conjunct1)
  next
    have "M(b,a)" using 'D(b) ∧ M(b,a)' by (rule conjunct2)
    have "B(a) ∧ (∀y. M(y,a) → B(y))" using a by (rule conjunct2)
    hence "∀y. M(y,a) → B(y)" by (rule conjunct2)
    hence "M(b,a) → B(b)" by (rule allE)
    thus "B(b)" using 'M(b,a)' by (rule mp)
  qed
  thus "∃x. D(x) ∧ B(x)" by (rule exI)
qed

text {* -----
  Ejercicio 18. Formalizar, y decidir la corrección, del siguiente
  argumento
    Las relaciones de parentesco verifican la siguientes propiedades
    generales:
    * Si x es hermano de y, entonces y es hermano de x.
    * Todo el mundo es hijo de alguien.
    * Nadie es hijo del hermano de su padre.
    * Cualquier padre de una persona es también padre de todos los
      hermanos de esa persona.
    * Nadie es hijo ni hermano de sí mismo.
    Tenemos los siguientes miembros de la familia Peláez: Don Antonio,
    Don Luis, Antoñito y Manolito y sabemos que Don Antonio y Don Luis
    son hermanos, Antoñito y Manolito son hermanos, y Antoñito es hijo
    de Don Antonio. Por tanto, Don Luis no es el padre de Manolito.
    Usar A      para Don Antonio
      He(x,y) para x es hermano de y
      Hi(x,y) para x es hijo de y
      L      para Don Luis
      a      para Antoñito
      m      para Manolito
  ----- *}

-- "La demostración automática es"
lemma ejercicio_18a:
  assumes "∀x y. He(x,y) → He(y,x)"
    "∀x. ∃y. Hi(x,y)"

```

```

    "∀x y z. Hi(x,y) ∧ He(z,y) → ¬Hi(x,z)"
    "∀x y. Hi(x,y) → (∀z. He(z,x) → Hi(z,y))"
    "∀x. ¬Hi(x,x) ∧ ¬He(x,x)"
    "He(A,L)"
    "He(a,m)"
    "Hi(a,A)"
  shows "¬Hi(m,L)"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_18b:
  assumes "∀x y. He(x,y) → He(y,x)"
    "∀x. ∃y. Hi(x,y)"
    "∀x y z. Hi(x,y) ∧ He(z,y) → ¬Hi(x,z)"
    "∀x y. Hi(x,y) → (∀z. He(z,x) → Hi(z,y))"
    "∀x. ¬Hi(x,x) ∧ ¬He(x,x)"
    "He(A,L)"
    "He(a,m)"
    "Hi(a,A)"
  shows "¬Hi(m,L)"
proof
  assume "Hi(m,L)"
  have "He(L,A)"
  proof -
    have "∀y. He(A,y) → He(y,A)" using assms(1) ..
    hence "He(A,L) → He(L,A)" ..
    thus "He(L,A)" using assms(6) ..
  qed
  have "Hi(a,L)"
  proof -
    have "∀y. Hi(m,y) → (∀z. He(z,m) → Hi(z,y))" using assms(4) ..
    hence "Hi(m,L) → (∀z. He(z,m) → Hi(z,L))" ..
    hence "∀z. He(z,m) → Hi(z,L)" using 'Hi(m,L)' ..
    hence "He(a,m) → Hi(a,L)" ..
    thus "Hi(a,L)" using assms(7) ..
  qed
  have "¬Hi(a,L)"
  proof -
    have "Hi(a,A) ∧ He(L,A)" using assms(8) 'He(L,A)' ..
    have "∀y z. Hi(a,y) ∧ He(z,y) → ¬Hi(a,z)" using assms(3) ..
    hence "∀z. Hi(a,A) ∧ He(z,A) → ¬Hi(a,z)" ..
    hence "Hi(a,A) ∧ He(L,A) → ¬Hi(a,L)" ..
    thus "¬Hi(a,L)" using 'Hi(a,A) ∧ He(L,A)' ..
  qed

```

```

    thus False using 'Hi(a,L)' ..
qed

-- "La demostración detallada es"
lemma ejercicio_18c:
  assumes "∀x y. He(x,y) → He(y,x)"
    "∀x. ∃y. Hi(x,y)"
    "∀x y z. Hi(x,y) ∧ He(z,y) → ¬Hi(x,z)"
    "∀x y. Hi(x,y) → (∀z. He(z,x) → Hi(z,y))"
    "∀x. ¬Hi(x,x) ∧ ¬He(x,x)"
    "He(A,L)"
    "He(a,m)"
    "Hi(a,A)"
  shows "¬Hi(m,L)"
proof (rule notI)
  assume "Hi(m,L)"
  have "He(L,A)"
  proof -
    have "∀y. He(A,y) → He(y,A)" using assms(1) by (rule allE)
    hence "He(A,L) → He(L,A)" by (rule allE)
    thus "He(L,A)" using assms(6) by (rule mp)
  qed
  have "Hi(a,L)"
  proof -
    have "∀y. Hi(m,y) → (∀z. He(z,m) → Hi(z,y))"
      using assms(4) by (rule allE)
    hence "Hi(m,L) → (∀z. He(z,m) → Hi(z,L))" by (rule allE)
    hence "∀z. He(z,m) → Hi(z,L)" using 'Hi(m,L)' by (rule mp)
    hence "He(a,m) → Hi(a,L)" by (rule allE)
    thus "Hi(a,L)" using assms(7) by (rule mp)
  qed
  have "¬Hi(a,L)"
  proof -
    have "Hi(a,A) ∧ He(L,A)" using assms(8) 'He(L,A)' by (rule conjI)
    have "∀y z. Hi(a,y) ∧ He(z,y) → ¬Hi(a,z)"
      using assms(3) by (rule allE)
    hence "∀z. Hi(a,A) ∧ He(z,A) → ¬Hi(a,z)" by (rule allE)
    hence "Hi(a,A) ∧ He(L,A) → ¬Hi(a,L)" by (rule allE)
    thus "¬Hi(a,L)" using 'Hi(a,A) ∧ He(L,A)' by (rule mp)
  qed
  thus False using 'Hi(a,L)' by (rule notE)
qed

text {* -----
  Ejercicio 19. [Problema del apisonador de Schubert (en inglés,
```

"Schubert's steamroller")] Formalizar, y decidir la corrección, del siguiente argumento

Si uno de los miembros del club afeita a algún otro (incluido a sí mismo), entonces todos los miembros del club lo han afeitado a él (aunque no necesariamente al mismo tiempo). Guido, Lorenzo, Petruccio y Cesare pertenecen al club de barberos. Guido ha afeitado a Cesare. Por tanto, Petruccio ha afeitado a Lorenzo.

Usar g para Guido
 l para Lorenzo
 p para Petruccio
 c para Cesare
 $B(x)$ para x es un miembro del club de barberos
 $A(x,y)$ para x ha afeitado a y

----- *

-- "La demostración automática es"

lemma ejercicio_19a:

assumes " $\forall x. B(x) \wedge (\exists y. B(y) \wedge A(x,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,x))$ "
 $"B(g)"$
 $"B(l)"$
 $"B(p)"$
 $"B(c)"$
 $"A(g,c)"$
shows " $A(p,l)"$

using assms

by meson

-- "La demostración estructurada es"

lemma ejercicio_19b:

assumes " $\forall x. B(x) \wedge (\exists y. B(y) \wedge A(x,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,x))$ "
 $"B(g)"$
 $"B(l)"$
 $"B(p)"$
 $"B(c)"$
 $"A(g,c)"$
shows " $A(p,l)"$

proof -

have " $A(l,g)"$

proof -

have " $B(g) \wedge (\exists y. B(y) \wedge A(g,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,g))$ "

using assms(1) ..

have " $B(c) \wedge A(g,c)"$ **using** assms(5,6) ..

hence " $(\exists y. B(y) \wedge A(g,y))$ " ..

with assms(2) **have** " $B(g) \wedge (\exists y. B(y) \wedge A(g,y))$ " ..

with ' $B(g) \wedge (\exists y. B(y) \wedge A(g,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,g))$ '

```

    have "( $\forall z. B(z) \rightarrow A(z,g)$ )" ..
    hence " $B(l) \rightarrow A(l,g)$ " ..
    thus " $A(l,g)$ " using assms(3) ..
qed
have " $B(l) \wedge (\exists y. B(y) \wedge A(l,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,l))$ "
  using assms(1) ..
have " $B(g) \wedge A(l,g)$ " using assms(2) 'A(l,g)' ..
hence " $(\exists y. B(y) \wedge A(l,y))$ " ..
with assms(3) have " $B(l) \wedge (\exists y. B(y) \wedge A(l,y))$ " ..
with 'B(l)  $\wedge (\exists y. B(y) \wedge A(l,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,l))$ '
have " $(\forall z. B(z) \rightarrow A(z,l))$ " ..
hence " $B(p) \rightarrow A(p,l)$ " ..
thus " $A(p,l)$ " using assms(4) ..
qed

-- "La demostración detallada es"
lemma ejercicio_19c:
  assumes " $\forall x. B(x) \wedge (\exists y. B(y) \wedge A(x,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,x))$ "
    " $B(g)$ "
    " $B(l)$ "
    " $B(p)$ "
    " $B(c)$ "
    " $A(g,c)$ "
  shows " $A(p,l)$ "
proof -
  have " $A(l,g)$ "
  proof -
    have " $B(g) \wedge (\exists y. B(y) \wedge A(g,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,g))$ "
      using assms(1) by (rule allE)
    have " $B(c) \wedge A(g,c)$ " using assms(5,6) by (rule conjI)
    hence " $(\exists y. B(y) \wedge A(g,y))$ " by (rule exI)
    with assms(2) have " $B(g) \wedge (\exists y. B(y) \wedge A(g,y))$ " by (rule conjI)
    with 'B(g)  $\wedge (\exists y. B(y) \wedge A(g,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,g))$ '
    have " $(\forall z. B(z) \rightarrow A(z,g))$ " by (rule mp)
    hence " $B(l) \rightarrow A(l,g)$ " by (rule allE)
    thus " $A(l,g)$ " using assms(3) by (rule mp)
  qed
  have " $B(l) \wedge (\exists y. B(y) \wedge A(l,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,l))$ "
    using assms(1) by (rule allE)
  have " $B(g) \wedge A(l,g)$ " using assms(2) 'A(l,g)' by (rule conjI)
  hence " $(\exists y. B(y) \wedge A(l,y))$ " by (rule exI)
  with assms(3) have " $B(l) \wedge (\exists y. B(y) \wedge A(l,y))$ " by (rule conjI)
  with 'B(l)  $\wedge (\exists y. B(y) \wedge A(l,y)) \rightarrow (\forall z. B(z) \rightarrow A(z,l))$ '
  have " $(\forall z. B(z) \rightarrow A(z,l))$ " by (rule mp)
  hence " $B(p) \rightarrow A(p,l)$ " by (rule allE)

```

```

    thus "A(p,l)" using assms(4) by (rule mp)
qed

text {* -----
  Ejercicio 20. Formalizar, y decidir la corrección, del siguiente
  argumento
    Carlos afeita a todos los habitantes de Las Chinas que no se
    afeitan a sí mismo y sólo a ellos. Carlos es un habitante de las
    Chinas. Por consiguiente, Carlos no afeita a nadie.
  Usar A(x,y) para x afeita a y
    C(x)   para x es un habitante de Las Chinas
    c       para Carlos
  ----- *}

-- "La demostración automática es"
lemma ejercicio_20a:
  assumes "∀x. A(c,x) ↔ C(x) ∧ ¬A(x,x)"
    "C(c)"
  shows "¬(∃x. A(c,x))"
using assms
by blast

-- "La demostración estructurada es"
lemma ejercicio_20b:
  assumes "∀x. A(c,x) ↔ C(x) ∧ ¬A(x,x)"
    "C(c)"
  shows "¬(∃x. A(c,x))"
proof -
  have 1: "A(c,c) ↔ C(c) ∧ ¬A(c,c)" using assms(1) ..
  have "A(c,c)"
  proof (rule ccontr)
    assume "¬A(c,c)"
    with assms(2) have "C(c) ∧ ¬A(c,c)" ..
    with 1 have "A(c,c)" ..
    with '¬A(c,c)' show False ..
  qed
  have "¬A(c,c)"
  proof -
    have "C(c) ∧ ¬A(c,c)" using 1 'A(c,c)' ..
    thus "¬A(c,c)" ..
  qed
  thus "¬(∃x. A(c,x))" using 'A(c,c)' ..
qed

-- "La demostración detallada es"

```

```

lemma ejercicio_20c:
  assumes "∀x. A(c,x) ↔ C(x) ∧ ¬A(x,x)"
          "C(c)"
  shows   "¬(∃x. A(c,x))"
proof -
  have 1: "A(c,c) ↔ C(c) ∧ ¬A(c,c)" using assms(1) by (rule allE)
  have "A(c,c)"
proof (rule ccontr)
  assume "¬A(c,c)"
  with assms(2) have "C(c) ∧ ¬A(c,c)" by (rule conjI)
  with 1 have "A(c,c)" by (rule iffD2)
  with '¬A(c,c)' show False by (rule notE)
qed
have "¬A(c,c)"
proof -
  have "C(c) ∧ ¬A(c,c)" using 1 'A(c,c)' by (rule iffD1)
  thus "¬A(c,c)" by (rule conjunct2)
qed
thus "¬(∃x. A(c,x))" using 'A(c,c)' by (rule notE)
qed

text {* -----
  Ejercicio 21. Formalizar, y decidir la corrección, del siguiente
  argumento
    Quien desprecia a todos los fanáticos desprecia también a todos los
    políticos. Alguien no desprecia a un determinado político. Por
    consiguiente, hay un fanático al que no todo el mundo desprecia.
  Usar D(x,y) para x desprecia a y
        F(x)   para x es fanático
        P(x)   para x es político
  ----- *}

-- "La demostración automática es"
lemma ejercicio_21a:
  assumes "∀x. (∀y. F(y) → D(x,y)) → (∀y. P(y) → D(x,y))"
          "∃x y. P(y) ∧ ¬D(x,y)"
  shows   "∃x. F(x) ∧ ¬(∀y. D(y,x))"
using assms
by blast

-- "La demostración semiautomática es"
lemma ejercicio_21b:
  assumes "∀x. (∀y. F(y) → D(x,y)) → (∀y. P(y) → D(x,y))"
          "∃x y. P(y) ∧ ¬D(x,y)"
  shows   "∃x. F(x) ∧ ¬(∀y. D(y,x))"

```

```

proof -
  obtain a where "∃y. P(y) ∧ ¬D(a,y)" using assms(2) ..
  then obtain b where "P(b) ∧ ¬D(a,b)" ..
  hence "¬(∀y. P(y) → D(a,y))" by auto
  have "(∀y. F(y) → D(a,y)) → (∀y. P(y) → D(a,y))" using assms(1) ..
  hence "¬(∀y. F(y) → D(a,y))" using '¬(∀y. P(y) → D(a,y))' by (rule mt)
  hence "∃y. ¬(F(y) → D(a,y))" by (rule no_para_todo)
  then obtain c where "¬(F(c) → D(a,c))" ..
  hence "F(c) ∧ ¬(∀y. D(y,c))" by auto
  thus "∃x. F(x) ∧ ¬(∀y. D(y,x))" ..
qed

-- "En la demostración estructurada usaremos el siguiente lema"
lemma no_implica:
  assumes "¬(p → q)"
  shows "p ∧ ¬q"
using assms
by auto

-- "La demostración estructurada del lema es"
lemma no_implica_1:
  assumes "¬(p → q)"
  shows "p ∧ ¬q"
proof
  show "p"
  proof (rule ccontr)
    assume "¬p"
    have "p → q"
    proof
      assume "p"
      with '¬p' show "q" ..
    qed
    with assms show False ..
  qed
next
  show "¬q"
  proof
    assume "q"
    have "p → q"
    proof
      assume "p"
      show "q" using 'q' .
    qed
    with assms show False ..
  qed

```

qed

-- "La demostración detallada del lema es"

```
lemma no_implica_2:
  assumes "¬(p → q)"
  shows "p ∧ ¬q"
proof (rule conjI)
  show "p"
proof (rule ccontr)
  assume "¬p"
  have "p → q"
  proof (rule impI)
    assume "p"
    with '¬p' show "q" by (rule notE)
  qed
  with assms show False by (rule notE)
qed
next
show "¬q"
proof (rule notI)
  assume "q"
  have "p → q"
  proof (rule impI)
    assume "p"
    show "q" using 'q' by this
  qed
  with assms show False by (rule notE)
qed
qed
```

-- "La demostración estructurada es"

```
lemma ejercicio_21c:
  assumes "∀x. (∀y. F(y) → D(x,y)) → (∀y. P(y) → D(x,y))"
  shows "∃x. F(x) ∧ ¬(∀y. D(y,x))"
proof -
  obtain a where "∃y. P(y) ∧ ¬D(a,y)" using assms(2) ..
  then obtain b where b: "P(b) ∧ ¬D(a,b)" ..
  have "¬(∀y. P(y) → D(a,y))"
  proof
    assume "∀y. P(y) → D(a,y)"
    hence "P(b) → D(a,b)" ..
    have "P(b)" using b ..
    with 'P(b) → D(a,b)' have "D(a,b)" ..
    have "¬D(a,b)" using b ..
  qed
qed
```

```

    thus False using 'D(a,b)' ..
qed
have "( $\forall y. F(y) \rightarrow D(a,y)$ )  $\rightarrow$  ( $\forall y. P(y) \rightarrow D(a,y)$ )" using assms(1) ..
hence " $\neg(\forall y. F(y) \rightarrow D(a,y))$ " using ' $\neg(\forall y. P(y) \rightarrow D(a,y))$ ' by (rule mt)
hence " $\exists y. \neg(F(y) \rightarrow D(a,y))$ " by (rule no_para_todo)
then obtain c where c: " $\neg(F(c) \rightarrow D(a,c))$ " ..
have " $F(c) \wedge \neg(\forall y. D(y,c))$ "
proof
  have " $F(c) \wedge \neg D(a,c)$ " using c by (rule no_implica)
  thus "F(c)" ..
next
  show " $\neg(\forall y. D(y,c))$ "
  proof
    assume " $\forall y. D(y,c)$ "
    hence " $D(a,c)$ " ..
    have " $F(c) \wedge \neg D(a,c)$ " using c by (rule no_implica)
    hence " $\neg D(a,c)$ " ..
    thus False using 'D(a,c)' ..
  qed
qed
qed
thus " $\exists x. F(x) \wedge \neg(\forall y. D(y,x))$ " ..
qed

-- "La demostración detallada es"
lemma ejercicio_21d:
  assumes " $\forall x. (\forall y. F(y) \rightarrow D(x,y)) \rightarrow (\forall y. P(y) \rightarrow D(x,y))$ "
    " $\exists x y. P(y) \wedge \neg D(x,y)$ "
  shows " $\exists x. F(x) \wedge \neg(\forall y. D(y,x))$ "
proof -
  obtain a where " $\exists y. P(y) \wedge \neg D(a,y)$ " using assms(2) by (rule exE)
  then obtain b where b: " $P(b) \wedge \neg D(a,b)$ " by (rule exE)
  have " $\neg(\forall y. P(y) \rightarrow D(a,y))$ "
  proof (rule notI)
    assume " $\forall y. P(y) \rightarrow D(a,y)$ "
    hence " $P(b) \rightarrow D(a,b)$ " by (rule allE)
    have " $P(b)$ " using b by (rule conjunct1)
    with 'P(b)  $\rightarrow$  D(a,b)' have "D(a,b)" by (rule mp)
    have " $\neg D(a,b)$ " using b by (rule conjunct2)
    thus False using 'D(a,b)' by (rule notE)
  qed
  have "( $\forall y. F(y) \rightarrow D(a,y)$ )  $\rightarrow$  ( $\forall y. P(y) \rightarrow D(a,y)$ )"
    using assms(1) by (rule allE)
  hence " $\neg(\forall y. F(y) \rightarrow D(a,y))$ " using ' $\neg(\forall y. P(y) \rightarrow D(a,y))$ ' by (rule mt)
  hence " $\exists y. \neg(F(y) \rightarrow D(a,y))$ " by (rule no_para_todo)
  then obtain c where c: " $\neg(F(c) \rightarrow D(a,c))$ " by (rule exE)

```

```

have "F(c) ∧ ¬(∀y. D(y,c))"
proof (rule conjI)
  have "F(c) ∧ ¬D(a,c)" using c by (rule no_implica)
  thus "F(c)" by (rule conjunct1)
next
show "¬(∀y. D(y,c))"
proof (rule notI)
  assume "∀y. D(y,c)"
  hence "D(a,c)" by (rule allE)
  have "F(c) ∧ ¬D(a,c)" using c by (rule no_implica)
  hence "¬D(a,c)" by (rule conjunct2)
  thus False using 'D(a,c)' by (rule notE)
qed
qed
thus "∃x. F(x) ∧ ¬(∀y. D(y,x))" by (rule exI)
qed

text {* -----
  Ejercicio 22. Formalizar, y decidir la corrección, del siguiente
  argumento
    El hombre puro ama todo lo que es puro. Por tanto, el hombre puro
    se ama a sí mismo.
  Usar A(x,y) para x ama a y
    H(x)    para x es un hombre
    P(x)    para x es puro
  ----- *}

-- "La demostración automática es"
lemma ejercicio_22a:
  assumes "∀x. H(x) ∧ P(x) → (∀y. P(y) → A(x,y))"
  shows   "∀x. H(x) ∧ P(x) → A(x,x)"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_22b:
  assumes "∀x. H(x) ∧ P(x) → (∀y. P(y) → A(x,y))"
  shows   "∀x. H(x) ∧ P(x) → A(x,x)"
proof
  fix b
  show "H(b) ∧ P(b) → A(b,b)"
  proof
    assume "H(b) ∧ P(b)"
    hence "P(b)" ..
    have "H(b) ∧ P(b) → (∀y. P(y) → A(b,y))" using assms ..

```

```

    hence "∀y. P(y) → A(b,y)" using 'H(b) ∧ P(b)' ..
    hence "P(b) → A(b,b)" ..
    thus "A(b,b)" using 'P(b)' ..
qed
qed

-- "La demostración detallada es"
lemma ejercicio_22c:
  assumes "∀x. H(x) ∧ P(x) → (∀y. P(y) → A(x,y))"
  shows   "∀x. H(x) ∧ P(x) → A(x,x)"
proof (rule allI)
  fix b
  show "H(b) ∧ P(b) → A(b,b)"
proof (rule impI)
  assume "H(b) ∧ P(b)"
  hence "P(b)" by (rule conjunct2)
  have "H(b) ∧ P(b) → (∀y. P(y) → A(b,y))"
    using assms by (rule allE)
  hence "∀y. P(y) → A(b,y)" using 'H(b) ∧ P(b)' by (rule mp)
  hence "P(b) → A(b,b)" by (rule allE)
  thus "A(b,b)" using 'P(b)' by (rule mp)
qed
qed

text {* -----
  Ejercicio 23. Formalizar, y decidir la corrección, del siguiente
  argumento
    Ningún socio del club está en deuda con el tesorero del club. Si
    un socio del club no paga su cuota está en deuda con el tesorero
    del club. Por tanto, si el tesorero del club es socio del club,
    entonces paga su cuota.
  Usar P(x) para x es socio del club
      Q(x) para x paga su cuota
      R(x) para x está en deuda con el tesorero
      a   para el tesorero del club
  ----- *}

-- "La demostración automática es"
lemma ejercicio_23a:
  assumes "¬(∃x. P(x) ∧ R(x))"
          "∀x. P(x) ∧ ¬Q(x) → R(x)"
  shows   "P(a) → Q(a)"
using assms
by auto

```

```
-- "La demostración estructurada es"
lemma ejercicio_23b:
  assumes "¬(∃x. P(x) ∧ R(x))"
           "∀x. P(x) ∧ ¬Q(x) → R(x)"
  shows   "P(a) → Q(a)"
proof
  assume "P(a)"
  show "Q(a)"
  proof (rule ccontr)
    assume "¬Q(a)"
    with 'P(a)' have "P(a) ∧ ¬Q(a)" ..
    have "P(a) ∧ ¬Q(a) → R(a)" using assms(2) ..
    hence "R(a)" using 'P(a) ∧ ¬Q(a)' ..
    with 'P(a)' have "P(a) ∧ R(a)" ..
    hence "∃x. P(x) ∧ R(x)" ..
    with assms(1) show False ..
  qed
qed

-- "La demostración detallada es"
lemma ejercicio_23c:
  assumes "¬(∃x. P(x) ∧ R(x))"
           "∀x. P(x) ∧ ¬Q(x) → R(x)"
  shows   "P(a) → Q(a)"
proof (rule impI)
  assume "P(a)"
  show "Q(a)"
  proof (rule ccontr)
    assume "¬Q(a)"
    with 'P(a)' have "P(a) ∧ ¬Q(a)" by (rule conjI)
    have "P(a) ∧ ¬Q(a) → R(a)" using assms(2) by (rule allE)
    hence "R(a)" using 'P(a) ∧ ¬Q(a)' by (rule mp)
    with 'P(a)' have "P(a) ∧ R(a)" by (rule conjI)
    hence "∃x. P(x) ∧ R(x)" by (rule exI)
    with assms(1) show False by (rule notE)
  qed
qed

text {* -----
  Ejercicio 24. Formalizar, y decidir la corrección, del siguiente
  argumento
  1. Los lobos, zorros, pájaros, orugas y caracoles son animales y
     existen algunos ejemplares de estos animales.
  2. También hay algunas semillas y las semillas son plantas.
  3. A todo animal le gusta o bien comer todo tipo de plantas o bien
```

le gusta comerse a todos los animales más pequeños que él mismo que gustan de comer algunas plantas.

4. *Las orugas y los caracoles son mucho más pequeños que los pájaros, que son mucho más pequeños que los zorros que a su vez son mucho más pequeños que los lobos.*
5. *A los lobos no les gusta comer ni zorros ni semillas, mientras que a los pájaros les gusta comer orugas pero no caracoles.*
6. *Las orugas y los caracoles gustan de comer algunas plantas.*
7. *Luego, existe un animal al que le gusta comerse un animal al que le gusta comer semillas.*

Usar $A(x)$ para x es un animal

$Ca(x)$ para x es un caracol

$Co(x,y)$ para x le gusta comerse a y

$L(x)$ para x es un lobo

$M(x,y)$ para x es más pequeño que y

$Or(x)$ para x es una oruga

$Pa(x)$ para x es un pájaro

$Pl(x)$ para x es una planta

$S(x)$ para x es una semilla

$Z(x)$ para x es un zorro

----- *

-- "La demostración automática es"

lemma ejercicio_24a:

assumes

(* 1 *) " $\forall x. L(x) \rightarrow A(x)$ "

(* 2 *) " $\forall x. Z(x) \rightarrow A(x)$ "

(* 3 *) " $\forall x. Pa(x) \rightarrow A(x)$ "

(* 4 *) " $\forall x. Or(x) \rightarrow A(x)$ "

(* 5 *) " $\forall x. Ca(x) \rightarrow A(x)$ "

(* 6 *) " $\exists x. L(x)$ "

(* 7 *) " $\exists x. Z(x)$ "

(* 8 *) " $\exists x. Pa(x)$ "

(* 9 *) " $\exists x. Or(x)$ "

(* 10 *) " $\exists x. Ca(x)$ "

(* 11 *) " $\exists x. S(x)$ "

(* 12 *) " $\forall x. S(x) \rightarrow Pl(x)$ "

(* 13 *) " $\forall x. A(x) \rightarrow$

$(\forall y. Pl(y) \rightarrow Co(x,y)) \vee$

$(\forall y. A(y) \wedge M(y,x) \wedge (\exists z. Pl(z) \wedge Co(y,z)) \rightarrow Co(x,y))$ "

(* 14 *) " $\forall x y. Pa(y) \wedge (Ca(x) \vee Or(x)) \rightarrow M(x,y)$ "

(* 15 *) " $\forall x y. Pa(x) \wedge Z(y) \rightarrow M(x,y)$ "

(* 16 *) " $\forall x y. Z(x) \wedge L(y) \rightarrow M(x,y)$ "

(* 17 *) " $\forall x y. L(x) \wedge (Z(y) \vee S(y)) \rightarrow \neg Co(x,y)$ "

(* 18 *) " $\forall x y. Pa(x) \wedge Or(y) \rightarrow Co(x,y)$ "

```

(* 19 *) "∀x y. Pa(x) ∧ Ca(y) → ¬Co(x,y)"
(* 20 *) "∀x. Or(x) ∨ Ca(x) → (∃y. Pl(y) ∧ Co(x,y))"
shows
  "∃x y. A(x) ∧ A(y) ∧ (∃z. S(z) ∧ Co(y,z) ∧ Co(x,y))"
using assms
by meson

-- "La demostración semiautomática es"
lemma ejercicio_24b:
  assumes
    (* 1 *) "∀x. L(x) → A(x)"
    (* 2 *) "∀x. Z(x) → A(x)"
    (* 3 *) "∀x. Pa(x) → A(x)"
    (* 4 *) "∀x. Or(x) → A(x)"
    (* 5 *) "∀x. Ca(x) → A(x)"
    (* 6 *) "∃x. L(x)"
    (* 7 *) "∃x. Z(x)"
    (* 8 *) "∃x. Pa(x)"
    (* 9 *) "∃x. Or(x)"
    (* 10 *) "∃x. Ca(x)"
    (* 11 *) "∃x. S(x)"
    (* 12 *) "∀x. S(x) → Pl(x)"
    (* 13 *) "∀x. A(x) →
      (∀y. Pl(y) → Co(x,y)) ∨
      (∀y. A(y) ∧ M(y,x) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(x,y))"
    (* 14 *) "∀x y. Pa(y) ∧ (Ca(x) ∨ Or(x)) → M(x,y)"
    (* 15 *) "∀x y. Pa(x) ∧ Z(y) → M(x,y)"
    (* 16 *) "∀x y. Z(x) ∧ L(y) → M(x,y)"
    (* 17 *) "∀x y. L(x) ∧ (Z(y) ∨ S(y)) → ¬Co(x,y)"
    (* 18 *) "∀x y. Pa(x) ∧ Or(y) → Co(x,y)"
    (* 19 *) "∀x y. Pa(x) ∧ Ca(y) → ¬Co(x,y)"
    (* 20 *) "∀x. Or(x) ∨ Ca(x) → (∃y. Pl(y) ∧ Co(x,y))"
  shows
    "∃x y. A(x) ∧ A(y) ∧ (∃z. S(z) ∧ Co(y,z) ∧ Co(x,y))"
proof -
  obtain l where l: "L(l)" using assms(6) ..
  obtain z where z: "Z(z)" using assms(7) ..
  obtain p where p: "Pa(p)" using assms(8) ..
  obtain c where c: "Ca(c)" using assms(10) ..
  obtain s where s: "S(s)" using assms(11) ..
  have 1: "Co(p,s)" using p c s assms(3,5,12,13,14,19,20) by meson
  have 2: "¬Co(z,s)" using z l s assms(1,2,12,16,17,13) by meson
  have 3: "M(p,z)" using p z assms(15) by auto
  have 4: "Co(z,p)" using z p s 1 2 3 assms(2,3,12,13) by meson
  hence "Co(z,p) ∧ Co(p,s)" using 1 ..

```

```

    thus "∃x y. A(x) ∧ A(y) ∧ (∃z. S(z) ∧ Co(y,z) ∧ Co(x,y))"
      using z p s assms(2,3) by meson
qed

```

```

lemma instancia:
  assumes "∀x. P(x) → Q(x)"
    "P(a)"
  shows "Q(a)"
proof -
  have "P(a) → Q(a)" using assms(1) ..
  thus "Q(a)" using assms(2) ..
qed

```

```

lemma mt2:
  assumes "p ∧ q ∧ r → s"
    "p"
    "q"
    "¬s"
  shows "¬r"
proof
  assume "r"
  with 'q' have "q ∧ r" ..
  with 'p' have "p ∧ q ∧ r" ..
  with assms(1) have "s" ..
  with '¬s' show False ..
qed

```

```

lemma mp3:
  assumes "p ∧ q ∧ r → s"
    "p"
    "q"
    "r"
  shows "s"
proof -
  have "q ∧ r" using assms(3,4) ..
  with 'p' have "p ∧ q ∧ r" ..
  with assms(1) show "s" ..
qed

```

```

lemma conjI3:
  assumes "p"
    "q"
    "r"
  shows "p ∧ q ∧ r"
proof -

```

```

have "q ∧ r" using assms(2,3) ..
with 'p' show "p ∧ q ∧ r" ..
qed

-- "La demostración estructurada es"
lemma ejercicio_24c:
  assumes
    (* 1 *) "∀x. L(x) → A(x)"
    (* 2 *) "∀x. Z(x) → A(x)"
    (* 3 *) "∀x. Pa(x) → A(x)"
    (* 4 *) "∀x. Or(x) → A(x)"
    (* 5 *) "∀x. Ca(x) → A(x)"
    (* 6 *) "∃x. L(x)"
    (* 7 *) "∃x. Z(x)"
    (* 8 *) "∃x. Pa(x)"
    (* 9 *) "∃x. Or(x)"
    (* 10 *) "∃x. Ca(x)"
    (* 11 *) "∃x. S(x)"
    (* 12 *) "∀x. S(x) → Pl(x)"
    (* 13 *) "∀x. A(x) →
      (∀y. Pl(y) → Co(x,y)) ∨
      (∀y. A(y) ∧ M(y,x) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(x,y))"
    (* 14 *) "∀x y. Pa(y) ∧ (Ca(x) ∨ Or(x)) → M(x,y)"
    (* 15 *) "∀x y. Pa(x) ∧ Z(y) → M(x,y)"
    (* 16 *) "∀x y. Z(x) ∧ L(y) → M(x,y)"
    (* 17 *) "∀x y. L(x) ∧ (Z(y) ∨ S(y)) → ¬Co(x,y)"
    (* 18 *) "∀x y. Pa(x) ∧ Or(y) → Co(x,y)"
    (* 19 *) "∀x y. Pa(x) ∧ Ca(y) → ¬Co(x,y)"
    (* 20 *) "∀x. Or(x) ∨ Ca(x) → (∃y. Pl(y) ∧ Co(x,y))"
  shows
    "∃x y. A(x) ∧ A(y) ∧ (∃z. S(z) ∧ Co(y,z) ∧ Co(x,y))"
proof -
  obtain l where l: "L(l)" using assms(6) ..
  obtain z where z: "Z(z)" using assms(7) ..
  obtain p where p: "Pa(p)" using assms(8) ..
  obtain c where c: "Ca(c)" using assms(10) ..
  obtain s where s: "S(s)" using assms(11) ..
  have 1: "Co(p,s)"
  proof -
    have "¬Co(p,c)"
    proof -
      have "Pa(p) ∧ Ca(c)" using p c ..
      have "∀y. Pa(p) ∧ Ca(y) → ¬Co(p,y)" using assms(19) ..
      hence "Pa(p) ∧ Ca(c) → ¬Co(p,c)" ..
      thus "¬Co(p,c)" using 'Pa(p) ∧ Ca(c)' ..
    end
  end

```

```

qed
have "∃y. Pl(y) ∧ Co(c,y)"
proof -
  have "Or(c) ∨ Ca(c)" using c ..
  have "Or(c) ∨ Ca(c) → (∃y. Pl(y) ∧ Co(c,y))" using assms(20) ..
  thus "∃y. Pl(y) ∧ Co(c,y)" using 'Or(c) ∨ Ca(c)' ..
qed
have "M(c,p)"
proof -
  have "∀y. Pa(y) ∧ (Ca(c) ∨ Or(c)) → M(c,y)" using assms(14) ..
  hence "Pa(p) ∧ (Ca(c) ∨ Or(c)) → M(c,p)" ..
  have "Ca(c) ∨ Or(c)" using c ..
  with p have "Pa(p) ∧ (Ca(c) ∨ Or(c))" ..
  with 'Pa(p) ∧ (Ca(c) ∨ Or(c)) → M(c,p)' show "M(c,p)" ..
qed
show "Co(p,s)"
proof -
  have "A(p)" using assms(3) p by (rule instancia)
  have "A(p) → (∀y. Pl(y) → Co(p,y)) ∨
    (∀y. A(y) ∧ M(y,p) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(p,y))"
    using assms(13) ..
  hence "(∀y. Pl(y) → Co(p,y)) ∨
    (∀y. A(y) ∧ M(y,p) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(p,y))"
    using 'A(p)' ..
  thus "Co(p,s)"
proof
  assume "∀y. Pl(y) → Co(p,y)"
  hence "Pl(s) → Co(p,s)" ..
  have "Pl(s)" using assms(12) s by (rule instancia)
  with 'Pl(s) → Co(p,s)' show "Co(p,s)" ..
next
  assume "∀y. A(y) ∧ M(y,p) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(p,y)"
  hence "A(c) ∧ M(c,p) ∧ (∃z. Pl(z) ∧ Co(c,z)) → Co(p,c)" ..
  have "Co(p,c)"
  proof -
    have "A(c)" using assms(5) c by (rule instancia)
    have "M(c,p) ∧ (∃z. Pl(z) ∧ Co(c,z))"
      using 'M(c,p)' '∃z. Pl(z) ∧ Co(c,z)' ..
    with 'A(c)' have "A(c) ∧ M(c,p) ∧ (∃z. Pl(z) ∧ Co(c,z))" ..
    with 'A(c) ∧ M(c,p) ∧ (∃z. Pl(z) ∧ Co(c,z)) → Co(p,c)'
      show "Co(p,c)" ..
  qed
  with '¬Co(p,c)' show "Co(p,s)" ..
qed
qed
qed

```

```

qed
have 2: "¬Co(z,s)"
proof -
  have "M(z,l)"
  proof -
    have "Z(z) ∧ L(l)" using z l ..
    have "∀y. Z(z) ∧ L(y) → M(z,y)" using assms(16) ..
    hence "Z(z) ∧ L(l) → M(z,l)" ..
    thus "M(z,l)" using 'Z(z) ∧ L(l)' ..
  qed
  have "¬Co(l,z)"
  proof -
    have "∀y. L(l) ∧ (Z(y) ∨ S(y)) → ¬Co(l,y)" using assms(17) ..
    hence "L(l) ∧ (Z(z) ∨ S(z)) → ¬Co(l,z)" ..
    have "Z(z) ∨ S(z)" using z ..
    with l have "L(l) ∧ (Z(z) ∨ S(z))" ..
    with 'L(l) ∧ (Z(z) ∨ S(z)) → ¬Co(l,z)' show "¬Co(l,z)" ..
  qed
  have "¬Co(l,s)"
  proof -
    have "∀y. L(l) ∧ (Z(y) ∨ S(y)) → ¬Co(l,y)" using assms(17) ..
    hence "L(l) ∧ (Z(s) ∨ S(s)) → ¬Co(l,s)" ..
    have "Z(s) ∨ S(s)" using s ..
    with l have "L(l) ∧ (Z(s) ∨ S(s))" ..
    with 'L(l) ∧ (Z(s) ∨ S(s)) → ¬Co(l,s)' show "¬Co(l,s)" ..
  qed
  show "¬Co(z,s)"
  proof -
    have "A(l)" using assms(1) l by (rule instancia)
    have "A(l) → (∀y. Pl(y) → Co(l,y)) ∨
      (∀y. A(y) ∧ M(y,l) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(l,y))"
      using assms(13) ..
    hence "(∀y. Pl(y) → Co(l,y)) ∨
      (∀y. A(y) ∧ M(y,l) ∧ (∃z. Pl(z) ∧ Co(y,z)) → Co(l,y))"
      using 'A(l)' ..
    thus "¬Co(z,s)"
  proof
    assume "∀y. Pl(y) → Co(l,y)"
    hence "Pl(s) → Co(l,s)" ..
    have "Pl(s)" using assms(12) s by (rule instancia)
    with 'Pl(s) → Co(l,s)' have "Co(l,s)" ..
    with '¬Co(l,s)' show "¬Co(z,s)" ..
  next
    assume "∀y. A(y) ∧ M(y,l) ∧ (∃u. Pl(u) ∧ Co(y,u)) → Co(l,y)"
    hence z1: "A(z) ∧ M(z,l) ∧ (∃u. Pl(u) ∧ Co(z,u)) → Co(l,z)" ..

```

```

have "A(z)" using assms(2) z by (rule instancia)
have "¬(∃u. Pl(u) ∧ Co(z,u))"
  using z1 'A(z)' 'M(z,l)' '¬Co(l,z)' by (rule mt2)
show "¬Co(z,s)"
proof
  assume "Co(z,s)"
  have "Pl(s)" using assms(12) s by (rule instancia)
  hence "Pl(s) ∧ Co(z,s)" using 'Co(z,s)' ..
  hence "∃u. Pl(u) ∧ Co(z,u)" ..
  with '¬(∃u. Pl(u) ∧ Co(z,u))' show False ..
qed
qed
qed
qed
have 3: "M(p,z)"
proof -
  have "Pa(p) ∧ Z(z)" using p z ..
  have "∀y. Pa(p) ∧ Z(y) → M(p,y)" using assms(15) ..
  hence "Pa(p) ∧ Z(z) → M(p,z)" ..
  thus "M(p,z)" using 'Pa(p) ∧ Z(z)' ..
qed
have 4: "Co(z,p)"
proof -
  have "A(z)" using assms(2) z by (rule instancia)
  have "A(z) → (∀y. Pl(y) → Co(z,y)) ∨
    (∀y. A(y) ∧ M(y,z) ∧ (∃u. Pl(u) ∧ Co(y,u)) → Co(z,y))"
    using assms(13) ..
  hence "(∀y. Pl(y) → Co(z,y)) ∨
    (∀y. A(y) ∧ M(y,z) ∧ (∃u. Pl(u) ∧ Co(y,u)) → Co(z,y))"
    using 'A(z)' ..
  thus "Co(z,p)"
proof
  assume "∀y. Pl(y) → Co(z,y)"
  hence "Pl(s) → Co(z,s)" ..
  have "Pl(s)" using assms(12) s by (rule instancia)
  with 'Pl(s) → Co(z,s)' have "Co(z,s)" ..
  with '¬Co(z,s)' show "Co(z,p)" ..
next
  assume "∀y. A(y) ∧ M(y,z) ∧ (∃u. Pl(u) ∧ Co(y,u)) → Co(z,y)"
  hence pz: "A(p) ∧ M(p,z) ∧ (∃u. Pl(u) ∧ Co(p,u)) → Co(z,p)" ..
  have "A(p)" using assms(3) p by (rule instancia)
  have "∃u. Pl(u) ∧ Co(p,u)"
  proof -
    have "Pl(s)" using assms(12) s by (rule instancia)
    hence "Pl(s) ∧ Co(p,s)" using 'Co(p,s)' ..

```

```

    thus "∃u. Pl(u) ∧ Co(p,u)" ..
  qed
  show "Co(z,p)" using pz 'A(p)' 'M(p,z)' '∃u. Pl(u) ∧ Co(p,u)'
    by (rule mp3)
  qed
  qed
  hence "Co(z,p) ∧ Co(p,s)" using 1 ..
  show "∃x y. A(x) ∧ A(y) ∧ (∃u. S(u) ∧ Co(y,u) ∧ Co(x,y))"
  proof -
    have "A(z)" using assms(2) z by (rule instancia)
    have "A(p)" using assms(3) p by (rule instancia)
    have "S(s) ∧ Co(p,s) ∧ Co(z,p)" using s 'Co(p,s)' 'Co(z,p)'
      by (rule conjI3)
    hence "∃u. S(u) ∧ Co(p,u) ∧ Co(z,p)" ..
    have "A(z) ∧ A(p) ∧ (∃u. S(u) ∧ Co(p,u) ∧ Co(z,p))"
      using 'A(z)' 'A(p)' '∃u. S(u) ∧ Co(p,u) ∧ Co(z,p)'
      by (rule conjI3)
    hence "∃y. A(z) ∧ A(y) ∧ (∃u. S(u) ∧ Co(y,u) ∧ Co(z,y))" ..
    thus "∃x y. A(x) ∧ A(y) ∧ (∃u. S(u) ∧ Co(y,u) ∧ Co(x,y))" ..
  qed
  qed
end

```

4.4. Ejercicios: Argumentación lógica de primer orden con igualdad

chapter {* T4R3: Argumentación en lógica de primer orden con igualdad *}

```

theory T4R3
imports Main
begin

```

```

text {*

```

 El objetivo de esta relación es formalizar y decidir la corrección de los argumentos. En el caso de que sea correcto, demostrarlo usando sólo las reglas básicas de deducción natural de la lógica de primer orden (sin usar el método auto). En el caso de que sea incorrecto, calcular un contraejemplo con QuickCheck.

Las reglas básicas de la deducción natural son las siguientes:

· conjI: $\llbracket P; Q \rrbracket \Longrightarrow P \wedge Q$

```

· conjunct1:  $P \wedge Q \implies P$ 
· conjunct2:  $P \wedge Q \implies Q$ 
· notnotD:  $\neg\neg P \implies P$ 
· notnotI:  $P \implies \neg\neg P$ 
· mp:  $\llbracket P \rightarrow Q; P \rrbracket \implies Q$ 
· mt:  $\llbracket F \rightarrow G; \neg G \rrbracket \implies \neg F$ 
· impI:  $(P \implies Q) \implies P \rightarrow Q$ 
· disjI1:  $P \implies P \vee Q$ 
· disjI2:  $Q \implies P \vee Q$ 
· disjE:  $\llbracket P \vee Q; P \implies R; Q \implies R \rrbracket \implies R$ 
· FalseE:  $\text{False} \implies P$ 
· notE:  $\llbracket \neg P; P \rrbracket \implies R$ 
· notI:  $(P \implies \text{False}) \implies \neg P$ 
· iffI:  $\llbracket P \implies Q; Q \implies P \rrbracket \implies P = Q$ 
· iffD1:  $\llbracket Q = P; Q \rrbracket \implies P$ 
· iffD2:  $\llbracket P = Q; Q \rrbracket \implies P$ 
· ccontr:  $(\neg P \implies \text{False}) \implies P$ 
· excluded_middle:  $\neg P \vee P$ 

· allI:  $\llbracket \forall x. P\ x; P\ x \implies R \rrbracket \implies R$ 
· allE:  $(\bigwedge x. P\ x) \implies \forall x. P\ x$ 
· exI:  $P\ x \implies \exists x. P\ x$ 
· exE:  $\llbracket \exists x. P\ x; \bigwedge x. P\ x \implies Q \rrbracket \implies Q$ 

· refl:  $t = t$ 
· subst:  $\llbracket s = t; P\ s \rrbracket \implies P\ t$ 

· trans:  $\llbracket r = s; s = t \rrbracket \implies r = t$ 
· sym:  $s = t \implies t = s$ 
· not_sym:  $t \neq s \implies s \neq t$ 
· ssubst:  $\llbracket t = s; P\ s \rrbracket \implies P\ t$ 
· box_equals:  $\llbracket a = b; a = c; b = d \rrbracket \implies a = d$ 
· arg_cong:  $x = y \implies f\ x = f\ y$ 
· fun_cong:  $f = g \implies f\ x = g\ x$ 
· cong:  $\llbracket f = g; x = y \rrbracket \implies f\ x = g\ y$ 

```

```

*}

```

```

text {*

```

Se usarán, además, siguientes reglas que demostramos a continuación.

```

*}

```

```

text {* -----

```

Ejercicio 1. Formalizar, y decidir la corrección, del siguiente argumento

Rosa ama a Curro. Paco no simpatiza con Ana. Quien no simpatiza con Ana ama a Rosa. Si una persona ama a otra, la segunda ama a la primera. Hay como máximo una persona que ama a Rosa. Por tanto, Paco es Curro.

Usar $A(x,y)$ para x ama a y

$S(x,y)$ para x simpatiza con y

a para Ana

c para Curro

p para Paco

r para Rosa

```

----- *}

-- "La demostración automática es"
lemma ejercicio_1a:
  assumes "A(r,c)"
    "¬S(p,a)"
    "∀x. ¬S(x,a) → A(x,r)"
    "∀x y. A(x,y) → A(y,x)"
    "∀x y. A(x,r) ∧ A(y,r) → x=y"
  shows "p = c"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_1b:
  assumes 1: "A(r,c)" and
    2: "¬S(p,a)" and
    3: "∀x. ¬S(x,a) → A(x,r)" and
    4: "∀x y. A(x,y) → A(y,x)" and
    5: "∀x y. A(x,r) ∧ A(y,r) → x=y"
  shows "p = c"
proof -
  have "∀ y. A(p,r) ∧ A(y,r) → p=y" using 5 ..
  hence "A(p,r) ∧ A(c,r) → p=c" ..
  moreover
  have "A(p,r) ∧ A(c,r)"
  proof
    have "¬S(p,a) → A(p,r)" using 3 ..
    thus "A(p,r)" using 2 ..
  next
    have "∀y. A(r,y) → A(y,r)" using 4 ..
    hence "A(r,c) → A(c,r)" ..
    thus "A(c,r)" using 1 ..
  qed
  ultimately show "p=c" ..

```

qed

-- "La demostración detallada es"

lemma ejercicio_1c:

assumes 1: "A(r,c)" and
 2: " $\neg S(p,a)$ " and
 3: " $\forall x. \neg S(x,a) \rightarrow A(x,r)$ " and
 4: " $\forall x y. A(x,y) \rightarrow A(y,x)$ " and
 5: " $\forall x y. A(x,r) \wedge A(y,r) \rightarrow x=y$ "

shows "p = c"

proof -

have " $\forall y. A(p,r) \wedge A(y,r) \rightarrow p=y$ " using 5 by (rule allE)

hence " $A(p,r) \wedge A(c,r) \rightarrow p=c$ " by (rule allE)

moreover

have " $A(p,r) \wedge A(c,r)$ "

proof (rule conjI)

have " $\neg S(p,a) \rightarrow A(p,r)$ " using 3 by (rule allE)

thus " $A(p,r)$ " using 2 by (rule mp)

next

have " $\forall y. A(r,y) \rightarrow A(y,r)$ " using 4 by (rule allE)

hence " $A(r,c) \rightarrow A(c,r)$ " by (rule allE)

thus " $A(c,r)$ " using 1 by (rule mp)

qed

ultimately show "p=c" by (rule mp)

qed

text {* -----

Ejercicio 2. Formalizar, y decidir la corrección, del siguiente argumento

Sólo hay un sofista que enseña gratuitamente, y éste es Sócrates. Sócrates argumenta mejor que ningún otro sofista. Platón argumenta mejor que algún sofista que enseña gratuitamente. Si una persona argumenta mejor que otra segunda, entonces la segunda no argumenta mejor que la primera. Por consiguiente, Platón no es un sofista.

Usar $G(x)$ *para* x *enseña gratuitamente*

$M(x,y)$ *para* x *argumenta mejor que* y

$S(x)$ *para* x *es un sofista*

p *para* Platón

s *para* Sócrates

----- *}

-- "La demostración automática es"

lemma ejercicio_2a:

assumes 1: " $\exists y. (\forall x. S(x) \wedge G(x) \leftrightarrow x=y) \wedge y=s$ " and

```

2: "∀x. S(x) ∧ x ≠ s → M(s,x)" and
3: "∃x. S(x) ∧ G(x) ∧ M(p,x)" and
4: "∀x y. M(x,y) → ¬M(y,x)"
shows "¬S(p)"
using assms
by metis

-- "La demostración semiautomática es"
lemma ejercicio_2b:
  assumes 1: "∃y. (∀x. S(x) ∧ G(x) ↔ x = y) ∧ y = s" and
    2: "∀x. S(x) ∧ x ≠ s → M(s,x)" and
    3: "∃x. S(x) ∧ G(x) ∧ M(p,x)" and
    4: "∀x y. M(x,y) → ¬M(y,x)"
  shows "¬S(p)"
proof
  assume "S(p)"
  obtain a where a: "(∀x. S(x) ∧ G(x) ↔ x = a) ∧ a = s" using 1 ..
  hence s: "∀x. S(x) ∧ G(x) ↔ x = s" by metis
  obtain b where b: "S(b) ∧ G(b) ∧ M(p,b)" using 3 ..
  hence "b = s" using s by metis
  hence "M(p,s)" using b by metis
  hence "¬M(s,p)" using 4 by metis
  hence "p = s" using 2 'S(p)' by metis
  hence "M(s,s)" using 'M(p,s)' by auto
  have "¬M(s,s)" using 'p = s' '¬M(s,p)' by auto
  thus False using 'M(s,s)' by auto
qed

-- "La demostración estructurada es"
lemma ejercicio_2c:
  assumes 1: "∃y. (∀x. S(x) ∧ G(x) ↔ x = y) ∧ y = s" and
    2: "∀x. S(x) ∧ x ≠ s → M(s,x)" and
    3: "∃x. S(x) ∧ G(x) ∧ M(p,x)" and
    4: "∀x y. M(x,y) → ¬M(y,x)"
  shows "¬S(p)"
proof
  assume "S(p)"
  obtain a where a: "(∀x. S(x) ∧ G(x) ↔ x = a) ∧ a = s" using 1 ..
  hence s: "∀x. S(x) ∧ G(x) ↔ x = s"
  proof
    have "a = s" using a ..
    have "∀x. S(x) ∧ G(x) ↔ x = a" using a ..
    with 'a = s' show "∀x. S(x) ∧ G(x) ↔ x = s" by (rule subst)
  qed
  obtain b where b: "S(b) ∧ G(b) ∧ M(p,b)" using 3 ..

```

```

hence "b = s"
proof
  have "S(b) ∧ G(b)"
  proof
    show "S(b)" using b ..
  next
    have "G(b) ∧ M(p,b)" using b ..
    thus "G(b)" ..
  qed
  have "S(b) ∧ G(b) ↔ b = s" using s ..
  thus "b = s" using 'S(b) ∧ G(b)' ..
qed
have "M(p,s)"
proof -
  have "G(b) ∧ M(p,b)" using b ..
  hence "M(p,b)" ..
  with 'b = s' show "M(p,s)" by (rule subst)
qed
have "¬M(s,p)"
proof -
  have "∀y. M(p,y) → ¬M(y,p)" using 4 ..
  hence "M(p,s) → ¬M(s,p)" ..
  thus "¬M(s,p)" using 'M(p,s)' ..
qed
have "p = s"
proof (rule ccontr)
  assume "p ≠ s"
  with 'S(p)' have "S(p) ∧ p ≠ s" ..
  have "S(p) ∧ p ≠ s → M(s,p)" using 2 ..
  hence "M(s,p)" using 'S(p) ∧ p ≠ s' ..
  with '¬M(s,p)' show False ..
qed
hence "M(s,s)" using 'M(p,s)' by (rule subst)
have "¬M(s,s)" using 'p = s' '¬M(s,p)' by (rule subst)
thus False using 'M(s,s)' ..
qed

-- "La demostración detallada es"
lemma ejercicio_2d:
  assumes 1: "∃y. (∀x. S(x) ∧ G(x) ↔ x = y) ∧ y = s" and
    2: "∀x. S(x) ∧ x ≠ s → M(s,x)" and
    3: "∃x. S(x) ∧ G(x) ∧ M(p,x)" and
    4: "∀x y. M(x,y) → ¬M(y,x)"
  shows "¬S(p)"
proof

```

```

assume "S(p)"
obtain a where a: "( $\forall x. S(x) \wedge G(x) \leftrightarrow x = a$ )  $\wedge a = s$ " using 1 by (rule exE)
have s: " $\forall x. S(x) \wedge G(x) \leftrightarrow x = s$ "
proof -
  have "a = s" using a ..
  have " $\forall x. S(x) \wedge G(x) \leftrightarrow x = a$ " using a ..
  with 'a = s' show " $\forall x. S(x) \wedge G(x) \leftrightarrow x = s$ " by (rule subst)
qed
obtain b where b: "S(b)  $\wedge$  G(b)  $\wedge$  M(p,b)" using 3 ..
have "b = s"
proof -
  have "S(b)  $\wedge$  G(b)"
  proof (rule conjI)
    show "S(b)" using b by (rule conjunct1)
  next
    have "G(b)  $\wedge$  M(p,b)" using b by (rule conjunct2)
    thus "G(b)" by (rule conjunct1)
  qed
  have "S(b)  $\wedge$  G(b)  $\leftrightarrow$  b = s" using s by (rule allE)
  thus "b = s" using 'S(b)  $\wedge$  G(b)' by (rule iffD1)
qed
have "M(p,s)"
proof -
  have "G(b)  $\wedge$  M(p,b)" using b ..
  hence "M(p,b)" ..
  with 'b = s' show "M(p,s)" by (rule subst)
qed
have " $\neg$ M(s,p)"
proof -
  have " $\forall y. M(p,y) \rightarrow \neg M(y,p)$ " using 4 ..
  hence "M(p,s)  $\rightarrow$   $\neg$ M(s,p)" ..
  thus " $\neg$ M(s,p)" using 'M(p,s)' ..
qed
have "p = s"
proof (rule ccontr)
  assume "p  $\neq$  s"
  with 'S(p)' have "S(p)  $\wedge$  p  $\neq$  s" ..
  have "S(p)  $\wedge$  p  $\neq$  s  $\rightarrow$  M(s,p)" using 2 ..
  hence "M(s,p)" using 'S(p)  $\wedge$  p  $\neq$  s' ..
  with ' $\neg$ M(s,p)' show False ..
qed
hence "M(s,s)" using 'M(p,s)' by (rule subst)
have " $\neg$ M(s,s)" using 'p = s' ' $\neg$ M(s,p)' by (rule subst)
thus False using 'M(s,s)' by (rule notE)
qed

```

```

text {* -----
  Ejercicio 3. Formalizar, y decidir la corrección, del siguiente
  argumento
    Todos los filósofos se han preguntado qué es la filosofía. Los que
    se preguntan qué es la filosofía se vuelven locos. Nietzsche es
    filósofo. El maestro de Nietzsche no acabó loco. Por tanto,
    Nietzsche y su maestro son diferentes personas.
  Usar  $F(x)$  para  $x$  es filósofo
     $L(x)$  para  $x$  se vuelve loco
     $P(x)$  para  $x$  se ha preguntado qué es la filosofía.
     $m$  para el maestro de Nietzsche
     $n$  para Nietzsche
  ----- *}

-- "La demostración automática es"
lemma ejercicio_3a:
  assumes "∀x. F(x) → P(x)"
    "∀x. P(x) → L(x)"
    "F(n)"
    "¬L(m)"
  shows "n ≠ m"
using assms
by auto

-- "En las siguientes demostraciones se usará este lema"
lemma para_todo_implica:
  assumes "∀x. P(x) → Q(x)"
    "P(a)"
  shows "Q(a)"
proof -
  have "P(a) → Q(a)" using assms(1) ..
  thus "Q(a)" using assms(2) ..
qed

-- "La demostración estructurada es"
lemma ejercicio_3b:
  assumes "∀x. F(x) → P(x)"
    "∀x. P(x) → L(x)"
    "F(n)"
    "¬L(m)"
  shows "n ≠ m"
proof
  assume "n = m"
  hence "F(m)" using assms(3) by (rule subst)

```

```

with assms(1) have "P(m)" by (rule para_todo_implica)
with assms(2) have "L(m)" by (rule para_todo_implica)
with assms(4) show False ..
qed

-- "La demostración detallada es"
lemma ejercicio_3c:
  assumes "∀x. F(x) → P(x)"
          "∀x. P(x) → L(x)"
          "F(n)"
          "¬L(m)"
  shows   "n ≠ m"
proof (rule notI)
  assume "n = m"
  hence "F(m)" using assms(3) by (rule subst)
  with assms(1) have "P(m)" by (rule para_todo_implica)
  with assms(2) have "L(m)" by (rule para_todo_implica)
  with assms(4) show False by (rule notE)
qed

text {* -----
  Ejercicio 4. Formalizar, y decidir la corrección, del siguiente
  argumento
    Los padres son mayores que los hijos. Juan es el padre de Luis. Por
    tanto, Juan es mayor que Luis.
  Usar M(x,y) para x es mayor que y
    p(x)   para el padre de x
    j       para Juan
    l       para Luis
  ----- *}

-- "La demostración automática es"
lemma ejercicio_4a:
  assumes "∀x. M(p(x),x)"
          "j = p(l)"
  shows   "M(j,l)"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_4b:
  assumes "∀x. M(p(x),x)"
          "j = p(l)"
  shows   "M(j,l)"
proof -

```

```

    have "M(p(l),l)" using assms(1) ..
    with assms(2) show "M(j,l)" by (rule ssubst)
qed

-- "La demostración detallada es"
lemma ejercicio_4c:
  assumes "∀x. M(p(x),x)"
    "j = p(l)"
  shows "M(j,l)"
proof -
  have "M(p(l),l)" using assms(1) by (rule allE)
  with assms(2) show "M(j,l)" by (rule ssubst)
qed

text {* -----
  Ejercicio 5. Formalizar, y decidir la corrección, del siguiente
  argumento
    El esposo de la hermana de Toni es Roberto. La hermana de Toni es
    María. Por tanto, el esposo de María es Roberto.
  Usar e(x) para el esposo de x
    h    para la hermana de Toni
    m    para María
    r    para Roberto
  ----- *}

-- "La demostración automática es"
lemma ejercicio_5a:
  assumes "e(h) = r"
    "h = m"
  shows "e(m) = r"
using assms
by auto

-- "La demostración detallada es"
lemma ejercicio_5b:
  assumes "e(h) = r"
    "h = m"
  shows "e(m) = r"
using assms(2,1)
by (rule subst)

text {* -----
  Ejercicio 6. Formalizar, y decidir la corrección, del siguiente
  argumento
    Luis y Jaime tienen el mismo padre. La madre de Rosa es

```

```

    Eva. Eva ama a Carlos. Carlos es el padre de Jaime. Por tanto,
    la madre de Rosa ama al padre de Luis.
Usar A(x,y) para x ama a y
    m(x)    para la madre de x
    p(x)    para el padre de x
    c       para Carlos
    e       para Eva
    j       para Jaime
    l       para Luis
    r       para Rosa
----- *}

-- "La demostración automática es"
lemma ejercicio_6a:
  assumes "p(l) = p(j)"
          "m(r) = e"
          "A(e,c)"
          "c = p(j)"
  shows   "A(m(r),p(l))"
using assms
by auto

-- "La demostración detallada es"
lemma ejercicio_6b:
  assumes "p(l) = p(j)"
          "m(r) = e"
          "A(e,c)"
          "c = p(j)"
  shows   "A(m(r),p(l))"
proof -
  have "A(m(r),c)" using assms(2,3) by (rule ssubst)
  with assms(4) have "A(m(r),p(j))" by (rule subst)
  with assms(1) show "A(m(r),p(l))" by (rule ssubst)
qed

text {* -----
Ejercicio 7. Formalizar, y decidir la corrección, del siguiente
argumento
    Si dos personas son hermanos, entonces tienen la misma madre y el
    mismo padre. Juan es hermano de Luis. Por tanto, la madre del padre
    de Juan es la madre del padre de Luis.
Usar H(x,y) para x es hermano de y
    m(x)    para la madre de x
    p(x)    para el padre de x
    j       para Juan

```

```

      l      para Luis
      ----- *}

-- "La demostración automática es"
lemma ejercicio_7a:
  assumes "∀x y. H(x,y) → m(x) = m(y) ∧ p(x) = p(y)"
    "H(j,l)"
  shows "m(p(j)) = m(p(l))"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_7b:
  assumes "∀x y. H(x,y) → m(x) = m(y) ∧ p(x) = p(y)"
    "H(j,l)"
  shows "m(p(j)) = m(p(l))"
proof -
  have "∀y. H(j,y) → m(j) = m(y) ∧ p(j) = p(y)" using assms(1) ..
  hence "H(j,l) → m(j) = m(l) ∧ p(j) = p(l)" ..
  hence "m(j) = m(l) ∧ p(j) = p(l)" using assms(2) ..
  hence "p(j) = p(l)" ..
  have "m(p(j)) = m(p(j))" by (rule refl)
  with 'p(j) = p(l)' show "m(p(j)) = m(p(l))" by (rule subst)
qed

-- "La demostración detallada es"
lemma ejercicio_7c:
  assumes "∀x y. H(x,y) → m(x) = m(y) ∧ p(x) = p(y)"
    "H(j,l)"
  shows "m(p(j)) = m(p(l))"
proof -
  have "∀y. H(j,y) → m(j) = m(y) ∧ p(j) = p(y)"
    using assms(1) by (rule allE)
  hence "H(j,l) → m(j) = m(l) ∧ p(j) = p(l)" by (rule allE)
  hence "m(j) = m(l) ∧ p(j) = p(l)" using assms(2) by (rule mp)
  hence "p(j) = p(l)" by (rule conjunct2)
  have "m(p(j)) = m(p(j))" by (rule refl)
  with 'p(j) = p(l)' show "m(p(j)) = m(p(l))" by (rule subst)
qed

text {* -----
  Ejercicio 8. Formalizar, y decidir la corrección, del siguiente
  argumento
  Todos los miembros del claustro son asturianos. El secretario forma
  parte del claustro. El señor Martínez es el secretario. Por tanto,

```

```

    el señor Martínez es asturiano.
    Usar  $C(x)$  para  $x$  es miembro del claustro
     $A(x)$  para  $x$  es asturiano
     $s$  para el secretario
     $m$  para el señor Martínez
    ----- *}

-- "La demostración automática es"
lemma ejercicio_8a:
  assumes "∀x. C(x) → A(x)"
    "C(s)"
    "m = s"
  shows "A(m)"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_8b:
  assumes "∀x. C(x) → A(x)"
    "C(s)"
    "m = s"
  shows "A(m)"
proof -
  have "C(s) → A(s)" using assms(1) ..
  hence "A(s)" using assms(2) ..
  with assms(3) show "A(m)" by (rule ssubst)
qed

-- "La demostración detallada es"
lemma ejercicio_8c:
  assumes "∀x. C(x) → A(x)"
    "C(s)"
    "m = s"
  shows "A(m)"
proof -
  have "C(s) → A(s)" using assms(1) by (rule allE)
  hence "A(s)" using assms(2) by (rule mp)
  with assms(3) show "A(m)" by (rule ssubst)
qed

text {* -----
  Ejercicio 10. Formalizar, y decidir la corrección, del siguiente
  argumento
  Eduardo pudo haber visto al asesino. Antonio fue el primer testigo
  de la defensa. O Eduardo estaba en clase o Antonio dio falso

```

```

    testimonio. Nadie en clase pudo haber visto al asesino. Luego, el
    primer testigo de la defensa dio falso testimonio.
  Usar  $C(x)$  para  $x$  estaba en clase
     $F(x)$  para  $x$  dio falso testimonio
     $V(x)$  para  $x$  pudo haber visto al asesino
     $a$  para Antonio
     $e$  para Eduardo
     $p$  para el primer testigo de la defensa
  ----- *}

-- "La demostración automática es"
lemma ejercicio_10a:
  assumes "V(e)"
    "a = p"
    "C(e) v F(a)"
    " $\forall x. C(x) \rightarrow \neg V(x)$ "
  shows "F(p)"
using assms
by auto

-- "La demostración estructurada es"
lemma ejercicio_10b:
  assumes "V(e)"
    "a = p"
    "C(e) v F(a)"
    " $\forall x. C(x) \rightarrow \neg V(x)$ "
  shows "F(p)"
proof -
  have "C(e) v F(a)" using assms(3) .
  hence "F(a)"
  proof
    assume "C(e)"
    have "C(e)  $\rightarrow \neg V(e)$ " using assms(4) ..
    hence " $\neg V(e)$ " using 'C(e)' ..
    thus "F(a)" using assms(1) ..
  next
    assume "F(a)"
    thus "F(a)" .
  qed
  with assms(2) show "F(p)" by (rule subst)
qed

-- "La demostración detallada es"
lemma ejercicio_10c:
  assumes "V(e)"

```

```

    "a = p"
    "C(e) ∨ F(a)"
    "∀x. C(x) → ¬V(x)"
  shows "F(p)"
proof -
  have "C(e) ∨ F(a)" using assms(3) by this
  hence "F(a)"
  proof (rule disjE)
    assume "C(e)"
    have "C(e) → ¬V(e)" using assms(4) by (rule allE)
    hence "¬V(e)" using 'C(e)' by (rule mp)
    thus "F(a)" using assms(1) by (rule notE)
  next
    assume "F(a)"
    thus "F(a)" by this
  qed
  with assms(2) show "F(p)" by (rule subst)
qed

text {* -----
  Ejercicio 11. Formalizar, y decidir la corrección, del siguiente
  argumento
    La luna hoy es redonda. La luna de hace dos semanas tenía forma de
    cuarto creciente. Luna no hay más que una, es decir, siempre es la
    misma. Luego existe algo que es a la vez redondo y con forma de
    cuarto creciente.
  Usar L(x) para la luna del momento x
    R(x) para x es redonda
    C(x) para x tiene forma de cuarto creciente
    h    para hoy
    d    para hace dos semanas
  ----- *}

-- "La demostración automática es"
lemma ejercicio_11a:
  assumes "R(l(h))"
    "C(l(d))"
    "∀x y. l(x) = l(y)"
  shows "∃x. R(x) ∧ C(x)"
using assms
by metis

-- "La demostración estructurada es"
lemma ejercicio_11b:
  assumes "R(l(h))"

```

```

    "C(l(d))"
    "∀x y. l(x) = l(y)"
  shows "∃x. R(x) ∧ C(x)"
proof -
  have "R(l(h)) ∧ C(l(d))" using assms(1,2) ..
  have "∀y. l(d) = l(y)" using assms(3) ..
  hence "l(d) = l(h)" ..
  hence "R(l(h)) ∧ C(l(h))" using 'R(l(h)) ∧ C(l(d))' by (rule subst)
  thus "∃x. R(x) ∧ C(x)" ..
qed

-- "La demostración detallada es"
lemma ejercicio_11c:
  assumes "R(l(h))"
    "C(l(d))"
    "∀x y. l(x) = l(y)"
  shows "∃x. R(x) ∧ C(x)"
proof -
  have "R(l(h)) ∧ C(l(d))" using assms(1,2) by (rule conjI)
  have "∀y. l(d) = l(y)" using assms(3) by (rule allE)
  hence "l(d) = l(h)" by (rule allE)
  hence "R(l(h)) ∧ C(l(h))" using 'R(l(h)) ∧ C(l(d))' by (rule subst)
  thus "∃x. R(x) ∧ C(x)" by (rule exI)
qed

text {* -----
  Ejercicio 12. Formalizar, y decidir la corrección, del siguiente
  argumento
    Juana sólo tiene un marido. Juana está casada con Tomás. Tomás es
    delgado y Guillermo no. Luego, Juana no está casada con Guillermo.
  Usar D(x) para x es delgado
    C(x,y) para x está casada con y
    g para Guillermo
    j para Juana
    t para Tomás
  ----- *}

-- "La demostración automática es"
lemma ejercicio_12a:
  assumes "∃x. ∀y. C(j,y) ↔ y = x"
    "C(j,t)"
    "D(t) ∧ ¬D(g)"
  shows "¬C(j,g)"
using assms
by auto

```

```
-- "La demostración estructurada es"
lemma ejercicio_12b:
  assumes "∃x. ∀y. C(j,y) ↔ y = x"
    "C(j,t)"
    "D(t) ∧ ¬D(g)"
  shows "¬C(j,g)"
proof
  assume "C(j,g)"
  obtain a where a: "∀y. C(j,y) ↔ y = a" using assms(1) ..
  hence "C(j,t) ↔ t = a" ..
  hence "t = a" using assms(2) ..
  have "C(j,g) ↔ g = a" using a ..
  hence "g = a" using 'C(j,g)' ..
  hence "t = g" using 't = a' by (rule ssubst)
  hence "D(g) ∧ ¬D(g)" using assms(3) by (rule subst)
  hence "D(g)" ..
  have "¬D(g)" using 'D(g) ∧ ¬D(g)' ..
  thus False using 'D(g)' ..
qed

-- "La demostración detallada es"
lemma ejercicio_12c:
  assumes "∃x. ∀y. C(j,y) ↔ y = x"
    "C(j,t)"
    "D(t) ∧ ¬D(g)"
  shows "¬C(j,g)"
proof (rule notI)
  assume "C(j,g)"
  obtain a where a: "∀y. C(j,y) ↔ y = a" using assms(1) by (rule exE)
  hence "C(j,t) ↔ t = a" by (rule allE)
  hence "t = a" using assms(2) by (rule iffD1)
  have "C(j,g) ↔ g = a" using a by (rule allE)
  hence "g = a" using 'C(j,g)' by (rule iffD1)
  hence "t = g" using 't = a' by (rule ssubst)
  hence "D(g) ∧ ¬D(g)" using assms(3) by (rule subst)
  hence "D(g)" by (rule conjunct1)
  have "¬D(g)" using 'D(g) ∧ ¬D(g)' by (rule conjunct2)
  thus False using 'D(g)' by (rule notE)
qed

text {* -----
  Ejercicio 13. Formalizar, y decidir la corrección, del siguiente
  argumento
    Sultán no es Chitón. Sultán no obtendrá un plátano a menos que
```

pueda resolver cualquier problema. Si el chimpancé Chitón trabaja más que Sultán resolverá problemas que Sultán no puede resolver. Todos los chimpancés distintos de Sultán trabajan más que Sultán. Por consiguiente, Sultán no obtendrá un plátano.

Usar $Pl(x)$ para x obtiene el plátano

$Pr(x)$ para x es un problema

$R(x,y)$ para x resuelve y

$T(x,y)$ para x trabaja más que y

c para Chitón

s para Sultán

```
----- *}

-- "La demostración automática es"
lemma ejercicio_13a:
  assumes "s ≠ c"
    "Pl(s) → (∀x. Pr(x) → R(s,x))"
    "T(c,s) → (∃x. Pr(x) ∧ R(c,x) ∧ ¬R(s,x))"
    "∀x. x ≠ s → T(x,s)"
  shows "¬Pl(s)"
using assms
by metis

-- "La demostración estructurada es"
lemma ejercicio_13b:
  assumes "s ≠ c"
    "Pl(s) → (∀x. Pr(x) → R(s,x))"
    "T(c,s) → (∃x. Pr(x) ∧ R(c,x) ∧ ¬R(s,x))"
    "∀x. x ≠ s → T(x,s)"
  shows "¬Pl(s)"
proof
  assume "Pl(s)"
  with assms(2) have "∀x. Pr(x) → R(s,x)" ..
  have "c ≠ s" using assms(1) by (rule not_sym)
  have "c ≠ s → T(c,s)" using assms(4) ..
  hence "T(c,s)" using 'c ≠ s' ..
  with assms(3) have "∃x. Pr(x) ∧ R(c,x) ∧ ¬R(s,x)" ..
  then obtain a where a: "Pr(a) ∧ R(c,a) ∧ ¬R(s,a)" ..
  have "R(s,a)"
  proof -
    have "Pr(a)" using a ..
    have "Pr(a) → R(s,a)" using '∀x. Pr(x) → R(s,x)' ..
    thus "R(s,a)" using 'Pr(a)' ..
  qed
  have "¬R(s,a)"
  proof -
```

```

    have "R(c,a)  $\wedge$   $\neg$ R(s,a)" using a ..
    thus " $\neg$ R(s,a)" ..
qed
thus False using 'R(s,a)' ..
qed

-- "La demostración detallada es"
lemma ejercicio_13c:
  assumes "s  $\neq$  c"
    "Pl(s)  $\rightarrow$  ( $\forall$ x. Pr(x)  $\rightarrow$  R(s,x))"
    "T(c,s)  $\rightarrow$  ( $\exists$ x. Pr(x)  $\wedge$  R(c,x)  $\wedge$   $\neg$ R(s,x))"
    " $\forall$ x. x  $\neq$  s  $\rightarrow$  T(x,s)"
  shows " $\neg$ Pl(s)"
proof (rule notI)
  assume "Pl(s)"
  with assms(2) have " $\forall$ x. Pr(x)  $\rightarrow$  R(s,x)" by (rule mp)
  have "c  $\neq$  s" using assms(1) by (rule not_sym)
  have "c  $\neq$  s  $\rightarrow$  T(c,s)" using assms(4) by (rule allE)
  hence "T(c,s)" using 'c  $\neq$  s' by (rule mp)
  with assms(3) have " $\exists$ x. Pr(x)  $\wedge$  R(c,x)  $\wedge$   $\neg$ R(s,x)" by (rule mp)
  then obtain a where a: "Pr(a)  $\wedge$  R(c,a)  $\wedge$   $\neg$ R(s,a)" by (rule exE)
  have "R(s,a)"
  proof -
    have "Pr(a)" using a by (rule conjunct1)
    have "Pr(a)  $\rightarrow$  R(s,a)" using ' $\forall$ x. Pr(x)  $\rightarrow$  R(s,x)' by (rule allE)
    thus "R(s,a)" using 'Pr(a)' by (rule mp)
  qed
  have " $\neg$ R(s,a)"
  proof -
    have "R(c,a)  $\wedge$   $\neg$ R(s,a)" using a by (rule conjunct2)
    thus " $\neg$ R(s,a)" by (rule conjunct2)
  qed
  thus False using 'R(s,a)' by (rule notE)
qed
end

```


Capítulo 5

Resumen de Isabelle/Isar y de la lógica

```
chapter {* Tema 5: Resumen del lenguaje Isabelle/Isar y las reglas de la lógica *
```

```
theory T5
imports Main
begin
```

```
section {* Sintaxis (simplificada) de Isabelle/Isar *
```

```
text {*
  Representación de lemas (y teoremas)
  · Un lema (o teorema) comienza con una etiqueta seguida por algunas
    premisas y una conclusión.
  · Las premisas se introducen con la palabra "assumes" y se separan
    con "and".
  · Cada premisa puede etiquetarse para referenciarse en la demostración.
  · La conclusión se introduce con la palabra "shows".
```

Gramática (simplificada) de las demostraciones en Isabelle/Isar

```
<demostración> ::= proof <método> <declaración>* qed
                  | by <método>
<declaración>   ::= fix <variable>+
                  | assume <proposición>+
                  | (from <hecho>+)? have <proposición>+ <demostración>
                  | (from <hecho>+)? show <proposición>+ <demostración>
<proposición>   ::= (<etiqueta>:)? <cadena>
hecho            ::= <etiqueta>
método          ::= -
                  | this
                  | rule <hecho>
```

```

| simp
| blast
| auto
| induct <variable>

```

La declaración "show" demuestra la conclusión de la demostración mientras que la declaración "have" demuestra un resultado intermedio.

```

*}

```

```

section {* Atajos de Isabelle/Isar *}

```

```

text {*

```

Isar tiene muchos atajos, como los siguientes:

<i>this</i>	<i>éste</i>	<i>el hecho probado en la declaración anterior</i>
<i>then</i>	<i>entonces</i>	<i>from this</i>
<i>hence</i>	<i>por lo tanto</i>	<i>then have</i>
<i>thus</i>	<i>de esta manera</i>	<i>then show</i>
<i>with hecho+</i>	<i>con</i>	<i>from hecho+ and this</i>
<i>.</i>	<i>por ésto</i>	<i>by this</i>
<i>..</i>	<i>trivialmente</i>	<i>by regla (Isabelle adivina la regla)</i>

```

*}

```

```

section {* Resumen de las reglas de la lógica *}

```

```

text {*

```

Resumen de reglas proposicionales:

<i>· TrueI:</i>	<i>True</i>
<i>· conjI:</i>	$\llbracket P; Q \rrbracket \Longrightarrow P \wedge Q$
<i>· conjunct1:</i>	$P \wedge Q \Longrightarrow P$
<i>· conjunct2:</i>	$P \wedge Q \Longrightarrow Q$
<i>· conjE:</i>	$\llbracket P \wedge Q; \llbracket P; Q \rrbracket \Longrightarrow R \rrbracket \Longrightarrow R$
<i>· disjI1:</i>	$P \Longrightarrow P \vee Q$
<i>· disjI2:</i>	$Q \Longrightarrow P \vee Q$
<i>· disjE:</i>	$\llbracket P \vee Q; P \Longrightarrow R; Q \Longrightarrow R \rrbracket \Longrightarrow R$
<i>· notI:</i>	$(P \Longrightarrow \text{False}) \Longrightarrow \neg P$
<i>· notE:</i>	$\llbracket \neg P; P \rrbracket \Longrightarrow R$
<i>· FalseE:</i>	$\text{False} \Longrightarrow P$
<i>· notnotD:</i>	$\neg \neg P \Longrightarrow P$
<i>· impI:</i>	$(P \Longrightarrow Q) \Longrightarrow P \rightarrow Q$
<i>· impE:</i>	$\llbracket P \rightarrow Q; P; Q \Longrightarrow R \rrbracket \Longrightarrow R$
<i>· mp:</i>	$\llbracket P \rightarrow Q; P \rrbracket \Longrightarrow Q$
<i>· iff:</i>	$(P \rightarrow Q) \rightarrow (Q \rightarrow P) \rightarrow P = Q$
<i>· iffI:</i>	$\llbracket P \Longrightarrow Q; Q \Longrightarrow P \rrbracket \Longrightarrow P = Q$
<i>· iffD1:</i>	$\llbracket Q = P; Q \rrbracket \Longrightarrow P$
<i>· iffD2:</i>	$\llbracket P = Q; Q \rrbracket \Longrightarrow P$

```

· iffE:       $\llbracket P = Q; \llbracket P \rightarrow Q; Q \rightarrow P \rrbracket \Longrightarrow R \rrbracket \Longrightarrow R$ 
· ccontr:     $(\neg P \Longrightarrow \text{False}) \Longrightarrow P$ 
· classical:  $(\neg P \Longrightarrow P) \Longrightarrow P$ 
· excluded_middle:  $\neg P \vee P$ 
· disjCI:     $(\neg Q \Longrightarrow P) \Longrightarrow P \vee Q$ 
· impCE:      $\llbracket P \rightarrow Q; \neg P \Longrightarrow R; Q \Longrightarrow R \rrbracket \Longrightarrow R$ 
· iffCE:      $\llbracket P = Q; \llbracket P; Q \rrbracket \Longrightarrow R; \llbracket \neg P; \neg Q \rrbracket \Longrightarrow R \rrbracket \Longrightarrow R$ 
· swap:       $\llbracket \neg P; \neg R \Longrightarrow P \rrbracket \Longrightarrow R$ 
*}

```

```

text {*
  Resumen de reglas de cuantificadores:
  · allE:     $\llbracket \forall x. P\ x; P\ y \Longrightarrow R \rrbracket \Longrightarrow R$ 
  · allI:     $(\bigwedge x. P\ x) \Longrightarrow \forall x. P\ x$ 
  · exI:      $P\ a \Longrightarrow \exists x. P\ x$ 
  · exE:      $\llbracket \exists x. P\ x; \bigwedge x. P\ x \Longrightarrow Q \rrbracket \Longrightarrow Q$ 
*}

```

```

text {*
  Resumen de reglas de la igualdad:
  · refl:     $t = t$ 
  · subst:    $\llbracket s = t; P\ s \rrbracket \Longrightarrow P\ t$ 
  · trans:    $\llbracket r = s; s = t \rrbracket \Longrightarrow r = t$ 
  · sym:      $s = t \Longrightarrow t = s$ 
  · not_sym:  $t \neq s \Longrightarrow s \neq t$ 
  · ssubst:   $\llbracket t = s; P\ s \rrbracket \Longrightarrow P\ t$ 
  · box_equals:  $\llbracket a = b; a = c; b = d \rrbracket \Longrightarrow c = d$ 
  · arg_cong:  $x = y \Longrightarrow f\ x = f\ y$ 
  · fun_cong:  $f = g \Longrightarrow f\ x = g\ x$ 
  · cong:     $\llbracket f = g; x = y \rrbracket \Longrightarrow f\ x = g\ y$ 
*}

```

```

text {*
  Nota: Más información sobre las reglas de inferencia se encuentra en
  la sección 2.2 de "Isabelle's Logics: HOL" http://goo.gl/ZwdUu
  (página 8).
*}

```

```

end

```


Capítulo 6

Razonamiento por casos y por inducción

```
chapter {* Tema 6: Razonamiento por casos y por inducción *}
```

```
theory T6
imports Main Parity
begin
```

```
section {* Razonamiento por distinción de casos *}
```

```
subsection {* Distinción de casos booleanos *}
```

```
text {*
  Lema. [Demostración por distinción de casos booleanos]
     $\neg A \vee A$ 
*}
```

```
-- "La demostración estructurada es"
```

```
lemma " $\neg A \vee A$ "
```

```
proof cases
```

```
  assume "A"
```

```
  thus ?thesis ..
```

```
next
```

```
  assume " $\neg A$ "
```

```
  thus ?thesis ..
```

```
qed
```

```
-- "La demostración detallada es"
```

```
lemma " $\neg A \vee A$ "
```

```
proof cases
```

```
  assume "A"
```

```

    thus ?thesis by (rule disjI2)
next
  assume "¬A"
  thus ?thesis by (rule disjI1)
qed

-- "La demostración automática es"
lemma "¬A ∨ A"
by auto

text {*
  Lema. [Demostración por distinción de casos booleanos nominados]
  ¬A ∨ A
*}

-- "La demostración estructurada es"
lemma "¬A ∨ A"
proof (cases "A")
  case True
  thus ?thesis ..
next
  case False
  thus ?thesis ..
qed

-- "La demostración detallada es"
lemma "¬A ∨ A"
proof (cases "A")
  case True
  thus ?thesis by (rule disjI2)
next
  case False
  thus ?thesis by (rule disjI1)
qed

text {*
  El método "cases" sobre una fórmula:
  · El método (cases F) es una abreviatura de la aplicación de la regla
     $\llbracket F \implies Q; \neg F \implies Q \rrbracket \implies Q$ 
  · La expresión "case True" es una abreviatura de F.
  · La expresión "case False" es una abreviatura de ¬F.
  · Ventajas de "cases" con nombre:
    · reduce la escritura de la fórmula y
    · es independiente del orden de los casos.
*}

```

```

subsection {* Distinción de casos sobre otros tipos de datos *}

text {*
  Lema. [Distinción de casos sobre listas]
  La longitud del resto de una lista es la longitud de la lista menos 1.
*}

-- "La demostración detallada es"
lemma "length(tl xs) = length xs - 1"
proof (cases xs)
  case Nil thus ?thesis by simp
next
  case Cons thus ?thesis by simp
qed

-- "La demostración automática es"
lemma "length(tl xs) = length xs - 1"
by auto

text {*
  Distinción de casos sobre listas:
  · El método de distinción de casos se activa con (cases xs) donde xs
    es del tipo lista.
  · "case Nil" es una abreviatura de
    "assume Nil: xs = []".
  · "case Cons" es una abreviatura de
    "fix ? ?? assume Cons: xs = ? # ??"
    donde ? y ?? son variables anónimas.
*}

text {*
  Lema. [Ejemplo de análisis de casos]
  El resultado de eliminar los n+1 primeros elementos de xs es el mismo
  que eliminar los n primeros elementos del resto de xs.
*}

-- "La demostración detallada es"
lemma "drop (n + 1) xs = drop n (tl xs)"
proof (cases xs)
  case Nil thus "drop (n + 1) xs = drop n (tl xs)" by simp
next
  case Cons thus "drop (n + 1) xs = drop n (tl xs)" by simp
qed

```

```
-- "La demostración automática es"
lemma "drop (n + 1) xs = drop n (tl xs)"
by (cases xs, auto)

text {*
  La función drop está definida en la teoría List de forma que
  (drop n xs) la lista obtenida eliminando en xs los n primeros
  elementos. Su definición es la siguiente
  drop_Nil: "drop n [] = []" |
  drop_Cons: "drop n (x#xs) = (case n of
                                0 => x#xs |
                                Suc(m) => drop m xs)"
*}

section {* Inducción matemática *}

text {*
  [Principio de inducción matemática]
  Para demostrar una propiedad P para todos los números naturales basta
  probar que el 0 tiene la propiedad P y que si n tiene la propiedad P,
  entonces n+1 también la tiene.
  
$$\llbracket P\ 0; \bigwedge n. P\ n \implies P\ (\text{Suc}\ n) \rrbracket \implies P\ m$$


  En Isabelle el principio de inducción matemática está formalizado en
  el teorema nat.induct y puede verse con
  thm nat.induct

  Ejemplo de demostración por inducción: Usaremos el principio de
  inducción matemática para demostrar que
  
$$1 + 3 + \dots + (2n-1) = n^2$$


  Definición. [Suma de los primeros impares]
  (suma_impares n) la suma de los n números impares.
*}

fun suma_impares :: "nat  $\Rightarrow$  nat" where
  "suma_impares 0 = 0"
| "suma_impares (Suc n) = (2*(Suc n) - 1) + suma_impares n"

text {*
  La suma de los 3 primeros números impares se puede calcular mediante "value".
*}

value "suma_impares 3"
```

```

text {*
  Lema. [Ejemplo de demostración por inducción matemática]
  La suma de los  $n$  primeros números impares es  $n^2$ .
*}

-- "La demostración automática es"
lemma "suma_impares n = n * n"
by (induct n) simp_all

text {*
  En la demostración "by (induct n) simp_all" se aplica inducción en  $n$  y
  los dos casos se prueban por simplificación.
*}

-- " Demostración del lema anterior usando patrones"
lemma "suma_impares n = n * n" (is "?P n")
proof (induct n)
  show "?P 0" by simp
next
  fix n assume "?P n"
  thus "?P (Suc n)" by simp
qed

text {*
  Patrones: Cualquier fórmula seguida de (is patrón) equipara el patrón
  con la fórmula.
*}

-- "Demostración del lema anterior con patrones y razonamiento ecuacional"
lemma "suma_impares n = n * n" (is "?P n")
proof (induct n)
  show "?P 0" by simp
next
  fix n assume HI: "?P n"
  have "suma_impares (Suc n) = (2 * (Suc n) - 1) + suma_impares n" by simp
  also have "... = (2 * (Suc n) - 1) + n * n" using HI by simp
  also have "... = n * n + 2 * n + 1" by simp
  finally show "?P (Suc n)" by simp
qed

-- "Demostración del lema anterior por inducción y razonamiento ecuacional"
lemma "suma_impares n = n * n"
proof (induct n)
  show "suma_impares 0 = 0 * 0" by simp
next

```

```

fix n assume HI: "suma_impares n = n * n"
have "suma_impares (Suc n) = (2 * (Suc n) - 1) + suma_impares n" by simp
also have "... = (2 * (Suc n) - 1) + n * n" using HI by simp
also have "... = n * n + 2 * n + 1" by simp
finally show "suma_impares (Suc n) = (Suc n) * (Suc n)" by simp
qed

```

```

text {*
  Definición. Un número natural  $n$  es par si existe un natural  $m$  tal que
   $n=m+m$ .
*}

```

```

definition par :: "nat  $\Rightarrow$  bool" where
  "par n  $\equiv \exists m. n=m+m$ "

```

```

text {*
  Lema. [Ejemplo de inducción y existenciales] Para todo número natural
   $n$ , se verifica que  $n*(n+1)$  par.
*}

```

```

lemma
  fixes n :: "nat"
  shows "par (n*(n+1))"
proof (induct n)
  show "par (0*(0+1))" by (simp add: par_def)
next
  fix n assume "par (n*(n+1))"
  hence " $\exists m. n*(n+1) = m+m$ " by (simp add: par_def)
  then obtain m where m: " $n*(n+1) = m+m$ " by (rule exE)
  hence "(Suc n)*((Suc n)+1) = (m+n+1)+(m+n+1)" by auto
  hence " $\exists m. (Suc n)*((Suc n)+1) = m+m$ " by (rule exI)
  thus "par ((Suc n)*((Suc n)+1))" by (simp add: par_def)
qed

```

```

text {*
  En Isabelle puede demostrarse de manera más simple un lema equivalente
  usando en lugar de la función "par" la función "even" definida en la
  teoría Parity por
   $even\ x \leftrightarrow x \bmod 2 = 0$ 
*}

```

```

lemma
  fixes n :: "nat"
  shows "even (n*(n+1))"
by auto

```

```
text {*
  Para completar la demostración basta demostrar la equivalencia de las
  funciones "par" y "even".
*}
```

```
lemma
  fixes n :: "nat"
  shows "par n = even n"
proof -
  have "par n = (∃m. n = m+m)" by (simp add:par_def)
  thus "par n = even n" by presburger
qed
```

```
text {*
  En la demostración anterior hemos usado la táctica "presburger" que
  corresponde a la aritmética de Presburger.
*}
```

```
section {* Inducción estructural *
```

```
text {*
  Inducción estructural]
  · En Isabelle puede hacerse inducción estructural sobre cualquier tipo
    recursivo.
  · La inducción matemática es la inducción estructural sobre el tipo de
    los naturales.
  · El esquema de inducción estructural sobre listas es
    · list.induct:  $\llbracket P []; \bigwedge x \text{ ys. } P \text{ ys} \implies P (x \# \text{ys}) \rrbracket \implies P \text{ zs}$ 
  · Para demostrar una propiedad para todas las listas basta demostrar
    que la lista vacía tiene la propiedad y que al añadir un elemento a una
    lista que tiene la propiedad se obtiene una lista que también tiene la
    propiedad.
  · En Isabelle el principio de inducción sobre listas está formalizado
    mediante el teorema list.induct que puede verse con
    thm list.induct
```

Concatenación de listas:

En la teoría List.thy está definida la concatenación de listas (que se representa por @) como sigue

```
append_Nil: "[]@ys = ys"
append_Cons: "(x#xs)@ys = x#(xs@ys)"
```

Lema. [Ejemplo de inducción sobre listas]
La concatenación de listas es asociativa.

```

*}

-- "La demostración automática es"
lemma conc_asociativa_1: "xs @ (ys @ zs) = (xs @ ys) @ zs"
by (induct xs) simp_all

-- "La demostración estructurada es"
lemma conc_asociativa: "xs @ (ys @ zs) = (xs @ ys) @ zs"
proof (induct xs)
  show "[ ] @ (ys @ zs) = ([ ] @ ys) @ zs"
  proof -
    have "[ ] @ (ys @ zs) = ys @ zs" by simp
    also have "... = ([ ] @ ys) @ zs" by simp
    finally show ?thesis .
  qed
next
  fix x xs
  assume HI: "xs @ (ys @ zs) = (xs @ ys) @ zs"
  show "(x#xs) @ (ys @ zs) = ((x#xs) @ ys) @ zs"
  proof -
    have "(x#xs) @ (ys @ zs) = x#(xs @ (ys @ zs))" by simp
    also have "... = x#((xs @ ys) @ zs)" using HI by simp
    also have "... = (x#(xs @ ys)) @ zs" by simp
    also have "... = ((x#xs) @ ys) @ zs" by simp
    finally show ?thesis .
  qed
qed

text {*
  Ejemplo [Árboles binarios]
  Definir un tipo de dato para los árboles binarios.
*}

datatype 'a arbolB = Hoja "'a"
                  | Nodo "'a" "'a arbolB" "'a arbolB"

text {*
  Ejemplo. [Imagen especular]
  Definir la función "espejo" que aplicada a un árbol devuelve su imagen
  especular.
*}

fun espejo :: "'a arbolB ⇒ 'a arbolB" where
  "espejo (Hoja a) = (Hoja a)"
| "espejo (Nodo f x y) = (Nodo f (espejo y) (espejo x))"

```

```

text {*
  Ejemplo [La imagen especular es involutiva]
  Demostrar que la función "espejo" involutiva; es decir, para cualquier
  árbol t, se tiene que
    espejo (espejo(t)) = t.
*}

-- "La demostración automática es"
lemma espejo_involutiva_1: "espejo(espejo(t)) = t"
by (induct t) auto

-- "La demostración estructurada es"
lemma espejo_involutiva: "espejo(espejo(t)) = t" (is "?P t")
proof (induct t)
  fix x :: 'a show "?P (Hoja x)" by simp
next
  fix t1 :: "'a arbolB" assume h1: "?P t1"
  fix t2 :: "'a arbolB" assume h2: "?P t2"
  fix x :: 'a
  show "?P (Nodo x t1 t2)"
  proof -
    have "espejo(espejo(Nodo x t1 t2)) = espejo(Nodo x (espejo t2) (espejo t1))"
    by simp
    also have "... = Nodo x (espejo (espejo t1)) (espejo (espejo t2))" by simp
    also have "... = Nodo x t1 t2" using h1 h2 by simp
    finally show ?thesis .
  qed
qed

text {*
  Ejemplo. [Aplanamiento de árboles]
  Definir la función "aplana" que aplane los árboles recorriéndolos en
  orden infijo.
*}

fun aplanar :: "'a arbolB ⇒ 'a list" where
  "aplana (Hoja a) = [a]"
| "aplana (Nodo x t1 t2) = (aplana t1)@[x]@(aplana t2)"

text {*
  Ejemplo. [Aplanamiento de la imagen especular] Demostrar que
    aplanar (espejo t) = rev (aplana t)
*}

```

```

-- "La demostración automática es"
lemma "aplana (espejo t) = rev (aplana t)"
by (induct t) auto

-- "La demostración estructurada es"
lemma "aplana (espejo t) = rev (aplana t)" (is "?P t")
proof (induct t)
  fix x :: 'a
  show "?P (Hoja x)" by simp
next
  fix t1 :: "'a arbolB" assume h1: "?P t1"
  fix t2 :: "'a arbolB" assume h2: "?P t2"
  fix x :: 'a
  show "?P (Nodo x t1 t2)"
proof -
  have "aplana (espejo (Nodo x t1 t2)) =
    aplana (Nodo x (espejo t2) (espejo t1))" by simp
  also have "... = (aplana (espejo t2))@[x]@(aplana (espejo t1))" by simp
  also have "... = (rev (aplana t2))@[x]@(rev (aplana t1))" using h1 h2 by simp
  also have "... = rev ((aplana t1)@[x]@(aplana t2))" by simp
  also have "... = rev (aplana (Nodo x t1 t2))" by simp
  finally show ?thesis .
qed
qed

section {* Heurísticas para la inducción *}

text {
  Definición. [Definición recursiva de inversa]
  (inversa xs) la inversa de la lista xs. Por ejemplo,
    inversa [a,b,c] = [c,b,a]
}

fun inversa :: "'a list ⇒ 'a list" where
  "inversa [] = []"
| "inversa (x#xs) = (inversa xs) @ [x]"

value "inversa [a,b,c]"

text {
  Definición. [Definición de inversa con acumuladores]
  (inversaAc xs) es la inversa de la lista xs calculada con
  acumuladores. Por ejemplo,
    inversaAc [a,b,c]          = [c,b,a]
    inversaAcAux [a,b,c] [] = [c,b,a]
}

```

```

*}

fun inversaAcAux :: "'a list  $\Rightarrow$  'a list  $\Rightarrow$  'a list" where
  "inversaAcAux [] ys = ys"
| "inversaAcAux (x#xs) ys = inversaAcAux xs (x#ys)"

definition inversaAc :: "'a list  $\Rightarrow$  'a list" where
  "inversaAc xs  $\equiv$  inversaAcAux xs []"

value "inversaAcAux [a,b,c] []"
value "inversaAc [a,b,c]"

text {*
  Lema. [Ejemplo de equivalencia entre las definiciones]
  La inversa de [a,b,c] es lo mismo calculada con la primera definición
  que con la segunda.
*}

lemma "inversaAc [a,b,c] = inversa [a,b,c]"
by (simp add: inversaAc_def)

text {*
  Nota. [Ejemplo fallido de demostración por inducción]
  El siguiente intento de demostrar que para cualquier lista xs, se
  tiene que "inversaAc xs = inversa xs" falla.
*}

lemma "inversaAc xs = inversa xs"
proof (induct xs)
  show "inversaAc [] = inversa []" by (simp add: inversaAc_def)
next
  fix a xs assume HI: "inversaAc xs = inversa xs"
  have "inversaAc (a#xs) = inversaAcAux (a#xs) []" by (simp add: inversaAc_def)
  also have "... = inversaAcAux xs [a]" by simp
  also have "... = inversa (a#xs)"
  -- "Problema: la hipótesis de inducción no es aplicable."
oops

text {*
  Nota. [Heurística de generalización]
  Cuando se use demostración estructural, cuantificar universalmente las
  variables libres (o, equivalentemente, considerar las variables libres
  como variables arbitrarias).

  Lema. [Lema con generalización]

```

```

    Para toda lista ys se tiene
      inversaAcAux xs ys = (inversa xs) @ ys
*}

-- "La demostración automática es"
lemma inversaAcAux_es_inversa_1:
  "inversaAcAux xs ys = (inversa xs)@ys"
by (induct xs arbitrary: ys) auto

-- "La demostración estructurada es"
lemma inversaAcAux_es_inversa:
  "inversaAcAux xs ys = (inversa xs)@ys"
proof (induct xs arbitrary: ys)
  show "\ys. inversaAcAux [] ys = (inversa [])@ys" by simp
next
  fix a xs
  assume HI: "\ys. inversaAcAux xs ys = inversa xs@ys"
  show "\ys. inversaAcAux (a#xs) ys = inversa (a#xs)@ys"
  proof -
    fix ys
    have "inversaAcAux (a#xs) ys = inversaAcAux xs (a#ys)" by simp
    also have "... = inversa xs@(a#ys)" using HI by simp
    also have "... = inversa (a#xs)@ys" by simp
    finally show "inversaAcAux (a#xs) ys = inversa (a#xs)@ys" by simp
  qed
qed

text {*
  Corolario. Para cualquier lista xs, se tiene que
    inversaAc xs = inversa xs
*}

corollary "inversaAc xs = inversa xs"
by (simp add: inversaAcAux_es_inversa inversaAc_def)

text {*
  Nota. En el paso "inversa xs@(a#ys) = inversa (a#xs)@ys" se usan
  lemas de la teoría List. Se puede observar, activando "Trace
  Simplifier" y D"|Trace Rules", que los lemas usados son
  · append_assoc:      (xs @ ys) @ zs = xs @ (ys @ zs)
  · append.append_Cons: (x#xs)@ys = x#(xs@ys)
  · append.append_Nil:  []@ys = ys
  Los dos últimos son las ecuaciones de la definición de append.

  En la siguiente demostración se detallan los lemas utilizados.

```

```

*}

lemma "(inversa xs)@(a#ys) = (inversa (a#xs))@ys"
proof -
  have "(inversa xs)@(a#ys) = (inversa xs)@(a#([]@ys))"
  by (simp only:append.append_Nil)
  also have "... = (inversa xs)@([a]@ys)" by (simp only:append.append_Cons)
  also have "... = ((inversa xs)[a])@ys" by (simp only:append_assoc)
  also have "... = (inversa (a#xs))@ys" by (simp only:inversa.simps(2))
  finally show ?thesis .
qed

section {* Recursión general. La función de Ackermann *}

text {*
  El objetivo de esta sección es mostrar el uso de las definiciones
  recursivas generales y sus esquemas de inducción. Como ejemplo se usa la
  función de Ackermann (se puede consultar información sobre dicha función en
  http://en.wikipedia.org/wiki/Ackermann\_function).

  Definición. La función de Ackermann se define por
  
$$A(m,n) = \begin{cases} n+1, & \text{si } m=0, \\ A(m-1,1), & \text{si } m>0 \text{ y } n=0, \\ A(m-1,A(m,n-1)), & \text{si } m>0 \text{ y } n>0 \end{cases}$$

  para todo los números naturales.

  La función de Ackermann es recursiva, pero no es primitiva recursiva.
  *}

fun ack :: "nat  $\Rightarrow$  nat  $\Rightarrow$  nat" where
  "ack 0 n = n+1"
| "ack (Suc m) 0 = ack m 1"
| "ack (Suc m) (Suc n) = ack m (ack (Suc m) n)"

text {*
  Nota. [Ejemplo de cálculo]
  El cálculo del valor de la función de Ackermann para 2 y 3 se realiza
  mediante "value"
  *}

value "ack 2 3" (* devuelve 9 *)

text {*
  Nota. [Definiciones recursivas generales]
  · Las definiciones recursivas generales se identifican mediante "fun".

```

- Al definir una función recursiva general se genera una regla de inducción. En la definición anterior, la regla generada es

ack.induct:

$$\begin{aligned} & \llbracket \bigwedge n. P \ 0 \ n; \\ & \quad \bigwedge m. P \ m \ 1 \implies P \ (Suc \ m) \ 0; \\ & \quad \bigwedge m \ n. \llbracket P \ (Suc \ m) \ n; P \ m \ (ack \ (Suc \ m) \ n) \rrbracket \implies P \ (Suc \ m) \ (Suc \ n) \rrbracket \\ & \implies P \ a \ b \end{aligned}$$

Lema. Para todos m y n, A(m,n) > n.

**/*

text *{**

El lema anterior se puede demostrar automáticamente, como sigue.

**/*

-- "La demostración automática es"

lemma "ack m n > n"

by (induct m n rule: ack.induct) simp_all

-- "La demostración detallada es"

lemma "ack m n > n"

proof (induct m n rule: ack.induct)

fix n :: "nat"

show "ack 0 n > n" **by** simp

next

fix m **assume** "ack m 1 > 1"

thus "ack (Suc m) 0 > 0" **by** simp

next

fix m n

assume "n < ack (Suc m) n" **and**

"ack (Suc m) n < ack m (ack (Suc m) n)"

thus "Suc n < ack (Suc m) (Suc n)" **by** simp

qed

text *{**

Nota. [Inducción sobre recursión]

El formato para iniciar una demostración por inducción en la regla inductiva correspondiente a la definición recursiva de la función f m n es

proof (induct m n rule:f.induct)

**/*

section *{* Recursión mutua e inducción **

text *{**

Nota. [Ejemplo de definición de tipos mediante recursión cruzada]

- Un árbol de tipo a es una hoja o un nodo de tipo a junto con un bosque de tipo a .
- Un bosque de tipo a es el boque vacío o un bosque contruido añadiendo un árbol de tipo a a un bosque de tipo a .

**}*

```
datatype 'a arbol = Hoja | Nodo "'a" "'a bosque"
and          'a bosque = Vacio | ConsB "'a arbol" "'a bosque"
```

text *{**

Nota. [Regla de inducción correspondiente a la recursión cruzada]

La regla de inducción sobre árboles y bosques es arbol_bosque.induct:

```
[[P1 Hoja;
   $\bigwedge x b. P2 b \implies P1 \text{ (Nodo } x \text{ } b)$ ;
  P2 Vacio;
   $\bigwedge a b. [[P1 a; P2 b]] \implies P2 \text{ (ConsB } a \text{ } b)]$ 
 $\implies P1 a \wedge P2 b$ 
```

Nota. [Ejemplos de definición por recursión cruzada]

- $\text{aplanar_arbol } a$) es la lista obtenida aplanando el árbol a .
- $\text{aplanar_bosque } b$) es la lista obtenida aplanando el bosque b .
- $\text{map_arbol } a \text{ } h$) es el árbol obtenido aplicando la función h a todos los nodos del árbol a .
- $\text{map_bosque } b \text{ } h$) es el bosque obtenido aplicando la función h a todos los nodos del bosque b .

**}*

fun

```
aplanar_arbol :: "'a arbol  $\Rightarrow$  'a list" and
aplanar_bosque :: "'a bosque  $\Rightarrow$  'a list" where
  "aplanar_arbol Hoja = []"
| "aplanar_arbol (Nodo x b) = x#(aplanar_bosque b)"
| "aplanar_bosque Vacio = []"
| "aplanar_bosque (ConsB a b) = (aplanar_arbol a) @ (aplanar_bosque b)"
```

fun

```
map_arbol :: "'a arbol  $\Rightarrow$  ('a  $\Rightarrow$  'b)  $\Rightarrow$  'b arbol" and
map_bosque :: "'a bosque  $\Rightarrow$  ('a  $\Rightarrow$  'b)  $\Rightarrow$  'b bosque" where
  "map_arbol Hoja h = Hoja"
| "map_arbol (Nodo x b) h = Nodo (h x) (map_bosque b h)"
| "map_bosque Vacio h = Vacio"
| "map_bosque (ConsB a b) h = ConsB (map_arbol a h) (map_bosque b h)"
```

text *{**

```

Lema. [Ejemplo de inducción cruzada]
· aplana_arbol (map_arbol a h) = map h (aplana_arbol a)
· aplana_bosque (map_bosque b h) = map h (aplana_bosque b)
*}

-- "La demostración automática es"
lemma "aplana_arbol (map_arbol a h) = map h (aplana_arbol a)
      ∧ aplana_bosque (map_bosque b h) = map h (aplana_bosque b)"
by (induct_tac a and b) auto

-- "La demostración detallada es"
lemma "aplana_arbol (map_arbol a h) = map h (aplana_arbol a)
      ∧ aplana_bosque (map_bosque b h) = map h (aplana_bosque b)"
proof (induct_tac a and b)
  show "aplana_arbol (map_arbol Hoja h) = map h (aplana_arbol Hoja)" by simp
next
  fix x b
  assume HI: "aplana_bosque (map_bosque b h) = map h (aplana_bosque b)"
  have "aplana_arbol (map_arbol (Nodo x b) h)
        = aplana_arbol (Nodo (h x) (map_bosque b h))" by simp
  also have "... = (h x) # (aplana_bosque (map_bosque b h))" by simp
  also have "... = (h x) # (map h (aplana_bosque b))" using HI by simp
  also have "... = map h (aplana_arbol (Nodo x b))" by simp
  finally show "aplana_arbol (map_arbol (Nodo x b) h)
                = map h (aplana_arbol (Nodo x b))" .
next
  show "aplana_bosque (map_bosque Vacio h) = map h (aplana_bosque Vacio)"
    by simp
next
  fix a b
  assume HI1: "aplana_arbol (map_arbol a h) = map h (aplana_arbol a)"
    and HI2: "aplana_bosque (map_bosque b h) = map h (aplana_bosque b)"
  have "aplana_bosque (map_bosque (ConsB a b) h)
        = aplana_bosque (ConsB (map_arbol a h) (map_bosque b h))" by simp
  also have "... = aplana_arbol (map_arbol a h) @ aplana_bosque (map_bosque b h)"
    by simp
  also have "... = (map h (aplana_arbol a)) @ (map h (aplana_bosque b))"
    using HI1 HI2 by simp
  also have "... = map h (aplana_bosque (ConsB a b))" by simp
  finally show "aplana_bosque (map_bosque (ConsB a b) h)
                = map h (aplana_bosque (ConsB a b))" by simp
qed

end

```

6.1. Ejercicios de inducción sobre listas

6.1.1. Cons inverso y cuantificadores sobre listas

```
chapter {* T6R1a: Cons inverso y cuantificadores sobre listas *
```

```
theory T6R1a
```

```
imports Main
```

```
begin
```

```
section {* Cons inverso *
```

```
text {*
```

```
-----
Ejercicio 1. Definir recursivamente la función
  snoc :: "'a list ⇒ 'a ⇒ 'a list"
tal que (snoc xs a) es la lista obtenida al añadir el elemento a al
final de la lista xs. Por ejemplo,
  value "snoc [2,5] (3::int)" == [2,5,3]
```

```
Nota: No usar @.
```

```
-----
*}
```

```
fun snoc :: "'a list ⇒ 'a ⇒ 'a list" where
```

```
  "snoc [] a = [a]"
```

```
| "snoc (x#xs) a = x # (snoc xs a)"
```

```
text {*
```

```
-----
Ejercicio 2. Demostrar el siguiente teorema
  snoc xs a = xs @ [a]
```

```
-----
*}
```

```
-- "La demostración automática del lema es"
```

```
lemma snoc_append: "snoc xs a = xs @ [a]"
```

```
by (induct xs) auto
```

```
-- "La demostración semiautomática del lema es"
```

```
lemma snoc_append_2: "snoc xs a = xs @ [a]"
```

```
proof (induct xs)
```

```
  show "snoc [] a = [] @ [a]" by auto
```

```
next
```

```
  fix b xs assume "snoc xs a = xs @ [a]"
```

```

    thus "snoc (b # xs) a = (b # xs) @ [a]" by auto
qed

-- "La demostración estructurada del lema es"
lemma snoc_append_3: "snoc xs a = xs @ [a]"
proof (induct "xs")
  show "snoc [] a = [] @ [a]"
  proof -
    have "snoc [] a = [a]" by simp
    also have "... = [] @ [a]" by simp
    finally show "snoc [] a = [] @ [a]" .
  qed
next
fix b xs assume HI: "snoc xs a = xs @ [a]"
show "snoc (b # xs) a = (b # xs) @ [a]"
proof -
  have "snoc (b # xs) a = b # (snoc xs a)" by simp
  also have "... = b # (xs @ [a])" using HI by simp
  also have "... = (b # xs) @ [a]" by simp
  finally show "snoc (b # xs) a = (b # xs) @ [a]" .
qed
qed

text {*
  -----
  Ejercicio 3. Demostrar el siguiente teorema
  rev (x # xs) = snoc (rev xs) x"
  -----
  *}

-- "La demostración automática del teorema es"
theorem rev_cons: "rev (x # xs) = snoc (rev xs) x"
by (auto simp add: snoc_append)

-- "La demostración estructurada del teorema es"
theorem rev_cons_2: "rev (x # xs) = snoc (rev xs) x"
proof -
  have "rev (x # xs) = (rev xs) @ [x]" by simp
  also have "... = snoc (rev xs) x" by (simp add: snoc_append)
  finally show "rev (x # xs) = snoc (rev xs) x" .
qed

section {* Cuantificadores sobre listas *}

text {*
```

Ejercicio 4. Definir la función

todos :: ('a ⇒ bool) ⇒ 'a list ⇒ bool
tal que (todos p xs) se verifica si todos los elementos de la lista
xs cumplen la propiedad p. Por ejemplo, se verifica
todos (λx. 1 < length x) [[2,1,4],[1,3]]
¬todos (λx. 1 < length x) [[2,1,4],[3]]

Nota: La función todos es equivalente a la predefinida list_all.

*}

```
fun todos :: "('a ⇒ bool) ⇒ 'a list ⇒ bool" where
  "todos p [] = True"
| "todos p (y#ys) = ((p y) ∧ (todos p ys))"
```

```
value "todos (λx. 1 < length x) [[2,1,4],[1,3]]" -- "= True"
```

```
value "todos (λx. 1 < length x) [[2,1,4],[3]]" -- "= False"
```

```
text {*
```

Ejercicio 5. Definir la función

algunos :: ('a ⇒ bool) ⇒ 'a list ⇒ bool
tal que (algunos p xs) se verifica si algunos elementos de la lista
xs cumplen la propiedad p. Por ejemplo, se verifica
algunos (λx. 1 < length x) [[2,1,4],[3]]
¬algunos (λx. 1 < length x) [],[3]]"

Nota: La función algunos es equivalente a la predefinida list_ex.

*}

```
fun algunos :: "('a ⇒ bool) ⇒ 'a list ⇒ bool" where
  "algunos p [] = False"
| "algunos p (x#xs) = ((p x) ∨ (algunos p xs))"
```

```
value "algunos (λx. 1 < length x) [[2,1,4],[3]]" -- "= True"
```

```
value "algunos (λx. 1 < length x) [],[3]]" -- "= False"
```

```
text {*
```

Ejercicio 6. Demostrar o refutar:

todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)

*}

```

-- "La demostración automática es"
lemma "todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma "todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)"
proof (induct xs)
  show "todos (λx. P x ∧ Q x) [] = (todos P [] ∧ todos Q [])" by simp
next
  fix a xs
  assume "todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)"
  thus "todos (λx. P x ∧ Q x) (a#xs) = (todos P (a#xs) ∧ todos Q (a#xs))"
    by auto
qed

-- "La demostración detallada es"
lemma "todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)"
proof (induct xs)
  show "todos (λx. P x ∧ Q x) [] = (todos P [] ∧ todos Q [])" by simp
next
  fix a xs
  assume HI: "todos (λx. P x ∧ Q x) xs = (todos P xs ∧ todos Q xs)"
  show "todos (λx. P x ∧ Q x) (a#xs) = (todos P (a#xs) ∧ todos Q (a#xs))"
  proof -
    have "todos (λx. P x ∧ Q x) (a#xs) =
      ((P a) ∧ (Q a) ∧ todos (λx. P x ∧ Q x) xs)" by simp
    also have "... = ((P a) ∧ (Q a) ∧ todos P xs ∧ todos Q xs)" using HI by simp
    also have "... = (((P a) ∧ todos P xs) ∧ ((Q a) ∧ todos Q xs))" by auto
    also have "... = (todos P (a#xs) ∧ todos Q (a#xs))" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 7. Demostrar o refutar:
    todos P (x @ y) = (todos P x ∧ todos P y)
  -----
  *}

-- "La demostración automática es"
lemma todos_append [simp]:
  "todos P (x @ y) = (todos P x ∧ todos P y)"
by (induct x) simp_all

```

```
-- "La demostración estructurada es"
lemma todos_append_2:
  "todos P (x @ y) = (todos P x ∧ todos P y)"
proof (induct x)
  show "todos P ([] @ y) = (todos P [] ∧ todos P y)" by simp
next
  fix a x
  assume "todos P (x @ y) = (todos P x ∧ todos P y)"
  thus "todos P ((a#x) @ y) = (todos P (a#x) ∧ todos P y)"
    by auto
qed

-- "La demostración detallada es"
lemma todos_append_3:
  "todos P (x @ y) = (todos P x ∧ todos P y)"
proof (induct x)
  show "todos P ([] @ y) = (todos P [] ∧ todos P y)" by simp
next
  fix a x
  assume HI: "todos P (x @ y) = (todos P x ∧ todos P y)"
  show "todos P ((a#x) @ y) = (todos P (a#x) ∧ todos P y)"
  proof -
    have "todos P ((a#x) @ y) = ((P a) ∧ todos P (x@y))" by simp
    also have "... = ((P a) ∧ todos P x ∧ todos P y)" using HI by simp
    also have "... = (todos P (a#x) ∧ todos P y)" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 8. Demostrar o refutar:
    todos P (rev xs) = todos P xs
  -----
  *}

-- "La demostración automática es"
lemma "todos P (rev xs) = todos P xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma "todos P (rev xs) = todos P xs"
proof (induct xs)
  show "todos P (rev []) = todos P []" by simp
```

```

next
  fix a xs
  assume "todos P (rev xs) = todos P xs"
  thus "todos P (rev (a#xs)) = todos P (a#xs)" by auto
qed

-- "La demostración detallada es"
lemma "todos P (rev xs) = todos P xs"
proof (induct xs)
  show "todos P (rev []) = todos P []" by simp
next
  fix a xs
  assume HI: "todos P (rev xs) = todos P xs"
  show "todos P (rev (a#xs)) = todos P (a#xs)"
  proof -
    have "todos P (rev (a#xs)) = todos P ((rev xs)@[a])" by simp
    also have "... = (todos P (rev xs) ∧ todos P [a])" by simp
    also have "... = (todos P xs ∧ todos P [a])" using HI by simp
    also have "... = (todos P [a] ∧ todos P xs)" by auto
    also have "... = (P a ∧ todos P xs)" by simp
    also have "... = todos P (a#xs)" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 9. Demostrar o refutar:
  algunos (λx. P x ∧ Q x) xs = (algunos P xs ∧ algunos Q xs)
  -----
*}

-- "Se busca un contraejemplo con nitpick"
lemma "algunos (λx. P x ∧ Q x) xs = (algunos P xs ∧ algunos Q xs)"
nitpick
oops

text {*
  El contraejemplo encontrado es
  Nitpick found a counterexample for card 'a = 3:

  Free variables:
  P = (λx. _)(a<^bsub>1<^esub> := True, a<^bsub>2<^esub> := True, a<^bsub>3<^esub> := False)
  Q = (λx. _)(a<^bsub>1<^esub> := False, a<^bsub>2<^esub> := False, a<^bsub>3<^esub> := True)
  xs = [a<^bsub>3<^esub>, a<^bsub>2<^esub>]

```

```

*}

text {*
-----
  Ejercicio 10. Demostrar o refutar:
    algunos P (map f xs) = algunos (P ◦ f) xs
-----
*}

-- "La demostración automática es"
lemma "algunos P (map f xs) = algunos (P ◦ f) xs"
by (induct xs) simp_all

-- "La demostración estructurada es"
lemma "algunos P (map f xs) = algunos (P ◦ f) xs"
proof (induct xs)
  show "algunos P (map f []) = algunos (P ◦ f) []" by simp
next
  fix a xs
  assume "algunos P (map f xs) = algunos (P ◦ f) xs"
  thus "algunos P (map f (a#xs)) = algunos (P ◦ f) (a#xs)" by auto
qed

-- "La demostración detallada es"
lemma "algunos P (map f xs) = algunos (P ◦ f) xs"
proof (induct xs)
  show "algunos P (map f []) = algunos (P ◦ f) []" by simp
next
  fix a xs
  assume HI: "algunos P (map f xs) = algunos (P ◦ f) xs"
  show "algunos P (map f (a#xs)) = algunos (P ◦ f) (a#xs)"
  proof -
    have "algunos P (map f (a#xs)) = algunos P ((f a)#(map f xs))" by simp
    also have "... = ((P (f a)) v (algunos P (map f xs)))" by simp
    also have "... = (((P ◦ f) a) v (algunos (P ◦ f) xs))" using HI by simp
    also have "... = algunos (P ◦ f) (a#xs)" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
  Ejercicio 11. Demostrar o refutar:
    algunos P (xs @ ys) = (algunos P xs v algunos P ys)
-----

```

```

*}

-- "La demostración automática es"
lemma algunos_append:
  "algunos P (xs @ ys) = (algunos P xs v algunos P ys)"
by (induct xs) simp_all

-- "La demostración estructurada es"
lemma algunos_append_2:
  "algunos P (xs @ ys) = (algunos P xs v algunos P ys)"
proof (induct xs)
  show "algunos P ([] @ ys) = (algunos P [] v algunos P ys)" by simp
next
  fix a xs
  assume "algunos P (xs @ ys) = (algunos P xs v algunos P ys)"
  thus "algunos P ((a#xs) @ ys) = (algunos P (a#xs) v algunos P ys)"
    by auto
qed

-- "La demostración detallada es"
lemma algunos_append_3:
  "algunos P (xs @ ys) = (algunos P xs v algunos P ys)"
proof (induct xs)
  show "algunos P ([] @ ys) = (algunos P [] v algunos P ys)" by simp
next
  fix a xs
  assume HI: "algunos P (xs @ ys) = (algunos P xs v algunos P ys)"
  show "algunos P ((a#xs) @ ys) = (algunos P (a#xs) v algunos P ys)"
  proof -
    have "algunos P ((a#xs) @ ys) = algunos P (a#(xs @ ys))" by simp
    also have "... = ((P a) v algunos P (xs @ ys))" by simp
    also have "... = ((P a) v algunos P xs v algunos P ys)" using HI by simp
    also have "... = (algunos P (a#xs) v algunos P ys)" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 12. Demostrar o refutar:
    algunos P (rev xs) = algunos P xs
  -----
*}

-- "La demostración automática es"

```

```

lemma "algunos P (rev xs) = algunos P xs"
by (induct xs) (auto simp add: algunos_append)

-- "La demostración estructurada es"
lemma "algunos P (rev xs) = algunos P xs"
proof (induct xs)
  show "algunos P (rev []) = algunos P []" by simp
next
  fix a xs
  assume "algunos P (rev xs) = algunos P xs"
  thus "algunos P (rev (a#xs)) = algunos P (a#xs)"
    by (auto simp add: algunos_append)
qed

-- "La demostración detallada es"
lemma "algunos P (rev xs) = algunos P xs"
proof (induct xs)
  show "algunos P (rev []) = algunos P []" by simp
next
  fix a xs
  assume HI: "algunos P (rev xs) = algunos P xs"
  show "algunos P (rev (a#xs)) = algunos P (a#xs)"
  proof -
    have "algunos P (rev (a#xs)) = algunos P ((rev xs) @ [a])" by simp
    also have "... = (algunos P (rev xs) v algunos P [a])"
      by (simp add: algunos_append)
    also have "... = (algunos P xs v algunos P [a])" using HI by simp
    also have "... = (algunos P xs v P a)" by simp
    also have "... = (P a v algunos P xs)" by auto
    also have "... = algunos P (a#xs)" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
Ejercicio 13. Encontrar un término no trivial Z tal que sea cierta la
siguiente ecuación:
    algunos (λx. P x v Q x) xs = Z
-----
*}

text {*
Solución: La ecuación se verifica eligiendo como Z el término
    algunos P xs v algunos Q xs

```

```

    En efecto,
  *}

lemma "algunos (λx. P x v Q x) xs = (algunos P xs v algunos Q xs)"
by (induct xs) auto

-- "De forma estructurada"
lemma "algunos (λx. P x v Q x) xs = (algunos P xs v algunos Q xs)"
proof (induct xs)
  show "algunos (λx. P x v Q x) [] = (algunos P [] v algunos Q [])" by simp
next
  fix a xs
  assume "algunos (λx. (P x v Q x)) xs = (algunos P xs v algunos Q xs)"
  thus "algunos (λx. P x v Q x) (a#xs) = (algunos P (a#xs) v algunos Q (a#xs))"
    by auto
qed

-- "De forma detallada"
lemma "algunos (λx. P x v Q x) xs = (algunos P xs v algunos Q xs)"
proof (induct xs)
  show "algunos (λx. P x v Q x) [] = (algunos P [] v algunos Q [])" by simp
next
  fix a xs
  assume HI: "algunos (λx. (P x v Q x)) xs = (algunos P xs v algunos Q xs)"
  show "algunos (λx. P x v Q x) (a#xs) = (algunos P (a#xs) v algunos Q (a#xs))"
  proof -
    have "algunos (λx. P x v Q x) (a#xs) =
      ((P a) v (Q a) v algunos (λx. P x v Q x) xs)" by simp
    also have "... = ((P a) v (Q a) v algunos P xs v algunos Q xs)"
      using HI by simp
    also have "... = (((P a) v algunos P xs) v ((Q a) v algunos Q xs))" by auto
    also have "... = (algunos P (a#xs) v algunos Q (a#xs))" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 14. Demostrar o refutar:
    algunos P xs = (¬ todos (λx. (¬ P x)) xs)
  -----
  *}

-- "La demostración automática es"
lemma "algunos P xs = (¬ todos (λx. (¬ P x)) xs)"

```

```

by (induct xs) simp_all

-- "La demostración estructurada es"
lemma "algunos P xs = (¬ todos (λx. (¬ P x)) xs)"
proof (induct xs)
  show "algunos P [] = (¬ todos (λx. (¬ P x)) [])" by simp
next
  fix a xs
  assume "algunos P xs = (¬ todos (λx. (¬ P x)) xs)"
  thus "algunos P (a#xs) = (¬ todos (λx. (¬ P x)) (a#xs))"
    by auto
qed

-- "La demostración detallada es"
lemma "algunos P xs = (¬ todos (λx. (¬ P x)) xs)"
proof (induct xs)
  show "algunos P [] = (¬ todos (λx. (¬ P x)) [])" by simp
next
  fix a xs
  assume HI: "algunos P xs = (¬ todos (λx. (¬ P x)) xs)"
  show "algunos P (a#xs) = (¬ todos (λx. (¬ P x)) (a#xs))"
  proof -
    have "algunos P (a#xs) = ((P a) ∨ algunos P xs)" by simp
    also have "... = ((P a) ∨ ¬ todos (λx. (¬ P x)) xs)" using HI by simp
    also have "... = (¬ (¬ (P a) ∧ todos (λx. (¬ P x)) xs))" by simp
    also have "... = (¬ todos (λx. (¬ P x)) (a#xs))" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
Ejercicio 15. Definir la función primitiva recursiva
estaEn :: 'a ⇒ 'a list ⇒ bool
tal que (estaEn x xs) se verifica si el elemento x está en la lista
xs. Por ejemplo,
estaEn (2::nat) [3,2,4] = True
estaEn (1::nat) [3,2,4] = False
-----
*}

fun estaEn :: "'a ⇒ 'a list ⇒ bool" where
  "estaEn x []      = False"
| "estaEn x (a#xs) = (x=a ∨ estaEn x xs)"

```

```

value "estaEn (2::nat) [3,2,4]" -- "= True"
value "estaEn (1::nat) [3,2,4]" -- "= False"

text {*
-----
Ejercicio 16. Expresar la relación existente entre estaEn y algunos.
-----
*}

text {*
  Solución: La relación es
    estaEn y xs = algunos (λx. x=y) xs
  En efecto,
*}

lemma estaEn_algunos:
  "estaEn y xs = algunos (λx. x=y) xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma estaEn_algunos_2:
  "estaEn y xs = algunos (λx. x=y) xs"
proof (induct xs)
  show "estaEn y [] = algunos (λx. x=y) []" by simp
next
  fix a xs
  assume "estaEn y xs = algunos (λx. x=y) xs"
  thus "estaEn y (a#xs) = algunos (λx. x=y) (a#xs)" by auto
qed

-- "La demostración detallada es"
lemma estaEn_algunos_3:
  "estaEn y xs = algunos (λx. x=y) xs"
proof (induct xs)
  show "estaEn y [] = algunos (λx. x=y) []" by simp
next
  fix a xs
  assume HI: "estaEn y xs = algunos (λx. x=y) xs"
  show "estaEn y (a#xs) = algunos (λx. x=y) (a#xs)"
  proof -
    have "estaEn y (a#xs) = (y=a ∨ estaEn y xs)" by simp
    also have "... = (y=a ∨ algunos (λx. x=y) xs)" using HI by simp
    also have "... = (a=y ∨ algunos (λx. x=y) xs)" by auto
    also have "... = algunos (λx. x=y) (a#xs)" by simp
    finally show ?thesis .
  end
end

```

```

    qed
  qed

text {*
  -----
  Ejercicio 17. Definir la función primitiva recursiva
    sinDuplicados :: 'a list ⇒ bool
  tal que (sinDuplicados xs) se verifica si la lista xs no contiene
  duplicados. Por ejemplo,
    sinDuplicados [1::nat,4,2]   = True
    sinDuplicados [1::nat,4,2,4] = False
  -----
*}

fun sinDuplicados :: "'a list ⇒ bool" where
  "sinDuplicados [] = True"
| "sinDuplicados (a#xs) = ((¬ estaEn a xs) ∧ sinDuplicados xs)"

value "sinDuplicados [1::nat,4,2]" -- "= True"
value "sinDuplicados [1::nat,4,2,4]" -- "= False"

text {*
  -----
  Ejercicio 18. Definir la función primitiva recursiva
    borraDuplicados :: 'a list ⇒ bool
  tal que (borraDuplicados xs) es la lista obtenida eliminando los
  elementos duplicados de la lista xs. Por ejemplo,
    borraDuplicados [1::nat,2,4,2,3] = [1,4,2,3]

  Nota: La función borraDuplicados es equivalente a la predefinida remdups.
  -----
*}

fun borraDuplicados :: "'a list ⇒ 'a list" where
  "borraDuplicados []      = []"
| "borraDuplicados (a#xs) = (if estaEn a xs
                             then borraDuplicados xs
                             else (a#borraDuplicados xs))"

value "borraDuplicados [1::nat,2,4,2,3]" -- "= [1,4,2,3]"

text {*
  -----
  Ejercicio 19. Demostrar o refutar:
    length (borraDuplicados xs) ≤ length xs
  -----

```

```

*}

-- "La demostración automática es"
lemma length_borraDuplicados:
  "length (borraDuplicados xs) ≤ length xs"
by (induct xs) simp_all

-- "La demostración estructurada es"
lemma length_borraDuplicados_2:
  "length (borraDuplicados xs) ≤ length xs"
proof (induct xs)
  show "length (borraDuplicados []) ≤ length []" by simp
next
  fix a xs
  assume HI: "length (borraDuplicados xs) ≤ length xs"
  thus "length (borraDuplicados (a#xs)) ≤ length (a#xs)"
  proof (cases)
    assume "estaEn a xs"
    thus "length (borraDuplicados (a#xs)) ≤ length (a#xs)"
      using HI by auto
  next
    assume "(¬ estaEn a xs)"
    thus "length (borraDuplicados (a#xs)) ≤ length (a#xs)"
      using HI by auto
  qed
qed

-- "La demostración detallada es"
lemma length_borraDuplicados_3:
  "length (borraDuplicados xs) ≤ length xs"
proof (induct xs)
  show "length (borraDuplicados []) ≤ length []" by simp
next
  fix a xs
  assume HI: "length (borraDuplicados xs) ≤ length xs"
  show "length (borraDuplicados (a#xs)) ≤ length (a#xs)"
  proof (cases)
    assume "estaEn a xs"
    hence "length (borraDuplicados (a#xs)) = length (borraDuplicados xs)"
      by simp
    also have "... ≤ length xs" using HI by simp
    also have "... ≤ length (a#xs)" by simp
    finally show ?thesis .
  next

```

```

    assume "(¬ estaEn a xs)"
    hence "length (borraDuplicados (a#xs)) = length (a#(borraDuplicados xs))"
      by simp
    also have "... = 1 + length (borraDuplicados xs)" by simp
    also have "... ≤ 1 + length xs" using HI by simp
    also have "... = length (a#xs)" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
  Ejercicio 20. Demostrar o refutar:
    estaEn a (borraDuplicados xs) = estaEn a xs
-----
*}

-- "La demostración automática es"
lemma estaEn_borraDuplicados:
  "estaEn a (borraDuplicados xs) = estaEn a xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma estaEn_borraDuplicados_2:
  "estaEn a (borraDuplicados xs) = estaEn a xs"
proof (induct xs)
  show "estaEn a (borraDuplicados []) = estaEn a []" by simp
next
  fix b xs
  assume HI: "estaEn a (borraDuplicados xs) = estaEn a xs"
  show "estaEn a (borraDuplicados (b#xs)) = estaEn a (b#xs)"
  proof (rule iffI)
    assume c1: "estaEn a (borraDuplicados (b#xs))"
    show "estaEn a (b#xs)"
    proof (cases)
      assume "estaEn b xs"
      thus "estaEn a (b#xs)" using c1 HI by auto
    next
      assume "¬ estaEn b xs"
      thus "estaEn a (b#xs)" using c1 HI by auto
    qed
  next
    assume c2: "estaEn a (b#xs)"
    show "estaEn a (borraDuplicados (b#xs))"
    proof (cases)

```

```

    assume "a=b"
    thus "estaEn a (borraDuplicados (b#xs))" using HI by auto
  next
    assume "a≠b"
    thus "estaEn a (borraDuplicados (b#xs))" using 'a≠b' c2 HI by auto
  qed
qed
qed

-- "La demostración detallada es"
lemma estaEn_borraDuplicados_3:
  "estaEn a (borraDuplicados xs) = estaEn a xs"
proof (induct xs)
  show "estaEn a (borraDuplicados []) = estaEn a []" by simp
next
  fix b xs
  assume HI: "estaEn a (borraDuplicados xs) = estaEn a xs"
  show "estaEn a (borraDuplicados (b#xs)) = estaEn a (b#xs)"
  proof (rule iffI)
    assume c1: "estaEn a (borraDuplicados (b#xs))"
    show "estaEn a (b#xs)"
    proof (cases)
      assume "estaEn b xs"
      hence "estaEn a (borraDuplicados xs)" using c1 by simp
      hence "estaEn a xs" using HI by simp
      thus "estaEn a (b#xs)" by simp
    next
      assume "¬ estaEn b xs"
      hence "estaEn a (b#(borraDuplicados xs))" using c1 by simp
      hence "a=b v (estaEn a (borraDuplicados xs))" by simp
      hence "a=b v (estaEn a xs)" using HI by simp
      thus "estaEn a (b#xs)" by simp
    qed
  next
    assume c2: "estaEn a (b#xs)"
    show "estaEn a (borraDuplicados (b#xs))"
    proof (cases)
      assume "a=b"
      thus "estaEn a (borraDuplicados (b#xs))" using HI by auto
    next
      assume "a≠b"
      hence "estaEn a xs" using c2 by simp
      hence "estaEn a (borraDuplicados xs)" using HI by simp
      thus "estaEn a (borraDuplicados (b#xs))" using 'a≠b' by simp
    qed
  qed

```

```

    qed
  qed

text {*
  -----
  Ejercicio 21. Demostrar o refutar:
    sinDuplicados (borraDuplicados xs)
  -----
*}

-- "La demostración automática"
lemma sinDuplicados_borraDuplicados:
  "sinDuplicados (borraDuplicados xs)"
by (induct xs) (auto simp add: estaEn_borraDuplicados)

-- "La demostración estructurada es"
lemma sinDuplicados_borraDuplicados_2:
  "sinDuplicados (borraDuplicados xs)"
proof (induct xs)
  show "sinDuplicados (borraDuplicados [])" by simp
next
  fix a xs
  assume HI: "sinDuplicados (borraDuplicados xs)"
  show "sinDuplicados (borraDuplicados (a#xs))"
  proof (cases)
    assume "estaEn a xs"
    thus "sinDuplicados (borraDuplicados (a#xs))" using HI by simp
  next
    assume "¬ estaEn a xs"
    thus "sinDuplicados (borraDuplicados (a#xs))"
      using '¬ estaEn a xs' HI
      by (auto simp add: estaEn_borraDuplicados)
  qed
qed

-- "La demostración detallada es"
lemma sinDuplicados_borraDuplicados_3:
  "sinDuplicados (borraDuplicados xs)"
proof (induct xs)
  show "sinDuplicados (borraDuplicados [])" by simp
next
  fix a xs
  assume HI: "sinDuplicados (borraDuplicados xs)"
  show "sinDuplicados (borraDuplicados (a#xs))"
  proof (cases)

```

```

    assume "estaEn a xs"
    thus "sinDuplicados (borraDuplicados (a#xs))" using HI by simp
next
  assume "¬ estaEn a xs"
  hence "¬ (estaEn a xs) ∧ sinDuplicados (borraDuplicados xs)"
    using HI by simp
  hence "¬ estaEn a (borraDuplicados xs) ∧
    sinDuplicados (borraDuplicados xs)"
    by (simp add: estaEn_borraDuplicados)
  hence "sinDuplicados (a#borraDuplicados xs)" by simp
  thus "sinDuplicados (borraDuplicados (a#xs))"
    using '¬ estaEn a xs' by simp
qed
qed

text {*
  -----
  Ejercicio 22. Demostrar o refutar:
  borraDuplicados (rev xs) = rev (borraDuplicados xs)
  -----
  *}

-- "Se busca un contraejemplo con"
lemma "borraDuplicados (rev xs) = rev (borraDuplicados xs)"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  xs = [3, 2, 3]
  En efecto,
  borraDuplicados (rev xs) = borraDuplicados (rev [3,2,3]) = [2,3]
  rev (borraDuplicados xs) = rev (borraDuplicados [3,2,3]) = [3,2]
  *}

end

```

6.1.2. Sustitución, inversión y eliminación

```
chapter {* T6R1b: Sustitución, inversión y eliminación *
```

```

theory T6R1b
imports Main
begin

```

```
section {* Sustitución, inversión y eliminación *
```

```
text {*
```

```
-----
Ejercicio 1. Definir la función
```

```
sust :: "'a ⇒ 'a ⇒ 'a list ⇒ 'a list"
```

```
tal que (sust x y zs) es la lista obtenida sustituyendo cada
ocurrencia de x por y en la lista zs. Por ejemplo,
```

```
sust (1::nat) 2 [1,2,3,4,1,2,3,4] = [2,2,3,4,2,2,3,4]
```

```
-----
*}
```

```
fun sust :: "'a ⇒ 'a ⇒ 'a list ⇒ 'a list" where
```

```
"sust x y [] = []"
```

```
| "sust x y (z#zs) = (if z=x then y else z)#(sust x y zs)"
```

```
value "sust (1::nat) 2 [1,2,3,4,1,2,3,4]" -- "= [2,2,3,4,2,2,3,4]"
```

```
text {*
```

```
-----
Ejercicio 2. Demostrar o refutar:
```

```
sust x y (xs@ys) = (sust x y xs)@(sust x y ys)"
```

```
-----
*}
```

```
-- "La demostración automática es"
```

```
lemma sust_append:
```

```
"sust x y (xs@ys) = (sust x y xs)@(sust x y ys)"
```

```
by (induct xs) auto
```

```
-- "La demostración estructurada es"
```

```
lemma sust_append_2:
```

```
"sust x y (xs @ ys) = (sust x y xs)@(sust x y ys)"
```

```
proof (induct xs)
```

```
show "sust x y ([]@ys) = (sust x y [])@(sust x y ys)" by simp
```

```
next
```

```
fix a xs
```

```
assume HI: "sust x y (xs@ys) = (sust x y xs)@(sust x y ys)"
```

```
show "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)"
```

```
proof (cases)
```

```
assume "x=a"
```

```
thus "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)"
```

```
using HI by auto
```

```
next
```

```
assume "x≠a"
```

```

    thus "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)"
      using HI by auto
  qed
qed

-- "La demostración detallada es"
lemma sust_append_3:
  "sust x y (xs @ ys) = (sust x y xs)@(sust x y ys)"
proof (induct xs)
  show "sust x y ([]@ys) = (sust x y [])@(sust x y ys)" by simp
next
  fix a xs
  assume HI: "sust x y (xs@ys) = (sust x y xs)@(sust x y ys)"
  show "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)"
  proof (cases)
    assume "x=a"
    hence "sust x y ((a#xs)@ys) = sust x y (a#(xs@ys))" by simp
    also have "... = y#(sust x y (xs@ys))" using 'x=a' by simp
    also have "... = y#((sust x y xs)@(sust x y ys))" using HI by simp
    also have "... = (y#(sust x y xs))@(sust x y ys)" by simp
    also have "... = (sust x y (a#xs))@(sust x y ys)" using 'x=a' by simp
    finally show "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)" .
  next
    assume "x≠a"
    hence "sust x y ((a#xs)@ys) = sust x y (a#(xs@ys))" by simp
    also have "... = a#(sust x y (xs@ys))" using 'x≠a' by simp
    also have "... = a#((sust x y xs)@(sust x y ys))" using HI by simp
    also have "... = (a#(sust x y xs))@(sust x y ys)" by simp
    also have "... = (sust x y (a#xs))@(sust x y ys)" using 'x≠a' by simp
    finally show "sust x y ((a#xs)@ys) = (sust x y (a#xs))@(sust x y ys)" .
  qed
qed

text {*
  -----
  Ejercicio 3. Demostrar o refutar:
    rev (sust x y zs) = sust x y (rev zs)
  -----
  *}

-- "La demostración automática es"
theorem rev_sust:
  "rev(sust x y zs) = sust x y (rev zs)"
by (induct zs) (simp_all add: sust_append)

```

```
-- "La demostración estructurada es"
theorem rev_sust_2:
  "rev (sust x y zs) = sust x y (rev zs)"
proof (induct zs)
  show "rev (sust x y []) = sust x y (rev [])" by simp
next
  fix a zs
  assume HI: "rev (sust x y zs) = sust x y (rev zs)"
  show "rev (sust x y (a#zs)) = sust x y (rev (a#zs))"
    using HI by (auto simp add: sust_append)
qed

-- "La demostración detallada es"
theorem rev_sust_3:
  "rev (sust x y zs) = sust x y (rev zs)"
proof (induct zs)
  show "rev (sust x y []) = sust x y (rev [])" by simp
next
  fix a zs
  assume HI: "rev (sust x y zs) = sust x y (rev zs)"
  show "rev (sust x y (a#zs)) = sust x y (rev (a#zs))"
  proof -
    have "rev (sust x y (a#zs)) = rev ((if x=a then y else a)#(sust x y zs))"
      by simp
    also have "... = (rev (sust x y zs))@[if x=a then y else a]" by simp
    also have "... = (sust x y (rev zs))@[if x=a then y else a]" using HI by simp
    also have "... = (sust x y (rev zs))@(sust x y [a])" by simp
    also have "... = sust x y ((rev zs)@[a])" by (simp add: sust_append)
    also have "... = sust x y (rev (a#zs))" by simp
    finally show "rev (sust x y (a#zs)) = sust x y (rev (a#zs))" .
  qed
qed

text {*
  -----
  Ejercicio 4. Demostrar o refutar:
    sust x y (sust u v zs) = sust u v (sust x y zs)
  -----
  *}

theorem "sust x y (sust u v zs) = sust u v (sust x y zs)"
quickcheck
oops

text {*
```

El contraejemplo encontrado es:

$u = -1$

$v = 0$

$x = -1$

$y = 1$

$zs = [-1]$

Efectivamente,

$\text{sust } (-1) \ 1 \ (\text{sust } (-1) \ 0 \ [(-1)]) = [0]$

$\text{sust } (-1) \ 0 \ (\text{sust } (-1) \ 1 \ [(-1)]) = [1]$

**/*

text {*

Ejercicio 5. Demostrar o refutar:

$\text{sust } y \ z \ (\text{sust } x \ y \ zs) = \text{sust } x \ z \ zs$

**/*

theorem "sust y z (sust x y zs) = sust x z zs"

quickcheck

oops

text {*

El contraejemplo encontrado es:

$x = 0$

$y = 1$

$z = 0$

$zs = [1]$

En efecto,

$\text{sust } 1 \ 0 \ (\text{sust } 0 \ 1 \ [1]) = [0]$

$\text{sust } 0 \ 0 \ [1] = [1]$

**/*

text {*

Ejercicio 6. Definir la función

$\text{borra} :: "'a \Rightarrow 'a \text{ list} \Rightarrow 'a \text{ list}"$

*tal que (borra x ys) es la lista obtenida borrando la primera
ocurrencia del elemento x en la lista ys. Por ejemplo,*

$\text{borra } (2::\text{nat}) \ [1,2,3,2] = [1,3,2]$

Nota: La función borra es equivalente a la predefinida remove1.

**/*

```

fun borra :: "'a ⇒ 'a list ⇒ 'a list" where
  "borra x [] = []"
| "borra x (y#ys) = (if x=y then ys else (y#(borra x ys)))"

```

```

value "borra (2::nat) [1,2,3,2]" -- "= [1,3,2]"

```

```

text {*

```

```

-----
Ejercicio 7. Definir la función

```

```

  borraTodas :: "'a ⇒ 'a list ⇒ 'a list"
  tal que (borraTodas x ys) es la lista obtenida borrando todas las
  ocurrencias del elemento x en la lista ys. Por ejemplo,
  borraTodas (2::nat) [1,2,3,2] = [1,3]
-----

```

```

*}

```

```

fun borraTodas :: "'a ⇒ 'a list ⇒ 'a list" where
  "borraTodas x [] = []"
| "borraTodas x (y#ys) =
  (if x=y
   then (borraTodas x ys)
   else (y#(borraTodas x ys)))"

```

```

value "borraTodas (2::nat) [1,2,3,2]" -- "= [1,3]"

```

```

text {*

```

```

-----
Ejercicio 8. Demostrar o refutar:

```

```

  borra x (borraTodas x xs) = borraTodas x xs
-----

```

```

*}

```

```

-- "La demostración automática es"

```

```

theorem "borra x (borraTodas x xs) = borraTodas x xs"

```

```

by (induct xs arbitrary: x) simp_all

```

```

-- "La demostración estructurada es"

```

```

theorem "borra x (borraTodas x xs) = borraTodas x xs"

```

```

proof (induct xs arbitrary: x)

```

```

  fix x

```

```

  show "borra x (borraTodas x []) = borraTodas x []" by simp

```

```

next

```

```

  fix a x :: "'b" and xs :: "'b list"

```

```

  assume HI: "∧x. borra x (borraTodas x xs) = borraTodas x xs"

```

```

  show "borra x (borraTodas x (a#xs)) = borraTodas x (a#xs)"

```

```

proof (cases)
  assume "x=a"
  hence "borra x (borraTodas x (a#xs)) = borra x (borraTodas x xs)" by simp
  also have "... = borraTodas x xs" using HI by simp
  also have "... = borraTodas x (a#xs)" using 'x=a' by simp
  finally show "borra x (borraTodas x (a#xs)) = borraTodas x (a#xs)" .
next
  assume "x≠a"
  hence "borra x (borraTodas x (a#xs)) = borra x (a#(borraTodas x xs))"
    by simp
  also have "... = a#(borra x (borraTodas x xs))" using 'x≠a' by simp
  also have "... = a#(borraTodas x xs)" using HI by simp
  also have "... = borraTodas x (a#xs)" using 'x≠a' by simp
  finally show "borra x (borraTodas x (a#xs)) = borraTodas x (a#xs)" .
qed

text {*
  -----
  Ejercicio 9. Demostrar o refutar:
    borraTodas x (borraTodas x xs) = borraTodas x xs
  -----
*}

-- "La demostración automática es"
theorem "borraTodas x (borraTodas x xs) = borraTodas x xs"
by (induct xs arbitrary: x) simp_all

-- "La demostración estructurada es"
theorem "borraTodas x (borraTodas x xs) = borraTodas x xs"
proof (induct xs arbitrary: x)
  fix x
  show "borraTodas x (borraTodas x []) = borraTodas x []" by simp
next
  fix a x :: "'b" and xs :: "'b list"
  assume HI: "∧x. borraTodas x (borraTodas x xs) = borraTodas x xs"
  show "borraTodas x (borraTodas x (a#xs)) = borraTodas x (a#xs)"
  proof (cases)
    assume "x=a"
    hence "borraTodas x (borraTodas x (a#xs)) = borraTodas x (borraTodas x xs)"
      by simp
    also have "... = borraTodas x xs" using HI by simp
    also have "... = borraTodas x (a#xs)" using 'x=a' by simp
    finally show "borraTodas x (borraTodas x (a#xs)) = borraTodas x (a#xs)" .
  next

```

```

    assume "x≠a"
    hence "borraTodas x (borraTodas x (a#xs)) =
      borraTodas x (a#(borraTodas x xs))" by simp
    also have "... = a#(borraTodas x (borraTodas x xs))" using 'x≠a' by simp
    also have "... = a#(borraTodas x xs)" using HI by simp
    also have "... = borraTodas x (a#xs)" using 'x≠a' by simp
    finally show "borraTodas x (borraTodas x (a#xs)) = borraTodas x (a#xs)" .
  qed
qed

text {*
  -----
  Ejercicio 10. Demostrar o refutar:
    borraTodas x (borra x xs) = borraTodas x xs
  -----
*}

-- "La demostración automática es"
theorem "borraTodas x (borra x xs) = borraTodas x xs"
by (induct xs) simp_all

-- "La demostración estructurada es"
theorem borraTodas_borra:
  "borraTodas x (borra x xs) = borraTodas x xs"
proof (induct xs)
  show "borraTodas x (borra x []) = borraTodas x []" by simp
next
  fix a :: "'a" and xs :: "'a list"
  assume HI: "borraTodas x (borra x xs) = borraTodas x xs"
  show "borraTodas x (borra x (a#xs)) = borraTodas x (a#xs)"
  proof (cases)
    assume "x=a"
    hence "borraTodas x (borra x (a#xs)) = borraTodas x xs" by simp
    also have "... = borraTodas x (a#xs)" using 'x=a' by simp
    finally show "borraTodas x (borra x (a#xs)) = borraTodas x (a#xs)" .
  next
    assume "x≠a"
    hence "borraTodas x (borra x (a#xs)) = borraTodas x (a# (borra x xs))"
      by simp
    also have "... = a#(borraTodas x (borra x xs))" using 'x≠a' by simp
    also have "... = a#(borraTodas x xs)" using HI by simp
    also have "... = borraTodas x (a#xs)" using 'x≠a' by simp
    finally show "borraTodas x (borra x (a#xs)) = borraTodas x (a#xs)" .
  qed
qed

```

```

text {*
-----
Ejercicio 11. Demostrar o refutar:
    borra x (borra y xs) = borra y (borra x xs)
-----
*}

-- "La demostración automática es"
theorem "borra x (borra y xs) = borra y (borra x xs)"
by (induct xs) simp_all

-- "La demostración estructurada es"
theorem "borra x (borra y xs) = borra y (borra x xs)"
proof (induct xs)
  show "borra x (borra y []) = borra y (borra x [])" by simp
next
  fix a xs
  assume HI: "borra x (borra y xs) = borra y (borra x xs)"
  show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))"
  proof (cases)
    assume "x=y"
    thus "borra x (borra y (a # xs)) = borra y (borra x (a # xs))" by simp
  next
    assume "x≠y"
    show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))"
    proof (cases)
      assume "y=a"
      hence "x≠a" using 'x≠y' by simp
      have "borra x (borra y (a # xs)) = borra x xs" using 'y=a' by simp
      also have "... = borra y (a # (borra x xs))" using 'y=a' by simp
      also have "... = borra y (borra x (a # xs))" using 'x≠a' by simp
      finally show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))" .
    next
      assume "y≠a"
      show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))"
      proof (cases)
        assume "x=a"
        have "borra x (borra y (a # xs)) = borra x (a#(borra y xs))"
          using 'y≠a' by simp
        also have "... = borra y xs" using 'x=a' by simp
        also have "... = borra y (borra x (a#xs))" using 'x=a' by simp
        finally show "borra x (borra y (a#xs)) = borra y (borra x (a#xs))" .
      next
        assume "x≠a"

```

```

    have "borra x (borra y (a # xs)) = borra x (a#(borra y xs))"
      using 'y≠a' by simp
    also have "... = a#(borra x (borra y xs))" using 'x≠a' by simp
    also have "... = a#(borra y (borra x xs))" using HI by simp
    also have "... = borra y (a#(borra x xs))" using 'y≠a' by simp
    also have "... = borra y (borra x (a#xs))" using 'x≠a' by simp
    finally show "borra x (borra y (a#xs)) = borra y (borra x (a#xs))" .
  qed
qed
qed
qed

-- "La demostración puede simplificarse como se muestra a continuación"
theorem "borra x (borra y xs) = borra y (borra x xs)"
proof (induct xs)
  show "borra x (borra y []) = borra y (borra x [])" by simp
next
  fix a xs
  assume HI: "borra x (borra y xs) = borra y (borra x xs)"
  show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))"
  proof (cases)
    assume "y=a" thus ?thesis by simp
  next
    assume "y≠a" thus ?thesis
    proof (cases)
      assume "x=a" thus ?thesis by simp
    next
      assume "x≠a" thus ?thesis
      proof -
        have "borra x (borra y (a#xs)) = a#(borra x (borra y xs))"
          using 'x≠a' 'y≠a' by simp
        also have "... = a#(borra y (borra x xs))" using HI by simp
        also have "... = borra y (borra x (a#xs))"
          using 'x≠a' 'y≠a' by simp
        finally show "borra x (borra y (a # xs)) = borra y (borra x (a # xs))".
      qed
    qed
  qed
qed
qed
qed

```

text {*

Ejercicio 12. Demostrar o refutar el teorema:

borraTodas x (borra y xs) = borra y (borraTodas x xs)

```

*}

-- "La demostración automática es"
theorem "borraTodas x (borra y xs) = borra y (borraTodas x xs)"
by (induct xs) (auto simp add: borraTodas_borra)

-- "La demostración estructurada es"
theorem "borraTodas x (borra y xs) = borra y (borraTodas x xs)"
proof (induct xs)
  show "borraTodas x (borra y []) = borra y (borraTodas x [])" by simp
next
  fix a xs
  assume HI: "borraTodas x (borra y xs) = borra y (borraTodas x xs)"
  show "borraTodas x (borra y (a#xs)) = borra y (borraTodas x (a#xs))"
  proof (cases)
    assume "x=y"
    show "borraTodas x (borra y (a#xs)) = borra y (borraTodas x (a#xs))"
    proof (cases)
      assume "y=a"
      hence "borraTodas x (borra y (a # xs)) = borraTodas x xs" by simp
      also have "... = borraTodas x (borra x xs)"
      by (rule borraTodas_borra[symmetric])
      also have "... = borraTodas x (borra y xs)" using 'x=y' by simp
      also have "... = borra y (borraTodas x xs)" using HI by simp
      also have "... = borra y (borraTodas x (a#xs))" using 'x=y' 'y=a' by simp
      finally show ?thesis .
    next
      assume "y≠a"
      hence "borraTodas x (borra y (a # xs)) = a#(borraTodas x (borra y xs))"
      using 'x=y' by simp
      also have "... = a#borra y (borraTodas x xs)" using HI by simp
      also have "... = borra y (borraTodas x (a # xs))" using 'x=y' 'y≠a' by simp
      finally show ?thesis .
    qed
  next
    assume "x≠y"
    show "borraTodas x (borra y (a # xs)) = borra y (borraTodas x (a # xs))"
    proof (cases)
      assume "x=a"
      hence "borraTodas x (borra y (a#xs)) = borraTodas x (a#(borra y xs))"
      using 'x≠y' by simp
      also have "... = borraTodas x (borra y xs)" using 'x=a' by simp
      also have "... = borra y (borraTodas x xs)" using HI by simp
      also have "... = borra y (borraTodas x (a#xs))" using 'x=a' by simp
      finally show ?thesis .
    end
  end
end

```

```

next
  assume "x ≠ a"
  show "borraTodas x (borra y (a # xs)) = borra y (borraTodas x (a # xs))"
  proof (cases)
    assume "y = a"
    hence "borraTodas x (borra y (a#xs)) = borraTodas x xs" by simp
    also have "... = borra y (a#(borraTodas x xs))" using 'x≠a' 'y=a' by simp
    also have "... = borra y (borraTodas x (a#xs))" using 'x≠a' by simp
    finally show ?thesis .
  next
    assume "y ≠ a"
    hence "borraTodas x (borra y (a#xs)) = (a#(borraTodas x (borra y xs)))"
      using 'x≠a' 'y≠a' by simp
    also have "... = (a#borra y (borraTodas x xs))" using HI by simp
    also have "... = borra y (borraTodas x (a#xs))" using 'x≠a' 'y≠a' by simp
    finally show ?thesis .
  qed
qed
qed
qed

text {*
  -----
  Ejercicio 13. Demostrar o refutar:
    borra y (sust x y xs) = borra x xs
  -----
*}

theorem "borra y (sust x y xs) = borra x xs"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  x = 1
  xs = [0]
  y = 0
  En efecto,
  borra y (sust x y xs) = borra 0 (sust 1 0 [0]) = []
  borra x xs = borra 1 [0] = [0]
*}

text {*
  -----
  Ejercicio 14. Demostrar o refutar:

```

```

    borraTodas y (sust x y xs) = borraTodas x xs"
-----
*}

theorem "borraTodas y (sust x y xs) = borraTodas x xs"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  x = -1
  xs = [1]
  y = 1
  En efecto,
  borraTodas y (sust x y xs) = borraTodas 1 (sust -1 1 [1]) = []
  borraTodas x xs = borraTodas -1 [1] = [1]
*}

text {*
-----
  Ejercicio 15. Demostrar o refutar:
  sust x y (borraTodas x zs) = borraTodas x zs
-----
*}

-- "La demostración automática es"
theorem "sust x y (borraTodas x zs) = borraTodas x zs"
by (induct zs) auto

-- "La demostración estructurada es"
theorem sust_borraTodas:
  "sust x y (borraTodas x zs) = borraTodas x zs"
proof (induct zs)
  show "sust x y (borraTodas x []) = borraTodas x []" by simp
next
  fix z zs
  assume HI: "sust x y (borraTodas x zs) = borraTodas x zs"
  show "sust x y (borraTodas x (z#zs)) = borraTodas x (z#zs)"
  proof (cases)
    assume "x=z"
    hence "sust x y (borraTodas x (z#zs)) = sust x y (borraTodas x zs)" by simp
    also have "... = borraTodas x zs" using HI by simp
    also have "... = borraTodas x (z#zs)" using 'x=z' by simp
    finally show ?thesis .
  next

```

```

    assume "x≠z"
    hence "sust x y (borraTodas x (z#zs)) = sust x y (z#(borraTodas x zs))"
      by simp
    also have "... = z#(sust x y (borraTodas x zs))" using 'x≠z' by simp
    also have "... = z#(borraTodas x zs)" using HI by simp
    also have "... = borraTodas x (z#zs)" using 'x≠z' by simp
    finally show ?thesis .
  qed
qed

```

```

text {*
-----
  Ejercicio 16. Demostrar o refutar:
    sust x y (borraTodas z zs) = borraTodas z (sust x y zs)
-----
*}

```

```

theorem "sust x y (borraTodas z zs) = borraTodas z (sust x y zs)"
quickcheck
oops

```

```

text {*
  El contraejemplo encontrado es
    x = 0
    y = 1
    z = 0
    zs = [0]
  En efecto,
    sust x y (borraTodas z zs) = sust 0 1 (borraTodas 0 [0]) = []
    borraTodas z (sust x y zs) = borraTodas 0 (sust 0 1 [0]) = [1]
*}

```

```

text {*
-----
  Ejercicio 17. Demostrar o refutar:
    rev (borra x xs) = borra x (rev xs)
-----
*}

```

```

theorem "rev (borra x xs) = borra x (rev xs)"
quickcheck
oops

```

```

text {*
  El contraejemplo encontrado es

```

```

    x = -4
    xs = [-4, 0, -4]
    En efecto,
    rev (borra x xs) = rev (borra -4 [-4,0,-4]) = [-4,0]
    borra x (rev xs) = borra -4 (rev [-4,0,-4]) = [0,-4]
*}

text {*
-----
Ejercicio 18. Demostrar o refutar el teorema:
    borraTodas x (xs@ys) = (borraTodas x xs)@(borraTodas x ys)
-----
*}

-- "La demostración automática es"
lemma "borraTodas x (xs@ys) = (borraTodas x xs)@(borraTodas x ys)"
by (induct xs arbitrary: x) simp_all

-- "La demostración estructurada es"
lemma borraTodas_append:
  "borraTodas x (xs@ys) = (borraTodas x xs)@(borraTodas x ys)"
proof (induct xs arbitrary: x)
  show "\x. borraTodas x ([]@ys) = (borraTodas x [])@(borraTodas x ys)" by simp
next
  fix a x xs
  assume HI: "\x. borraTodas x (xs@ys) = (borraTodas x xs)@(borraTodas x ys)"
  show "borraTodas x ((a#xs)@ys) = (borraTodas x (a#xs))@(borraTodas x ys)"
  proof (cases)
    assume "x=a"
    hence "borraTodas x ((a#xs)@ys) = borraTodas x (xs@ys)" by simp
    also have "... = (borraTodas x xs)@(borraTodas x ys)" using HI by simp
    also have "... = (borraTodas x (a#xs))@(borraTodas x ys)" using 'x=a' by simp
    finally show ?thesis .
  next
    assume "x≠a"
    hence "borraTodas x ((a#xs)@ys) = a#(borraTodas x (xs@ys))" by simp
    also have "... = a#((borraTodas x xs)@(borraTodas x ys))" using HI by simp
    also have "... = (borraTodas x (a#xs))@(borraTodas x ys)" using 'x≠a' by simp
    finally show ?thesis .
  qed
qed

text {*
-----
Ejercicio 19. Demostrar o refutar el teorema:

```

```

    rev (borraTodas x xs) = borraTodas x (rev xs)
-----
*}

-- "La demostración automática es"
theorem "rev (borraTodas x xs) = borraTodas x (rev xs)"
by (induct xs) (auto simp add: borraTodas_append)

-- "La demostración estructurada es"
theorem "rev (borraTodas x xs) = borraTodas x (rev xs)"
proof (induct xs)
  show "rev (borraTodas x []) = borraTodas x (rev [])" by simp
next
  fix a xs
  assume HI: "rev (borraTodas x xs) = borraTodas x (rev xs)"
  show "rev (borraTodas x (a#xs)) = borraTodas x (rev (a#xs))"
  proof (cases)
    assume "x=a"
    hence "rev (borraTodas x (a#xs)) = rev (borraTodas x xs)" by simp
    also have "... = borraTodas x (rev xs)" using HI by simp
    also have "... = (borraTodas x (rev xs))@(borraTodas x [a])"
      using 'x=a' by simp
    also have "... = borraTodas x ((rev xs)@[a])"
      by (simp add: borraTodas_append)
    also have "... = borraTodas x (rev (a#xs))" by simp
    finally show ?thesis .
  next
    assume "x≠a"
    hence "rev (borraTodas x (a#xs)) = rev (a#(borraTodas x xs))" by simp
    also have "... = (rev (borraTodas x xs))@[a]" by simp
    also have "... = (borraTodas x (rev xs))@[a]" using HI by simp
    also have "... = (borraTodas x (rev xs))@(borraTodas x [a])"
      using 'x≠a' by simp
    also have "... = borraTodas x ((rev xs)@[a])"
      by (simp add: borraTodas_append)
    also have "... = borraTodas x (rev (a#xs))" by simp
    finally show ?thesis .
  qed
qed
end

```

6.1.3. Menor posición válida

```
chapter {* T6R1c: Menor posición válida *}

theory T6R1c
imports Main
begin

section {* Menor posición válida *}

text {*
-----
Ejercicio 1. Definir la función
  menorValida :: "('a ⇒ bool) ⇒ 'a list ⇒ nat"
tal que (menorValida p xs) es el índice del primer elemento de una
lista xs que satisface el predicado p y es la longitud de xs si
ningún elemento satisface el predicado p. Por ejemplo,
  menorValida (λx. 4<x) [1::nat, 3, 5, 3, 1]      = 2
  menorValida (λx. 6<x) [1::nat, 3, 5, 3, 1]      = 5
  menorValida (λx. 1<length x) [[], [1, 2], [3]] = 1
-----
*}

fun menorValida :: "('a ⇒ bool) ⇒ 'a list ⇒ nat" where
  "menorValida P []      = 0"
| "menorValida P (x#xs) = (if P x then 0 else 1+(menorValida P xs))"

value "menorValida (λx. 4<x) [1::nat, 3, 5, 3, 1]"      -- "= 2"
value "menorValida (λx. 6<x) [1::nat, 3, 5, 3, 1]"      -- "= 5"
value "menorValida (λx. 1<length x) [[], [1, 2], [3::nat]]" -- "= 1"

text {*
-----
Ejercicio 2. Demostrar que menorValida devuelve la longitud de la
lista syss ningún elemento satisface el predicado dado.
-----
*}

-- "La demostración automática es"
lemma "(menorValida P xs = length xs) = list_all (λ x. (¬ P x)) xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma menorValida_es_length:
  "(menorValida P xs = length xs) = list_all (λ x. ¬ P x) xs"
proof (induct xs)
```

```

show "(menorValida P [] = length []) = list_all (λ x. ¬ P x) []"
  by simp
next
fix a xs
assume HI: "(menorValida P xs = length xs) = list_all (λ x. ¬ P x) xs"
show "(menorValida P (a#xs) = length (a#xs)) = list_all (λ x. ¬ P x) (a#xs)"
proof (cases)
  assume "P a"
  thus "(menorValida P (a#xs) = length (a#xs)) =
    list_all (λ x. ¬ P x) (a#xs)" by simp
next
  assume "¬(P a)"
  hence "(menorValida P (a#xs) = length (a#xs)) =
    (1 + menorValida P xs = 1 + length xs)" by simp
  also have "... = (menorValida P xs = length xs)" by simp
  also have "... = list_all (λ x. ¬(P x)) xs" using HI by simp
  also have "... = list_all (λ x. ¬(P x)) (a#xs)" using '¬(P a)' by simp
  finally show ?thesis .
qed
qed

text {*
  -----
  Ejercicio 3. Demostrar si  $n$  es el valor de  $(\text{menorValida } P \text{ } xs)$ ,
  entonces ninguno de los primeros  $n$  elementos de la lista  $xs$  verifica
  la propiedad  $P$ .
  -----
  *}

-- "La demostración automática es"
lemma "list_all (λ x. ¬ P x) (take (menorValida P xs) xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma "list_all (λ x. ¬ P x) (take (menorValida P xs) xs)"
proof (induct xs)
  show "list_all (λ x. ¬ P x) (take (menorValida P []) [])" by simp
next
  fix a xs
  assume HI: "list_all (λ x. ¬ P x) (take (menorValida P xs) xs)"
  show "list_all (λ x. ¬ P x) (take (menorValida P (a#xs)) (a#xs))"
  proof (cases)
    assume "P a"
    hence "menorValida P (a#xs) = 0" by simp
    hence "take (menorValida P (a#xs)) (a#xs) = []" by simp

```

```

      thus "list_all (λx. ¬ P x) (take (menorValida P (a#xs)) (a#xs))"
        by simp
    next
      assume "¬(P a)"
      hence "menorValida P (a#xs) = 1 + menorValida P xs" by simp
      hence "take (menorValida P (a#xs)) (a#xs) =
        a#(take (menorValida P xs) xs)" by simp
      thus "list_all (λx. ¬ P x) (take (menorValida P (a#xs)) (a#xs))"
        using HI '¬(P a)' by simp
    qed
  qed

text {*
-----
Ejercicio 4. ¿Cómo se puede relacionar
  "menorValida (λx. P x ∨ Q x) xs"
con
  "menorValida P xs" y "menorValida Q xs"?
¿Se puede decir algo parecido con la conjunción de P y Q?
Prueba tus conjeturas.
-----
*}

text {*
La relación es la siguiente: La menor posición de los elementos que
verifican la propiedad "P ∨ Q" es el mínimo de la menor posición de
los elementos que verifican P y de la menor posición de los elementos
que verifican Q.
*}

-- "La demostración automática es"
lemma menorValida_diyuncion_auto:
  "menorValida (λx. P x ∨ Q x) xs =
    min (menorValida P xs) (menorValida Q xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma menorValida_diyuncion:
  "menorValida (λ x. P x ∨ Q x) xs =
    min (menorValida P xs) (menorValida Q xs)"
proof (induct xs)
  show "menorValida (λ x. P x ∨ Q x) [] =
    min (menorValida P []) (menorValida Q [])" by simp
next
  fix a xs

```

```

assume HI: "menorValida ( $\lambda x. P x \vee Q x$ ) xs =
             min (menorValida P xs) (menorValida Q xs)"
show "menorValida ( $\lambda x. P x \vee Q x$ ) (a#xs) =
       min (menorValida P (a#xs)) (menorValida Q (a#xs))"
proof (cases)
  assume "P a"
  hence "menorValida ( $\lambda x. P x \vee Q x$ ) (a#xs) = 0" by simp
  also have "... = min 0 (menorValida Q (a#xs))" by simp
  also have "... = min (menorValida P (a#xs)) (menorValida Q (a#xs))"
    using 'P a' by simp
  finally show ?thesis .
next
  assume " $\neg P a$ "
  show "menorValida ( $\lambda x. P x \vee Q x$ ) (a#xs) =
       min (menorValida P (a#xs)) (menorValida Q (a#xs))"
  proof (cases)
    assume "Q a"
    hence "menorValida ( $\lambda x. P x \vee Q x$ ) (a#xs) = 0" by simp
    also have "... = min (menorValida P (a#xs)) 0" by simp
    also have "... = min (menorValida P (a#xs)) (menorValida Q (a#xs))"
      using 'Q a' by simp
    finally show ?thesis .
  next
    assume " $\neg Q a$ "
    hence "menorValida ( $\lambda x. P x \vee Q x$ ) (a#xs) =
          1 + (menorValida ( $\lambda x. P x \vee Q x$ ) xs)" using ' $\neg P a$ ' by simp
    also have "... = 1 + min (menorValida P xs) (menorValida Q xs)"
      using HI by simp
    also have "... = min (menorValida P (a#xs)) (menorValida Q (a#xs))"
      using ' $\neg P a$ ' ' $\neg Q a$ ' by simp
    finally show ?thesis .
  qed
qed
qed

text {*
  La menor posición de los elementos que verifican la propiedad " $P \wedge Q$ "
  es mayor o igual que el máximo de la menor posición de los elementos
  que verifican P y de la menor posición de los elementos que verifican Q.
*}

-- "La demostración automática es"
lemma menorValida_conjuncion_auto:
  "max (menorValida P xs) (menorValida Q xs)  $\leq$  menorValida ( $\lambda x. P x \wedge Q x$ ) xs"
by (induct xs) auto

```

```

-- "La demostración estructurada es"
lemma menorValida_conjuncion:
  "max (menorValida P xs) (menorValida Q xs) ≤ menorValida (λ x. P x ∧ Q x) xs"
proof (induct xs)
  show "max (menorValida P []) (menorValida Q []) ≤
    menorValida (λ x. P x ∧ Q x) []" by simp
next
  fix a xs
  assume HI: "max (menorValida P xs) (menorValida Q xs) ≤
    menorValida (λ x. P x ∧ Q x) xs"
  show "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
    menorValida (λ x. P x ∧ Q x) (a#xs)"
  proof (cases)
    assume "P a"
    show "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
      menorValida (λ x. P x ∧ Q x) (a#xs)"
    proof -
      have "menorValida P (a#xs) = 0" using 'P a' by simp
      hence f1: "max (menorValida P (a#xs)) (menorValida Q (a#xs)) =
        menorValida Q (a#xs)" by simp
      have "menorValida Q (a#xs) ≤ menorValida (λ x. P x ∧ Q x) (a#xs)"
      proof (cases)
        assume "Q a"
        hence "menorValida Q (a#xs) = 0" by simp
        also have "... ≤ menorValida (λ x. P x ∧ Q x) (a#xs)" by simp
        finally show "menorValida Q (a#xs) ≤
          menorValida (λ x. P x ∧ Q x) (a#xs)" .
      next
        assume "¬ Q a"
        hence "menorValida Q (a#xs) = 1+(menorValida Q xs)" by simp
        also have "... ≤ 1+(max (menorValida P xs) (menorValida Q xs))" by simp
        also have "... ≤ 1+(menorValida (λ x. P x ∧ Q x) xs)" using HI by simp
        also have "... = menorValida (λ x. P x ∧ Q x) (a#xs)"
          using '¬ Q a' by simp
        finally show "menorValida Q (a#xs) ≤
          menorValida (λ x. P x ∧ Q x) (a#xs)" .
      qed
    qed
    thus "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
      menorValida (λ x. P x ∧ Q x) (a#xs)" using f1 by simp
  qed
next
  assume "¬ P a"
  show "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
    menorValida (λ x. P x ∧ Q x) (a#xs)"

```

```

proof (cases)
  assume "Q a"
  hence "menorValida Q (a#xs) = 0" by simp
  hence "max (menorValida P (a#xs)) (menorValida Q (a#xs)) =
    menorValida P (a#xs)" by simp
  also have "... = 1+(menorValida P xs)" using '¬ P a' by simp
  also have "... ≤ 1+(max (menorValida P xs) (menorValida Q xs))" by simp
  also have "... ≤ 1+(menorValida (λ x. P x ∧ Q x) xs)" using HI by simp
  also have "... = menorValida (λ x. P x ∧ Q x) (a#xs)"
    using '¬ P a' by simp
  finally show "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
    menorValida (λ x. P x ∧ Q x) (a#xs)" .

next
  assume "¬ Q a"
  hence "max (menorValida P (a#xs)) (menorValida Q (a#xs)) =
    max (1+(menorValida P xs)) (1+(menorValida Q xs))"
    using '¬ P a' by simp
  also have "... = 1+(max (menorValida P xs) (menorValida Q xs))" by simp
  also have "... ≤ 1+(menorValida (λ x. P x ∧ Q x) xs)" using HI by simp
  also have "... = menorValida (λ x. P x ∧ Q x) (a#xs)"
    using '¬ P a' by simp
  finally show "max (menorValida P (a#xs)) (menorValida Q (a#xs)) ≤
    menorValida (λ x. P x ∧ Q x) (a#xs)" .

qed
qed
qed

text {*
  La desigualdad complementaria no es válida.
*}

lemma
  "menorValida (λ x. P x ∧ Q x) xs ≤ max (menorValida P xs) (menorValida Q xs)"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  P = {a|<^isub>1}
  Q = {a|<^isub>2}
  xs = [a|<^isub>1, a|<^isub>2]
*}

text {*
  -----

```

Ejercicio 5. Si P implica Q , ¿qué relación puede deducirse entre "menorValida P xs " y "menorValida Q xs "?

```

-----
*}

text {*
  Se deduce que
    "menorValida  $Q$   $xs \leq$  menorValida  $P$   $xs$ ".
*}

-- "La demostración automática es"
lemma menorValida_si_implica_auto:
  "( $\forall x. P\ x \rightarrow Q\ x$ )  $\rightarrow$  menorValida  $Q$   $xs \leq$  menorValida  $P$   $xs$ "
by (induct xs) auto

-- "La demostración estructurada es"
lemma menorValida_si_implica:
  assumes "( $\forall x. P\ x \rightarrow Q\ x$ )"
  shows "menorValida  $Q$   $xs \leq$  menorValida  $P$   $xs$ "
proof (induct xs)
  show "menorValida  $Q$  []  $\leq$  menorValida  $P$  []" by simp
next
  fix a xs
  assume HI: "menorValida  $Q$   $xs \leq$  menorValida  $P$   $xs$ "
  show "menorValida  $Q$  ( $a\#xs$ )  $\leq$  menorValida  $P$  ( $a\#xs$ )"
  proof (cases)
    assume "Q a"
    hence "menorValida  $Q$  ( $a\#xs$ ) = 0" by simp
    also have "...  $\leq$  menorValida  $P$  ( $a\#xs$ )" by simp
    finally show ?thesis .
  next
    assume " $\neg Q\ a$ "
    hence " $\neg P\ a$ " using assms by blast
    have "menorValida  $Q$  ( $a\#xs$ ) = 1+(menorValida  $Q$   $xs$ )"
      using ' $\neg Q\ a$ ' by simp
    also have "...  $\leq$  1+(menorValida  $P$   $xs$ )" using HI by simp
    also have "... = menorValida  $P$  ( $a\#xs$ )" using ' $\neg P\ a$ ' by simp
    finally show ?thesis .
  qed
qed
end

```

6.1.4. Número de elementos válidos

```
chapter {* T6R1d: Número de elementos válidos *}

theory T6R1d
imports Main
begin

section {* Número de elementos válidos *}

text {*
-----
Ejercicio 1. Definir la función
  cuentaP :: "('a ⇒ bool) ⇒ 'a list ⇒ nat"
tal que (cuentaP Q xs) es el número de elementos de la lista xs que
satisfacen el predicado Q. Por ejemplo,
  cuentaP (λx. 2<x) [1,3,4,0,5] = 3
-----
*}

fun cuentaP :: "('a ⇒ bool) ⇒ 'a list ⇒ nat" where
  "cuentaP Q [] = 0"
| "cuentaP Q (x#xs) = (if Q x then (1 + cuentaP Q xs) else cuentaP Q xs)"

value "cuentaP (λx. 2<x) [1::nat,3,4,0,5]" -- "= 3"

text {*
-----
Ejercicio 2. Demostrar o refutar:
  cuentaP P (xs @ ys) = cuentaP P xs + cuentaP P ys*
-----
*}

-- "La demostración automática es"
lemma cuentaP_append_auto:
  "cuentaP P (xs @ ys) = cuentaP P xs + cuentaP P ys"
by (induct xs) auto

-- "La demostración estructurada es"
lemma cuentaP_append:
  "cuentaP P (xs @ ys) = cuentaP P xs + cuentaP P ys"
proof (induct xs)
  show "cuentaP P ([] @ ys) = cuentaP P [] + cuentaP P ys" by simp
next
  fix a xs
  assume HI: "cuentaP P (xs @ ys) = cuentaP P xs + cuentaP P ys"
```

```

show "cuentaP P ((a#xs) @ ys) = cuentaP P (a#xs) + cuentaP P ys"
proof (cases)
  assume "P a"
  hence "cuentaP P ((a#xs) @ ys) = 1 + cuentaP P (xs @ ys)" by simp
  also have "... = 1 + cuentaP P xs + cuentaP P ys" using HI by simp
  also have "... = cuentaP P (a#xs) + cuentaP P ys" using 'P a' by simp
  finally show ?thesis .
next
  assume "¬ P a"
  hence "cuentaP P ((a#xs) @ ys) = cuentaP P (xs @ ys)" by simp
  also have "... = cuentaP P xs + cuentaP P ys" using HI by simp
  also have "... = cuentaP P (a#xs) + cuentaP P ys" using '¬ P a' by simp
  finally show ?thesis .
qed
qed

text {*
  -----
  Ejercicio 3. Demostrar que el número de elementos de una lista que
  cumplen una determinada propiedad es el mismo que el de esa lista
  invertida.
  -----
*}

-- "La demostración automática es"
lemma cuentaP_rev_auto:
  "cuentaP P xs = cuentaP P (rev xs)"
by (induct xs) (simp_all add: cuentaP_append)

-- "La demostración estructurada es"
lemma cuentaP_rev:
  "cuentaP P xs = cuentaP P (rev xs)"
proof (induct xs)
  show "cuentaP P [] = cuentaP P (rev [])" by simp
next
  fix a xs
  assume HI: "cuentaP P xs = cuentaP P (rev xs)"
  show "cuentaP P (a#xs) = cuentaP P (rev (a#xs))"
  proof -
    have "cuentaP P (a#xs) = cuentaP P xs + cuentaP P [a]" by simp
    also have "... = cuentaP P (rev xs) + cuentaP P [a]" using HI by simp
    also have "... = cuentaP P ((rev xs) @ [a])" by (simp add: cuentaP_append)
    also have "... = cuentaP P (rev (a#xs))" by simp
    finally show ?thesis .
  qed
qed

```

qed

text {*

*Ejercicio 4. Encontrar y demostrar una relación entre las funciones
 filter y cuentaP.*

*)

text {*

*Solución: La relación existente es
 length (filter P xs) = cuentaP P xs*

*)

-- "La demostración automática es"

lemma length_filter:

"length (filter P xs) = cuentaP P xs"

by (induct xs) simp_all

-- "La demostración estructurada es"

lemma length_filter_2:

"length (filter P xs) = cuentaP P xs"

proof (induct xs)

show "length (filter P []) = cuentaP P []" by simp

next

fix a xs

assume HI: "length (filter P xs) = cuentaP P xs"

show "length (filter P (a#xs)) = cuentaP P (a#xs)"

proof (cases)

assume "P a"

hence "length (filter P (a#xs)) = length (a#(filter P xs))" by simp

also have "... = 1 + length (filter P xs)" by simp

also have "... = 1 + cuentaP P xs" using HI by simp

also have "... = cuentaP P (a#xs)" using 'P a' by simp

finally show ?thesis .

next

assume "¬ P a"

hence "length (filter P (a#xs)) = length (filter P xs)" by simp

also have "... = cuentaP P xs" using HI by simp

also have "... = cuentaP P (a#xs)" using '¬ P a' by simp

finally show ?thesis .

qed

qed

end

6.1.5. Contador de ocurrencias

```
chapter {* T6R1e: Contador de ocurrencias *}

theory T6R1e
imports Main
begin

section {* Contador de ocurrencias *}

text {*
-----
Ejercicio 1. Definir la función
veces :: "'a ⇒ 'a list ⇒ nat"
tal que (veces x ys) es el número de ocurrencias del elemento x en la
lista ys. Por ejemplo,
veces (2::nat) [2,1,2,5,2] = 3
-----
*}

fun veces :: "'a ⇒ 'a list ⇒ nat" where
  "veces x [] = 0"
| "veces x (y#xs) = (if x=y then (1+(veces x xs)) else (veces x xs))"

value "veces (2::nat) [2,1,2,5,2]" -- "= 3"

text {*
-----
Ejercicio 2. Demostrar o refutar:
veces a (xs @ ys) = veces a xs + veces a ys
-----
*}

-- "La demostración automática es"
lemma veces_append [simp]:
  "veces a (xs @ ys) = veces a xs + veces a ys"
by (induct xs) auto

-- "La demostración estructurada es"
lemma veces_append_2:
  "veces a (xs @ ys) = veces a xs + veces a ys"
proof (induct xs)
  show "veces a ([] @ ys) = veces a [] + veces a ys" by simp
next
  fix b xs
```

```

assume HI: "veces a (xs @ ys) = veces a xs + veces a ys"
show "veces a ((b#xs) @ ys) = veces a (b#xs) + veces a ys"
proof (cases "a=b")
  assume "a=b"
  have "veces a ((b#xs) @ ys) = veces a (b#(xs@ys))" by simp
  also have "... = 1 + veces a (xs@ys)" using 'a=b' by simp
  also have "... = 1 + veces a xs + veces a ys" using HI by simp
  also have "... = veces a (b#xs) + veces a ys" using 'a=b' by simp
  finally show ?thesis .
next
  assume "a≠b"
  have "veces a ((b#xs) @ ys) = veces a (b#(xs@ys))" by simp
  also have "... = veces a (xs@ys)" using 'a≠b' by simp
  also have "... = veces a xs + veces a ys" using HI by simp
  also have "... = veces a (b#xs) + veces a ys" using 'a≠b' by simp
  finally show ?thesis .
qed
qed

text {*
-----
Ejercicio 3. Demostrar o refutar:
  veces a xs = veces a (rev xs)
-----
*}

-- "La demostración automática es"
lemma veces_rev:
  "veces a xs = veces a (rev xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma veces_rev_2:
  "veces a xs = veces a (rev xs)"
proof (induct xs)
  show "veces a [] = veces a (rev [])" by simp
next
  fix b xs
  assume HI: "veces a xs = veces a (rev xs)"
  show "veces a (b#xs) = veces a (rev (b#xs))"
  proof -
    have "veces a (b#xs) = veces a [b] + veces a xs" by simp
    also have "... = veces a (rev xs) + veces a [b]" using HI by simp
    also have "... = veces a ((rev xs) @ [b])" by simp
    also have "... = veces a (rev (b#xs))" by simp
  end
end

```

```

    finally show ?thesis .
qed
qed

text {*
-----
Ejercicio 4. Demostrar o refutar:
"veces a xs ≤ length xs".
-----
*}

-- "La demostración automática es"
lemma veces_le_length_auto:
  "veces a xs ≤ length xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma veces_le_length:
  "veces a xs ≤ length xs"
proof (induct xs)
  show "veces a [] ≤ length []" by simp
next
  fix b xs
  assume HI: "veces a xs ≤ length xs"
  show "veces a (b#xs) ≤ length (b#xs)"
  proof -
    have "veces a (b#xs) ≤ 1 + veces a xs" by simp
    also have "... ≤ 1 + length xs" using HI by simp
    also have "... = length (b#xs)" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
Ejercicio 5. Sabiendo que la función map aplica una función a todos
los elementos de una lista:
map f [x\<^isub>1,...,x\<^isub>n] = [f x\<^isub>1,...,f x\<^isub>n],
demostrar o refutar
veces a (map f xs) = veces (f a) xs
-----
*}

-- "La búsqueda de un contraejemplo con quickcheck es"
lemma veces_map:

```

```

"veces a (map f xs) = veces (f a) xs"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  a  = a\<^isub>2
  f  = (λ_. _)(a\<^isub>1 := a\<^isub>1, a\<^isub>2 := a\<^isub>1)
  xs = [a\<^isub>1]
  En efecto,
  veces a (map f xs) = veces a\<^isub>2 (map f [a\<^isub>1]) = veces a\<^isub>2 [a\<^isub>1]
  veces (f a\<^isub>2) [a\<^isub>1] = veces (f a\<^isub>2) [a\<^isub>1] = veces a\<^isub>2 [a\<^isub>1]
*}

text {*
  -----
  Ejercicio 6. La función
    filter :: "('a ⇒ bool) ⇒ 'a list ⇒ 'a list"
  está definida por
    filter P []          = []
    filter P (x # xs) = (if P x then x # filter P xs else filter P xs)
  Encontrar una expresión e que no contenga filter tal que se verifique
  la siguiente propiedad:
    veces a (filter P xs) = e
  -----
*}

text {*
  Solución: "La expresión es
    (if (P a) then veces a xs else 0)
  como se prueba a continuación.
*}

-- "La demostración automática es"
lemma veces_filter:
  "veces a (filter P xs) = (if (P a) then veces a xs else 0)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma veces_filter_2:
  "veces a (filter P xs) = (if (P a) then veces a xs else 0)"
proof (induct xs)
  show "veces a (filter P []) = (if (P a) then veces a [] else 0)" by simp
next
  fix b xs

```

```

assume HI: "veces a (filter P xs) = (if (P a) then veces a xs else 0)"
show "veces a (filter P (b#xs)) = (if (P a) then veces a (b#xs) else 0)"
proof (cases)
  assume "P b"
  hence "veces a (filter P (b#xs)) = veces a (b#(filter P xs))" by simp
  also have "... = veces a ([b]@(filter P xs))" by simp
  also have "... = (veces a [b]) + (veces a (filter P xs))" by simp
  also have "... = (veces a [b]) + (if (P a) then veces a xs else 0)"
    using HI by simp
  also have "... = (if (P a) then ((veces a [b])+(veces a xs)) else 0)"
    using 'P b' by simp
  also have "... = (if (P a) then (veces a ([b]@xs)) else 0)" by simp
  also have "... = (if (P a) then (veces a (b#xs)) else 0)" by simp
  finally show ?thesis .
next
  assume "¬ P b"
  hence "veces a (filter P (b#xs)) = veces a (filter P xs)" by simp
  also have "... = (if (P a) then veces a xs else 0)" using HI by simp
  also have "... = (if (P a) then (veces a (b#xs)) else 0)"
    using '¬ P b' by simp
  finally show ?thesis .
qed
qed

text {*
  -----
  Ejercicio 7. Usando veces, definir la función
    borraDups :: "'a list ⇒ bool"
  tal que (borraDups xs) es la lista obtenida eliminando los elementos
  duplicados de la lista xs. Por ejemplo,
    borraDups [1::nat,2,4,2,3] = [1,4,2,3]

  Nota: La función borraDups es equivalente a la predefinida remdups.
  -----
*}

fun borraDups :: "'a list ⇒ 'a list" where
  "borraDups []      = []"
| "borraDups (x#xs) = (if 0<(veces x xs) then (borraDups xs)
                      else (x#borraDups xs))"

value "borraDups [1::nat,2,4,2,3]" -- "= [1,4,2,3]"

text {*
  -----

```

Ejercicio 8. Encontrar una expresión e que no contenga la función `borraDups` tal que se verifique

$$\text{veces } x \text{ (borraDups } xs) = e$$

```

-----
*}

text {*
  Solución: La expresión es
    (if 0<(veces x xs) then 1 else 0)
  comp se prueba a continuación.
*}

-- "La demostración automática es"
lemma veces_borraDups:
  "veces x (borraDups xs) = (if 0<veces x xs then 1 else 0)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma veces_borraDups_2:
  "veces x (borraDups xs) = (if 0<(veces x xs) then 1 else 0)"
proof (induct xs)
  show "veces x (borraDups []) = (if (0<veces x []) then 1 else 0)" by simp
next
  fix a xs
  assume HI: "veces x (borraDups xs) = (if (0<veces x xs) then 1 else 0)"
  show "veces x (borraDups (a#xs)) = (if (0<veces x (a#xs)) then 1 else 0)"
  proof (cases)
    assume c1: "0<(veces a xs)"
    hence "veces x (borraDups (a#xs)) = veces x (borraDups xs)" by simp
    also have "... = (if (0<veces x xs) then 1 else 0)" using HI by simp
    also have "... = (if (0<veces x (a#xs)) then 1 else 0)" using c1 by simp
    finally show ?thesis .
  next
    assume c2: "¬ 0<(veces a xs)"
    hence "veces x (borraDups (a#xs)) = veces x (a#(borraDups xs))" by simp
    also have "... = (veces x [a]) + (veces x (borraDups xs))" by simp
    also have "... = (veces x [a]) + (if (0<veces x xs) then 1 else 0)"
      using HI by simp
    also have "... = (if (0<veces x (a#xs)) then 1 else 0)" using c2 by simp
    finally show ?thesis .
  qed
qed

text {*
  -----

```

Ejercicio 9. Usando la función "veces", definir la función

distintos :: "'a list ⇒ bool"

tal que (distintos xs) se verifica si cada elemento de xs aparece tan solo una vez. Por ejemplo,

distintos [1,4,3]

¬ distintos [1,4,1]

Nota: La función "distintos" es equivalente a la predefinida "distinct".

**}*

```
fun distintos :: "'a list ⇒ bool" where
  "distintos [] = True"
| "distintos (x#xs) = ((veces x xs = 0) ∧ distintos xs)"
```

```
value "distintos [1::nat,4,3]" -- "= True"
```

```
value "¬ distintos [1::nat,4,1]" -- "= True"
```

```
fun distintos' :: "'a list ⇒ bool" where
  "distintos' [] = True"
| "distintos' (x#xs) = (¬ (elem x xs) ∧ distintos' xs)"
```

```
value "distintos' [1::nat,4,3]" -- "= True"
```

```
value "¬ distintos' [1::nat,4,1]" -- "= True"
```

```
text {*
```

Ejercicio 10. Demostrar que el valor de "borraDups" verifica "distintos".

**}*

```
-- "La demostración automática es"
```

```
lemma distintos_borraDups:
```

```
  "distintos (borraDups xs)"
```

```
by (induct xs) (auto simp add: veces_borraDups)
```

```
-- "La demostración estructurada es"
```

```
lemma distintos_borraDups_2:
```

```
  "distintos (borraDups xs)"
```

```
proof (induct xs)
```

```
  show "distintos (borraDups [])" by simp
```

```
next
```

```
  fix a xs
```

```
  assume HI: "distintos (borraDups xs)"
```

```
  show "distintos (borraDups (a#xs))"
```

```

proof (cases)
  assume "0 < veces a xs"
  hence "distintos (borraDups (a#xs)) = distintos (borraDups xs)" by simp
  thus ?thesis using HI by simp
next
  assume "¬ 0 < veces a xs"
  hence "veces a (borraDups xs) = 0" by (simp add: veces_borraDups)
  hence "(veces a (borraDups xs) = 0) ∧ (distintos (borraDups xs))"
    using HI by simp
  hence "distintos (a#borraDups xs)" using '¬ 0 < veces a xs' by simp
  thus "distintos (borraDups (a#xs))" using '¬ 0 < veces a xs' by simp
qed
qed

end

```

6.1.6. Suma y aplanamiento de listas

```
chapter {* T6R1f: Suma y aplanamiento de listas *
```

```
theory T6R1f
```

```
imports Main T6R1_1-Cons_inverso_y_cuantificadores_sobre_listas
```

```
begin
```

```
section {* Suma y aplanamiento de listas *
```

```
text {*
```

```
-----
Ejercicio 1. Definir la función
```

```
suma :: "nat list ⇒ nat"
```

```
tal que (suma xs) es la suma de los elementos de la lista de números
naturales xs. Por ejemplo,
```

```
suma [3::nat,2,4] = 9
```

```
-----
*}
```

```
fun suma :: "nat list ⇒ nat" where
```

```
  "suma [] = 0"
```

```
| "suma (x#xs) = x + suma xs"
```

```
value "suma [3::nat,2,4]" -- "= 9"
```

```
text {*
```

```
-----
Ejercicio 2. Definir la función
```

```

    aplana :: "'a list list  $\Rightarrow$  'a list"
    tal que (aplana xss) es la obtenida concatenando los miembros de la
    lista de listas "xss". Por ejemplo,
    aplana [[2,3], [4,5], [7,9]] = [2,3,4,5,7,9]
    -----
*}

fun aplana :: "'a list list  $\Rightarrow$  'a list" where
  "aplana []      = []"
| "aplana (x#xs) = x @ aplana xs"

value "aplana [[2::nat,3], [4,5], [7,9]]" -- "= [2,3,4,5,7,9]"

text {*
    -----
    Ejercicio 3. Demostrar o refutar
    length (aplana xs) = suma (map length xs)
    -----
*}

-- "La demostración automática es"
lemma length_aplana:
  "length (aplana xs) = suma (map length xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma length_aplana_2:
  "length (aplana xs) = suma (map length xs)"
proof (induct xs)
  show "length (aplana []) = suma (map length [])" by simp
next
  fix a xs
  assume HI: "length (aplana xs) = suma (map length xs)"
  show "length (aplana (a#xs)) = suma (map length (a#xs))"
  proof -
    have "length (aplana (a#xs)) = length (a @ (aplana xs))" by simp
    also have "... = length a + length (aplana xs)" by simp
    also have "... = length a + suma (map length xs)" using HI by simp
    also have "... = suma (map length (a#xs))" by simp
    finally show ?thesis .
  qed
qed

text {*
    -----

```

Ejercicio 4. Demostrar o refutar

suma (xs @ ys) = suma xs + suma ys

**)*

-- "La demostración automática es:"

lemma suma_append:

"suma (xs @ ys) = suma xs + suma ys"

by (induct xs) auto

-- "La demostración estructurada es"

lemma suma_append_2:

"suma (xs @ ys) = suma xs + suma ys"

proof (induct xs)

show "suma ([]) @ ys = suma [] + suma ys" **by** simp

next

fix a xs

assume HI: "suma (xs @ ys) = suma xs + suma ys"

show "suma ((a#xs) @ ys) = suma (a#xs) + suma ys"

proof -

have "suma ((a#xs) @ ys) = suma (a#(xs@ys))" **by** simp

also have "... = a + suma (xs@ys)" **by** simp

also have "... = a + suma xs + suma ys" **using** HI **by** simp

also have "... = suma (a#xs) + suma ys" **by** simp

finally show ?thesis .

qed

qed

text {*

Ejercicio 5. Demostrar o refutar

aplana (xs @ ys) = (aplana xs) @ (aplana ys)

**)*

-- "La demostración automática es"

lemma aplana_append:

"aplana (xs @ ys) = (aplana xs) @ (aplana ys)"

by (induct xs) auto

-- "La demostración estructurada es"

lemma aplana_append_2:

"aplana (xs @ ys) = (aplana xs) @ (aplana ys)"

proof (induct xs)

show "aplana ([]) @ ys = (aplana []) @ (aplana ys)" **by** simp

```

next
  fix a xs
  assume HI: "aplana (xs @ ys) = (aplana xs) @ (aplana ys)"
  show "aplana ((a#xs) @ ys) = (aplana (a#xs)) @ (aplana ys)"
  proof -
    have "aplana ((a#xs) @ ys) = aplana (a#(xs@ys))" by simp
    also have "... = a @ (aplana (xs@ys))" by simp
    also have "... = a @ aplana xs @ aplana ys" using HI by simp
    also have "... = aplana (a#xs) @ aplana ys" by simp
    finally show ?thesis .
  qed
qed

text {*
  -----
  Ejercicio 6. Demostrar o refutar
    aplana (map rev (rev xs)) = rev (aplana xs)
  -----
*}

-- "La demostración automática es"
lemma aplana_map_rev_rev:
  "aplana (map rev (rev xs)) = rev (aplana xs)"
by (induct xs) (auto simp add: aplana_append)

-- "La demostración estructurada es"
lemma aplana_map_rev_rev_2:
  "aplana (map rev (rev xs)) = rev (aplana xs)"
proof (induct xs)
  show "aplana (map rev (rev [])) = rev (aplana [])" by simp
next
  fix a xs
  assume HI: "aplana (map rev (rev xs)) = rev (aplana xs)"
  show "aplana (map rev (rev (a#xs))) = rev (aplana (a#xs))"
  proof -
    have "aplana (map rev (rev (a#xs))) = aplana (map rev ((rev xs)@[a]))"
      by simp
    also have "... = aplana ((map rev (rev xs))@(map rev [a]))" by simp
    also have "... = (aplana (map rev (rev xs))@(aplana (map rev [a])))"
      by (simp add: aplana_append)
    also have "... = (rev (aplana xs))@(aplana (map rev [a]))" using HI by simp
    also have "... = (rev (aplana xs))@(rev (aplana [a]))" by simp
    also have "... = rev ((aplana [a])@(aplana xs))" by simp
    also have "... = rev (aplana (a#xs))" by simp
    finally show ?thesis .
  qed

```

```

    qed
qed

text {*
-----
Ejercicio 7. Demostrar o refutar
    aplana (rev (map rev xs)) = rev (aplana xs)
-----
*}

-- "La demostración automática es"
lemma aplana_rev_map_rev:
    "aplana (rev (map rev xs)) = rev (aplana xs)"
by (induct xs) (auto simp add: aplana_append)

text {*
-----
Ejercicio 8. Demostrar o refutar
    list_all (list_all P) xs = list_all P (aplana xs)
-----
*}

-- "La demostración automática es"
lemma list_all_list_all:
    "list_all (list_all P) xs = list_all P (aplana xs)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma list_all_list_all_2:
    "list_all (list_all P) xs = list_all P (aplana xs)"
proof (induct xs)
    show "list_all (list_all P) [] = list_all P (aplana [])" by simp
next
    fix a xs
    assume HI: "list_all (list_all P) xs = list_all P (aplana xs)"
    show "list_all (list_all P) (a#xs) = list_all P (aplana (a#xs))"
    proof -
        have "list_all (list_all P) (a#xs) =
            ((list_all P a) ∧ (list_all (list_all P) xs))" by simp
        also have "... = ((list_all P a) ∧ (list_all P (aplana xs)))"
            using HI by simp
        also have "... = list_all P (a@(aplana xs))" by simp
        also have "... = list_all P (aplana (a#xs))" by simp
        finally show ?thesis .
    qed
qed

```

qed

```

text {*
-----
  Ejercicio 9. Demostrar o refutar
    aplana (rev xs) = aplana xs
-----
*}

-- "La búsqueda de un contraejemplo con QuickCheck es"
lemma aplana_rev:
  "aplana (rev xs) = aplana xs"
quickcheck
oops

text {*
  El contraejemplo encontrado es
  xs = [[a\<^isub>1], [a\<^isub>2]]
*}

text {*
-----
  Ejercicio 10. Demostrar o refutar
    suma (rev xs) = suma xs
-----
*}

-- "La demostración automática es"
lemma suma_rev:
  "suma (rev xs) = suma xs"
by (induct xs) (auto simp add: suma_append)

-- "La demostración estructurada es"
lemma suma_rev_2:
  "suma (rev xs) = suma xs"
proof (induct xs)
  show "suma (rev []) = suma []" by simp
next
  fix a xs
  assume HI: "suma (rev xs) = suma xs"
  show "suma (rev (a#xs)) = suma (a#xs)"
  proof -
    have "suma (rev (a#xs)) = suma ((rev xs)@[a])" by simp
    also have "... = (suma (rev xs)) + (suma [a])" by (simp add: suma_append)
    also have "... = (suma xs) + (suma [a])" using HI by simp
  
```

```

    also have "... = a + (suma xs)" by simp
    also have "... = suma (a#xs)" by simp
    finally show ?thesis .
qed
qed

text {*
-----
Ejercicio 11. Buscar un predicado  $P$  para que se verifique la siguiente
propiedad
 $list\_all\ P\ xs \rightarrow length\ xs \leq suma\ xs$ 
-----
*}

text {*
Solución: El predicado es  $(\lambda x. 1 \leq x)$ , como se demuestra a
continuación
*}

-- "La demostración automática es"
lemma "list_all ( $\lambda x. 1 \leq x$ ) xs  $\rightarrow$  length xs  $\leq$  suma xs"
by (induct xs) auto

-- "La demostración estructurada es"
lemma "list_all ( $\lambda x. 1 \leq x$ ) xs  $\rightarrow$  length xs  $\leq$  suma xs"
proof (induct xs)
  show "list_all ( $\lambda x. 1 \leq x$ ) []  $\rightarrow$  length []  $\leq$  suma []" by simp
next
  fix a xs
  assume HI: "list_all ( $\lambda x. 1 \leq x$ ) xs  $\rightarrow$  length xs  $\leq$  suma xs"
  show "list_all ( $\lambda x. 1 \leq x$ ) (a#xs)  $\rightarrow$  length (a#xs)  $\leq$  suma (a#xs)"
  proof
    assume c1: "list_all ( $\lambda x. 1 \leq x$ ) (a#xs)"
    hence c2: "1  $\leq$  a" by simp
    have c3: "list_all ( $\lambda x. 1 \leq x$ ) xs" using c1 by simp
    have "length (a#xs)  $\leq$  a + (length xs)" using c2 by simp
    also have "...  $\leq$  a + (suma xs)" using HI c3 by simp
    also have "... = suma (a#xs)" by simp
    finally show "length (a#xs)  $\leq$  suma (a#xs)" .
  qed
qed

text {*
-----
Ejercicio 12. Demostrar o refutar

```

```

    algunos (algunos P) xs = algunos P (aplana xs)
-----
*}

-- "La demostración automática es"
lemma algunos_algunos:
  "algunos (algunos P) xs = algunos P (aplana xs)"
by (induct xs) (auto simp add: algunos_append)

-- "La demostración estructurada es"
lemma algunos_algunos_2:
  "algunos (algunos P) xs = algunos P (aplana xs)"
proof (induct xs)
  show "algunos (algunos P) [] = algunos P (aplana [])" by simp
next
  fix a xs
  assume HI: "algunos (algunos P) xs = algunos P (aplana xs)"
  show "algunos (algunos P) (a#xs) = algunos P (aplana (a#xs))"
  proof -
    have "algunos (algunos P) (a#xs) =
      ((algunos P a) v (algunos (algunos P) xs))" by simp
    also have "... = ((algunos P a) v (algunos P (aplana xs)))" using HI by simp
    also have "... = algunos P (a @ (aplana xs))" by (simp add: algunos_append)
    also have "... = algunos P (aplana (a#xs))" by simp
    finally show ?thesis .
  qed
qed

text {*
-----
Ejercicio 13. Redefinir, usando la función list_all, la función
algunos. Llamar la nueva función algunos2 y demostrar que es
equivalente a algunos.
-----
*}

fun algunos2 :: "('a ⇒ bool) ⇒ ('a list ⇒ bool)" where
  "algunos2 P xs = (¬ list_all (λx. ¬ P x) xs)"

-- "La demostración automática es"
lemma algunos2_algunos:
  "algunos2 P xs = algunos P xs"
by (induct xs) auto

-- "La demostración estructurada es"

```

```

lemma algunos2_algunos_2:
  "algunos2 P xs = algunos P xs"
proof (induct xs)
  show "algunos2 P [] = algunos P []" by simp
next
  fix a xs
  assume HI: "algunos2 P xs = algunos P xs"
  show "algunos2 P (a#xs) = algunos P (a#xs)"
  proof -
    have "algunos2 P (a#xs) = (¬ list_all (λx. ¬ P x) (a#xs))"
      by simp
    also have "... = (¬ ((¬ P a) ∧ (list_all (λx. ¬ P x) xs)))" by simp
    also have "... = ((P a) ∨ ¬ (list_all (λx. ¬ P x) xs))" by simp
    also have "... = ((P a) ∨ (algunos2 P xs))" by simp
    also have "... = ((P a) ∨ (algunos P xs))" using HI by simp
    also have "... = algunos P (a#xs)" by simp
    finally show ?thesis .
  qed
qed
end

```

6.1.7. Conjuntos mediante listas

```
chapter {* T6R1g: Conjuntos mediante listas *}
```

```

theory T6R1g
imports Main
begin

```

```

text {*
  Los conjuntos finitos se pueden representar mediante listas. En esta
  relación se define la unión y se demuestran algunas de sus
  propiedades. Análogamente se puede hacer con la intersección y
  diferencia. *}

```

```

text {*
  -----
  Ejercicio 1. Definir la función
    union_l :: "'a list ⇒ 'a list ⇒ 'a list"
  tal que (union_l xs ys) es la unión de las listas xs e ys. Por
  ejemplo,
    union_l [1,2] [2,3] = [1,2,3]
  -----

```

```

*}

fun union_l :: "'a list ⇒ 'a list ⇒ 'a list" where
  "union_l [] ys      = ys"
| "union_l (x#xs) ys = (if x ∈ set ys
                        then (union_l xs ys)
                        else x#(union_l xs ys))"

value "union_l [1::nat,2] [2,3]" -- "[1,2,3]"

text {*
-----
  Ejercicio 2. Demostrar o refutar
    set (union_l xs ys) = set xs ∪ set ys
-----
*}

lemma set_union_l: "set (union_l xs ys) = set xs ∪ set ys"
by (induct "xs") auto

text {*
-----
  Ejercicio 3. Demostrar que si xs e ys no tienen elementos repetidos,
  entonces (union_l xs ys) tampoco los tiene.

  Indicación: Usar la función "distinctc" de la teoría List.thy
-----
*}

lemma [rule_format]:
  "distinct xs → distinct ys → (distinct (union_l xs ys))"
by (induct "xs") (auto simp add: set_union_l)

section {* Cuantificación sobre conjuntos *}

text {*
-----
  Ejercicio 4. Definir un conjunto S para que se verifique que
    (∀x∈A. P x) ∧ (∀x∈ B. P x) → (∀x∈A∪B. P x)
-----
*}

text {* S = A ∪ B *}
lemma "(∀x∈A. P x) ∧ (∀x∈ B. P x) → (∀x∈A∪B. P x)"
by auto

```

```

text {*
-----
Ejercicio 5. Definir una propiedad P para que se verifique que
 $\forall x \in A. P(f\ x) \implies \forall y \in f' A. Q\ y$ 
-----
*}

text {* Q = P *}
lemma "∀x∈A. Q (f x) ⇒ ∀y∈f 'A. Q y"
by auto

end

```

6.1.8. Suma de listas y recursión final

```

chapter {* T6R1h: Suma de listas y recursión final *}

theory T6R1h
imports Main
begin

text {*
-----
Ejercicio 1. Definir la función
suma :: "nat list ⇒ nat"
tal que (suma xs) es la suma de los elementos de xs. Por ejemplo,
suma [2,3,5] = 10
-----
*}

fun suma :: "nat list ⇒ nat" where
  "suma [] = 0"
| "suma (x#xs) = x + suma xs"

value "suma [2,3,5]" -- "= 10"

text {*
-----
Ejercicio 2. Demostrar que la suma de la concatenación de dos listas
es la suma de las sumas de cada lista.
-----
*}

lemma suma_append [simp]:

```

```

"suma (xs @ ys) = suma xs + suma ys"
by (induct xs) auto

text {*
-----
Ejercicio 3. Demostrar que la suma de los  $n$  primeros números naturales
es  $n*(n+1)/2$ ; es decir,

$$0 + 1 + 2 + 3 + \dots + n = n*(n+1)/2$$

Indicación: Usar la notación  $[0..<n+1]$  para la lista de los  $n$  primeros
números naturales.
-----
*}

lemma "2 * suma [0..<n+1] = n * (n + 1)"
by (induct n) auto

text {*
-----
Ejercicio 4. Demostrar que la suma de la lista formada por  $n$  copias
del número  $a$  es  $n*a$ .
Indicación: Usar la expresión  $(\text{replicate } n \ a)$  para la lista formada
por  $n$  copias del número  $a$ 
-----
*}

lemma "suma (replicate n a) = n * a"
by (induct n) auto

text {*
-----
Ejercicio 5. Definir, mediante recursión final, la función
 $\text{sumaF} :: \text{"nat list"} \Rightarrow \text{nat}$ 
tal que  $(\text{sumaF } xs)$  es la suma de los elementos de  $xs$ . Por ejemplo,
 $\text{sumaF } [2,3,5] = 10$ 
-----
*}

fun sumaFAux :: "nat list  $\Rightarrow$  nat  $\Rightarrow$  nat" where
  "sumaFAux []      n = n"
| "sumaFAux (x#xs) n = sumaFAux xs (x + n)"

definition sumaF :: "nat list  $\Rightarrow$  nat" where
  "sumaF xs  $\equiv$  sumaFAux xs 0"

value "sumaF [2,3,5]" -- "= 10"

```

```

text {*
-----
Ejercicio 6. Demostrar que las funciones suma y sumaF son
equivalentes.
-----
*}

lemma sumaFAux_add:
  "sumaFAux xs (a+b) = a + sumaFAux xs b"
by (induct xs arbitrary: a b) auto

lemma [simp]: "sumaF [] = 0"
by (auto simp add: sumaF_def)

lemma [simp]: "sumaF (x#xs) = x + sumaF xs"
by (auto simp add: sumaF_def sumaFAux_add[THEN sym])

-- "Una demostración estructurada del lema anterior es"
lemma "sumaF (x#xs) = x + sumaF xs"
proof -
  have "sumaF (x#xs) = sumaFAux (x#xs) 0" by (simp add: sumaF_def)
  also have "... = sumaFAux xs (x + 0)" by simp
  also have "... = x + sumaFAux xs 0"
    using sumaFAux_add[THEN sym] by simp
  also have "... = x + sumaF xs" by (simp add: sumaF_def)
  finally show ?thesis .
qed

theorem "sumaF xs = suma xs"
by (induct xs) auto

end

```

6.1.9. Intercalación de listas

```
chapter {* T6R1i: Intercalación de listas *
```

```
theory T6R1i
imports Main
begin

```

```

text {*
-----
Ejercicio 1. Definir la función

```

```

    intercala :: "'a list ⇒ 'a list ⇒ 'a list"
    tal que (intercala xs ys) es la lista obtenida intercalando los
    elementos de xs e ys. Por ejemplo,
    value "intercala [a,b] [u,v]"      -- "= [a, u, b, v]"
    value "intercala [a,b] [u,v,x,y,z]" -- "= [a, u, b, v, x, y, z]"
    value "intercala [a,b,c,d,e] [u,v]" -- "= [a, u, b, v, c, d, e]"
    -----
*}

fun intercala :: "'a list ⇒ 'a list ⇒ 'a list" where
  "intercala [] ys = ys"
| "intercala xs [] = xs"
| "intercala (x#xs) (y#ys) = x # y # intercala xs ys"

value "intercala [a,b] [u,v]"      -- "= [a, u, b, v]"
value "intercala [a,b] [u,v,x,y,z]" -- "= [a, u, b, v, x, y, z]"
value "intercala [a,b,c,d,e] [u,v]" -- "= [a, u, b, v, c, d, e]"

text {*
    -----
    Ejercicio 2. Demostrar la propiedad distributiva de la intercalación
    sobre la concatenación.
    -----
*}

lemma
  assumes "length p = length u"
          "length q = length v"
  shows "intercala (p@q) (u@v) = intercala p u @ intercala q v"
using assms
by (induct p arbitrary: q u v) (simp, case_tac u, auto)

-- "Demostración estructurada del lema anterior"
lemma
  assumes "length p = length u"
          "length q = length v"
  shows "intercala (p@q) (u@v) = intercala p u @ intercala q v"
using assms
proof (induct p arbitrary: q u v)
  fix q u v :: "'a list"
  assume "length [] = length u" and "length q = length v"
  thus "intercala ([]@q) (u@v) = intercala [] u @ intercala q v"
  by auto
next
  fix a::"'a" and p q u v :: "'a list"

```

```

assume
  "∧q u v. [length p = length u; length q = length v]
           ⇒ intercala (p @ q) (u @ v) = intercala p u @ intercala q v"
  "length (a # p) = length u"
  "length q = length v"
thus "intercala ((a#p)@q) (u@v) = intercala (a#p) u @ intercala q v"
by (case_tac u, auto)
qed

end

```

6.1.10. Ordenación de listas por inserción

```
chapter {* T6R1j: Ordenación de listas por inserción *
```

```
theory T6R1j_Ordenacion_de_listas_por_insercion
imports Main
begin

```

```

text {
  En esta relación de ejercicios se define el algoritmo de ordenación de
listas por inserción y se demuestra que es correcto.
*}

```

```

text {
  -----
  Ejercicio 1. Definir la función
  inserta :: nat ⇒ nat list ⇒ nat list
tal que (inserta a xs) es la lista obtenida insertando a delante del
primer elemento de xs que es mayor o igual que a. Por ejemplo,
inserta 3 [2,5,1,7] = [2,3,5,1,7]
  ----- *}

```

```

fun inserta :: "nat ⇒ nat list ⇒ nat list" where
  "inserta a []      = [a]"
| "inserta a (x#xs) = (if a ≤ x then a#x#xs else x # inserta a xs)"

```

```
value "inserta 3 [2,5,1,7]" -- "= [2,3,5,1,7]"
```

```

text {
  -----
  Ejercicio 2. Definir la función
  ordena :: nat list ⇒ nat list
tal que (ordena xs) es la lista obtenida ordenando xs por inserción.

```

Por ejemplo,

```

----- *}

fun ordena :: "nat list ⇒ nat list" where
  "ordena []      = []"
| "ordena (x#xs) = inserta x (ordena xs)"

value "ordena [3,2,5,3]" -- "[2,3,3,5]"

text {*
-----
  Ejercicio 3. Definir la función
    menor :: nat ⇒ nat list ⇒ bool
  tal que (menor a xs) se verifica si a es menor o igual que todos los
  elementos de xs. Por ejemplo,
    menor 2 [3,2,5] = True
    menor 2 [3,0,5] = False
----- *}

fun menor :: "nat ⇒ nat list ⇒ bool" where
  "menor a []      = True"
| "menor a (x#xs) = (a ≤ x ∧ menor a xs)"

value "menor 2 [3,2,5]" -- "= True"
value "menor 2 [3,0,5]" -- "= False"

text {*
-----
  Ejercicio 4. Definir la función
    ordenada :: nat list ⇒ bool
  tal que (ordenada xs) se verifica si xs es una lista ordenada de
  manera creciente. Por ejemplo,
    ordenada [2,3,3,5] = True
    ordenada [2,4,3,5] = False
----- *}

fun ordenada :: "nat list ⇒ bool" where
  "ordenada []      = True"
| "ordenada (x#xs) = (menor x xs & ordenada xs)"

value "ordenada [2,3,3,5]" -- "= True"
value "ordenada [2,4,3,5]" -- "= False"

text {*
```

```

-----
Ejercicio 5. Demostrar que si y es una cota inferior de xs y  $x \leq y$ ,
entonces x es una cota inferior de xs.
----- *}

-- "La demostración automática es"
lemma menor_menor:
  assumes "x ≤ y"
  shows "menor y xs → menor x xs"
using assms
by (induct xs) auto

-- "La demostración estructurada es"
lemma menor_menor_2:
  assumes "x ≤ y"
  shows "menor y xs → menor x xs"
proof (induct xs)
  show "menor y [] → menor x []" by simp
next
  fix z zs
  assume HI: "menor y zs → menor x zs"
  show "menor y (z # zs) → menor x (z # zs)"
  proof
    assume sup: "menor y (z # zs)"
    show "menor x (z # zs)"
    proof (simp only: menor.simps(2))
      show "x ≤ z ∧ menor x zs"
      proof
        have "x ≤ y" using assms .
        also have "y ≤ z" using sup by simp
        finally show "x ≤ z" by simp
      next
        have "menor y zs" using sup by simp
        with HI show "menor x zs" by simp
      qed
    qed
  qed
qed

text {*
-----
Ejercicio 6. Demostrar el siguiente teorema de corrección: x es una
cota inferior de la lista obtenida insertando y en zs syss  $x \leq y$  y x
es una cota inferior de zs.
----- *}

```

```

-- "La demostración automática es"
lemma menor_inserta:
  "menor x (inserta y zs) = (x ≤ y ∧ menor x zs)"
by (induct zs) auto

-- "La demostración estructurada es"
lemma menor_inserta_2:
  "menor x (inserta y zs) = (x ≤ y ∧ menor x zs)"
proof (induct zs)
  show "menor x (inserta y []) = (x ≤ y ∧ menor x [])" by simp
next
  fix z zs
  assume HI: "menor x (inserta y zs) = (x ≤ y ∧ menor x zs)"
  show "menor x (inserta y (z#zs)) = (x ≤ y ∧ menor x (z#zs))"
  proof (cases "y ≤ z")
    assume "y ≤ z"
    hence "menor x (inserta y (z#zs)) = menor x (y#z#zs)" by simp
    also have "... = (x ≤ y ∧ menor x (z#zs))" by simp
    finally show ?thesis by simp
  next
    assume "¬(y ≤ z)"
    hence "menor x (inserta y (z#zs)) =
      menor x (z # inserta y zs)" by simp
    also have "... = (x ≤ z ∧ menor x (inserta y zs))" by simp
    also have "... = (x ≤ z ∧ x ≤ y ∧ menor x zs)" using HI by simp
    also have "... = (x ≤ y ∧ menor x (z#zs))" by auto
    finally show ?thesis by simp
  qed
qed

text {*
  -----
  Ejercicio 6. Demostrar que al insertar un elemento la lista obtenida
  está ordenada syss lo estaba la original.
  ----- *}

-- "La demostración automática es"
lemma ordenada_inserta:
  "ordenada (inserta a xs) = ordenada xs"
by (induct xs) (auto simp add: menor_menor menor_inserta)

-- "La demostración estructurada es"
lemma ordenada_inserta_2:
  "ordenada (inserta a xs) = ordenada xs"

```

```

proof (induct xs)
  show "ordenada (inserta a []) = ordenada []" by simp
next
  fix x xs
  assume HI: "ordenada (inserta a xs) = ordenada xs"
  show "ordenada (inserta a (x # xs)) = ordenada (x # xs)"
  proof (cases "a ≤ x")
    assume "a ≤ x"
    hence "ordenada (inserta a (x # xs)) =
      ordenada (a # x # xs)" by simp
    also have "... = (menor a (x#xs) ∧ ordenada (x # xs))" by simp
    also have "... = ordenada (x # xs)"
      using 'a ≤ x' by (auto simp add: menor_menor)
    finally show "ordenada (inserta a (x # xs)) = ordenada (x # xs)"
      by simp
  next
    assume "¬(a ≤ x)"
    hence "ordenada (inserta a (x # xs)) =
      ordenada (x # inserta a xs)" by simp
    also have "... = (menor x (inserta a xs) ∧ ordenada (inserta a xs))"
      by simp
    also have "... = (menor x (inserta a xs) ∧ ordenada xs)"
      using HI by simp
    also have "... = (menor x xs ∧ ordenada xs)"
      using '¬(a ≤ x)' by (simp add: menor_inserta)
    also have "... = ordenada (x # xs)" by simp
    finally show "ordenada (inserta a (x # xs)) = ordenada (x # xs)"
      by simp
  qed
qed

text {*
  -----
  Ejercicio 7. Demostrar que, para toda lista xs, (ordena xs) está
  ordenada.
  ----- *}

-- "La demostración automática es"
theorem ordenada_ordena:
  "ordenada (ordena xs)"
by (induct xs) (auto simp add: ordenada_inserta)

-- "La demostración estructurada es"
theorem ordenada_ordena_2:
  "ordenada (ordena xs)"

```

```

proof (induct xs)
  show "ordenada (ordena [])" by simp
next
  fix x xs
  assume "ordenada (ordena xs)"
  hence "ordenada (inserta x (ordena xs))"
    by (simp add: ordenada_inserta)
  thus "ordenada (ordena (x # xs))" by simp
qed

text {*
-----
Nota. El teorema anterior no garantiza que ordena sea correcta, ya que
puede que (ordena xs) no tenga los mismos elementos que xs. Por
ejemplo, si se define (ordena xs) como [] se tiene que (ordena xs)
está ordenada pero no es una ordenación de xs. Para ello, definimos la
función cuenta.
----- *}

text {*
-----
Ejercicio 8. Definir la función
  cuenta :: nat list => nat => nat
tal que (cuenta xs y) es el número de veces que aparece el elemento y
en la lista xs. Por ejemplo,
  cuenta [1,3,4,3,5] 3 = 2
----- *}

fun cuenta :: "nat list => nat => nat" where
  "cuenta [] y = 0"
| "cuenta (x#xs) y = (if x=y then Suc(cuenta xs y) else cuenta xs y)"

value "cuenta [1,3,4,3,5] 3" -- "= 2"

text {*
-----
Ejercicio 9. Demostrar que el número de veces que aparece y en
(inserta x xs) es
* uno más el número de veces que aparece en xs, si y = x;
* el número de veces que aparece en xs, si y ≠ x;
----- *}

-- "La demostración automática es"
lemma cuenta_inserta:
  "cuenta (inserta x xs) y =

```

```

    (if x=y then Suc (cuenta xs y) else cuenta xs y)"
by (induct xs) auto

-- "La demostración estructurada es"
lemma cuenta_inserta_2:
  "cuenta (inserta x xs) y =
   (if x=y then Suc (cuenta xs y) else cuenta xs y)"
  (is "?P x y xs")
proof (induct xs)
  show "?P x y []" by simp
next
  fix a xs
  assume "?P x y xs"
  thus "?P x y (a#xs)" by auto
qed

text {*
  -----
  Ejercicio 10. Demostrar que el número de veces que aparece y en
  (ordena xs) es el número de veces que aparece en xs.
  ----- *}

-- "La demostración automática es"
theorem cuenta_ordena:
  "cuenta (ordena xs) y = cuenta xs y"
by (induct xs) (auto simp add: cuenta_inserta)

-- "La demostración estructurada es"
theorem cuenta_ordena_2:
  "cuenta (ordena xs) y = cuenta xs y"
proof (induct xs)
  show "cuenta (ordena []) y = cuenta [] y" by simp
next
  fix x xs
  assume HI: "cuenta (ordena xs) y = cuenta xs y"
  show "cuenta (ordena (x # xs)) y = cuenta (x # xs) y"
  proof (cases "x = y")
    assume "x = y"
    have "cuenta (ordena (x # xs)) y = cuenta (inserta x (ordena xs)) y"
      by simp
    also have "... = Suc (cuenta (ordena xs) y)"
      using 'x = y' by (simp add: cuenta_inserta)
    also have "... = Suc (cuenta xs y)" using HI by simp
    also have "... = cuenta (x # xs) y" using 'x = y' by simp
    finally show "cuenta (ordena (x # xs)) y = cuenta (x # xs) y"
  end
end

```

```

    by simp
  next
    assume "x ≠ y"
    have "cuenta (ordena (x # xs)) y = cuenta (inserta x (ordena xs)) y"
      by simp
    also have "... = cuenta (ordena xs) y"
      using 'x ≠ y' by (simp add: cuenta_inserta)
    also have "... = cuenta xs y" using HI by simp
    also have "... = cuenta (x # xs) y" using 'x ≠ y' by simp
    finally show "cuenta (ordena (x # xs)) y = cuenta (x # xs) y"
      by simp
  qed
qed

end

```

6.1.11. Ordenación de listas por mezcla

```
chapter {* T6R1k: Ordenación de listas por mezcla *
```

```
theory T6R1k_Ordenacion_de_listas_por_mezcla
```

```
imports Main
```

```
begin
```

```
text {*
```

En esta relación de ejercicios se define el algoritmo de ordenación de listas por mezcla y se demuestra que es correcto.

```
*)
```

```
section {* Ordenación de listas *
```

```
text {*
```

Ejercicio 1. Definir la función

menor :: nat ⇒ nat list ⇒ bool

tal que (menor a xs) se verifica si a es menor o igual que todos los elementos de xs. Por ejemplo,

menor 2 [3,2,5] = True

menor 2 [3,0,5] = False

*----- *)*

```
fun menor :: "nat ⇒ nat list ⇒ bool" where
```

```
  "menor a [] = True"
```

```
| "menor a (x#xs) = (a ≤ x ∧ menor a xs)"
```

```

value "menor 2 [3,2,5]" -- "= True"
value "menor 2 [3,0,5]" -- "= False"

text {*
-----
Ejercicio 2. Definir la función
ordenada :: nat list => bool
tal que (ordenada xs) se verifica si xs es una lista ordenada de
manera creciente. Por ejemplo,
ordenada [2,3,3,5] = True
ordenada [2,4,3,5] = False
----- *}

fun ordenada :: "nat list => bool" where
  "ordenada [] = True"
| "ordenada (x#xs) = (menor x xs & ordenada xs)"

value "ordenada [2,3,3,5]" -- "= True"
value "ordenada [2,4,3,5]" -- "= False"

text {*
-----
Ejercicio 3. Definir la función
cuenta :: nat list => nat => nat
tal que (cuenta xs y) es el número de veces que aparece el elemento y
en la lista xs. Por ejemplo,
cuenta [1,3,4,3,5] 3 = 2
----- *}

fun cuenta :: "nat list => nat => nat" where
  "cuenta [] y = 0"
| "cuenta (x#xs) y = (if x=y then Suc(cuenta xs y) else cuenta xs y)"

value "cuenta [1,3,4,3,5] 3" -- "= 2"

section {* Ordenación por mezcla *}

text {*
-----
Ejercicio 4. Definir la función
mezcla :: nat list => nat list => nat list
tal que (mezcla xs ys) es la lista obtenida mezclando las listas
ordenadas xs e ys. Por ejemplo,
mezcla [1,2,5] [3,5,7] = [1,2,3,5,5,7]
----- *}

```

```

fun mezcla :: "nat list  $\Rightarrow$  nat list  $\Rightarrow$  nat list" where
  "mezcla [] ys = ys"
| "mezcla xs [] = xs"
| "mezcla (x # xs) (y # ys) = (if x  $\leq$  y
                                then x # mezcla xs (y # ys)
                                else y # mezcla (x # xs) ys)"

value "mezcla [1,2,5] [3,5,7]" -- "= [1,2,3,5,5,7]"

text {*
-----
Ejercicio 5. Definir la función
  ordenaM :: nat list  $\Rightarrow$  nat list
tal que (ordenaM xs) es la lista obtenida ordenando la lista xs
mediante mezclas; es decir, la divide en dos mitades, las ordena y las
mezcla. Por ejemplo,
----- *}

fun ordenaM :: "nat list  $\Rightarrow$  nat list" where
  "ordenaM [] = []"
| "ordenaM [x] = [x]"
| "ordenaM xs =
    (let mitad = length xs div 2 in
     mezcla (ordenaM (take mitad xs))
            (ordenaM (drop mitad xs)))"

value "ordenaM [3,2,5,2]" -- "= [2,2,3,5]"

text {*
-----
Ejercicio 6. Sea  $x \leq y$ . Si y es menor o igual que todos los elementos
de xs, entonces x es menor o igual que todos los elementos de xs
----- *}

lemma menor_menor:
  "x  $\leq$  y  $\implies$  menor y xs  $\rightarrow$  menor x xs"
by (induct xs) auto

text {*
-----
Ejercicio 7. Demostrar que el número de veces que aparece n en la
mezcla de dos listas es igual a la suma del número de apariciones en
cada una de las listas
-----

```

```

----- *}

lemma cuenta_mezcla:
  "cuenta (mezcla xs ys) n = cuenta xs n + cuenta ys n"
by (induct xs ys rule: mezcla.induct) auto

text {*
-----
  Ejercicio 8. Demostrar que  $x$  es menor que todos los elementos de  $ys$  y
  de  $zs$ , entonces también lo es de su mezcla.
----- *}

lemma menor_mezcla:
  assumes "menor x ys"
    "menor x zs"
  shows "menor x (mezcla ys zs)"
using assms
by (induct ys zs rule: mezcla.induct) simp_all

text {*
-----
  Ejercicio 9. Demostrar que la mezcla de dos listas ordenadas es una
  lista ordenada.
  Indicación: Usar los siguientes lemas
  · linorder_not_le:  $(\neg x \leq y) = (y < x)$ 
  · order_less_le:  $(x < y) = (x \leq y \wedge x \neq y)$ 
----- *}

lemma ordenada_mezcla:
  assumes "ordenada xs"
    "ordenada ys"
  shows "ordenada (mezcla xs ys)"
using assms
by (induct xs ys rule: mezcla.induct)
  (auto simp add: menor_mezcla
    menor_menor
    linorder_not_le
    order_less_le)

text {*
-----
  Ejercicio 10. Demostrar que si  $x$  es mayor que 1, entonces el mínimo de
   $x$  y su mitad es menor que  $x$ .
-----

```

Indicación: Usar los siguientes lemas

```
· min_def:      min a b = (if a ≤ b then a else b)
· linorder_not_le: (¬ x ≤ y) = (y < x)
----- *
```

lemma min_mitad:

"1 < x $\implies$ min x (x div 2::nat) < x"

by (simp add: min_def linorder_not_le)

text {*

Ejercicio 11. Demostrar que si x es mayor que 1, entonces x menos su mitad es menor que x.

----- *

lemma menos_mitad:

"1 < x $\implies$ x - x div (2::nat) < x"

by arith

text {*

Ejercicio 11. Demostrar que (ordenaM xs) está ordenada.

----- *

theorem ordenada_ordenaM:

"ordenada (ordenaM xs)"

by (induct xs rule: ordenaM.induct)

(auto simp add: ordenada_mezcla)

text {*

Ejercicio 12. Demostrar que el número de apariciones de un elemento en la concatenación de dos listas es la suma del número de apariciones en cada una.

----- *

lemma cuenta_conc:

"cuenta (xs @ ys) x = cuenta xs x + cuenta ys x"

by (induct xs) auto

text {*

Ejercicio 13. Demostrar que las listas xs y (ordenaM xs) tienen los mismos elementos.

----- *

```

theorem cuenta_ordenaM:
  "cuenta (ordenaM xs) x = cuenta xs x"
by (induct xs rule: ordenaM.induct)
   (auto simp add: cuenta_mezcla
                  cuenta_conc [symmetric])

end

```

6.2. Ejercicios sobre árboles y otros tipos de datos inductivos

6.2.1. Recorridos de árboles

```
chapter {* T6R2a: Recorridos de árboles *
```

```

theory T6R2a
imports Main
begin

```

```

text {
  -----
  Ejercicio 1. Definir el tipo de datos arbol para representar los
  árboles binarios que tiene información en los nodos y en las hojas.
  Por ejemplo, el árbol
      e
     / \
    /   \
   /     \
  c       g
 / \     / \
a  d f   h
se representa por "N e (N c (H a) (H d)) (N g (H f) (H h))".
  -----
*}

```

```
datatype 'a arbol = H "'a" | N "'a" "'a arbol" "'a arbol"
```

```
value "N e (N c (H a) (H d)) (N g (H f) (H h))"
```

```

text {
  -----
  Ejercicio 2. Definir la función
  preOrden :: "'a arbol ⇒ 'a list"

```

tal que (preOrden a) es el recorrido pre orden del árbol a. Por ejemplo,

```
preOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))
= [e,c,a,d,g,f,h]
```

```
-----
*}
```

```
fun preOrden :: "'a arbol => 'a list" where
```

```
  "preOrden (H x)      = [x]"
```

```
| "preOrden (N x i d) = x#((preOrden i)@(preOrden d))"
```

```
value "preOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))"
```

```
-- "[e,c,a,d,g,f,h]"
```

```
text {*
```

```
-----
Ejercicio 3. Definir la función
```

```
  postOrden :: "'a arbol => 'a list"
```

tal que (postOrden a) es el recorrido post orden del árbol a. Por ejemplo,

```
postOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))
= [e,c,a,d,g,f,h]
```

```
-----
*}
```

```
fun postOrden :: "'a arbol => 'a list" where
```

```
  "postOrden (H x)      = [x]"
```

```
| "postOrden (N x i d) = (postOrden i)@(postOrden d)@[x]"
```

```
value "postOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))"
```

```
-- "[a,d,c,f,h,g,e]"
```

```
text {*
```

```
-----
Ejercicio 4. Definir la función
```

```
  inOrden :: "'a arbol => 'a list"
```

tal que (inOrden a) es el recorrido in orden del árbol a. Por ejemplo,

```
inOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))
= [a,c,d,e,f,g,h]
```

```
-----
*}
```

```
fun inOrden :: "'a arbol => 'a list" where
```

```
  "inOrden (H x)      = [x]"
```

```
| "inOrden (N x i d) = (inOrden i)@[x]@(inOrden d)"

value "inOrden (N e (N c (H a) (H d)) (N g (H f) (H h)))"
-- "[a,c,d,e,f,g,h]"

text {*
-----
Ejercicio 5. Definir la función
    espejo :: "'a arbol ⇒ 'a arbol"
tal que (espejo a) es la imagen especular del árbol a. Por ejemplo,
    espejo (N e (N c (H a) (H d)) (N g (H f) (H h)))
    = N e (N g (H h) (H f)) (N c (H d) (H a))
-----
*}

fun espejo :: "'a arbol ⇒ 'a arbol" where
    "espejo (H x)      = (H x)"
| "espejo (N x i d) = (N x (espejo d) (espejo i))"

value "espejo (N e (N c (H a) (H d)) (N g (H f) (H h)))"
-- "N e (N g (H h) (H f)) (N c (H d) (H a))"

text {*
-----
Ejercicio 6. Demostrar que
    preOrden (espejo a) = rev (postOrden a)
-----
*}

lemma "preOrden (espejo a) = rev (postOrden a)"
by (induct a) auto

text {*
-----
Ejercicio 7. Demostrar que
    postOrden (espejo a) = rev (preOrden a)
-----
*}

lemma "postOrden (espejo a) = rev (preOrden a)"
by (induct a) auto

text {*
-----
Ejercicio 8. Demostrar que
```

```

    inOrden (espejo a) = rev (inOrden a)
-----
*}

theorem "inOrden (espejo a) = rev (inOrden a)"
by (induct a) auto

text {*
-----
Ejercicio 9. Definir la función
    raiz :: "'a arbol ⇒ 'a"
tal que (raiz a) es la raíz del árbol a. Por ejemplo,
    raiz (N e (N c (H a) (H d)) (N g (H f) (H h))) = e
-----
*}

fun raiz :: "'a arbol ⇒ 'a" where
  "raiz (H x)      = x"
| "raiz (N x i d) = x"

value "raiz (N e (N c (H a) (H d)) (N g (H f) (H h)))" -- "= e"

text {*
-----
Ejercicio 10. Definir la función
    extremo_izquierda :: "'a arbol ⇒ 'a"
tal que (extremo_izquierda a) es el nodo más a la izquierda del árbol
a. Por ejemplo,
    extremo_izquierda (N e (N c (H a) (H d)) (N g (H f) (H h))) = a
-----
*}

fun extremo_izquierda :: "'a arbol ⇒ 'a" where
  "extremo_izquierda (H a)      = a"
| "extremo_izquierda (N f x y) = (extremo_izquierda x)"

value "extremo_izquierda (N e (N c (H a) (H d)) (N g (H f) (H h)))" -- "= a"

text {*
-----
Ejercicio 11. Definir la función
    extremo_derecha :: "'a arbol ⇒ 'a"
tal que (extremo_derecha a) es el nodo más a la derecha del árbol
a. Por ejemplo,
    extremo_derecha (N e (N c (H a) (H d)) (N g (H f) (H h))) = h
-----

```

```

*}

fun extremo_derecha :: "'a arbol ⇒ 'a" where
  "extremo_derecha (H x)      = x"
| "extremo_derecha (N x i d) = (extremo_derecha d)"

value "extremo_derecha (N e (N c (H a) (H d)) (N g (H f) (H h)))" -- "= h"

text {*
-----
Ejercicio 12. Demostrar o refutar
  last (inOrden a) = extremo_derecha a
-----
*}

lemma [simp]: "inOrden a ≠ []"
by (induct a) auto

lemma [simp]: "ys ≠ [] → last (xs@ys) = last ys"
by (induct xs) auto

theorem "last (inOrden a) = extremo_derecha a"
by (induct a) auto

text {*
-----
Ejercicio 13. Demostrar o refutar
  hd (inOrden a) = extremo_izquierda a
-----
*}

theorem "hd (inOrden a) = extremo_izquierda a"
by (induct a) auto

text {*
-----
Ejercicio 14. Demostrar o refutar
  hd (preOrden a) = last (postOrden a)
-----
*}

theorem "hd (preOrden a) = last (postOrden a)"
by (induct a) auto

```

```

text {*
-----
  Ejercicio 15. Demostrar o refutar
    hd (preOrden a) = raiz a
-----
*}

theorem "hd (preOrden a) = raiz a"
by (induct a) auto

text {*
-----
  Ejercicio 16. Demostrar o refutar
    hd (inOrden a) = raiz a
-----
*}

theorem "hd (inOrden a) = raiz a"
quickcheck
oops

text {*
  Quickcheck found a counterexample:
  a = N a\<^isub>1 (H a\<^isub>2) (H a\<^isub>1)

  Evaluated terms:
  hd (inOrden a) = a\<^isub>2
  raiz a = a\<^isub>1
*}

text {*
-----
  Ejercicio 17. Demostrar o refutar
    last (postOrden a) = raiz a
-----
*}

theorem "last (postOrden a) = raiz a"
by (induct a) auto

end

```

6.2.2. Plegados de listas y de árboles

```
chapter {* T6R2b: Plegados de listas y de árboles *}
```

```
theory T6R2b
```

```
imports Main
```

```
begin
```

```
section {* Nuevas funciones sobre listas *}
```

```
text {*
```

```
  Nota. En esta relación se usará la función suma tal que (suma xs) es
  la suma de los elementos de xs, definida por
*}
```

```
fun suma :: "nat list ⇒ nat" where
```

```
  "suma []          = 0"
```

```
| "suma (x # xs) = x + suma xs"
```

```
text {*
```

```
  Las funciones de plegado, foldr y foldl, están definidas en la teoría
  List.thy por
```

```
  foldr :: "('a ⇒ 'b ⇒ 'b) ⇒ 'a list ⇒ 'b ⇒ 'b"
```

```
  foldr f []          = id
```

```
  foldr f (x # xs) = f x ◦ foldr f xs
```

```
  foldl :: "('b ⇒ 'a ⇒ 'b) ⇒ 'b ⇒ 'a list ⇒ 'b"
```

```
  foldl f a []        = a
```

```
  foldl f a (x # xs) = foldl f (f a x) xs"
```

```
  Por ejemplo,
```

```
  foldr (op +) [a,b,c] d          = a + (b + (c + d))
```

```
  foldl (op +) d [a,b,c]          = ((d + a) + b) + c
```

```
  foldr (op -) [9,4,2] (0::int) = 7
```

```
  foldl (op -) (0::int) [9,4,2] = -15
```

```
*}
```

```
value "foldr (op +) [a,b,c] d"          -- "= a + (b + (c + d))"
```

```
value "foldl (op +) d [a,b,c]"          -- "= ((d + a) + b) + c"
```

```
value "foldr (op -) [9,4,2] (0::int)" -- "= 7"
```

```
value "foldl (op -) (0::int) [9,4,2]" -- "= -15"
```

```
text {*
```

```
  -----
  Ejercicio 1. Demostrar que
```

```
  suma xs = foldr (op +) xs 0
```

```

*}

```

```

lemma suma_foldr: "suma xs = foldr (op +) xs 0"
by (induct xs) auto

```

```

text {*

```

```

-----
Ejercicio 2. Demostrar que
  length xs = foldr (λ x res. 1 + res) xs 0
-----

```

```

*}

```

```

lemma length_foldr: "length xs = foldr (λ x res. 1 + res) xs 0"
by (induct xs) auto

```

```

text {*

```

```

-----
Ejercicio 3. La aplicación repetida de foldr y map tiene el
inconveniente de que la lista se recorre varias veces. Sin embargo, es
suficiente recorrerla una vez como se muestra en el siguiente ejemplo,
  suma (map (λx. x + 3) xs) = foldr h xs b
Determinar los valores de h y b para que se verifique la igualdad
anterior y demostrarla.
-----

```

```

*}

```

```

text {* Basta tomar (λ x y. x + y + 3) como h y 0 como b *}
lemma "suma (map (λx. x + 3) xs) = foldr (λ x y. x + y + 3) xs 0"
by (induct xs) auto

```

```

text {*

```

```

-----
Ejercicio 4. Generalizar el resultado anterior; es decir determinar
los valores de h y b para que se verifique la igualdad
  foldr g (map f xs) a = foldr h xs b
y demostrarla.
-----

```

```

*}

```

```

text {* Basta tomar (λx acc. g (f x) acc) como h y a como b *}
lemma "foldr g (map f xs) a = foldr (λx acc. g (f x) acc) xs a"
by (induct xs) auto

```

```

text {*

```

```

-----
Ejercicio 5. La siguiente función invierte una lista en tiempo lineal
  fun inversa_ac :: "'a list, 'a list] ⇒ 'a list" where
    "inversa_ac [] ys = ys"
  | "inversa_ac (x#xs) ys = (inversa_ac xs (x#ys))"

  definition inversa_ac :: "'a list ⇒ 'a list" where
    "inversa_ac xs ≡ inversa_ac_aux xs []"
Por ejemplo,
  inversa_ac [a,d,b,c] = [c, b, d, a]
Demostrar que inversa_ac se puede definir usando foldl.
-----
*}

fun inversa_ac_aux :: "'a list, 'a list] ⇒ 'a list" where
  "inversa_ac_aux [] ys = ys"
| "inversa_ac_aux (x#xs) ys = (inversa_ac_aux xs (x#ys))"

definition inversa_ac :: "'a list ⇒ 'a list" where
  "inversa_ac xs ≡ inversa_ac_aux xs []"

value "inversa_ac [a,d,b,c]" -- "= [c, b, d, a]"

lemma inversa_ac_aux_foldl_aux:
  "inversa_ac_aux xs a = foldl (λ ys x. x # ys) a xs"
by (induct xs arbitrary: a) auto

corollary inversa_ac_aux_foldl:
  "inversa_ac_aux xs a = foldl (λ ys x. x # ys) a xs"
by (rule inversa_ac_aux_foldl_aux)

text {*
-----
Ejercicio 6. Demostrar la siguiente propiedad distributiva de la suma
sobre la concatenación:
  suma (xs @ ys) = suma xs + suma ys
-----
*}

lemma suma_append [simp]:
  "suma (xs @ ys) = suma xs + suma ys"
by (induct xs) auto

text {*
-----

```

Ejercicio 7. Demostrar una propiedad similar para foldr
 $\text{foldr } f \text{ (xs @ ys) } a = f \text{ (foldr } f \text{ xs } a) \text{ (foldr } f \text{ ys } a)$
 En este caso, hay que restringir el resultado teniendo en cuenta
 propiedades algebraicas de f y a .

 *}

text *{* Definición de a es neutro por la izquierda de f **
definition neutro_izquierda :: "[a ⇒ 'b ⇒ 'b, 'a] ⇒ bool" **where**
 "neutro_izquierda f a ≡ (∀x. (f a x = x))"

text *{* Definición de f es asociativa **
definition asociativa :: "[a ⇒ 'a ⇒ 'a] ⇒ bool" **where**
 "asociativa f ≡ (∀ x y z. f (f x y) z = f x (f y z))"

lemma foldr_append:
assumes "neutro_izquierda f a"
 "asociativa f"
shows "foldr f (xs @ ys) a = f (foldr f xs a) (foldr f ys a)"
using assms
by (induct xs) (auto simp add: neutro_izquierda_def asociativa_def)

text *{**

Ejercicio 8. Definir, usando foldr, la función
 $\text{prod} :: \text{"nat list} \Rightarrow \text{nat}"$
tal que (prod xs) es el producto de los elementos de xs. Por ejemplo,

 *}

definition prod :: "nat list ⇒ nat" **where**
 "prod xs ≡ foldr (op *) xs 1"

value "prod [2::nat,3,5]" -- "= 30"

text *{**

Ejercicio 9. Demostrar directamente (es decir, sin inducción) que
 $\text{prod (xs @ ys)} = \text{prod xs} * \text{prod ys}$

 *}

lemma "prod (xs @ ys) = prod xs * prod ys"
unfolding prod_def
by (rule foldr_append,

```

    simp add: neutro_izquierda_def,
    simp add: asociativa_def)

-- "Una demostración aplicativa del lema anterior es"
lemma "prod (xs @ ys) = prod xs * prod ys"
  apply (simp only: prod_def)
  apply (rule foldr_append)
  apply (simp add: neutro_izquierda_def)
  apply (simp add: asociativa_def)
done

section {* Functions on Trees *}

text {*
  -----
  Ejercicio 10. Definir el tipo de datos arbol para representar los
  árboles binarios que tiene información sólo en los nodos. Por ejemplo,
  el árbol
      e
     /\
    /\ 
   c  g
  /\  /\
 .  . . .
  se representa por "N (N H c H) e (N H g H)"
  -----
*}

datatype 'a arbol = H | N "'a arbol" "'a" "'a arbol"

value "N (N H c H) e (N H g H)"

text {*
  -----
  Ejercicio 11. Definir la función
    preOrden :: "'a arbol ⇒ 'a list"
  tal que (preOrden a) es el recorrido pre orden del árbol a. Por
  ejemplo,
    preOrden (N (N H c H) e (N H g H))
      = [e,c,g]
  -----
*}

fun preOrden :: "'a arbol ⇒ 'a list" where
  "preOrden H          = []"
```

```

| "preOrden (N i x d) = x#((preOrden i)@(preOrden d))"

value "preOrden (N (N H c H) e (N H g H))"
-- "= [e,c,g]"

text {*
-----
Ejercicio 12. Definir la función
  postOrden :: "'a arbol ⇒ 'a list"
tal que (postOrden a) es el recorrido post orden del árbol a. Por
ejemplo,
  postOrden (N (N H c H) e (N H g H))
    = [c,g,e]
-----
*}

fun postOrden :: "'a arbol ⇒ 'a list" where
  "postOrden H          = []"
| "postOrden (N i x d) = (postOrden i)@(postOrden d)@[x]"

value "postOrden (N (N H c H) e (N H g H))"
-- "= [c,g,e]"

text {*
-----
Ejercicio 13. Definir, usando un acumulador, la función
  postOrdenA :: "'a arbol ⇒ 'a list"
tal que (postOrdenA a) es el recorrido post orden del árbol a. Por
ejemplo,
  postOrdenA (N (N H c H) e (N H g H))
    = [c,g,e]
-----
*}

fun postOrdenAux :: "[ 'a arbol, 'a list] ⇒ 'a list" where
  "postOrdenAux H          xs = xs"
| "postOrdenAux (N i x d) xs = (postOrdenAux i (postOrdenAux d (x#xs)))"

definition postOrdenA :: "'a arbol ⇒ 'a list" where
  "postOrdenA a ≡ postOrdenAux a []"

value "postOrdenA (N (N H c H) e (N H g H))"
-- "= [c,g,e]"

text {*

```

```

-----
Ejercicio 14. Demostrar que
  postOrdenAux a xs = (postOrden a) @ xs
-----
*}

lemma postOrdenA:
  "postOrdenAux a xs = (postOrden a) @ xs"
by (induct a arbitrary: xs) auto

corollary "postOrdenAux a xs = (postOrden a) @ xs"
by (rule postOrdenA)

text {*
-----
Ejercicio 15. Definir la función
  foldl_arbol :: "('b => 'a => 'b) => 'b => 'a arbol => 'b" where
  tal que (foldl_arbol f b a) es el plegado izquierdo del árbol a con la
  operación f y elemento inicial b.
-----
*}

fun foldl_arbol :: "('b => 'a => 'b) => 'b => 'a arbol => 'b" where
  "foldl_arbol f b H          = b"
| "foldl_arbol f b (N i x d) = (foldl_arbol f (foldl_arbol f (f b x) d) i)"

text {*
-----
Ejercicio 16. Demostrar que
  postOrdenAux t a = foldl_arbol (λ xs x. Cons x xs) a t
-----
*}

lemma "postOrdenAux t a = foldl_arbol (λ xs x. Cons x xs) a t"
by (induct t arbitrary: a) auto

text {*
-----
Ejercicio 17. Definir la función
  suma_arbol :: "nat arbol => nat"
  tal que (suma_arbol a) es la suma de los elementos del árbol de
  números naturales a.
-----
*}

```

```

fun suma_arbol :: "nat arbol  $\Rightarrow$  nat" where
  "suma_arbol H          = 0"
| "suma_arbol (N i x d) = (suma_arbol i) + x + (suma_arbol d)"

```

```

text {*
-----
  Ejercicio 18. Demostrar que
    suma_arbol a = suma (preOrden a)
-----
*}

```

```

lemma "suma_arbol a = suma (preOrden a)"
by (induct a) auto

```

```
end
```

6.2.3. Árboles binarios completos

```
chapter {* T6R2c: Árboles binarios completos *
```

```

theory T6R2c
imports Main
begin

```

```

text {*
-----
  Ejercicio 1. Definir el tipo de datos arbol para representar los
  árboles binarios que no tienen información ni en los nodos y ni en las
  hojas. Por ejemplo, el árbol
      .
     / \
    /   \
   .     .
  / \   / \
 .  . . . .
se representa por "N (N H H) (N H H)".
-----
*}

```

```
datatype arbol = H | N arbol arbol
```

```
value "N (N H H) (N H H)"
```

```

text {*
-----
Ejercicio 2. Definir la función
  hojas :: "arbol => nat"
tal que (hojas a) es el número de hojas del árbol a. Por ejemplo,
  hojas (N (N H H) (N H H)) = 4
-----
*}

fun hojas :: "arbol => nat" where
  "hojas H          = 1"
| "hojas (N i d) = hojas i + hojas d"

value "hojas (N (N H H) (N H H))" -- "= 4"

text {*
-----
Ejercicio 4. Definir la función
  profundidad :: "arbol => nat"
tal que (profundidad a) es la profundidad del árbol a. Por ejemplo,
  profundidad (N (N H H) (N H H)) = 2
-----
*}

fun profundidad :: "arbol => nat" where
  "profundidad H          = 0"
| "profundidad (N i d) = 1 + max (profundidad i) (profundidad d)"

value "profundidad (N (N H H) (N H H))" -- "= 2"

text {*
-----
Ejercicio 5. Definir la función
  abc :: "nat => arbol"
tal que (abc n) es el árbol binario completo de profundidad n. Por
ejemplo,
  abc 3 = N (N (N H H) (N H H)) (N (N H H) (N H H))
-----
*}

fun abc :: "nat => arbol" where
  "abc 0          = H"
| "abc (Suc n) = N (abc n) (abc n)"

value "abc 3" -- "= N (N (N H H) (N H H)) (N (N H H) (N H H))"

```

```

text {*
-----
Ejercicio 6. Un árbol binario  $a$  es completo respecto de la medida  $f$  si
 $a$  es una hoja o bien  $a$  es de la forma  $(N\ i\ d)$  y se cumple que tanto  $i$ 
como  $d$  son árboles binarios completos respecto de  $f$  y, además,
 $f(i) = f(r)$ .

Definir la función
    es_abc :: "(arbol => 'a) => arbol => bool
tal que (es_abc f a) se verifica si  $a$  es un árbol binario completo
respecto de  $f$ .
-----
*}

fun es_abc :: "(arbol => 'a) => arbol => bool" where
  "es_abc f H      = True"
| "es_abc f (N i d) = (es_abc f i ∧ es_abc f d ∧ f i = f d)"

text {*
-----
Nota. (size a) es el número de nodos del árbol  $a$ . Por ejemplo,
    size (N (N H H) (N H H)) = 3
-----
*}

value "size (N (N H H) (N H H))"
value "size (N (N (N H H) (N H H)) (N (N H H) (N H H)))"

text {*
-----
Nota. Tenemos 3 funciones de medida sobre los árboles: número de
hojas, número de nodos y profundidad. A cada una le corresponde un
concepto de completitud. En los siguientes ejercicios demostraremos
que los tres conceptos de completitud son iguales.
-----
*}

text {*
-----
Ejercicio 7. Demostrar que un árbol binario  $a$  es completo respecto de
la profundidad  $sys$  es completo respecto del número de hojas.
-----
*}

```

```

text {* Si a es un árbol completo respecto de la profundidad, entonces
    el número de hojas de a es 2 elevado a la profundidad de a. *}
lemma [simp]: "es_abc profundidad a → hojas a = 2 ^ (profundidad a)"
by (induct a) auto

theorem es_abc_profundidad_hojas: "es_abc profundidad a = es_abc hojas a"
by (induct a) auto

text {*
-----
Ejercicio 8. Demostrar que un árbol binario a es completo respecto del
número de hojas syss es completo respecto del número de nodos
-----
*}

text {* El número de hojas de un árbol binario a es igual al número de
    nodos da a más 1. *}
lemma [simp]: "hojas a = size a + 1"
by (induct a) auto

theorem es_abc_hojas_size: "es_abc hojas a = es_abc size a"
by (induct a) auto

text {*
-----
Ejercicio 9. Demostrar que un árbol binario a es completo respecto de
la profundidad syss es completo respecto del número de nodos
-----
*}

corollary es_abc_size_profundidad: "es_abc size a = es_abc profundidad a"
by (simp add: es_abc_profundidad_hojas es_abc_hojas_size)

text {*
-----
Ejercicio 10. Demostrar que (abc n) es un árbol binario completo.
-----
*}

lemma "es_abc f (abc n)"
by (induct n) auto

text {*
-----

```

Ejercicio 11. Demostrar que si a es un árbol binario completo respecto de la profundidad, entonces a es $(abc \text{ (profundidad } a))$.

```

*}

```

```

theorem "es_abc profundidad a  $\rightarrow$  a = abc (profundidad a)"
by (induct a) auto

```

```

text {*

```

Ejercicio 12. Encontrar una medida f tal que $(es_abc \ f)$ es distinto de $(es_abc \ size)$.

```

*}

```

```

text {* Basta tomar como  $f$  la función constante 0. *}

```

```

value "es_abc size (N H (N H H))" -- "= False"
value "es_abc ( $\lambda t. 0$ ) (N H (N H H))" -- "= True"

```

```

lemma "es_abc ( $\lambda t. 0::nat$ )  $\neq$  es_abc size"

```

```

proof

```

```

  assume "es_abc ( $\lambda t. 0::nat$ ) = es_abc size"

```

```

  hence "(es_abc ( $\lambda t. 0::nat$ ) (N H (N H H))) = (es_abc size (N H (N H H)))"

```

```

    by (simp add: fun_eq_iff)

```

```

  thus False by simp

```

```

qed

```

```

text {*

```

Referencia: Este ejercicio es una adaptación del de Tobias Nipkow "Complete Binary Trees" que se encuentra en <http://isabelle.in.tum.de/exercises/trees/complete/ex.pdf>

```

*}

```

```

end

```

6.2.4. Diagramas de decisión binarios

```

chapter {* T6R2d: Diagramas de decisión binarios *}

```

```

theory T6R2d

```

```

imports Main

```

```

begin

```


```

text {*

```

Las funciones booleanas se pueden representar mediante diagramas de decisión binarios (DDB). Por ejemplo, la función f definida por la tabla de la izquierda se representa por el DDB de la derecha

+-----+				
p	q	r	f(p,q,r)	
+-----+				
F	F	*	V	
F	V	*	F	
V	F	*	F	
V	V	F	F	
V	V	V	V	
+-----+				



Para cada variable, si su valor es falso se evalúa su hijo izquierdo y si es verdadero se evalúa su hijo derecho.

```
*}
```

```
text {*
```

```
-----
Ejercicio 1. Definir el tipo de datos ddb para representar los
diagramas de decisión binarios. Por ejemplo, el DDB anterior se
representa por
```

```
N (N (H True) (H False)) (N (H False) (N (H False) (H True)))
```

```
-----
*}
```

```
datatype ddb = H bool | N ddb ddb
```

```
value "N (N (H True) (H False)) (N (H False) (N (H False) (H True)))"
```

```
text {*
```

```
-----
Ejercicio 2. Definir ddb1 para representar el DDB del ejercicio 1.
-----
```

```
*}
```

```
abbreviation ddb1 :: ddb where
```

```
"ddb1 ≡ N (N (H True) (H False)) (N (H False) (N (H False) (H True)))"
```

```
text {*
```

```
-----
Ejercicio 3. Definir int1,..., int8 para representar las
interpretaciones del ejercicio 1.
-----
```

```
*}
```

```

abbreviation int1 :: "nat  $\Rightarrow$  bool" where
  "int1 x  $\equiv$  False"
abbreviation int2 :: "nat  $\Rightarrow$  bool" where
  "int2  $\equiv$  int1 (2 := True)"
abbreviation int3 :: "nat  $\Rightarrow$  bool" where
  "int3  $\equiv$  int1 (1 := True)"
abbreviation int4 :: "nat  $\Rightarrow$  bool" where
  "int4  $\equiv$  int1 (1 := True, 2 := True)"
abbreviation int5 :: "nat  $\Rightarrow$  bool" where
  "int5  $\equiv$  int1 (0 := True)"
abbreviation int6 :: "nat  $\Rightarrow$  bool" where
  "int6  $\equiv$  int1 (0 := True, 2 := True)"
abbreviation int7 :: "nat  $\Rightarrow$  bool" where
  "int7  $\equiv$  int1 (0 := True, 1 := True)"
abbreviation int8 :: "nat  $\Rightarrow$  bool" where
  "int8  $\equiv$  int1 (0 := True, 1 := True, 2 := True)"

```

```
text {*
```

Ejercicio 4. Definir la función

valor :: "(nat $\Rightarrow$ bool) $\Rightarrow$ nat $\Rightarrow$ ddb $\Rightarrow$ bool"

tal que (valor i n d) es el valor del DDB d en la interpretación i a partir de la variable de índice n. Por ejemplo,

valor int1 0 ddb1 = True

valor int2 0 ddb1 = True

valor int3 0 ddb1 = False

valor int4 0 ddb1 = False

valor int5 0 ddb1 = False

valor int6 0 ddb1 = False

valor int7 0 ddb1 = False

valor int8 0 ddb1 = True

```
*}
```

```

fun valor :: "(nat  $\Rightarrow$  bool)  $\Rightarrow$  nat  $\Rightarrow$  ddb  $\Rightarrow$  bool" where
  "valor i n (H x) = x"
| "valor i n (N b1 b2) =
  (if i n then valor i (Suc n) b2 else valor i (Suc n) b1)"

```

```
value "valor int1 0 ddb1" -- "= True"
```

```
value "valor int2 0 ddb1" -- "= True"
```

```
value "valor int3 0 ddb1" -- "= False"
```

```
value "valor int4 0 ddb1" -- "= False"
```

```
value "valor int5 0 ddb1" -- "= False"
```

```
value "valor int6 0 ddb1" -- "= False"
```

```

value "valor int7 0 ddb1" -- "= False"
value "valor int8 0 ddb1" -- "= True"

text {*
-----
Ejercicio 5. Definir la función
  ddb_op_un :: "(bool ⇒ bool) ⇒ ddb ⇒ ddb"
tal que (ddb_op_un f d) es el diagrama obtenido aplicando el operador
unitario f a cada hoja de DDB d de forma que se conserve el valor; es
decir,
  valor i n (ddb_op_un f d) = f (valor i n d)"
Por ejemplo,
  value "ddb_op_un (λx. ¬x) ddb1"
    = N (N (H False) (H True)) (N (H True) (N (H True) (H False)))
-----
*}

fun ddb_op_un :: "(bool ⇒ bool) ⇒ ddb ⇒ ddb" where
  "ddb_op_un f (H x) = H (f x)"
| "ddb_op_un f (N b1 b2) = N (ddb_op_un f b1) (ddb_op_un f b2)"

value "ddb_op_un (λx. ¬x) ddb1"
-- "= N (N (H False) (H True)) (N (H True) (N (H True) (H False)))"

text {*
-----
Ejercicio 6. Demostrar que la definición de ddb_op_un es correcta; es
decir,
  valor i n (ddb_op_un f d) = f (valor i n d)"
-----
*}

theorem ddb_op_un_correcto:
  "valor i n (ddb_op_un f d) = f (valor i n d)"
by (induct d arbitrary: i n) auto

text {*
-----
Ejercicio 7. Definir la función
  ddb_op_bin :: "(bool ⇒ bool ⇒ bool) ⇒ ddb ⇒ ddb ⇒ ddb"
tal que (ddb_op_bin f d1 d2) es el diagrama obtenido aplicando el
operador binario f a los DDB d1 y d2 de forma que se conserve el
valor; es decir,
  valor i n (ddb_op_bin f d1 d2) = f (valor i n d1) (valor i n d2)
-----

```

Por ejemplo,

```

ddb_op_bin (op ∧) ddb1 (N (H True) (H False))
= N (N (H True) (H False)) (N (H False) (N (H False) (H False)))
ddb_op_bin (op ∧) ddb1 (N (H False) (H True))
= N (N (H False) (H False)) (N (H False) (N (H False) (H True)))
ddb_op_bin (op ∨) ddb1 (N (H True) (H False))
= N (N (H True) (H True)) (N (H False) (N (H False) (H True)))
ddb_op_bin (op ∨) ddb1 (N (H False) (H True))
= N (N (H True) (H False)) (N (H True) (N (H True) (H True)))

```

*}

```

fun ddb_op_bin :: "(bool ⇒ bool ⇒ bool) ⇒ ddb ⇒ ddb ⇒ ddb" where
  "ddb_op_bin f (H x) d = ddb_op_un (f x) d"
| "ddb_op_bin f (N d1 d2) d = (case d of
    H x      ⇒ N (ddb_op_bin f d1 (H x)) (ddb_op_bin f d2 (H x))
  | N d1' d2' ⇒ N (ddb_op_bin f d1 d1') (ddb_op_bin f d2 d2'))"

```

```

value "ddb_op_bin (op ∧) ddb1 (N (H True) (H False))"
-- "N (N (H True) (H False)) (N (H False) (N (H False) (H False)))"
value "ddb_op_bin (op ∧) ddb1 (N (H False) (H True))"
-- "N (N (H False) (H False)) (N (H False) (N (H False) (H True)))"
value "ddb_op_bin (op ∨) ddb1 (N (H True) (H False))"
-- "N (N (H True) (H True)) (N (H False) (N (H False) (H True)))"
value "ddb_op_bin (op ∨) ddb1 (N (H False) (H True))"
-- "N (N (H True) (H False)) (N (H True) (N (H True) (H True)))"

```

text {*

```

-----
Ejercicio 8. Demostrar que la definición de ddb_op_bin es correcta;
es decir,
  valor i n (ddb_op_bin f d1 d2) = f (valor i n d1) (valor i n d2)
-----

```

*}

theorem ddb_op_bin_correcto:

```

  "valor i n (ddb_op_bin f d1 d2) = f (valor i n d1) (valor i n d2)"

```

proof (induct d1 arbitrary: n d2)

fix b n b2

```

show "valor i n (ddb_op_bin f (H b) b2) =
      f (valor i n (H b)) (valor i n b2)"

```

by (auto simp add: ddb_op_un_correcto)

next

fix d11 d12 n d2

```

assume "∧ n d2. valor i n (ddb_op_bin f d11 d2) =

```

```

        f (valor i n d11) (valor i n d2)"
    and "\n d2. valor i n (ddb_op_bin f d12 d2) =
        f (valor i n d12) (valor i n d2)"
    thus "valor i n (ddb_op_bin f (N d11 d12) d2) =
        f (valor i n (N d11 d12)) (valor i n d2)"
    by (cases d2) auto
qed

-- "Una demostración más detallada es"
theorem ddb_op_bin_correcto_2:
    "valor i n (ddb_op_bin f d1 d2) = f (valor i n d1) (valor i n d2)"
proof (induct d1 arbitrary: n d2)
    fix b n b2
    show "valor i n (ddb_op_bin f (H b) b2) =
        f (valor i n (H b)) (valor i n b2)"
    by (auto simp add: ddb_op_un_correcto)
next
    fix d11 d12 n d2
    assume HI1: "\n d2. valor i n (ddb_op_bin f d11 d2) =
        f (valor i n d11) (valor i n d2)"
    and HI2: "\n d2. valor i n (ddb_op_bin f d12 d2) =
        f (valor i n d12) (valor i n d2)"
    show "valor i n (ddb_op_bin f (N d11 d12) d2) =
        f (valor i n (N d11 d12)) (valor i n d2)"
    proof (cases d2)
        fix b
        assume "d2 = H b"
        thus "valor i n (ddb_op_bin f (N d11 d12) d2) =
            f (valor i n (N d11 d12)) (valor i n d2)"
            using HI1 HI2 by auto
    next
        fix d21 d22
        assume "d2 = N d21 d22"
        thus "valor i n (ddb_op_bin f (N d11 d12) d2) =
            f (valor i n (N d11 d12)) (valor i n d2)"
            using HI1 HI2 by auto
    qed
qed

text {*
    -----
    Ejercicio 9. Definir la función
    ddb_and :: "ddb ⇒ ddb ⇒ ddb"
    tal que (ddb_and d1 d2) es el diagrama correspondiente a la conjunción
    de los DDB d1 y d2 de forma que se conserva el valor; es decir,

```

```

    valor i n (ddb_and d1 d2) = (valor i n d1 ∧ valor i n d2)
Por ejemplo,
    ddb_and ddb1 (N (H True) (H False))
  = N (N (H True) (H False)) (N (H False) (N (H False) (H False)))
    ddb_and ddb1 (N (H False) (H True))
  = N (N (H False) (H False)) (N (H False) (N (H False) (H True)))
-----
*}

definition ddb_and :: "ddb ⇒ ddb ⇒ ddb" where
  "ddb_and ≡ ddb_op_bin (op ∧)"

value "ddb_and ddb1 (N (H True) (H False))"
-- "= N (N (H True) (H False)) (N (H False) (N (H False) (H False)))"
value "ddb_and ddb1 (N (H False) (H True))"
-- "= N (N (H False) (H False)) (N (H False) (N (H False) (H True)))"

text {*
-----
Ejercicio 10. Demostrar que la definición de ddb_and es correcta;
es decir,
    valor i n (ddb_and d1 d2) = (valor i n d1 ∧ valor i n d2)
-----
*}

theorem ddb_and_correcta:
  "valor i n (ddb_and d1 d2) = (valor i n d1 ∧ valor i n d2)"
by (auto simp add: ddb_and_def ddb_op_bin_correcto)

text {*
-----
Ejercicio 11. Definir la función
    ddb_or :: "ddb ⇒ ddb ⇒ ddb"
tal que (ddb_or d1 d2) es el diagrama correspondiente a la disyunción
de los DDB d1 y d2 de forma que se conserva el valor; es decir,
    valor i n (ddb_or d1 d2) = (valor i n d1 ∨ valor i n d2)
Por ejemplo,
    ddb_or ddb1 (N (H True) (H False))
  = N (N (H True) (H True)) (N (H False) (N (H False) (H True)))
    ddb_or ddb1 (N (H False) (H True))
  = N (N (H True) (H False)) (N (H True) (N (H True) (H True)))
-----
*}

definition ddb_or :: "ddb ⇒ ddb ⇒ ddb" where

```

```

"ddb_or ≡ ddb_op_bin (op v)"

value "ddb_or ddb1 (N (H True) (H False))"
-- "= N (N (H True) (H True)) (N (H False) (N (H False) (H True)))"
value "ddb_or ddb1 (N (H False) (H True))"
-- "= N (N (H True) (H False)) (N (H True) (N (H True) (H True)))"

text {*
-----
Ejercicio 12. Demostrar que la definición de ddb_or es correcta;
es decir,
    valor i n (ddb_or d1 d2) = (valor i n d1 v valor i n d2)
-----
*}

theorem ddb_or_correcta:
  "valor i n (ddb_or d1 d2) = (valor i n d1 v valor i n d2)"
by (auto simp add: ddb_or_def ddb_op_bin_correcto)

text {*
-----
Ejercicio 13. Definir la función
    ddb_not :: "ddb ⇒ ddb"
tal que (ddb_not d) es el diagrama correspondiente a la negación
del DDB d de forma que se conserva el valor; es decir,
    valor i n (ddb_or d1 d2) = (valor i n d1 v valor i n d2)
Por ejemplo,
    ddb_not ddb1
    = N (N (H False) (H True)) (N (H True) (N (H True) (H False)))
-----
*}

definition ddb_not :: "ddb ⇒ ddb" where
  "ddb_not ≡ ddb_op_un Not"

value "ddb_not ddb1"
-- "= N (N (H False) (H True)) (N (H True) (N (H True) (H False)))"

text {*
-----
Ejercicio 14. Demostrar que la definición de ddb_not es correcta;
es decir,
    valor i n (ddb_not d) = (¬ valor i n d)
-----
*}

```

```

theorem ddb_not_correcta:
  "valor i n (ddb_not d) = ( $\neg$  valor i n d)"
by (auto simp add: ddb_not_def ddb_op_un_correcto)

text {*
  -----
  Ejercicio 15. Definir la función
    xor :: "bool  $\Rightarrow$  bool  $\Rightarrow$  bool"
  tal que (xor x y) es la disyunción excluyente de x e y. Por ejemplo,
    xor True False = True
    xor True True  = False
  -----
  *}

definition xor :: "bool  $\Rightarrow$  bool  $\Rightarrow$  bool" where
  "xor x y  $\equiv$  (x  $\wedge$   $\neg$ y)  $\vee$  ( $\neg$ x  $\wedge$  y)"

value "xor True False" -- "= True"
value "xor True True"  -- "= False"

text {*
  -----
  Ejercicio 16. Definir la función
    ddb_xor :: "ddb  $\Rightarrow$  ddb  $\Rightarrow$  ddb"
  tal que (ddb_xor d1 d2) es el diagrama correspondiente a la disyunción
  excluyente de los DDB d1 y d2. Por ejemplo,
    ddb_xor ddb1 (N (H True) (H False))
    = N (N (H True) (H True)) (N (H False) (N (H False) (H True)))
    ddb_xor ddb1 (N (H False) (H True))
    = N (N (H True) (H False)) (N (H True) (N (H True) (H True)))
  -----
  *}

definition ddb_xor :: "ddb  $\Rightarrow$  ddb  $\Rightarrow$  ddb" where
  "ddb_xor  $\equiv$  ddb_op_bin xor"

value "ddb_xor ddb1 (N (H True) (H False))"
-- "= N (N (H False) (H True)) (N (H False) (N (H False) (H True)))"
value "ddb_xor ddb1 (N (H False) (H True))"
-- "= N (N (H True) (H False)) (N (H True) (N (H True) (H False)))"

text {*
  -----
  Ejercicio 17. Demostrar que la definición de ddb_xor es correcta;

```

```

    es decir,
      valor i n (ddb_xor d1 d2) = xor (valor i n d1) (valor i n d2)
    -----
  *}

theorem ddb_xor_correcta:
  "valor i n (ddb_xor d1 d2) = xor (valor i n d1) (valor i n d2)"
by (auto simp add: ddb_xor_def ddb_op_bin_correcto)

text {*
  -----
  Ejercicio 18. Definir la función
    ddb_var :: "nat ⇒ ddb" where
  tal que (ddb_var n) es el diagrama equivalente a la variable p(n). Por
  ejemplo,
    ddb_var 0
    = N (H False) (H True)
    ddb_var 1
    = N (N (H False) (H True)) (N (H False) (H True))
  -----
  *}

fun ddb_var :: "nat ⇒ ddb" where
  "ddb_var 0      = N (H False) (H True)"
| "ddb_var (Suc i) = N (ddb_var i) (ddb_var i)"

value "ddb_var 0"
-- "= N (H False) (H True)"
value "ddb_var 1"
-- "= N (N (H False) (H True)) (N (H False) (H True))"

text {*
  -----
  Ejercicio 19. Demostrar que la definición de ddb_var es correcta;
  es decir,
    "valor i 0 (ddb_var n) = i n
  -----
  *}

-- "Para probarlo, primero hay que generalizarlo"
theorem ddb_var_correcta:
  "valor i m (ddb_var n) = i (n+m)"
by (induct n arbitrary: m) auto

corollary ddb_var_correcta_2:

```

```

"valor i 0 (ddb_var n) = i n"
by (simp add: ddb_var_correcta)

text {*
-----
Ejercicio 20. Definir el tipo de las fórmulas proposicionales
contruidas con la constante T, las variables (Var n) y las conectivas
Not, And, Or y Xor.
-----
*}

datatype form = T
            | Var nat
            | Not form
            | And form form
            | Or form form
            | Xor form form

text {*
-----
Ejercicio 21. Definir la función
    valor_fla :: "(nat ⇒ bool) ⇒ form ⇒ bool"
tal que (valor_fla i f) es el valor de la fórmula f en la
interpretación i. Por ejemplo,
    valor_fla (λn. True) (Xor T T)          = False
    valor_fla (λn. False) (Xor T (Var 1)) = True
-----
*}

fun valor_fla :: "(nat ⇒ bool) ⇒ form ⇒ bool" where
  "valor_fla i T          = True"
| "valor_fla i (Var n)    = i n"
| "valor_fla i (Not f)    = (¬(valor_fla i f))"
| "valor_fla i (And f1 f2) = (valor_fla i f1 ∧ valor_fla i f2)"
| "valor_fla i (Or f1 f2)  = (valor_fla i f1 ∨ valor_fla i f2)"
| "valor_fla i (Xor f1 f2) = xor (valor_fla i f1) (valor_fla i f2)"

value "valor_fla (λn. True) (Xor T T)"          -- "= False"
value "valor_fla (λn. False) (Xor T (Var 1))" -- "= True"

text {*
-----
Ejercicio 22. Definir la función
    ddb_fla :: "form ⇒ ddb"
tal que (ddb_fla f) es el DDB equivalente a la fórmula f; es decir,

```

```

    valor i 0 (ddb_flg f) = valor_flg i f
-----
*}

fun ddb_flg :: "form  $\Rightarrow$  ddb" where
  "ddb_flg T          = H True"
| "ddb_flg (Var i)    = ddb_var i"
| "ddb_flg (Not f)     = ddb_not (ddb_flg f)"
| "ddb_flg (And f1 f2) = ddb_and (ddb_flg f1) (ddb_flg f2)"
| "ddb_flg (Or  f1 f2) = ddb_or  (ddb_flg f1) (ddb_flg f2)"
| "ddb_flg (Xor f1 f2) = ddb_xor (ddb_flg f1) (ddb_flg f2)"

text {*
-----
Ejercicio 23. Demostrar que la definición de ddb_flg es correcta; es decir,
    valor i 0 (ddb_flg f) = valor_flg i f
-----
*}

theorem ddb_flg_correcta:
  "valor e 0 (ddb_flg f) = valor_flg e f"
by (induct f) (auto simp add: ddb_var_correcta
                             ddb_and_correcta
                             ddb_or_correcta
                             ddb_not_correcta
                             ddb_xor_correcta)

text {*
Referencias:
  · S. Berghofer "Binary decision diagrams". En
    http://isabelle.in.tum.de/exercises/trees/bdd/ex.pdf
  · J.A. Alonso, F.J. Martín y J.L. Ruiz "Diagramas de decisión binarios". En
    http://www.cs.us.es/cursos/lp-2005/temas/tema-07.pdf
  · Wikipedia "Binary decision diagram". En
    http://en.wikipedia.org/wiki/Binary\_decision\_diagram
*}

end

```

6.2.5. Representación de fórmulas proposicionales mediante polinomios

```
chapter {* T6R2e: Representación de fórmulas proposicionales mediante
  polinomios *
```

```
theory T6R2e
```

```
imports Main
```

```
begin
```

```
text {*
```

```
  El objetivo de esta relación es definir un procedimiento para
  transformar fórmulas proposicionales (construidas con  $\top$ ,  $\wedge$  y  $\oplus$ ) en
  polinomios de la forma
```

```
     $(p_1 \wedge \dots \wedge p_n) \oplus \dots \oplus (q_1 \wedge \dots \wedge q_m)$ 
```

```
  y demostrar que, para cualquier interpretación  $I$ , el valor de las
  fórmulas coincide con la de su correspondiente polinomio. *}

```

```
text {*
```

```
  -----
  Ejercicio 1. Las fórmulas proposicionales pueden definirse mediante
  las siguientes reglas:
```

- $\top$ es una fórmula proposicional
- Las variables proposicionales $p_1, p_2, \dots$ son fórmulas proposicionales,
- Si F y G son fórmulas proposicionales, entonces $(F \wedge G)$ y $(F \oplus G)$ también lo son.

```
  donde  $\top$  es una fórmula que siempre es verdadera,  $\wedge$  es la conjunción y
   $\oplus$  es la disyunción exclusiva.
```

```
  Definir el tipo de datos form para representar las fórmulas
  proposicionales usando
```

- T en lugar de $\top$,
- $(Var\ i)$ en lugar de p_i ,
- $(And\ F\ G)$ en lugar de $(F \wedge G)$ y
- $(Xor\ F\ G)$ en lugar de $(F \oplus G)$.

```
  ----- *}

```

```
datatype form = T | Var nat | And form form | Xor form form
```

```
text {*
```

```
  -----
  Ejercicio 2. Los siguientes ejemplos de fórmulas
```

```
    form1 =  $p_0 \oplus \top$ 
```

```
    form2 =  $(p_0 \oplus p_1) \oplus (p_0 \wedge p_1)$ 
```

```

----- *}

abbreviation form1 :: "form" where
  "form1  $\equiv$  Xor (Var 0) T"
abbreviation form2 :: "form" where
  "form2  $\equiv$  Xor (Xor (Var 0) (Var 1)) (And (Var 0) (Var 1))"

text {*
-----
  Ejercicio 3. Definir la función
    xor :: bool  $\Rightarrow$  bool  $\Rightarrow$  bool
  tal que (xor p q) es el valor de la disyunción exclusiva de p y q. Por
  ejemplo,
    xor False True = True
----- *}

definition xor :: "bool  $\Rightarrow$  bool  $\Rightarrow$  bool" where
  "xor x y  $\equiv$  (x  $\wedge$   $\neg$ y)  $\vee$  ( $\neg$ x  $\wedge$  y)"

value "xor False True" -- "= True"

text {*
-----
  Ejercicio 4. Una interpretación es una aplicación de los naturales en
  los booleanos. Definir las siguientes interpretaciones
    | p0 | p1 | p2 | p3 | ...
  int1 | F | F | F | F | ...
  int2 | F | V | F | F | ...
  int3 | V | F | F | F | ...
  int3 | V | V | F | F | ...
----- *}

abbreviation int1 :: "nat  $\Rightarrow$  bool" where
  "int1 x  $\equiv$  False"
abbreviation int2 :: "nat  $\Rightarrow$  bool" where
  "int2  $\equiv$  int1 (1 := True)"
abbreviation int3 :: "nat  $\Rightarrow$  bool" where
  "int3  $\equiv$  int1 (0 := True)"
abbreviation int4 :: "nat  $\Rightarrow$  bool" where
  "int4  $\equiv$  int1 (0 := True, 1 := True)"

text {*
-----
  Ejercicio 5. Dada una interpretación I, el valor de de una fórmula F
  respecto de I, I(F), se define por

```

- T , si F es T ;
- $I(n)$, si F es p_n ;
- $I(G) \wedge I(H)$, si F es $(G \wedge H)$;
- $I(G) \oplus I(H)$, si F es $(G \oplus H)$.

Definir la función

$\text{valorF} :: (\text{nat} \Rightarrow \text{bool}) \Rightarrow \text{form} \Rightarrow \text{bool}$

tal que $(\text{valorF } i \ f)$ es el valor de la fórmula f respecto de la interpretación i . Por ejemplo,

$\text{valorF } \text{int1 } \text{form1} = \text{True}$

$\text{valorF } \text{int3 } \text{form1} = \text{False}$

$\text{valorF } \text{int1 } \text{form2} = \text{False}$

$\text{valorF } \text{int2 } \text{form2} = \text{True}$

$\text{valorF } \text{int3 } \text{form2} = \text{True}$

$\text{valorF } \text{int4 } \text{form2} = \text{True}$

----- *

```
fun valorF :: "(nat ⇒ bool) ⇒ form ⇒ bool" where
  "valorF i T = True"
| "valorF i (Var n) = i n"
| "valorF i (And f g) = (valorF i f ∧ valorF i g)"
| "valorF i (Xor f g) = xor (valorF i f) (valorF i g)"
```

```
value "valorF int1 form1" -- "= True"
value "valorF int3 form1" -- "= False"
value "valorF int1 form2" -- "= False"
value "valorF int2 form2" -- "= True"
value "valorF int3 form2" -- "= True"
value "valorF int4 form2" -- "= True"
```

text {*

Ejercicio 6. Un monomio es una lista de números naturales y se puede interpretar como la conjunción de variables proposicionales cuyos índices son los números de la lista. Por ejemplo, el monomio $[0,2,1]$ se interpreta como la fórmula $(p_0 \wedge p_2 \wedge p_1)$.

Definir la función

$\text{formM} :: \text{nat list} \Rightarrow \text{form}$

tal que $(\text{formM } m)$ es la fórmula correspondiente al monomio. Por ejemplo,

$\text{formM } [0,2,1] = \text{And } (\text{Var } 0) (\text{And } (\text{Var } 2) (\text{And } (\text{Var } 1) T))$

----- *

```
fun formM :: "nat list ⇒ form" where
```

```

"formM []      = T"
| "formM (n#ns) = And (Var n) (formM ns)"

value "formM [0,2,1]" -- "= And (Var 0) (And (Var 2) (And (Var 1) T))"

text {*
-----
Ejercicio 7. Definir, por recursión, la función
  valorM :: (nat ⇒ bool) ⇒ nat list ⇒ bool
tal que (valorM i m) es el valor de la fórmula representada por el
monomio m en la interpretación i. Por ejemplo,
  valorM int1 [0,2,1]                                = False
  valorM (int1(0:=True,1:=True,2:=True)) [0,2,1] = True
Demostrar que, para toda interpretación i y todo monomio m, se tiene
que
  valorM i m = valorF i (formM m)
----- *}

fun valorM :: "(nat ⇒ bool) ⇒ nat list ⇒ bool" where
  "valorM i []      = True"
| "valorM i (x # xs) = (i x ∧ valorM i xs)"

value "valorM int1 [0,2,1]" -- "= False"
value "valorM (int1(0:=True,1:=True,2:=True)) [0,2,1]" -- "= True"

-- "La demostración automática es"
lemma correccion_valorM:
  "valorM i m = valorF i (formM m)"
by (induct m) auto

-- "La demostración estructurada es"
lemma correccion_valorM2:
  "valorM i m = valorF i (formM m)"
proof (induct m)
  show "valorM i [] = valorF i (formM [])" by simp
next
  fix x xs assume HI: "valorM i xs = valorF i (formM xs)"
  have "valorM i (x#xs) = (i x ∧ valorM i xs)" by simp
  also have "... = (i x ∧ (valorF i (formM xs)))" using HI by simp
  also have "... = ((valorF i (Var x)) ∧ (valorF i (formM xs)))" by simp
  also have "... = valorF i (And (Var x) (formM xs))" by simp
  also have "... = valorF i (formM (x#xs))" by simp
  finally show "valorM i (x#xs) = valorF i (formM (x#xs))" by simp
qed

```

```

text {*
-----
Ejercicio 8. Un polinomio es una lista de monomios y se puede
interpretar como la disyunción exclusiva de los monomios. Por ejemplo,
el polinomio  $[[0,2,1],[1,3]]$  se interpreta como la fórmula
 $(p_0 \wedge p_2 \wedge p_1) \oplus (p_1 \wedge p_3)$ .

Definir la función
  formP :: nat list list  $\Rightarrow$  form
tal que (formP p) es la fórmula correspondiente al polinomio p. Por
ejemplo,
  formP  $[[1,2],[3]]$ 
    = Xor (And (Var 1) (And (Var 2) T)) (Xor (And (Var 3) T) (Xor T T))
----- *}

fun formP :: "nat list list  $\Rightarrow$  form" where
  "formP []      = (Xor T T)"
| "formP (m#ms) = (Xor (formM m) (formP ms))"

value "formP  $[[1,2],[3]]$ "
-- "= Xor (And (Var 1) (And (Var 2) T)) (Xor (And (Var 3) T) (Xor T T))"

text {*
-----
Ejercicio 9. Definir la función
  valorP :: (nat  $\Rightarrow$  bool)  $\Rightarrow$  nat list list  $\Rightarrow$  bool
tal que (valorP i p) es el valor de la fórmula representada por el
polinomio p en la interpretación i. Por ejemplo,
  valorP (int1(1:=True,3:=True))  $[[0,2,1],[1,3]]$  = True
Demostrar que, para toda interpretación i y todo polinomio p, se tiene
que
  valorM i p = valorF i (formP p)
----- *}

fun valorP :: "(nat  $\Rightarrow$  bool)  $\Rightarrow$  nat list list  $\Rightarrow$  bool" where
  "valorP i []      = False"
| "valorP i (m # ms) = xor (valorM i m) (valorP i ms)"

value "valorP (int1(1:=True,3:=True))  $[[0,2,1],[1,3]]$ " -- "= True"

-- "La demostración automática es"
lemma correccion_valorP:
  "valorP i p = valorF i (formP p)"
by (induct p) (auto simp add: xor_def correccion_valorM)

```

```
-- "La demostración estructurada es"
lemma correccion_valorP2:
  "valorP i p = valorF i (formP p)"
proof (induct p)
  show "valorP i [] = valorF i (formP [])" by (simp add: xor_def)
next
  fix m ms assume HI: "valorP i ms = valorF i (formP ms)"
  have "valorP i (m#ms) = xor (valorM i m) (valorP i ms)" by simp
  also have "... = xor (valorM i m) (valorF i (formP ms))" using HI by simp
  also have "... = xor (valorF i (formM m)) (valorF i (formP ms))"
    by (simp add: correccion_valorM)
  also have "... = valorF i (Xor (formM m) (formP ms))" by simp
  also have "... = valorF i (formP (m#ms))" by simp
  finally show "valorP i (m#ms) = valorF i (formP (m#ms))" by simp
qed

text {*
  -----
  Ejercicio 10. Definir la función
    productoM :: nat list ⇒ nat list list ⇒ nat list list
  tal que (productoM m p) es el producto del monomio p por el polinomio
  p. Por ejemplo,
    productoM [1,3] [[1,2,4],[7],[4,1]]
      = [[1,3,1,2,4],[1,3,7],[1,3,4,1]]
  ----- *}

fun productoM :: "nat list ⇒ nat list list ⇒ nat list list" where
  "productoM m [] = []"
| "productoM m (x#xs) = (m@x)#(productoM m xs)"

value "productoM [1,3] [[1,2,4],[7],[4,1]]"
-- "= [[1,3,1,2,4],[1,3,7],[1,3,4,1]]"

text {*
  -----
  Ejercicio 11. Demostrar que, en cualquier interpretación i, el valor
  de la concatenación de dos monomios es la conjunción de sus valores.
  ----- *}

-- "La demostración automática es"
lemma valorM_conc:
  "valorM i (xs @ ys) = (valorM i xs ∧ valorM i ys)"
by (induct xs) auto

-- "La demostración estructurada es"
```

```

lemma valorM_conc2:
  "valorM i (xs @ ys) = (valorM i xs ∧ valorM i ys)"
proof (induct xs)
  show "valorM i ([] @ ys) = (valorM i [] ∧ valorM i ys)" by simp
next
  fix x xs assume HI: "valorM i (xs @ ys) = (valorM i xs ∧ valorM i ys)"
  have "valorM i ((x#xs) @ ys) = valorM i (x#(xs@ys))" by simp
  also have "... = (i x ∧ (valorM i (xs@ys)))" by simp
  also have "... = (i x ∧ (valorM i xs ∧ valorM i ys))" using HI by simp
  also have "... = ((i x ∧ valorM i xs) ∧ valorM i ys)" by simp
  also have "... = ((valorM i (x#xs)) ∧ valorM i ys)" by simp
  finally show "valorM i ((x#xs) @ ys) =
    ((valorM i (x#xs)) ∧ valorM i ys)" by simp
qed

text {*
  -----
  Ejercicio 12. Demostrar que, en cualquier interpretación i, el valor
  del producto de un monomio por un polinomio es la conjunción de sus
  valores.
  ----- *}

-- "La demostración automática es"
lemma correccion_productoM:
  "valorP i (productoM m p) = (valorM i m ∧ valorP i p)"
by (induct p) (auto simp add: xor_def valorM_conc)

-- "La demostración estructurada es"
lemma correccion_productoM2:
  "valorP i (productoM m p) = (valorM i m ∧ valorP i p)"
proof (induct p)
  show "valorP i (productoM m []) = (valorM i m ∧ valorP i [])" by simp
next
  fix xs p
  assume HI: "valorP i (productoM m p) = (valorM i m ∧ valorP i p)"
  have "valorP i (productoM m (xs#p)) = valorP i ((m@xs)#(productoM m p))"
    by simp
  also have "... = xor (valorM i (m@xs)) (valorP i (productoM m p))"
    by simp
  also have "... = xor (valorM i (m@xs)) (valorM i m ∧ valorP i p)"
    using HI by simp
  also have "... = xor (valorM i m ∧ valorM i xs)
    (valorM i m ∧ valorP i p)"
    by (simp add: valorM_conc)
  also have "... = (valorM i m ∧ (xor (valorM i xs) (valorP i p)))"

```

```

    by (auto simp add: xor_def)
  also have "... = (valorM i m ∧ valorP i (xs#p))" by simp
  finally show "valorP i (productoM m (xs#p)) =
    (valorM i m ∧ valorP i (xs#p))" by simp
qed

text {*
-----
Ejercicio 13. Definir la función
  producto :: nat list list ⇒ nat list list ⇒ nat list list
tal que (producto p q) es el producto de los polinomios p y q. Por
ejemplo,
  producto [[1,3],[2]] [[1,2,4],[7],[4,1]]
    = [[1,3,1,2,4],[1,3,7],[1,3,4,1],[2,1,2,4],[2,7],[2,4,1]]
----- *}

fun producto :: "nat list list ⇒ nat list list ⇒ nat list list" where
  "producto [] q      = []"
| "producto (m # p) q = (productoM m q) @ (producto p q)"

value "producto [[1,3],[2]] [[1,2,4],[7],[4,1]]"
-- "= [[1,3,1,2,4],[1,3,7],[1,3,4,1],[2,1,2,4],[2,7],[2,4,1]]"

text {*
-----
Ejercicio 14. Demostrar que, en cualquier interpretación i, el valor
de la concatenación de dos polinomios es la disyunción exclusiva de
sus valores.
----- *}

-- "La demostración automática es"
lemma valorP_conc:
  "valorP i (xs @ ys) = (xor (valorP i xs) (valorP i ys))"
by (induct xs) (auto simp add: xor_def)

-- "La demostración estructurada es"
lemma valorP_conc2:
  "valorP i (xs @ ys) = (xor (valorP i xs) (valorP i ys))"
proof (induct xs)
  show "valorP i ([]@ys) = xor (valorP i []) (valorP i ys)"
    by (simp add: xor_def)
next
  fix x xs
  assume HI: "valorP i (xs@ys) = xor (valorP i xs) (valorP i ys)"
  have "valorP i ((x#xs)@ys) = valorP i (x#(xs@ys))" by simp

```

```

also have "... = xor (valorM i x) (valorP i (xs@ys))" by simp
also have "... = xor (valorM i x) (xor (valorP i xs) (valorP i ys))"
  using HI by simp
also have "... = xor (xor (valorM i x) (valorP i xs)) (valorP i ys)"
  by (auto simp add: xor_def)
also have "... = xor (valorP i (x#xs)) (valorP i ys)"
  by simp
finally show "valorP i ((x#xs)@ys) = xor (valorP i (x#xs)) (valorP i ys)"
  by simp
qed

text {*
-----
Ejercicio 15. Demostrar que, en cualquier interpretación i, el valor
del producto de dos polinomios es la conjunción de sus valores.
----- *}

-- "La demostración automática es"
lemma correccion_producto:
  "valorP i (producto p q) = (valorP i p ∧ valorP i q)"
by (induct p)
  (auto simp add: xor_def correccion_productoM valorP_conc)

-- "La demostración estructurada es"
lemma correccion_producto2:
  "valorP i (producto p q) = (valorP i p ∧ valorP i q)"
proof (induct p)
  show "valorP i (producto [] q) = (valorP i [] ∧ valorP i q)"
    by simp
next
  fix m p
  assume HI: "valorP i (producto p q) = (valorP i p ∧ valorP i q)"
  have "valorP i (producto (m # p) q) =
    valorP i ((productoM m q) @ (producto p q))" by simp
  also have "... = xor (valorP i (productoM m q))
    (valorP i (producto p q))"
    by (simp add: valorP_conc)
  also have "... = xor (valorM i m ∧ valorP i q)
    (valorP i (producto p q))"
    by (simp add: correccion_productoM)
  also have "... = xor (valorM i m ∧ valorP i q)
    (valorP i p ∧ valorP i q)"
    by (simp add: correccion_producto)
  also have "... = (xor (valorM i m) (valorP i p) ∧ valorP i q)"
    by (auto simp add: xor_def)

```

```

    also have "... = (valorP i (m # p) ∧ valorP i q)"
    by simp
    finally show "valorP i (producto (m # p) q) =
        (valorP i (m # p) ∧ valorP i q)" by simp
qed

text {*
-----
Ejercicio 16. Definir la función
    polinomio :: form ⇒ nat list list
tal que (polinomio f) es el polinomio que representa la fórmula f. Por
ejemplo,
    polinomio (Xor (Var 1) (Var 2))      = [[1],[2]]
    polinomio (And (Var 1) (Var 2))      = [[1,2]]
    polinomio (Xor (Var 1) T)            = [[1],[ ]]
    polinomio (And (Var 1) T)            = [[1]]
    polinomio (And (Xor (Var 1) (Var 2)) (Var 3)) = [[1,3],[2,3]]
    polinomio (Xor (And (Var 1) (Var 2)) (Var 3)) = [[1,2],[3]]
----- *}

fun polinomio :: "form ⇒ nat list list" where
  "polinomio T          = [ ]"
| "polinomio (Var i)    = [[i]]"
| "polinomio (Xor f1 f2) = polinomio f1 @ polinomio f2"
| "polinomio (And f1 f2) = producto (polinomio f1) (polinomio f2)"

value "polinomio (Xor (Var 1) (Var 2))"      -- "= [[1],[2]]"
value "polinomio (And (Var 1) (Var 2))"      -- "= [[1,2]]"
value "polinomio (Xor (Var 1) T)"            -- "= [[1],[ ]]"
value "polinomio (And (Var 1) T)"            -- "= [[1]]"
value "polinomio (And (Xor (Var 1) (Var 2)) (Var 3))" -- "= [[1,3],[2,3]]"
value "polinomio (Xor (And (Var 1) (Var 2)) (Var 3))" -- "= [[1,2],[3]]"

text {*
-----
Ejercicio 17. Demostrar que, en cualquier interpretación i, el valor
de f es igual que el de su polinomio.
----- *}

-- "La demostración automática es"
theorem correccion_polinomio:
  "valorF i f = valorP i (polinomio f)"
by (induct f)
   (auto simp add: xor_def correccion_producto valorP_conc)

```

```

-- "La demostración estructurada es"
theorem correccion_polinomio2:
  "valorF i f = valorP i (polinomio f)"
proof (induct f)
  show "valorF i T = valorP i (polinomio T)" by (simp add: xor_def)
next
  fix n
  show "valorF i (Var n) = valorP i (polinomio (Var n))"
    by (simp add: xor_def)
next
  fix f g
  assume HI1: "valorF i f = valorP i (polinomio f)" and
    HI2: "valorF i g = valorP i (polinomio g)"
  have "valorF i (And f g) = (valorF i f ∧ valorF i g)" by simp
  also have "... = (valorP i (polinomio f) ∧ valorP i (polinomio g))"
    using HI1 HI2 by simp
  also have "... = valorP i (producto (polinomio f) (polinomio g))"
    by (simp add: correccion_producto)
  also have "... = valorP i (polinomio (And f g))"
    by simp
  finally show "valorF i (And f g) = valorP i (polinomio (And f g))"
    by simp
next
  fix f g
  assume HI1: "valorF i f = valorP i (polinomio f)" and
    HI2: "valorF i g = valorP i (polinomio g)"
  have "valorF i (Xor f g) = xor (valorF i f) (valorF i g)"
    by simp
  also have "... = xor (valorP i (polinomio f)) (valorP i (polinomio g))"
    using HI1 HI2 by simp
  also have "... = valorP i ((polinomio f) @ (polinomio g))"
    by (simp add: valorP_conc)
  also have "... = valorP i (polinomio (Xor f g))"
    by simp
  finally show "valorF i (Xor f g) = valorP i (polinomio (Xor f g))"
    by simp
qed

end

```

6.2.6. Ordenación de listas mediante árboles ordenados

chapter *{* T6R2f: Ordenación de listas mediante árboles ordenados **}

theory T6R2f_Ordenacion_de_listas_mediante_arboles_ordenados

```

imports Main T6R1j_Ordenacion_de_listas_por_insercion
begin

text {*
  En esta relación de ejercicios se definen los árboles ordenados y se
  muestra cómo se puede ordenar listas mediante árboles. Para ello, se
  definen dos funciones
  * arbol_de que transforma lista en árboles ordenados y
  * lista_de que transforma árboles en listas
  Se demuestra que (lista_de (arbol_de xs)) es una lista ordenada con
  los mismos elementos que xs.
*}

section {* Árboles ordenados *}

text {*
  -----
  Ejercicio 1. Definir el tipo de datos arbol para representar los
  árboles binarios que tiene en los nodos números naturales. Por
  ejemplo, el árbol
      4
     / \
    /   \
   /     \
  1       6
 / \   / \
H  H H  H
se representa por "Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)".
  -----
*}

datatype arbol = Hoja | Nodo nat arbol arbol

value "Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)"

text {*
  -----
  Ejercicio 2. Definir la función
  menorA :: nat ⇒ arbol ⇒ bool
  tal que (menorA x a) se verifica si x es menor o igual que todos los
  elementos del árbol a. Por ejemplo,
  menorA 2 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) = True
  menorA 5 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) = False
  -----
*}

fun menorA :: "nat ⇒ arbol ⇒ bool" where

```

```

    "menorA x Hoja          = True"
| "menorA x (Nodo n a1 a2) = (x ≤ n ∧ menorA x a1 ∧ menorA x a2)"

value "menorA 2 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= True"
value "menorA 5 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= False"

text {*
-----
Ejercicio 3. Definir la función
    mayorA :: nat ⇒ arbol ⇒ bool
tal que (mayorA x a) se verifica si x es mayor o igual que todos los
elementos del árbol a. Por ejemplo,
    mayorA 7 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) = True
    mayorA 5 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) = False
----- *}

fun mayorA :: "nat ⇒ arbol ⇒ bool" where
    "mayorA x Hoja          = True"
| "mayorA x (Nodo n a1 a2) = (n ≤ x ∧ mayorA x a1 ∧ mayorA x a2)"

value "mayorA 7 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= True"
value "mayorA 5 (Nodo 4 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= False"

text {*
-----
Ejercicio 5. Un árbol está ordenado si cada nodo es mayor o igual que
todos los nodos de su subárbol izquierdo y menor o igual que todos los
nodos de su subárbol derecho.

Definir la función
    ordenadoA :: arbol ⇒ bool
tal que (ordenadoA x a) se verifica si a es un árbol ordenado. Por
ejemplo,
    ordenadoA (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))
    = True
    ordenadoA (Nodo 4 (Nodo 5 Hoja Hoja) (Nodo 6 Hoja Hoja))
    = False
    ordenadoA (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 3 Hoja Hoja))
    = False
----- *}

```

```

fun ordenadoA :: "arbol  $\Rightarrow$  bool" where
  "ordenadoA Hoja                = True"
| "ordenadoA (Nodo n a1 a2) = (ordenadoA a1  $\wedge$  ordenadoA a2  $\wedge$ 
                                mayorA n a1  $\wedge$  menorA n a2)"

value "ordenadoA (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= True"
value "ordenadoA (Nodo 4 (Nodo 5 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= False"
value "ordenadoA (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 3 Hoja Hoja))"
-- "= False"

text {*
-----
Ejercicio 6. Definir la función
  insertaA :: nat  $\Rightarrow$  arbol  $\Rightarrow$  arbol
tal que (inserta n a) es el árbol obtenido insertando n delante del
primer subárbol de a cuya raíz es mayor o igual que a. Por ejemplo,
  insertaA 4 Hoja
    = Nodo 4 Hoja Hoja
  insertaA 1 (Nodo 4 Hoja Hoja)
    = Nodo 4 (Nodo 1 Hoja Hoja) Hoja
  insertaA 6 (Nodo 4 (Nodo 1 Hoja Hoja) Hoja)
    = Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)
  insertaA 3 (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))
    = Nodo 4 (Nodo 1 Hoja (Nodo 3 Hoja Hoja)) (Nodo 6 Hoja Hoja)
  insertaA 5 (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))
    = Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 (Nodo 5 Hoja Hoja) Hoja)
----- *}

fun insertaA :: "nat  $\Rightarrow$  arbol  $\Rightarrow$  arbol" where
  "insertaA x Hoja                = Nodo x Hoja Hoja"
| "insertaA x (Nodo n a1 a2) = (if x  $\leq$  n
                                then Nodo n (insertaA x a1) a2
                                else Nodo n a1 (insertaA x a2))"

value "insertaA 4 Hoja"
-- "= Nodo 4 Hoja Hoja"
value "insertaA 1 (Nodo 4 Hoja Hoja)"
-- "= Nodo 4 (Nodo 1 Hoja Hoja) Hoja"
value "insertaA 6 (Nodo 4 (Nodo 1 Hoja Hoja) Hoja)"
-- "= Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)"
value "insertaA 3 (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= Nodo 4 (Nodo 1 Hoja (Nodo 3 Hoja Hoja)) (Nodo 6 Hoja Hoja)"
value "insertaA 5 (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))"

```

```
-- "= Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 (Nodo 5 Hoja Hoja) Hoja)"

text {*
-----
Ejercicio 7. Definir la función
  arbol_de :: nat list ⇒ arbol
tal que (arbol_de xs) es el árbol obtenido insertando n los elementos
de la lista xs. Por ejemplo,
  arbol_de [6,1,4]
  = Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)
----- *}

fun arbol_de :: "nat list ⇒ arbol" where
  "arbol_de []      = Hoja"
| "arbol_de (x#xs) = insertaA x (arbol_de xs)"

value "arbol_de [6,1,4]"
-- "= Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja)"

text {*
-----
Ejercicio 8. Demostrar que x es mayor o igual que los nodos del árbol
obtenido insertando y en el árbol a syss x es mayor o igual que y y x
es mayor o igual que los nodos del árbol a.
----- *}

-- "La demostración automática es"
lemma mayorA_insertaA:
  "mayorA x (insertaA y a) = (y ≤ x ∧ mayorA x a)"
by (induct a) auto

text {*
-----
Ejercicio 9. Demostrar que x es menor o igual que los nodos del árbol
obtenido insertando y en el árbol a syss x es menor o igual que y y x
es menor o igual que los nodos del árbol a.
----- *}

-- "La demostración automática es"
lemma menorA_insertaA:
  "menorA x (insertaA y a) = (x ≤ y ∧ menorA x a)"
by (induct a) auto

text {*
-----
```

Ejercicio 10. Demostrar que el árbol obtenido insertando x en el árbol a es ordenado syss el árbol a es ordenado.

```
----- *}

-- "La demostración automática es"
lemma ordenadoA_insertaA:
  "ordenadoA (insertaA x a) = ordenadoA a"
by (induct a) (auto simp add: mayorA_insertaA menorA_insertaA)

text {*
-----
Ejercicio 11. Demostrar que para cualquier lista  $xs$ , (arbol_de  $xs$ ) es un árbol ordenado.
----- *}

-- "La demostración automática es"
theorem ordenadoA_arbol_de:
  "ordenadoA (arbol_de xs)"
by (induct xs) (auto simp add: ordenadoA_insertaA)

text {*
-----
Ejercicio 12. Definir la función
  cuentaA :: arbol => nat => nat
tal que (cuentaA a y) es el número de veces que aparece y en el árbol a. Por ejemplo,
  cuentaA (Nodo 6 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) 6 = 2
----- *}

fun cuentaA :: "arbol => nat => nat" where
  "cuentaA Hoja y = 0"
| "cuentaA (Nodo x a1 a2) y = (if x=y
                               then Suc (cuentaA a1 y + cuentaA a2 y)
                               else cuentaA a1 y + cuentaA a2 y)"

value "cuentaA (Nodo 6 (Nodo 3 Hoja Hoja) (Nodo 6 Hoja Hoja)) 6"
-- "= 2"

text {*
-----
Ejercicio 13. Demostrar que el número de veces que aparece y en el árbol obtenido insertando  $x$  en el árbol  $a$  es
  * uno más el número de veces que aparece y en  $a$ , si  $y = x$ ,
  * el número de veces que aparece y en  $a$ , si  $y \neq x$ ,
----- *
```

```
-- "La demostración automática es"
lemma cuentaA_insertaA:
  "cuentaA (insertaA x a) y =
    (if x=y then Suc (cuentaA a y) else cuentaA a y)"
by (induct a) auto

text {*
  -----
  Ejercicio 14. Demostrar que el número de veces que aparece y en
  (arbol_de xs) es el mismo que el número de veces que aparece y en xs.
  ----- *}

-- "La demostración automática es"
theorem cuenta_arbol_de:
  "cuentaA (arbol_de xs) x = cuenta xs x"
by (induct xs) (auto simp add: cuentaA_insertaA)

section {* Ordenación de lista mediante árboles *}

text {*
  La función arbol_de transforma una lista en un árbol ordenado. Vamos a
  definir la función lista_de que transforme un árbol ordenado en una
  lista ordenada. Con ambas, tendremos un algoritmo de ordenación de
  listas.
  *}

text {*
  -----
  Ejercicio 15. Definir la función
    lista_de :: arbol ⇒ nat list
  tal que (lista_de a) es la lista obtenida recorriendo el árbol a de
  forma prefija. Por ejemplo,
    lista_de (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))
    = [1,4,6]
  ----- *}

fun lista_de :: "arbol ⇒ nat list" where
  "lista_de Hoja          = []"
| "lista_de (Nodo n a1 a2) = lista_de a1 @ [n] @ lista_de a2"

value "lista_de (Nodo 4 (Nodo 1 Hoja Hoja) (Nodo 6 Hoja Hoja))"
-- "= [1,4,6]"

text {*
```

```

-----
Ejercicio 16. Demostrar que  $x$  es menor o igual que todos elementos de la concatenación de dos listas  $syss$  es menor o igual que todos los elementos de cada una de las listas.
----- *}

-- "La demostración automática es"
lemma menor_conc:
  "menor x (a@b) = (menor x a  $\wedge$  menor x b)"
by (induct a) auto

text {*
-----
Ejercicio 17. Definir la función
  mayor :: nat  $\Rightarrow$  nat list  $\Rightarrow$  bool
tal que (mayor x ys) se verifica si  $x$  es mayor o igual que todos los elementos de ys. Por ejemplo,
----- *}

fun mayor :: "nat  $\Rightarrow$  nat list  $\Rightarrow$  bool" where
  "mayor a [] = True"
| "mayor a (x#xs) = (x  $\leq$  a  $\wedge$  mayor a xs)"

value "mayor 7 [1,5,2]" --"= True"
value "mayor 7 [1,9,2]" --"= False"

text {*
-----
Ejercicio 18. Demostrar que  $x$  es mayor o igual que todos elementos de la concatenación de dos listas  $syss$  es mayor o igual que todos los elementos de cada una de las listas.
----- *}

-- "La demostración automática es"
lemma mayor_conc:
  "mayor x (a@b) = (mayor x a  $\wedge$  mayor x b)"
by (induct a) auto

text {*
-----
Ejercicio 18. Demostrar que la lista (a@(x#b)) está ordenada syss se a y b está ordenada,  $x$  es mayor o igual que los elementos de a y es menor o igual que los elementos de b.
----- *}

```

```
-- "La demostración automática es"
lemma ordenada_conc:
  "ordenada (a@(x#b)) = (ordenada a  $\wedge$  ordenada b  $\wedge$  mayor x a  $\wedge$  menor x b)"
by (induct a) (auto simp add: menor_conc mayor_conc menor_menor)
```

```
text {*
-----
Ejercicio 19. Demostrar que n es mayor o igual que todos los elementos
de la lista de elementos del árbol a syss es mayor o igual que todos
los elementos de a.
----- *}

```

```
-- "La demostración automática es"
lemma mayor_lista_de:
  "mayor n (lista_de a) = mayorA n a"
by (induct a) (auto simp add: mayorA_insertaA mayor_conc)
```

```
text {*
-----
Ejercicio 20. Demostrar que n es menor o igual que todos los elementos
de la lista de elementos del árbol a syss es menor o igual que todos
los elementos de a.
----- *}

```

```
-- "La demostración automática es"
lemma menor_lista_de:
  "menor n (lista_de a) = menorA n a"
by (induct a) (auto simp add: menorA_insertaA menor_conc)
```

```
text {*
-----
Ejercicio 21. Demostrar que (lista_de a) está ordenada syss a es un
árbol ordenado.
----- *}

```

```
-- "La demostración automática es"
lemma ordenada_lista_de:
  "ordenada (lista_de a) = ordenadoA a"
by (induct a) (auto simp add: ordenada_conc mayor_lista_de menor_lista_de)
```

```
text {*
-----
Ejercicio 22. Demostrar que (lista_de (arbol_de xs)) está ordenada.
----- *}

```

```
-- "La demostración automática es"
theorem ordenada_lista_de_arbol_de:
  "ordenada (lista_de (arbol_de xs))"
by (auto simp add: ordenada_lista_de ordenadoA_arbol_de)

text {*
-----
Ejercicio 23. Demostrar que el número de veces que aparece un elemento
en la concatenación de dos listas es la suma del número de veces que
aparece en cada una.
----- *}

-- "La demostración automática es"
lemma cuenta_conc:
  "cuenta (a@b) n = cuenta a n + cuenta b n"
by (induct a) auto

text {*
-----
Ejercicio 24. Demostrar que el número de veces que aparece n en
(lista_de a) es el mismo que el número de veces que aparece n en el
árbol a.
----- *}

-- "La demostración automática es"
lemma cuenta_lista_de:
  "cuenta (lista_de a) n = cuentaA a n"
by (induct a) (auto simp add: cuenta_conc)

text {*
-----
Ejercicio 25. Demostrar que el número de veces que aparece n
(lista_de (arbol_de xs)) es el mismo número de veces que aparece n en
xs.
----- *}

-- "La demostración automática es"
theorem cuenta_lista_de_arbol_de:
  "cuenta (lista_de (arbol_de xs)) n = cuenta xs n"
by (induct xs) (auto simp add: cuenta_lista_de cuentaA_insertaA)

end
```

6.3. Ejercicios sobre aritmética

6.3.1. Potencias y sumatorios

```
chapter {* T6R3a: Potencias y sumatorios *}

theory T6R3a
imports Main
begin

text {*
  En esta relación se definen las potencias y sumatorios y se demuestran
  algunas de sus propiedades. *}

section {* Potencia *}

text {*
  -----
  Ejercicio 1. Definir la función
    potencia :: nat => nat => nat
  tal que (potencia x n) es x elevado a n. Por ejemplo,
    potencia 2 3 = 8
  ----- *}

fun potencia :: "nat => nat => nat" where
  "potencia x 0      = Suc 0"
| "potencia x (Suc n) = x * potencia x n"

value "potencia 2 3" -- "= 8"

text {*
  -----
  Ejercicio 2. Demostrar que
     $x^{m \cdot n} = (x^m)^n$ 
  Indicación: Demostrar antes un lema.
  ----- *}

-- "La demostración automática del lema es"
lemma potencia_suma:
  "potencia x (m + n) = potencia x m * potencia x n"
by (induct n) auto

-- "La demostración estructurada del lema es"
lemma potencia_suma2:
  "potencia x (m + n) = potencia x m * potencia x n"
```

```

proof (induct n)
  show "potencia x (m + 0) = potencia x m * potencia x 0" by simp
next
  fix n
  assume HI: "potencia x (m + n) = potencia x m * potencia x n"
  have "potencia x (m + Suc n) = potencia x (Suc (m + n))" by simp
  also have "... = x * potencia x (m + n)" by simp
  also have "... = x * (potencia x m * potencia x n)" using HI by simp
  also have "... = potencia x m * (x * potencia x n)" by simp
  also have "... = potencia x m * potencia x (Suc n)" by simp
  finally show "potencia x (m + Suc n) = potencia x m * potencia x (Suc n)"
    by simp
qed

```

-- "La demostración automática del teorema es"

theorem potencia_mult:

"potencia x (m * n) = potencia (potencia x m) n"

by (induct n) (auto simp add: potencia_suma)

-- "La demostración estructurada del teorema es"

theorem potencia_mult2:

"potencia x (m * n) = potencia (potencia x m) n"

proof (induct n)

show "potencia x (m * 0) = potencia (potencia x m) 0" by simp

next

fix n

assume HI: "potencia x (m * n) = potencia (potencia x m) n"

have "potencia x (m * Suc n) = potencia x (m + m * n)" by simp

also have "... = potencia x m * potencia x (m * n)"

by (simp add: potencia_suma)

also have "... = potencia x m * potencia (potencia x m) n"

using HI by simp

also have "... = potencia (potencia x m) (Suc n)"

by simp

finally show "potencia x (m * Suc n) = potencia (potencia x m) (Suc n)"

by simp

qed

section *{* Sumatorios **}

text *{**

Ejercicio 3. Definir la función

suma :: nat list => nat

tal que (suma ns) es la suma de los elementos de ns. Por ejemplo,

```

    suma [3,2,5] = 10
----- *}

fun suma :: "nat list => nat" where
  "suma []      = 0"
| "suma (n#ns) = n + suma ns"

value "suma [3,2,5]" -- "= 10"

text {*
-----
  Ejercicio 4. Demostrar que
    suma (xs @ ys) = suma xs + suma ys
  Indicación: Conviene demostrar un lema previo.
----- *}

-- "La demostración automática del lema es"
lemma suma_append:
  "suma (xs @ ys) = suma xs + suma ys"
by (induct xs) auto

-- "La demostración estructurada del lema es"
lemma suma_append2:
  "suma (xs @ ys) = suma xs + suma ys"
proof (induct xs)
  show "suma ([] @ ys) = suma [] + suma ys" by simp
next
  fix x xs
  assume HI: "suma (xs @ ys) = suma xs + suma ys"
  have "suma ((x#xs) @ ys) = suma (x # (xs @ ys))" by simp
  also have "... = x + suma (xs @ ys)" by simp
  also have "... = x + (suma xs + suma ys)" using HI by simp
  also have "... = (x + suma xs) + suma ys" by simp
  also have "... = suma (x#xs) + suma ys" by simp
  finally show "suma ((x#xs) @ ys) = suma (x#xs) + suma ys" by simp
qed

-- "La demostración automática del teorema es"
theorem suma_rev: "
  suma (rev ns) = suma ns"
by (induct ns) (auto simp add: suma_append)

-- "La demostración estructurada del teorema es"
theorem suma_rev2: "
  suma (rev ns) = suma ns"

```

```

proof (induct ns)
  show "suma (rev []) = suma []" by simp
next
  fix n ns
  assume HI: "suma (rev ns) = suma ns"
  have "suma (rev (n#ns)) = suma (rev ns @ [n])" by simp
  also have "... = suma (rev ns) + suma [n]" by (simp add: suma_append)
  also have "... = suma ns + suma [n]" using HI by simp
  also have "... = n + suma ns" by simp
  also have "... = suma (n#ns)" by simp
  finally show "suma (rev (n#ns)) = suma (n#ns)" by simp
qed

```

```

text {*
-----
  Ejercicio 5. Definir la función
    Suma :: (nat => nat) => nat => nat
  tal que (Suma f n) es la suma  $f(0) + \dots + f(n-1)$ . Por ejemplo,
    Suma ( $\lambda x. x+2$ ) 3 = 9
----- *}

```

```

fun Suma :: "(nat => nat) => nat => nat" where
  "Suma f 0      = 0"
| "Suma f (Suc n) = Suma f n + f n"

```

```

value "Suma ( $\lambda x. x+2$ ) 3" -- "= 9"

```

```

text {*
-----
  Ejercicio 6. Demostrar que
    Suma ( $\lambda i. f i + g i$ ) k = Suma f k + Suma g k
----- *}

```

```

-- "La demostración automática es"

```

```

lemma "Suma ( $\lambda i. f i + g i$ ) k = Suma f k + Suma g k"
by (induct k) auto

```

```

-- "La demostración estructurada es"

```

```

lemma "Suma ( $\lambda i. f i + g i$ ) k = Suma f k + Suma g k"

```

```

proof (induct k)

```

```

  show "Suma ( $\lambda i. f i + g i$ ) 0 = Suma f 0 + Suma g 0" by simp

```

```

next

```

```

  fix k

```

```

  assume HI: "Suma ( $\lambda i. f i + g i$ ) k = Suma f k + Suma g k"

```

```

  have "Suma ( $\lambda i. f i + g i$ ) (Suc k) =

```

```

      Suma (λi. f i + g i) k + (f k + g k)" by simp
also have "... = (Suma f k + Suma g k) + (f k + g k)"
  using HI by simp
also have "... = (Suma f k + f k) + (Suma g k + g k)" by simp
also have "... = Suma f (Suc k) + Suma g (Suc k)" by simp
finally show "Suma (λi. f i + g i) (Suc k) =
  Suma f (Suc k) + Suma g (Suc k)" by simp
qed

text {*
-----
Ejercicio 7. Determinar el valor de indefinida para que se verifique
la siguiente igualdad
  Suma f (k + l) = Suma f k + Suma indefinida l
y demostrarla.
----- *}

-- "El valor de indefinida es (λi. f (k + i))"

-- "La demostración automática es"
lemma "Suma f (k + l) = Suma f k + Suma (λi. f (k + i)) l"
by (induct l) auto

-- "La demostración estructurada es"
lemma "Suma f (k + l) = Suma f k + Suma (λi. f (k + i)) l"
proof (induct l)
  show "Suma f (k + 0) = Suma f k + Suma (λi. f (k + i)) 0" by simp
next
  fix l
  assume HI: "Suma f (k + l) = Suma f k + Suma (λi. f (k + i)) l"
  have "Suma f (k + Suc l) = Suma f (Suc (k + l))" by simp
  also have "... = Suma f (k + l) + f (k + l)" by simp
  also have "... = Suma f k + Suma (λi. f (k + i)) l + f (k + l)"
    using HI by simp
  also have "... = Suma f k + Suma (λi. f (k + i)) (Suc l)" by simp
  finally show "Suma f (k + Suc l) =
    Suma f k + Suma (λi. f (k + i)) (Suc l)" by simp
qed

text {*
-----
Ejercicio 8. Determinar el valor de indefinida para que se verifique
la siguiente igualdad
  Suma f k = suma indefinida
y demostrarla.
----- *}

```

```

----- *}

-- "El valor de indefinida es (map f [0..<k])"

-- "La demostración automática es"
lemma "Suma f k = suma (map f [0..<k])"
by (induct k) (auto simp add: suma_append)

-- "La demostración estructurada es"
lemma "Suma f k = suma (map f [0..<k])" (is "?P k")
proof (induct k)
  show "?P 0" by simp
next
  fix k assume HI: "?P k"
  have "Suma f (Suc k) = Suma f k + f k" by simp
  also have "... = suma (map f [0..<k]) + f k" using HI by simp
  also have "... = suma (map f [0..<k]) + suma [f k]" by simp
  also have "... = suma (map f [0..<k] @ [f k])"
    by (simp add: suma_append)
  also have "... = suma (map f [0..<Suc k])" by simp
  finally show "?P (Suc k)" by simp
qed

text {*
-----
  Ejercicio 9. Demostrar la fórmula de la suma de naturales:
    
$$1 + 2 + \dots + n = (n(n+1))/2$$

----- *}

-- "La demostración automática es"
lemma suma_de_naturales:
  fixes n :: "nat"
  shows "2 * ( $\sum_{i=1..n} i$ ) = n * (n + 1)"
by (induct n) simp_all

-- "La demostración estructurada es"
lemma suma_de_naturales_2:
  fixes n :: "nat"
  shows "2 * ( $\sum_{i=1..n} i$ ) = n * (n + 1)" (is "?P n")
proof (induct n)
  show "?P 0" by simp
next
  fix n assume HI: "?P n"
  have "2 *  $\sum_{i=1..Suc\ n}$  = 2 * ( $\sum_{i=1..n}$  + Suc n)" by simp
  also have "... = 2 *  $\sum_{i=1..n}$  + 2 * Suc n" by simp

```

```

also have "... = n * (n + 1) + 2 * Suc n" using HI by simp
also have "... = Suc n * (Suc n + 1)" by simp
finally show "?P (Suc n)" by simp
qed

text {*
-----
Ejercicio 11. Demostrar la fórmula de la suma de impares:

$$1 + 3 + \dots + (2*(n-1)+1) = n^2$$

----- *}

-- "La demostración automática es"
theorem suma_impares:
  " $(\sum i::\text{nat}=0..<n. 2 * i + 1) = n^2$ "
by (induct n) simp_all

-- "La demostración estructurada es"
lemma suma_impares_2:
  " $(\sum i::\text{nat}=0..<n. 2 * i + 1) = n^2$ "
  (is "?P n")
proof (induct n)
  show "?P 0" by simp
next
  fix n assume HI: "?P n"
  have " $(\sum i = 0..<\text{Suc } n. 2 * i + 1) =$ "
    " $(\sum i = 0..<n. 2 * i + 1) + 2 * n + 1$ " by simp
  also have "... =  $n^2 + 2 * n + 1$ " using HI by simp
  also have "... =  $(n + 1)^2$ " by simp
  finally show "?P (Suc n)" by simp
qed

text {*
Nota: En los siguientes ejercicios se usarán las siguientes
propiedades distributivas:
· add_mult_distrib:  $(m + n) * k = m * k + n * k$ 
· add_mult_distrib2:  $k * (m + n) = k * m + k * n$ 
que agruparemos en el lema distrib
*}

lemmas distrib = add_mult_distrib add_mult_distrib2

text {*
-----
Ejercicio 12. Demostrar la fórmula de la suma de cuadrados

$$1^2 + 2^2 + \dots + n^2 = \frac{n(n+1)(2n+1)}{6}$$

----- *}

```

```

----- *}

-- "La demostración automática es"
lemma suma_cuadrados:
  "6 * (∑ i::nat=0..n. i^Suc (Suc 0)) = n * (n + 1) * (2 * n + 1)"
by (induct n) (auto simp add: distrib)

-- "La demostración estructurada es"
lemma suma_cuadrados_2:
  "6 * (∑ i::nat=0..n. i^Suc (Suc 0)) = n * (n + 1) * (2 * n + 1)"
  (is "?P n")
proof (induct n)
  show "?P 0" by simp
next
  fix n :: "nat"
  assume HI: "6 * (∑ i = 0..n. i ^ Suc (Suc 0)) = n * (n + 1) * (2 * n + 1)"
  have "6 * (∑ i = 0..Suc n. i ^ Suc (Suc 0)) =
    6 * ((∑ i = 0..n. i ^ Suc (Suc 0)) + (Suc n)^Suc (Suc 0))"
  by simp
  also have "... = 6 * (∑ i = 0..n. i ^ Suc (Suc 0)) + 6 * (Suc n)^Suc (Suc 0)"
  by (simp add: distrib)
  also have "... = n * (n + 1) * (2 * n + 1) + 6 * (Suc n)^Suc (Suc 0)"
  using HI by simp
  also have "... = Suc n * (Suc n + 1) * (2 * Suc n + 1)"
  by (simp add: distrib)
  finally show "?P (Suc n)" by simp
qed

text {*
-----
  Ejercicio 12. Demostrar la fórmula de la suma de cubos
     $1 + 2^3 + \dots + n^3 = (n(n+1))^2/4$ 
  Indicación: Se pueden usar, además de distrib, el lema
    · power_eq_if:  $p^m = (\text{if } m = 0 \text{ then } 1 \text{ else } p * p^{m-1})$ 
----- *}

-- "La demostración automática es"
lemma suma_de_cubos:
  "4 * (∑ i::nat=0..n. i^3) = (n * (n + 1))^Suc (Suc 0)"
by (induct n) (auto simp add: distrib power_eq_if)

-- "La demostración estructurada es"
lemma suma_de_cubos_2:
  "4 * (∑ i::nat=0..n. i^3) = (n * (n + 1))^Suc (Suc 0)"
  (is "?P n" is "?S n = _")

```

```

proof (induct n)
  show "?P 0" by simp
next
  fix n
  assume HI: "?S n = (n * (n + 1))^Suc (Suc 0)"
  have "?S (n + 1) = ?S n + 4 * (n + 1)^3" by simp
  also have "... = (n * (n + 1))^Suc (Suc 0) + 4 * (n + 1)^3"
    using HI by simp
  also have "... = ((n + 1) * ((n + 1) + 1))^Suc (Suc 0)"
    by (simp add: power_eq_if distrib)
  finally show "?P (Suc n)" by simp
qed

end

```

6.3.2. Métodos de cálculo de cuadrados

```
chapter {* T6R3b: Métodos de cálculo de cuadrados *
```

```

theory T6R3b
imports Main
begin

```

```

text {*
  En esta relación demostraremos la corrección de tres métodos para
  calcular el cuadrado de números naturales que aparece en un libro de
  matemáticas védicas.

```

```

  La relación está basada en el trabajo de Farhad Mehta titulado
  "Magical methods (Computing with natural numbers)". *}

```

```

text {*
  -----
  Ejercicio 1. El primer método sirve para calcular el cuadrado cuyo
  predecesor tenga un cuadrado fácilmente calculable, Por ejemplo, para
  el 61 se tiene
     $61^2 = 60^2 + 60 + 61 = 3600 + 121 = 3721$ 
  Basado en la idea anterior, definir la función
    cuadrado :: nat ⇒ nat
  ----- *}

```

```

fun cuadrado :: "nat ⇒ nat" where
  "cuadrado 0 = 0"
| "cuadrado (Suc n) = (cuadrado n) + n + (Suc n)"

```

```

value "cuadrado 61"

text {*
-----
Ejercicio 2. Demostrar que cuadrado es correcta; es decir, que
cuadrado  $n = n * n$ 
----- *}

-- "La demostración automática es"
theorem cuadrado_correcta [simp]:
  "cuadrado  $n = n * n$ "
by (induct n) auto

-- "La demostración estructurada es"
theorem cuadrado_correcta2:
  "cuadrado  $n = n * n$ "
proof (induct n)
  show "cuadrado 0 = 0 * 0" by simp
next
  fix n assume HI: "cuadrado  $n = n * n$ "
  have "cuadrado (Suc n) = cuadrado n + n + Suc n" by simp
  also have "... =  $n * n + n + Suc n$ " using HI by simp
  also have "... =  $Suc n * Suc n$ " by simp
  finally show "cuadrado (Suc n) =  $Suc n * Suc n$ " by simp
qed

text {*
-----
Ejercicio 2. El segundo método se aplica a los números mayores que 100
que están próximos a 100. Por ejemplo,

$$102^2 = (102 + (102 - 100)) * 100 + (102 - 100)^2$$


$$= (102 + 2) * 100 + 4$$


$$= 10404$$

En general, si  $n \geq 100$ ,

$$n^2 = (n + (n - 100)) * 100 + (n - 100)^2$$

que se puede generalizar para todo  $m$ , si  $n \geq m$ 

$$n^2 = (n + (n - m)) * m + (n - m)^2$$

Demostrar la última propiedad.
----- *}

-- "La demostración automática es"
lemma cuadrado2_correcta_aux [rule_format]:
  " $\forall m \leq n. \text{cuadrado } n = (n + (n - m)) * m + \text{cuadrado } (n - m)$ "
by (induct n) (auto, case_tac m, auto)

```

```
-- "La demostración estructurada es"
lemma cuadrado2_correcta_aux2 [rule_format]:
  "∀m≤n. cuadrado n = (n+(n-m))*m + cuadrado (n-m)" (is "∀m≤n. ?P m n")
proof (induct n)
  show "∀m≤0. ?P m 0" by simp
next
  fix n
  assume HI: "∀m≤n. ?P m n"
  show "∀m ≤ Suc n. ?P m (Suc n)"
  proof
    fix m
    show "m ≤ Suc n → ?P m (Suc n)"
    proof
      assume "m ≤ Suc n"
      show "?P m (Suc n)"
      proof (cases "m")
        assume "m = 0"
        thus "?P m (Suc n)" by simp
      next
        fix x
        assume "m = Suc x"
        thus "?P m (Suc n)" using HI 'm ≤ Suc n' by auto
      qed
    qed
  qed
qed

-- "La demostración estructurada más detallada es"
lemma cuadrado2_correcta_aux3 [rule_format]:
  "∀m≤n. cuadrado n = (n+(n-m))*m + cuadrado (n-m)" (is "∀m≤n. ?P m n")
proof (induct n)
  show "∀m≤0. ?P m 0"
  proof
    fix m
    show "m ≤ 0 → ?P m 0"
    proof
      assume "m ≤ 0"
      hence "m = 0" by auto
      thus "?P m 0" by auto
    qed
  qed
next
  fix n
  assume HI: "∀m≤n. ?P m n"
  show "∀m ≤ Suc n. ?P m (Suc n)"
```

```

proof
  fix m
  show "m ≤ Suc n → ?P m (Suc n)"
  proof
    assume "m ≤ Suc n"
    show "?P m (Suc n)"
    proof (cases "m")
      assume "m = 0"
      thus "?P m (Suc n)" by simp
    next
      fix x
      assume "m = Suc x"
      show "?P m (Suc n)"
      proof -
        have "x ≤ n" using 'm ≤ Suc n' 'm = Suc x' by auto
        have "cuadrado (Suc n) = cuadrado n + n + Suc n" by simp
        also have "... = (n+(n-x))*x + cuadrado (n-x) + n + Suc n"
          using HI 'x ≤ n' by simp
        also have "... = (Suc n + (Suc n - m)) * m + cuadrado (Suc n - m)"
          using 'm = Suc x' 'm ≤ Suc n' by auto
        finally show "?P m (Suc n)" by simp
      qed
    qed
  qed
qed
qed
qed
qed

text {*
-----
Ejercicio 3. Demostrar que, si  $n \geq 100$ ,

$$n \backslash \langle^{sup} 2 = (n + (n - 100)) * 100 + (n - 100) \backslash \langle^{sup} 2$$

----- *}

-- "La demostración automática es"
lemma cuadrado2_correcta:
  "100 ≤ n ⇒ cuadrado n = (n + (n - 100)) * 100 + cuadrado (n - 100)"
by (rule cuadrado2_correcta_aux)

text {*
-----
Ejercicio 3. El tercer método se aplica a los números mayores que
terminan en 5. Por ejemplo,

$$85 \backslash \langle^{bsup} 2 \backslash \langle^{esup} = (8 * 9) * 100 + 25 = 7225$$


$$995 \backslash \langle^{bsup} 2 \backslash \langle^{esup} = (99 * 100) * 100 + 25 = 990025$$

Demostrar que

```

```

      ((10*n)+5)\<^sup>2 = (n*(n+1))*100 + 25
      ----- *}

-- "La demostración automática es"
lemma cuadrado3_correcta:
  "cuadrado((10 * n) + 5) = ((n * (Suc n)) * 100) + 25"
by (auto simp add: add_mult_distrib add_mult_distrib2)

-- "La demostración estructurada es"
lemma cuadrado3_correcta2:
  "cuadrado((10 * n) + 5) = ((n * (Suc n)) * 100) + 25"
proof -
  have "cuadrado((10 * n) + 5) = ((10 * n) + 5) * ((10 * n) + 5)"
  by (rule cuadrado_correcta)
  also have "... = ((n * (Suc n)) * 100) + 25"
  by (auto simp add: add_mult_distrib add_mult_distrib2)
  finally show "cuadrado((10 * n) + 5) = ((n * (Suc n)) * 100) + 25"
  by simp
qed

end

```

Capítulo 7

Caso de estudio: Compilación de expresiones

```
chapter {* Tema 7: Caso de estudio: Compilación de expresiones *}

theory T7
imports Main
begin

text {*
  El objetivo de este tema es contruir un compilador de expresiones
  genéricas (construidas con variables, constantes y operaciones
  binarias) a una máquina de pila y demostrar su corrección.
*}

section {* Las expresiones y el intérprete *}

text {*
  Definición. Las expresiones son las constantes, las variables
  (representadas por números naturales) y las aplicaciones de operadores
  binarios a dos expresiones.
*}

type_synonym 'v binop = "'v  $\Rightarrow$  'v  $\Rightarrow$  'v"

datatype 'v expr =
  Const 'v
| Var nat
| App "'v binop" "'v expr" "'v expr"

text {*
  Definición. [Intérprete]
```

La función "valor" toma como argumentos una expresión y un entorno (i.e. una aplicación de las variables en elementos del lenguaje) y devuelve el valor de la expresión en el entorno.

```
*}
```

```
fun valor :: "'v expr ⇒ (nat ⇒ 'v) ⇒ 'v" where
  "valor (Const b) ent = b"
| "valor (Var x) ent = ent x"
| "valor (App f e1 e2) ent = (f (valor e1 ent) (valor e2 ent))"
```

```
text {*
```

Ejemplo. A continuación mostramos algunos ejemplos de evaluación con el intérprete.

```
*}
```

```
lemma
```

```
"valor (Const 3) id = 3 ∧
  valor (Var 2) id = 2 ∧
  valor (Var 2) (λx. x+1) = 3 ∧
  valor (App (op +) (Const 3) (Var 2)) (λx. x+1) = 6 ∧
  valor (App (op +) (Const 3) (Var 2)) (λx. x+4) = 9"
```

```
by simp
```

```
section {* La máquina de pila *}
```

```
text {*
```

Nota. La máquina de pila tiene tres clases de instrucciones:

- *cargar en la pila una constante,*
- *cargar en la pila el contenido de una dirección y*
- *aplicar un operador binario a los dos elementos superiores de la pila.*

```
*}
```

```
datatype 'v instr =
```

```
  IConst 'v
```

```
| ILoad nat
```

```
| IApp "'v binop"
```

```
text {*
```

Definición. [Ejecución]

La ejecución de la máquina de pila se modeliza mediante la función "ejec" que toma una lista de instrucciones, una memoria (representada como una función de las direcciones a los valores, análogamente a los entornos) y una pila (representada como una lista) y devuelve la pila al final de la ejecución.

```
*}
```

```

fun ejec :: "'v instr list  $\Rightarrow$  (nat $\Rightarrow$ 'v)  $\Rightarrow$  'v list  $\Rightarrow$  'v list" where
  "ejec [] ent vs = vs"
| "ejec (i#is) ent vs =
  (case i of
    IConst v  $\Rightarrow$  ejec is ent (v#vs)
  | ILoad x  $\Rightarrow$  ejec is ent ((ent x)#vs)
  | IApp f  $\Rightarrow$  ejec is ent ((f (hd vs) (hd (tl vs)))#(tl(tl vs))))"

```

```

text {*
  A continuación se muestran ejemplos de ejecución.
*}

```

```

lemma
  "ejec [IConst 3] id [7] = [3,7]  $\wedge$ 
  ejec [ILoad 2, IConst 3] id [7] = [3,2,7]  $\wedge$ 
  ejec [ILoad 2, IConst 3] ( $\lambda x. x+4$ ) [7] = [3,6,7]  $\wedge$ 
  ejec [ILoad 2, IConst 3, IApp (op +)] ( $\lambda x. x+4$ ) [7] = [9,7]"
by simp

```

```

section {* El compilador *}

```

```

text {*
  Definición. El compilador "comp" traduce una expresión en una lista de
  instrucciones.
*}

```

```

fun comp :: "'v expr  $\Rightarrow$  'v instr list" where
  "comp (Const v) = [IConst v]"
| "comp (Var x) = [ILoad x]"
| "comp (App f e1 e2) = (comp e2) @ (comp e1) @ [IApp f]"

```

```

text {*
  A continuación se muestran ejemplos de compilación.
*}

```

```

lemma
  "comp (Const 3) = [IConst 3]  $\wedge$ 
  comp (Var 2) = [ILoad 2]  $\wedge$ 
  comp (App (op +) (Const 3) (Var 2)) = [ILoad 2, IConst 3, IApp (op +)]"
by simp

```

```

section {* Corrección del compilador *}

```

```

text {*

```

```

Para demostrar que el compilador es correcto, probamos que el
resultado de compilar una expresión y a continuación ejecutarla es lo
mismo que interpretarla; es decir,
*}

theorem "ejec (comp e) ent [] = [valor e ent]"
oops

text {*
  El teorema anterior no puede demostrarse por inducción en e. Para
  demostrarlo, lo generalizamos a
*}

theorem "∀vs. ejec (comp e) ent vs = (valor e ent)#vs"
oops

text {*
  En la demostración del teorema anterior usaremos el siguiente lema.
*}

lemma ejec_append:
  "∀ vs. ejec (xs@ys) ent vs = ejec ys ent (ejec xs ent vs)" (is "?P xs")
proof (induct xs)
  show "?P []" by simp
next
  fix a xs
  assume "?P xs"
  thus "?P (a#xs)" by (cases "a", auto)
qed

-- "La demostración detallada es"
lemma ejec_append_1:
  "∀ vs. ejec (xs@ys) ent vs = ejec ys ent (ejec xs ent vs)" (is "?P xs")
proof (induct xs)
  show "?P []" by simp
next
  fix a xs
  assume HI: "?P xs"
  thus "?P (a#xs)"
  proof (cases "a")
    case IConst thus ?thesis using HI by simp
  next
    case ILoad thus ?thesis using HI by simp
  next
    case IApp thus ?thesis using HI by simp
  end
end

```

```

qed
qed

text {*
  Una demostración más detallada del lema es la siguiente:
*}

lemma ejec_append_2:
  "∀vs. ejec (xs@ys) ent vs = ejec ys ent (ejec xs ent vs)" (is "?P xs")
proof (induct xs)
  show "?P []" by simp
next
  fix a xs
  assume HI: "?P xs"
  thus "?P (a#xs)"
  proof (cases "a")
    fix v assume C1: "a=IConst v"
    show "∀vs. ejec ((a#xs)@ys) ent vs = ejec ys ent (ejec (a#xs) ent vs)"
    proof
      fix vs
      have "ejec ((a#xs)@ys) ent vs = ejec (((IConst v)#xs)@ys) ent vs"
        using C1 by simp
      also have "... = ejec (xs@ys) ent (v#vs)" by simp
      also have "... = ejec ys ent (ejec xs ent (v#vs))" using HI by simp
      also have "... = ejec ys ent (ejec ((IConst v)#xs) ent vs)" by simp
      also have "... = ejec ys ent (ejec (a#xs) ent vs)" using C1 by simp
      finally show "ejec ((a#xs)@ys) ent vs =
        ejec ys ent (ejec (a#xs) ent vs)" .
    qed
  qed
next
  fix n assume C2: "a=ILoad n"
  show "∀vs. ejec ((a#xs)@ys) ent vs = ejec ys ent (ejec (a#xs) ent vs)"
  proof
    fix vs
    have "ejec ((a#xs)@ys) ent vs = ejec (((ILoad n)#xs)@ys) ent vs"
      using C2 by simp
    also have "... = ejec (xs@ys) ent ((ent n)#vs)" by simp
    also have "... = ejec ys ent (ejec xs ent ((ent n)#vs))" using HI by simp
    also have "... = ejec ys ent (ejec ((ILoad n)#xs) ent vs)" by simp
    also have "... = ejec ys ent (ejec (a#xs) ent vs)" using C2 by simp
    finally show "ejec ((a#xs)@ys) ent vs =
      ejec ys ent (ejec (a#xs) ent vs)" .
  qed
next
  fix f assume C3: "a=IApp f"

```

```

show "∀vs. ejec ((a#xs)@ys) ent vs = ejec ys ent (ejec (a#xs) ent vs)"
proof
  fix vs
  have "ejec ((a#xs)@ys) ent vs = ejec (((IApp f)#xs)@ys) ent vs"
    using C3 by simp
  also have "... = ejec (xs@ys) ent ((f (hd vs) (hd (tl vs)))#(tl(tl vs)))"
    by simp
  also have "... = ejec ys
    ent
    (ejec xs ent ((f (hd vs) (hd (tl vs)))#(tl(tl vs))))"
    using HI by simp
  also have "... = ejec ys ent (ejec ((IApp f)#xs) ent vs)" by simp
  also have "... = ejec ys ent (ejec (a#xs) ent vs)" using C3 by simp
  finally show "ejec ((a#xs)@ys) ent vs =
    ejec ys ent (ejec (a#xs) ent vs)" .

qed
qed
qed

text {*
  La demostración automática del teorema es
*}

theorem "∀vs. ejec (comp e) ent vs = (valor e ent)#vs"
by (induct e) (auto simp add:ejec_append)

text {*
  La demostración estructurada del teorema es
*}

theorem "∀vs. ejec (comp e) ent vs = (valor e ent)#vs"
proof (induct e)
  fix v
  show "∀vs. ejec (comp (Const v)) ent vs = (valor (Const v) ent)#vs" by simp
next
  fix x
  show "∀vs. ejec (comp (Var x)) ent vs = (valor (Var x) ent) # vs" by simp
next
  fix f e1 e2
  assume HI1: "∀vs. ejec (comp e1) ent vs = (valor e1 ent) # vs"
  and HI2: "∀vs. ejec (comp e2) ent vs = (valor e2 ent) # vs"
  show "∀vs. ejec (comp (App f e1 e2)) ent vs = (valor (App f e1 e2) ent) # vs"
  proof
    fix vs
    have "ejec (comp (App f e1 e2)) ent vs

```

```

      = ejec ((comp e2) @ (comp e1) @ [IApp f]) ent vs" by simp
also have "... = ejec ((comp e1) @ [IApp f]) ent (ejec (comp e2) ent vs)"
  using ejec_append by blast
also have "... = ejec [IApp f]
  ent
    (ejec (comp e1) ent (ejec (comp e2) ent vs))"
  using ejec_append by blast
also have "... = ejec [IApp f] ent (ejec (comp e1) ent ((valor e2 ent)#vs))"
  using HI2 by simp
also have "... = ejec [IApp f] ent ((valor e1 ent)#((valor e2 ent)#vs))"
  using HI1 by simp
also have "... = (f (valor e1 ent) (valor e2 ent))#vs" by simp
also have "... = (valor (App f e1 e2) ent) # vs" by simp
finally
show "ejec (comp (App f e1 e2)) ent vs = (valor (App f e1 e2) ent) # vs"
  by blast
qed
qed
end

```


Capítulo 8

Conjuntos, funciones y relaciones

```
chapter {* Tema 8: Conjuntos, funciones y relaciones *}
```

```
theory T8
imports Main
begin
```

```
section {* Conjuntos *}
```

```
subsection {* Operaciones con conjuntos *}
```

```
text {*
```

Nota. La teoría elemental de conjuntos es HOL/Set.thy.

Nota. En un conjunto todos los elementos son del mismo tipo (por ejemplo, del tipo τ) y el conjunto tiene tipo (en el ejemplo, " τ set").

Reglas de la intersección:

- *IntI:* $\llbracket c \in A; c \in B \rrbracket \implies c \in A \cap B$
- *IntD1:* $c \in A \cap B \implies c \in A$
- *IntD2:* $c \in A \cap B \implies c \in B$

Nota. Propiedades del complementario:

- *Compl_iff:* $(c \in - A) = (c \notin A)$
- *Compl_Un:* $-(A \cup B) = -A \cap -B$

Nota. El conjunto vacío se representa por $\{\}$ y el universal por UNIV.

Nota. Propiedades de la diferencia y del complementario:

- *Diff_disjoint:* $A \cap (B - A) = \{\}$

· *Compl_partition*: $A \cup -A = \text{UNIV}$

Nota. Reglas de la relación de subconjunto:

- *subsetI*: $(\bigwedge x. x \in A \implies x \in B) \implies A \subseteq B$
- *subsetD*: $\llbracket A \subseteq B; c \in A \rrbracket \implies c \in B$

Nota. Ejemplo trivial.

**/*

lemma "(A $\cup$ B $\subseteq$ C) = (A $\subseteq$ C $\wedge$ B $\subseteq$ C)"

by blast

text {*

Nota. Otro ejemplo trivial.

**/*

lemma "(A $\subseteq$ -B) = (B $\subseteq$ -A)"

by blast

text {*

Principio de extensionalidad de conjuntos:

- *set_ext*: $(\bigwedge x. (x \in A) = (x \in B)) \implies A = B$

Reglas de la igualdad de conjuntos:

- *equalityI*: $\llbracket A \subseteq B; B \subseteq A \rrbracket \implies A = B$
- *equalityE*: $\llbracket A = B; \llbracket A \subseteq B; B \subseteq A \rrbracket \implies P \rrbracket \implies P$

Lema. [Analogía entre intersección y conjunción]

" $x \in A \cap B$ " syss " $x \in A$ " y " $x \in B$ ".

**/*

lemma "(x $\in$ A $\cap$ B) = (x $\in$ A $\wedge$ x $\in$ B)"

by simp

text {*

Lema. [Analogía entre unión y disyunción]

" $x \in A \cup B$ " syss " $x \in A$ " ó " $x \in B$ ".

**/*

lemma "(x $\in$ A $\cup$ B) = (x $\in$ A $\vee$ x $\in$ B)"

by simp

text {*

Lema. [Analogía entre subconjunto e implicación]

"(A $\subseteq$ B)" syss para todo "x", si " $x \in A$ " entonces " $x \in B$ ".

```

*}

```

```

lemma "(A ⊆ B) = (∀ x. x ∈ A → x ∈ B)"

```

```

by auto

```

```

text {*

```

Lema. [Analogía entre complementario y negación]

x pertenece al complementario de A syss x no pertenece a A.

```

*}

```

```

lemma "(x ∈ -A) = (x ∉ A)"

```

```

by simp

```

```

subsection {* Notación de conjuntos finitos *}

```

```

text {*

```

Nota. La teoría de conjuntos finitos es HOL/Finite_Set.thy.

Nota. Los conjuntos finitos se definen por inducción a partir de las siguientes reglas inductivas:

- *El conjunto vacío es un conjunto finito.*
- *emptyI: "finite {}"*
- *Si se le añade un elemento a un conjunto finito se obtiene otro conjunto finito.*
- *insertI: "finite A ⇒ finite (insert a A)"*

A continuación se muestran ejemplos de conjuntos finitos.

```

*}

```

```

lemma

```

```

  "insert 2 {} = {2} ∧
   insert 3 {2} = {2,3} ∧
   insert 2 {2,3} = {2,3} ∧
   {2,3} = {3,2,3,2,2}"

```

```

by auto

```

```

text {*

```

Nota. Los conjuntos finitos se representan con la notación conjuntista habitual: los elementos entre llaves y separados por comas.

Nota. Lema trivial.

```

*}

```

```

lemma "{a,b} ∪ {c,d} = {a,b,c,d}"

```

```

by blast

```

```

text {*
  Nota. Conjetura falsa.
*}

lemma "{a,b} ∩ {b,c} = {b}"
refute
oops

text {*
  Nota. Conjetura corregida.
*}

lemma "{a,b} ∩ {b,c} = (if a=c then {a,b} else {b})"
by auto

text {*
  Sumas y productos de conjuntos finitos:
  · (setsum f A) es la suma de la aplicación de f a los elementos del
    conjunto finito A,
  · (setprod f A) es producto de la aplicación de f a los elementos del
    conjunto finito A,
  ·  $\sum A$  es la suma de los elementos del conjunto finito A,
  ·  $\prod A$  es el producto de los elementos del conjunto finito A.

  Ejemplos de definiciones recursivas sobre conjuntos finitos:
  Sea A un conjunto finito de números naturales.
  · sumaConj A es la suma de los elementos A.
  · productoConj A es el producto de los elementos de A.
  · sumaCuadradosconj A es la suma de los cuadrados de los elementos A.
*}

definition sumaConj :: "nat set ⇒ nat" where
  "sumaConj S ≡  $\sum S$ "

definition productoConj :: "nat set ⇒ nat" where
  "productoConj S ≡  $\prod S$ "

definition sumaCuadradosConj :: "nat set ⇒ nat" where
  "sumaCuadradosConj S ≡ setsum (λx. x*x) S"

text {*
  Nota. Para simplificar lo que sigue, declaramos las anteriores
  definiciones como reglas de simplificación.
*}

```

```

declare sumaConj_def[simp]
declare productoConj_def[simp]
declare sumaCuadradosConj_def[simp]

text {*
  Ejemplos de evaluación de las anteriores definiciones recursivas.
*}

lemma
  "sumaConj {1,2,3,4} = 10 ∧
   productoConj {1,2,3} = productoConj {3,2} ∧
   sumaCuadradosConj {1,2,3,4} = 30"
by simp

text {*
  Inducción sobre conjuntos finitos: Para demostrar que todos los
  conjuntos finitos tienen una propiedad P basta probar que
  · El conjunto vacío tiene la propiedad P.
  · Si a un conjunto finito que tiene la propiedad P se le añade un
    nuevo elemento, el conjunto obtenido sigue teniendo la propiedad P.
  En forma de regla
  · finite_induct: 
$$\begin{aligned} & \llbracket \text{finite } F; \\ & \quad P \ \{\} \rrbracket; \\ & \quad \bigwedge x \ F. \llbracket \text{finite } F; x \notin F; P \ F \rrbracket \implies P \ (\{x\} \cup F) \rrbracket \\ & \implies P \ F \end{aligned}$$


  Lema. [Ejemplo de inducción sobre conjuntos finitos]
  Sea S un conjunto finito de números naturales. Entonces todos los
  elementos de S son menores o iguales que la suma de los elementos de S.

  Demostración automática:
*}

lemma "finite S  $\implies \forall x \in S. x \leq \text{sumaConj } S$ "
by (induct rule: finite_induct) auto

text {*
  Demostración estructurada:
*}

lemma sumaConj_acota: "finite S  $\implies \forall x \in S. x \leq \text{sumaConj } S$ "
proof (induct rule: finite_induct)
  show " $\forall x \in \{\}. x \leq \text{sumaConj } \{\}$ " by simp
next

```

```

fix x and F
assume fF: "finite F"
  and xF: "x ∉ F"
  and HI: "∀ x ∈ F. x ≤ sumaConj F"
show "∀ y ∈ insert x F. y ≤ sumaConj (insert x F)"
proof
  fix y
  assume "y ∈ insert x F"
  show "y ≤ sumaConj (insert x F)"
  proof (cases "y = x")
    assume "y = x"
    hence "y ≤ x + (sumaConj F)" by simp
    also have "... = sumaConj (insert x F)" using fF xF by simp
    finally show ?thesis .
  next
    assume "y ≠ x"
    hence "y ∈ F" using 'y ∈ insert x F' by simp
    hence "y ≤ sumaConj F" using HI by blast
    also have "... ≤ x + (sumaConj F)" by simp
    also have "... = sumaConj (insert x F)" using fF xF by simp
    finally show ?thesis .
  qed
qed
qed

subsection {* Definiciones por comprensión *}

text {*
  El conjunto de los elementos que cumple la propiedad P se representa
  por {x. P}.

  Reglas de comprensión (relación entre colección y pertenencia):
  · mem_Collect_eq: (a ∈ {x. P x}) = P a
  · Collect_mem_eq: {x. x ∈ A} = A

  Dos lemas triviales.
*}

lemma "{x. P x ∨ x ∈ A} = {x. P x} ∪ A"
by blast

lemma "{x. P x → Q x} = -{x. P x} ∪ {x. Q x}"
by blast

text {*

```

Nota. Ejemplo con la sintaxis general de comprensión.

**/*

lemma

"{p*q | p q. p ∈ prime ∧ q ∈ prime} =
 {z. ∃p q. z = p*q ∧ p ∈ prime ∧ q ∈ prime}"

by blast

text {*

En HOL, la notación conjuntista es azúcar sintáctica:

- $x \in A$ es equivalente a $A(x)$.
- $\{x. P\}$ es equivalente a $\lambda x. P$.

**/*

text {*

Definición. [Ejemplo de definición por comprensión]

*El conjunto de los pares es el de los números n para los que existe un m tal que $n = 2*m$.*

**/*

definition Pares :: "nat set" **where**

"Pares $\equiv$ {n. ∃m. n = 2*m }"

text {*

Ejemplo. Los números 2 y 34 son pares.

**/*

lemma

"2 ∈ Pares ∧
 34 ∈ Pares"

by (simp add: Pares_def)

text {*

*Definición. El conjunto de los impares es el de los números n para los que existe un m tal que $n = 2*m + 1$.*

**/*

definition Impares :: "nat set" **where**

"Impares $\equiv$ {n. ∃m. n = 2*m + 1 }"

text {*

Lema. [Ejemplo con las reglas de intersección y comprensión]

El conjunto de los pares es disjunto con el de los impares.

**/*

```

lemma "x ∉ (Pares n Impares)"
proof
  fix x assume S: "x ∈ (Pares n Impares)"
  hence "x ∈ Pares" by (rule IntD1)
  hence "∃m. x = 2 * m" by (simp only: Pares_def mem_Collect_eq)
  then obtain p where p: "x = 2 * p" ..
  from S have "x ∈ Impares" by (rule IntD2)
  hence "∃ m. x = 2 * m + 1" by (simp only: Impares_def mem_Collect_eq)
  then obtain q where q: "x = 2 * q + 1" ..
  from p and q show "False" by arith
qed

```

subsection *{* Cuantificadores acotados **}

text *{**

Reglas de cuantificador universal acotado ("bounded"):

- *ballI*: $(\bigwedge x. x \in A \implies P\ x) \implies \forall x \in A. P\ x$
- *bspec*: $\llbracket \forall x \in A. P\ x; x \in A \rrbracket \implies P\ x$

Reglas de cuantificador existencial acotado ("bounded"):

- *bexI*: $\llbracket P\ x; x \in A \rrbracket \implies \exists x \in A. P\ x$
- *bexE*: $\llbracket \exists x \in A. P\ x; \bigwedge x. \llbracket x \in A; P\ x \rrbracket \implies Q \rrbracket \implies Q$

Reglas de la unión indexada:

- *UN_iff*: $(b \in (\bigcup x \in A. B\ x)) = (\exists x \in A. b \in B\ x)$
- *UN_I*: $\llbracket a \in A; b \in B\ a \rrbracket \implies b \in (\bigcup x \in A. B\ x)$
- *UN_E*: $\llbracket b \in (\bigcup x \in A. B\ x); \bigwedge x. \llbracket x \in A; b \in B\ x \rrbracket \implies R \rrbracket \implies R$

Reglas de la unión de una familia:

- *Union_def*: $\bigcup S = (\bigcup x \in S. x)$
- *Union_iff*: $(A \in \bigcup C) = (\exists X \in C. A \in X)$

Reglas de la intersección indexada:

- *INT_iff*: $(b \in (\bigcap x \in A. B\ x)) = (\forall x \in A. b \in B\ x)$
- *INT_I*: $(\bigwedge x. x \in A \implies b \in B\ x) \implies b \in (\bigcap x \in A. B\ x)$
- *INT_E*: $\llbracket b \in (\bigcap x \in A. B\ x); b \in B\ a \implies R; a \notin A \implies R \rrbracket \implies R$

Reglas de la intersección de una familia:

- *Inter_def*: $\bigcap S = (\bigcap x \in S. x)$
- *Inter_iff*: $(A \in \bigcap C) = (\forall X \in C. A \in X)$

Abreviaturas:

- "Collect P" es lo mismo que " $\{x. P\}$ ".
- "All P" es lo mismo que " $\forall x. P\ x$ ".
- "Ex P" es lo mismo que " $\exists x. P\ x$ ".

```

· "Ball A P" es lo mismo que " $\forall x \in A. P\ x$ ".
· "Bex A P" es lo mismo que " $\exists x \in A. P\ x$ ".
*}

```

subsection *{* Conjuntos finitos y cardinalidad **}

```

text {*
  El número de elementos de un conjunto finito A es el cardinal de A y
  se representa por "card A".

```

```

  Ejemplos de cardinales de conjuntos finitos.
*}

```

lemma

```

"card {} = 0 ∧
 card {4} = 1 ∧
 card {4,1} = 2 ∧
 x ≠ y ⇒ card {x,y} = 2"

```

```

by simp

```

text *{**

Propiedades de cardinales:

```

· Cardinal de la unión de conjuntos finitos:
  card_Un_Int:  $\llbracket \text{finite } A; \text{ finite } B \rrbracket$ 
                 $\implies \text{card } A + \text{card } B = \text{card } (A \cup B) + \text{card } (A \cap B)$ 
· Cardinal del conjunto potencia:
  card_Pow:  $\text{finite } A \implies \text{card } (\text{Pow } A) = 2^{\text{card } A}$ 

```

```

*}

```

section *{* Funciones **}

text *{**

La teoría de funciones es HOL/Fun.thy.

```

*}

```

subsection *{* Nociones básicas de funciones **}

text *{**

Principio de extensionalidad para funciones:

```

· ext:  $(\bigwedge x. f\ x = g\ x) \implies f = g$ 

```

Actualización de funciones

```

· fun_upd_apply:  $(f(x := y))\ z = (\text{if } z = x \text{ then } y \text{ else } f\ z)$ 
· fun_upd_upd:  $f(x := y, x := z) = f(x := z)$ 

```

Función identidad

· *id_def*: $id \equiv \lambda x. x$

Composición de funciones:

· *o_def*: $f \circ g \equiv (\lambda x. f (g x))$

Asociatividad de la composición:

· *o_assoc*: $f \circ (g \circ h) \equiv (f \circ g) \circ h$

**/*

subsection *{* Funciones inyectivas, suprayectivas y biyectivas **

text *{**

Función inyectiva sobre A:

· *inj_on_def*: $inj_on\ f\ A \equiv \forall x \in A. \forall y \in A. f\ x = f\ y \rightarrow x = y$

Nota. "inj f" es una abreviatura de "inj_on f UNIV".

Función suprayectiva:

· *surj_def*: $surj\ f \equiv \forall y. \exists x. y = f\ x$

Función biyectiva:

· *bij_def*: $bij\ f \equiv inj\ f \wedge surj\ f$

Propiedades de las funciones inversas:

· *inv_f_f*: $inj\ f \implies inv\ f\ (f\ x) = x$

· *surj_f_inv_f*: $surj\ f \implies f\ (inv\ f\ y) = y$

· *inv_inv_eq*: $bij\ f \implies inv\ (inv\ f) = f$

Igualdad de funciones (por extensionalidad):

· *fun_eq_iff*: $(f = g) \equiv (\forall x. f\ x = g\ x)$

Lema. Una función inyectiva puede cancelarse en el lado izquierdo de la composición de funciones.

**/*

lemma

assumes "inj f"

shows " $(f \circ g = f \circ h) \equiv (g = h)$ "

proof

assume " $f \circ g = f \circ h$ "

thus " $g = h$ " **using** 'inj f' **by** (simp add: fun_eq_iff inj_on_def)

next

assume " $g = h$ "

thus " $f \circ g = f \circ h$ " **by** auto

qed

```
text {*
  Una demostración más detallada es la siguiente
*}
```

lemma

```
  assumes "inj f"
  shows "(f ◦ g = f ◦ h) = (g = h)"
```

proof

```
  assume "f ◦ g = f ◦ h"
  show "g = h"
  proof
    fix x
    have "(f ◦ g)(x) = (f ◦ h)(x)" using 'f ◦ g = f ◦ h' by simp
    hence "f(g(x)) = f(h(x))" by simp
    thus "g(x) = h(x)" using 'inj f' by (simp add:inj_on_def)
```

qed

next

```
  assume "g = h"
  show "f ◦ g = f ◦ h"
  proof
    fix x
    have "(f ◦ g) x = f(g(x))" by simp
    also have "... = f(h(x))" using 'g = h' by simp
    also have "... = (f ◦ h) x" by simp
    finally show "(f ◦ g) x = (f ◦ h) x" by simp
```

qed

qed

subsubsection {* Función imagen *

text {*

Imagen de un conjunto mediante una función:

· *image_def:* $f' A = \{y. (\exists x \in A. y = f x)\}$

Propiedades de la imagen:

· *image_compose:* $(f \circ g)'r = f'g'r$
 · *image_Un:* $f'(A \cup B) = f'A \cup f'B$
 · *image_Int:* $\text{inj } f \implies f'(A \cap B) = f'A \cap f'B$

Ejemplos de demostraciones triviales de propiedades de la imagen.

*)

lemma "f'A ∪ g'A = (⋃x∈A. {f x, g x})"

by auto

lemma "f' {(x,y). P x y} = {f(x,y) | x y. P x y}"

by auto

text {*

El rango de una función ("range f") es la imagen del universo ("f'UNIV").

Imagen inversa de un conjunto:

· *vimage_def: $f^{-1} B \equiv \{x. f x : B\}$*

Propiedad de la imagen inversa de un conjunto:

· *vimage_Compl: $f^{-1} (-A) = -(f^{-1} A)$*

*)

section {* Relaciones *}

subsection {* Relaciones básicas *}

text {*

La teoría de relaciones es HOL/Relation.thy.

Las relaciones son conjuntos de pares.

Relación identidad:

· *Id_def: $Id \equiv \{p. \exists x. p = (x,x)\}$*

Composición de relaciones:

· *rel_comp_def: $r \circ s \equiv \{(x,z). \exists y. (x, y) \in r \wedge (y, z) \in s\}$*

Propiedades:

· *R_O_Id: $R \circ Id = R$*

· *rel_comp_mono: $\llbracket r' \subseteq r; s' \subseteq s \rrbracket \implies (r' \circ s') \subseteq (r \circ s)$*

Imagen inversa de una relación:

· *converse_iff: $((a,b) \in r \implies (b,a) \in r)$*

Propiedad de la imagen inversa de una relación:

· *converse_rel_comp: $(r \circ s)^{-1} = s^{-1} \circ r^{-1}$*

Imagen de un conjunto mediante una relación:

· *Image_iff: $(b \in r' A) = (\exists x:A. (x, b) \in r)$*

Dominio de una relación:

· *Domain_iff: $(a \in \text{Domain } r) = (\exists y. (a, y) \in r)$*

Rango de una relación:

· *Range_iff*: $(a \in \text{Range } r) = (\exists y. (y, a) \in r)$
 *}

subsection {** Clausura reflexiva y transitiva **}

text {***

La teoría de la clausura reflexiva y transitiva de una relación es HOL/Transitive_Closure.thy.

Potencias de relaciones:

· $R \wedge^0 = \text{Id}$
 · $R \wedge^{(\text{Suc } n)} = (R \wedge^n) \circ R$

La clausura reflexiva y transitiva de la relación r es la menor solución de la ecuación:

· *rtrancl_unfold*: $r \mid \wedge^{\sup} = \text{Id} \cup (r \mid \wedge^{\sup} \circ r)$

Propiedades básicas de la clausura reflexiva y transitiva:

· *rtrancl_refl*: $(a, a) \in r \mid \wedge^{\sup}$
 · *r_into_rtrancl*: $p \in r \implies p \in r \mid \wedge^{\sup}$
 · *rtrancl_trans*: $\llbracket (a, b) \in r \mid \wedge^{\sup}; (b, c) \in r \mid \wedge^{\sup} \rrbracket \implies (a, c) \in r \mid \wedge^{\sup}$

Inducción sobre la clausura reflexiva y transitiva

· *rtrancl_induct*: $\llbracket (a, b) \in r \mid \wedge^{\sup};$
 $P \ b;$
 $\bigwedge y \ z. \llbracket (y, z) \in r; (z, b) \in r \mid \wedge^{\sup}; P \ z \rrbracket \implies P \ y \rrbracket$
 $\implies P \ a'' \rrbracket$

Idempotencia de la clausura reflexiva y transitiva:

· *rtrancl_idemp*: $(r \mid \wedge^{\sup}) \mid \wedge^{\sup} = r \mid \wedge^{\sup}$

Reglas de introducción de la clausura transitiva:

· *r_into_trancl'*: $p \in r \implies p \in r \mid \wedge^{\sup+}$
 · *trancl_trans*: $\llbracket (a, b) \in r \mid \wedge^{\sup+}; (b, c) \in r \mid \wedge^{\sup+} \rrbracket \implies (a, c) \in r \mid \wedge^{\sup+}$

Ejemplo de propiedad:

· *trancl_converse*: $(r^{-1}) \mid \wedge^{\sup+} = (r \mid \wedge^{\sup+})^{-1}$
 *}

subsection {** Una demostración elemental **}

text {***

El teorema que se desea demostrar es que la clausura reflexiva y

```

    transitiva conmuta con la inversa (rtrancl_converse). Para
    demostrarlo introducimos dos lemas auxiliares: rtrancl_converseD y
    rtrancl_converseI.
*}

lemma rtrancl_converseD: "(x,y) ∈ (r⁻¹)\<sup>* ⇒ (y,x) ∈ r\<sup>*"
proof (induct rule:rtrancl_induct)
  show "(x,x) ∈ r\<sup>*" by (rule rtrancl_refl)
next
  fix y z
  assume "(x,y) ∈ (r⁻¹)\<sup>*" and "(y,z) ∈ r⁻¹" and "(y,x) ∈ r\<sup>*"
  show "(z,x) ∈ r\<sup>*"
  proof (rule rtrancl_trans)
    show "(z,y) ∈ r\<sup>*" using '(y,z) ∈ r⁻¹' by simp
  next
    show "(y,x) ∈ r\<sup>*" using '(y,x) ∈ r\<sup>*' by simp
  qed
qed

lemma rtrancl_converseI: "(y,x) ∈ r\<sup>* ⇒ (x,y) ∈ (r⁻¹)\<sup>*"
proof (induct rule:rtrancl_induct)
  show "(y,y) ∈ (r⁻¹)\<sup>*" by (rule rtrancl_refl)
next
  fix u z
  assume "(y,u) ∈ r\<sup>*" and "(u,z) ∈ r" and "(u,y) ∈ (r⁻¹)\<sup>*"
  show "(z,y) ∈ (r⁻¹)\<sup>*"
  proof (rule rtrancl_trans)
    show "(z,u) ∈ (r⁻¹)\<sup>*" using '(u,z) ∈ r' by auto
  next
    show "(u,y) ∈ (r⁻¹)\<sup>*" using '(u,y) ∈ (r⁻¹)\<sup>*' by simp
  qed
qed

theorem rtrancl_converse: "(r⁻¹)\<sup>* = (r\<sup>*)⁻¹"
proof
  show "(r⁻¹)\<sup>* ⊆ (r\<sup>*)⁻¹" by (auto simp add:rtrancl_converseD)
next
  show "(r\<sup>*)⁻¹ ⊆ (r⁻¹)\<sup>*" by (auto simp add:rtrancl_converseI)
qed

text {*
  Puede demostrarse de manera más corta como sigue:
*}

theorem "(r⁻¹)\<sup>* = (r\<sup>*)⁻¹"

```

by (auto intro: rtrancl_converseI dest: rtrancl_converseD)

section *{* Relaciones bien fundamentadas e inducción **}

text *{**

*La teoría de las relaciones bien fundamentadas es
HOL/Wellfounded_Relations.thy.*

*La relación-objeto "less_than" es el orden de los naturales que es
bien fundamentada:*

- *less_than_iff: $((x,y) \in \text{less_than}) = (x < y)$*
- *wf_less_than: wf less_than*

Notas sobre medidas:

- *Imagen inversa de una relación mediante una función:*
 - *inv_image_def: $\text{inv_image } r \ f \equiv \{(x,y). (f \ x, f \ y) \in r\}$*
- *Conservación de la buena fundamentación:*
 - *wf_inv_image: $\text{wf } r \implies \text{wf } (\text{inv_image } r \ f)$*
- *Definición de la medida:*
 - *measure_def: $\text{measure} \equiv \text{inv_image less_than}$*
- *Buena fundamentación de la medida:*
 - *wf_measure: $\text{wf } (\text{measure } f)$*

**}*

text *{**

Notas sobre el producto lexicográfico:

- *Definición del producto lexicográfico (lex_prod_def):*

$$ra <^* \text{lex}^* > rb \equiv \{((a,b), (a',b')). (a,a') \in ra \vee (a = a' \wedge (b,b') \in rb)\}$$
- *Conservación de la buena fundamentación:*
 - *wf_lex_prod: $\llbracket \text{wf } ra; \text{wf } rb \rrbracket \implies \text{wf } (ra <^* \text{lex}^* > rb)$*

El orden de multiconjuntos está en la teoría HOL/Library/Multiset.thy.

Inducción sobre relaciones bien fundamentadas:

- *wf_induct: $\llbracket \text{wf } r; \bigwedge x. (\bigwedge y. (y,x) \in r \implies P \ y) \implies P \ x \rrbracket \implies P \ a$*

**}*

end

8.1. Gramáticas libres de contexto

```
chapter {* T8R1: Gramáticas libres de contexto *}

```

```
theory T8R1
imports Main
begin

```

```
text {*
  En esta relación se definen dos gramáticas libres de contexto y se
  demuestra que son equivalentes. Además, se define por recursión una
  función para reconocer las palabras de la gramática y se demuestra que
  es correcta y completa. *}

```

```
text {*
  -----
  Ejercicio 1. Una gramática libre de contexto para las expresiones
  parentizadas es
     $S \rightarrow \varepsilon \mid ' ( ' S ' ) ' \mid SS$ 
  definir inductivamente la gramática  $S$  usando  $A$  y  $B$  para  $' ( ' y ' ) '$ ,
  respectivamente.
  ----- *}

```

```
datatype alfabeto = A | B

```

```
inductive_set S :: "alfabeto list set" where
  S1: "[ ] ∈ S"
| S2: " $w \in S \implies [A] @ w @ [B] \in S$ "
| S3: " $v \in S \implies w \in S \implies v @ w \in S$ "

```

```
text {*
  -----
  Ejercicio 2. Otra gramática libre de contexto para las expresiones
  parentizadas es
     $T \rightarrow \varepsilon \mid T ' ( ' T ' ) '$ 
  definir inductivamente la gramática  $T$  usando  $A$  y  $B$  para  $' ( ' y ' ) '$ ,
  respectivamente.
  ----- *}

```

```
inductive_set T :: "alfabeto list set" where
  T1: "[ ] ∈ T"
| T2: " $v \in T \implies w \in T \implies v @ [A] @ w @ [B] \in T$ "

```

```
text {*
  -----

```

Ejercicio 3. Demostrar que T está contenido en S .

----- *

```
lemma T_en_S:
  assumes "w ∈ T"
  shows   "w ∈ S"
using assms
proof (induct rule: T.induct)
  show "[ ] ∈ S" by (rule S1)
next
  fix v w
  assume "v ∈ T" and "v ∈ S" and "w ∈ T" and "w ∈ S"
  have "[A] @ w @ [B] ∈ S" using 'w ∈ S' by (rule S2)
  with 'v ∈ S' show "v @ [A] @ w @ [B] ∈ S" by (rule S3)
qed
```

text {*

Ejercicio 4. Demostrar que S está contenido en T .

----- *

text {*

Se usarán dos lemas auxiliares:

- $S_en_T_aux1: w \in T \implies [A] @ w @ [B] \in T$
- $S_en_T_aux2: v \in T \implies u \in T \implies u @ v \in T$

*)

-- "La demostración estructurada del primer lema es"

```
lemma S_en_T_aux1:
  assumes "w ∈ T"
  shows   "[A] @ w @ [B] ∈ T"
proof -
  have "[ ] ∈ T" by (rule T1)
  hence "[ ] @ [A] @ w @ [B] ∈ T" using assms by (rule T2)
  thus "[A] @ w @ [B] ∈ T" by simp
qed
```

-- "La demostración automática del primer lema es"

```
lemma S_en_T_aux1b:
  "w ∈ T  $\implies$  [A] @ w @ [B] ∈ T"
using T1 T2[where v = "[ ]"] by simp
```

-- "La demostración estructurada del segundo lema es"

```
lemma S_en_T_aux2:
  "v ∈ T  $\implies$  u ∈ T  $\implies$  u @ v ∈ T"
```

```

proof (induct rule: T.induct)
  assume "u ∈ T"
  thus "u @ [] ∈ T" by auto
next
  fix w1 w2
  assume "w1 ∈ T"
    "u ∈ T ⇒ u @ w1 ∈ T"
    "w2 ∈ T"
    "u ∈ T ⇒ u @ w2 ∈ T"
    "u ∈ T"
  hence "u @ w1 ∈ T" by simp
  hence "(u @ w1) @ [A] @ w2 @ [B] ∈ T" using 'w2 ∈ T' by (rule T2)
  thus "u @ w1 @ [A] @ w2 @ [B] ∈ T" by auto
qed

```

```

lemma S_en_T:
  "w ∈ S ⇒ w ∈ T"
proof (induct rule: S.induct)
  show "[] ∈ T" by (rule T1)
next
  fix w
  assume "w ∈ S" "w ∈ T"
  show "[A] @ w @ [B] ∈ T" using 'w ∈ T' by (rule S_en_T_aux1)
next
  fix v w
  assume "v ∈ S" "v ∈ T" "w ∈ S" "w ∈ T"
  show "v @ w ∈ T" using 'w ∈ T' 'v ∈ T' by (rule S_en_T_aux2)
qed

```

```

text {*
-----
Ejercicio 4. Demostrar que S y T son iguales.
----- *}

```

```

lemma S_igual_T:
  "S = T"
by (auto simp add: S_en_T T_en_S)

```

```

text {*
-----
Ejercicio 5. En lugar de una gramática, se puede usar el siguiente
procedimiento para determinar si la cadena es una sucesión de
paréntesis bien balanceada: se recorre la cadena de izquierda a
derecha contando cuántos paréntesis de necesitan para que esté bien
balanceada. Si el contador al final de la cadena es 0, la cadena está

```

bien balanceada.

Definir la función

balanceada :: alfabeto list $\Rightarrow$ bool

tal que (balanceada w) se verifica si w está bien balanceada. Por ejemplo,

balanceada [A,A,B,B] = True

balanceada [A,B,A,B] = True

balanceada [A,B,B,A] = False

Indicación: Definir balanceada usando la función auxiliar

balanceada_aux :: alfabeto list $\Rightarrow$ nat $\Rightarrow$ bool

tal que (balanceada_aux w 0) se verifica si w está bien balanceada.

*----- *}*

```
fun balanceada_aux :: "alfabeto list  $\Rightarrow$  nat  $\Rightarrow$  bool" where
  "balanceada_aux [] 0 = True"
| "balanceada_aux (A#w) n = balanceada_aux w (Suc n)"
| "balanceada_aux (B#w) (Suc n) = balanceada_aux w n"
| "balanceada_aux w n = False"
```

```
fun balanceada :: "alfabeto list  $\Rightarrow$  bool" where
  "balanceada w = balanceada_aux w 0"
```

```
value "balanceada [A,A,B,B]" -- "= True"
value "balanceada [A,B,A,B]" -- "= True"
value "balanceada [A,B,B,A]" -- "= False"
```

```
text {*
```

Ejercicio 6. Demostrar que balanceada es un reconocedor correcto de la gramática S; es decir,

$w \in S \implies \text{balanceada } w$

*----- *}*

```
text {*
```

Nota: En la demostración se usarán los siguientes lemas auxiliares:

- *balanceada_correcto_aux_1:*
 $\text{balanceada_aux } w \ n \implies \text{balanceada_aux } (w @ [B]) \ (Suc \ n)$
- *balanceada_correcto_aux_2:*
 $\llbracket \text{balanceada_aux } ?v \ ?n; \text{balanceada_aux } ?w \ 0 \rrbracket$
 $\implies \text{balanceada_aux } (?v @ ?w) \ ?n$
- *balanceada_correcto_aux_3:*
 $w \in S \implies \text{balanceada_aux } w \ 0 \quad \text{*}$

```

-- "La demostración automática del primer lema auxiliar es"
lemma balanceada_correcto_aux_1:
  "balanceada_aux w n  $\implies$  balanceada_aux (w @ [B]) (Suc n)"
by (induct w n rule: balanceada_aux.induct) simp_all

-- "La demostración estructurada del primer lema auxiliar es"
lemma balanceada_correcto_aux_1b:
  assumes "balanceada_aux w n"
  shows "balanceada_aux (w @ [B]) (Suc n)"
using assms
proof (induct w n rule: balanceada_aux.induct)
  assume "balanceada_aux [] 0"
  thus "balanceada_aux ([] @ [B]) (Suc 0)" by simp
next
  fix w n
  assume "balanceada_aux w (Suc n)  $\implies$  balanceada_aux (w @ [B]) (Suc (Suc n))"
    "balanceada_aux (A # w) n"
  thus "balanceada_aux ((A # w) @ [B]) (Suc n)" by simp
next
  fix w n
  assume "balanceada_aux w n  $\implies$  balanceada_aux (w @ [B]) (Suc n)"
    "balanceada_aux (B # w) (Suc n)"
  thus "balanceada_aux ((B # w) @ [B]) (Suc (Suc n))" by simp
next
  fix v
  assume "balanceada_aux (B # v) 0"
  thus "balanceada_aux ((B # v) @ [B]) (Suc 0)" by simp
next
  fix v
  assume "balanceada_aux [] (Suc v)"
  thus "balanceada_aux ([] @ [B]) (Suc (Suc v))" by simp
qed

-- "La demostración automática del segundo lema auxiliar es"
lemma balanceada_correcto_aux_2:
  assumes "balanceada_aux v n"
    "balanceada_aux w 0"
  shows "balanceada_aux (v @ w) n"
using assms
by (induct v n rule: balanceada_aux.induct) simp_all

-- "La demostración automática del tercer lema auxiliar es"
lemma balanceada_correcto_aux_3:
  "w  $\in$  S  $\implies$  balanceada_aux w 0"
by (induct rule: S.induct)

```

```

(auto simp add: balanceada_correcto_aux_1 balanceada_correcto_aux_2)

-- "La demostración estructurada del tercer lema auxiliar es"
lemma balanceada_correcto_aux_3b:
  "w ∈ S ⇒ balanceada_aux w 0"
proof (induct rule: S.induct)
  show "balanceada_aux [] 0" by simp
next
  fix w
  assume "w ∈ S"
    "balanceada_aux w 0"
  thus "balanceada_aux ([A] @ w @ [B]) 0"
    by (simp add: balanceada_correcto_aux_1)
next
  fix v w
  assume "v ∈ S"
    "balanceada_aux v 0"
    "w ∈ S"
    "balanceada_aux w 0"
  thus "balanceada_aux (v @ w) 0"
    by (simp add: balanceada_correcto_aux_2)
qed

-- "La demostración estructurada del tercer lema es"
lemma balanceada_correcto:
  "w ∈ S ⇒ balanceada w"
by (simp add: balanceada_correcto_aux_3)

text {
  -----
  Ejercicio 7. Demostrar que balanceada es un reconocedor completo de
  la gramática S; es decir,
  balanceada w ⇒ w ∈ S
  ----- *}

-- "La demostración automática del primer lema auxiliar es"
lemma balanceada_completo_aux_1:
  "[A,B] ∈ S"
using S1 S2 [where w = "[]"] by simp

-- "La demostración estructurada del primer lema auxiliar es"
lemma balanceada_completo_aux_1b:
  "[A,B] ∈ S"
proof -
  have "[] ∈ S" using S1 by simp

```

```

    hence "[A] @ [] @ [B] ∈ S" using S2 [where w = "[]"] by simp
    thus "[A,B] ∈ S" by simp
qed

-- "La demostración estructurada del segundo lema auxiliar es"
lemma balanceada_completo_aux_2:
  assumes "u ∈ S"
  shows "∧v w. u = v @ w ⇒ v @ A # B # w ∈ S"
using assms
proof (induct)
  fix v w :: "alfabeto list"
  assume "[ ] = v @ w"
  thus "v @ A # B # w ∈ S" by (simp add: balanceada_completo_aux_1)
next
  fix v u w :: "alfabeto list"
  assume uS: "u ∈ S" and
    HI: "∧v w. u = v @ w ⇒ v @ A # B # w ∈ S" and
    sup: "[A] @ u @ [B] = v @ w"
  show "v @ A # B # w ∈ S"
proof (cases v)
  case Nil
  hence "w = A # u @ [B]" using sup by simp
  hence "w ∈ S" using uS S2 by simp
  hence "[A,B] @ w ∈ S" using balanceada_completo_aux_1 S3 by blast
  thus ?thesis using Nil by simp
next
  case (Cons x v')
  show ?thesis
proof (cases w rule: rev_cases)
  case Nil
  have "A # u @ [B] ∈ S" using S2 uS by simp
  hence "(A # u @ [B]) @ [A,B] ∈ S"
    using balanceada_completo_aux_1 S3 by blast
  thus ?thesis using Nil Cons sup by auto
next
  case (snoc w' y)
  hence u: "u = v' @ w'" and [simp]: "x = A ∧ y = B"
    using Cons sup by auto
  from u have "v' @ A # B # w' ∈ S" by (rule HI)
  hence "A # (v' @ A # B # w') @ [B] ∈ S"
    using S2 [where w = "v' @ A # B # w'"] by simp
  thus ?thesis using Cons snoc by auto
qed
qed
next

```

```

fix v' w' v w
assume v'S: "v' ∈ S"
  and HIv: "∧v w. v' = v @ w ⇒ v @ A # B # w ∈ S"
  and w'S: "w' ∈ S"
  and HIw: "∧v w. w' = v @ w ⇒ v @ A # B # w ∈ S"
  and sup: "v' @ w' = v @ w"
then obtain r where "v' = v @ r ∧ r @ w' = w v v' @ r = v ∧ w' = r @ w"
  (is "?A v ?B")
  by (auto simp: append_eq_append_conv2)
thus "v @ A # B # w ∈ S"
proof
  assume A: ?A
  hence "v @ A # B # r ∈ S" using HIv by blast
  hence "(v @ A # B # r) @ w' ∈ S" using w'S by (rule S3)
  thus ?thesis using A by auto
next
  assume B: ?B
  hence "r @ A # B # w ∈ S" using HIw by blast
  with v'S have "v' @ (r @ A # B # w) ∈ S" by (rule S3)
  thus ?thesis using B by auto
qed
qed

-- "La demostración estructurada del tercer lema auxiliar es"
lemma balanceada_completo_aux_3:
  "balanceada_aux w n ⇒ replicate n A @ w ∈ S"
proof (induct w n rule: balanceada_aux.induct)
  assume "balanceada_aux [] 0"
  thus "replicate 0 A @ [] ∈ S" using S1 by simp
next
  fix w n
  assume "balanceada_aux w (Suc n) ⇒ replicate (Suc n) A @ w ∈ S"
  and "balanceada_aux (A # w) n"
  thus "replicate n A @ A # w ∈ S"
  by (simp add: replicate_app_Cons_same)
next
  fix w n
  assume "balanceada_aux w n ⇒ replicate n A @ w ∈ S"
  and "balanceada_aux (B # w) (Suc n)"
  thus "replicate (Suc n) A @ B # w ∈ S"
  by (simp add: balanceada_completo_aux_2
    replicate_app_Cons_same[symmetric])
next
  fix v
  assume "balanceada_aux (B # v) 0"

```

```
      thus "replicate 0 A @ B # v ∈ S" by simp
next
  fix v
  assume "balanceada_aux [] (Suc v)"
  thus "replicate (Suc v) A @ [] ∈ S" by simp
qed

-- "La demostración del lema es"
lemma balanceada_completo:
  assumes "balanceada w"
  shows   "w ∈ S"
proof -
  have "balanceada_aux w 0" using assms by simp
  hence "replicate 0 A @ w ∈ S" by (rule balanceada_completo_aux_3)
  thus "w ∈ S" by simp
qed

end
```